Anne Baanen Alexander Bentkamp Jasmin Blanchette Johannes Hölzl Jannis Limperg

# The Hitchhiker's Guide to Logical Verification

2021 Tablet Edition

(October 12, 2021)



lean-forward.github.io/ logical-verification/2021

# **Contents**

Co	Contents		
Pr	eface	<b>!</b>	ix
I	Bas	sics	1
1	Defi	nitions and Statements	3
	1.1	Types and Terms	4
	1.2	Type Definitions	12
	1.3	Function Definitions	20
	1.4	Lemma Statements	24
	1.5	Summary of New Lean Constructs	27
2	Вас	kward Proofs	29
	2.1	Tactic Mode	30
	2.2	Basic Tactics	33
	2.3	Reasoning about Connectives and Quantifiers	37
	2.4	Reasoning about Equality	43
	2.5	Rewriting Tactics	44
	2.6	Proofs by Mathematical Induction	46
	2.7	Induction Tactic	49
	2.8	Cleanup Tactics	49
	2.9	Summary of New Lean Constructs	50

3	Forw	vard Proofs	53
	3.1	Structured Proofs	54
	3.2	Structured Constructs	57
	3.3	Forward Reasoning about Connectives and Quantifiers	59
	3.4	Calculational Proofs	63
	3.5	Forward Reasoning with Tactics	64
	3.6	Dependent Types	66
	3.7	The PAT Principle	69
	3.8	Induction by Pattern Matching	73
	3.9	Summary of New Lean Constructs	76
II	Fun	ctional–Logic Programming	77
4	Func	tional Programming	79
	4.1	Inductive Types	79
	4.2	Structural Induction	81
	4.3	Structural Recursion	84
	4.4	Pattern Matching Expressions	86
	4.5	Structures	88
	4.6	Type Classes	91
	4.7	Lists	96
	4.8	Binary Trees	104
	4.9	Case Distinction and Induction Tactics	106
	4.10	Dependent Inductive Types	107
	4.11	Summary of New Lean Constructs	111
5		ictive Predicates	113
	5.1	Introductory Examples	113
	5.2	Logical Symbols	122
	5.3	Rule Induction	124
	5.4	Linear Arithmetic Tactic	129
	5.5	Elimination	130
	5.6	Further Examples	133
	5.7	Summary of New Lean Constructs	140

6	Mon	ads	141
	6.1	Introductory Example	141
	6.2	Two Operations and Three Laws	144
	6.3	A Type Class	148
	6.4	No Effects	149
	6.5	Basic Exceptions	150
	6.6	Mutable State	153
	6.7	Nondeterminism	157
	6.8	Tautology Tactic	159
	6.9	A Generic Algorithm: Iteration over a List	159
	6.10	Summary of New Lean Constructs	161
7	Meta	aprogramming	163
	7.1	Tactics and Tactic Combinators	164
	7.2	The Metaprogramming Monad	170
	7.3	Names, Expressions, Declarations, and Environments	175
	7.4	First Example: A Conjunction-Destructing Tactic	180
	7.5	Second Example: A Provability Advisor	182
	7.6	A Look at Two Predefined Tactics	185
	7.7	Miscellaneous Tactics	187
	7.8	Summary of New Lean Constructs	187
Ш	Pro	gram Semantics	189
8	0 <b>pe</b> 8.1	rational Semantics	<b>191</b> 191
	8.2	Formal Semantics	191 192
	8.3	Big-Step Semantics	192
	8.4	Properties of the Big-Step Semantics	194
	8.5	Small-Step Semantics	201
	8.6	Properties of the Small-Step Semantics	201
	8.7	Parallelism	205
	0./	1 arattetioni	200
Q	Hoa	re Logic	211

	9.1	Hoare Triples	211
	9.2	Hoare Rules	213
	9.3	A Semantic Approach to Hoare Logic	216
	9.4	First Program: Exchanging Two Variables	219
	9.5	Second Program: Adding Two Numbers	221
	9.6	Finish Tactic	224
	9.7	A Verification Condition Generator	224
	9.8	Second Program Revisited: Adding Two Numbers	227
	9.9	Hoare Triples for Total Correctness	228
	9.10	Summary of New Lean Constructs	229
10	Den	otational Semantics	231
	10.1	Compositionality	231
	10.2	A Relational Denotational Semantics	233
		Fixpoints	235
	10.4	Monotone Functions	236
		Complete Lattices	237
		Least Fixpoint	239
	10.7	A Relational Denotational Semantics, Continued	239
		Application to Program Equivalence	239
	10.9	A Simpler Approach Based on an Inductive Predicate .	242
11.7	N/1-1	:hematics	242
IV	mat	nematics	243
11		cal Foundations of Mathematics	245
		Universes	245
		The Peculiarities of Prop	247
		The Axiom of Choice	251
		Subtypes	253
		Quotient Types	258
	11.6	Summary of New Lean Constructs	266
12		c Mathematical Structures	269
	12.1	Type Classes over a Single Binary Operator	269

Bibliography					
	13.4	Summary of New Lean Constructs	29/		
		Final Exhortation			
	_	Real Numbers			
	13.1	Rational Numbers	286		
13		onal and Real Numbers	285		
	12.7	Summary of New Lean Constructs	283		
	12.6	Order Type Classes	280		
		Lists, Multisets, and Finite Sets			
	12.4	Normalization Tactics	278		
	12.3	Coercions	276		
	12.2	Type Classes over Two Binary Operators	273		

## **Preface**

Formal proof assistants are software tools designed to help their users carry out computer-checked proofs in a logical calculus. We usually call them *proof assistants*, or *interactive theorem provers*, but a mischievous student coined the phase "proof-preventing beasts," and dictation software occasionally misunderstands "theorem prover" as "fear improver." Consider yourself warned.

**Rigorous and Formal Proofs** Interactive theorem proving has its own terminology, already starting with the notion of "proof." A *formal proof* is a logical argument expressed in a logical formalism. In this context, "formal" means "logical" or "logic-based." Logicians—the mathematicians of logics—carried out formal proofs on papers decades before the advent of computers, but nowadays formal proofs are almost always carried out using a proof assistant.

In contrast, an *informal proof* is what a mathematician would normally call a proof. These are often carried out in MEX or on a blackboard, and are also called "pen-and-paper proofs." The level of detail can vary a lot, and phrases such as "it is obvious that," "clearly," and "without loss of generality" move some of the proof burden onto the reader. A *rigorous proof* is a very detailed informal proof.

The main strength of proof assistants is that they help develop highly trustworthy, unambiguous proofs of mathematical statements, using a precise logic. They can be used to prove arbitrarily advanced results, and not only toy examples or logic puzzles. Formal proofs also help students understand what constitutes a valid definition or a valid proof. To quote Scott Aaronson:

I still remember having to grade hundreds of exams where the students started out by assuming what had to be proved, or filled page after page with gibberish in the hope that, somewhere in the mess, they *might* accidentally have said something correct.

When we develop a new theory, formal proofs can help us explore it. They are useful when we generalize, refactor, or otherwise modify an existing proof, in much the same way as a compiler helps us develop correct programs. They give a high level of trustworthiness that makes it easier for others to review the proof. In addition, formal proofs can form the basis of verified computational tools (e.g., verified computer algebra systems).

**Success Stories** There have been a number of success stories in mathematics and computer science. Three landmark results in the formalization of mathematics have been the proof of the four-color theorem by Gonthier et al. [8], the proof of the odd-order theorem by Gonthier et al. [9], and the proof of the Kepler conjecture by Hales et al. [12]. The earliest work in this area was carried out by Nicolaas de Bruijn and his colleagues starting in the 1960s in a system called AUTOMATH.<sup>2</sup>

Today, few mathematicians use proof assistants, but this is slowly changing. For example, 29 participants of the Lean Together 2019 meeting in Amsterdam,<sup>3</sup> about formalization of mathematics, self-identified as mathematicians.

Most users of proof assistants today are computer scientists. A few companies, including AMD [32] and Intel [13], have been using

https://www.scottaaronson.com/teaching.pdf

<sup>2</sup>https://www.win.tue.nl/automath/

<sup>3</sup>https://lean-forward.github.io/lean-together/2019/index.html

proof assistants to verify their designs. In academia, some of the milestones are the verifications of the operating system kernels seL4 [16] and CertiKOS [11] and the development of the verified compilers CompCert [19], JinjaThreads [23], and CakeML [18].

**Proof Assistants** There are dozens of proof assistants in development or use across the world. A list of the main ones follows, classified by logical foundations:

- set theory: Isabelle/ZF, Metamath, Mizar;
- simple type theory: HOL4, HOL Light, Isabelle/HOL;
- dependent type theory: Agda, Coq, Lean, Matita, PVS;
- Lisp-like first-order logic: ACL2.

For a history of proof assistants and interactive theorem proving, we refer to Harrison, Urban, and Wiedijk's highly informative chapter [14].

**Lean** Lean is a new proof assistant developed primarily by Leonardo de Moura (Microsoft Research) since 2012. Its mathematical library, mathlib, was originally developed under the leadership of Jeremy Avigad (Carnegie Mellon University) but is now maintained and further extended by the users' community [24].<sup>4</sup>

We will use Lean community version 3.30.0, mathlib revision 48104c3a22ae, and a few extensions collected in a small library called LoVelib.<sup>5</sup> Although it is still a research project, with some rough edges, there are several reasons why Lean is a suitable vehicle to teach interactive theorem proving:

It has a highly expressive, and highly interesting, logic based on the calculus of inductive constructions, a dependent type theory.

<sup>4</sup>https://github.com/leanprover-community/mathlib

<sup>5</sup>https://github.com/blanchette/logical\_verification\_2021/raw/main/ lean/lovelib.lean

- It is extended with classical axioms and quotient types, making it convenient to verify mathematics.
- It includes a convenient metaprogramming framework, which can be used to program custom proof automation.
- It includes a modern user interface via a Visual Studio Code plugin.
- It has highly readable, fairly complete documentation.
- It is open source.

Lean's core library includes only basic algebraic definitions. More setup and proof automation are found in mathlib. It is more than a mathematical library; it provides a lot of basic automation on top of Lean's core library that one would expect from a modern proof assistant.

**This Guide** This guide is a companion to the MSc-level course Logical Verification (LoVe) taught at the Vrije Universiteit Amsterdam. Our primary aim is to teach interactive theorem proving. Lean is the vehicle, not an end of itself. As such, this guide is not designed to be a comprehensive Lean tutorial—for this, we recommend *Theorem Proving in Lean* [1]. The guide is also no substitute for attending class or doing the exercises and homework. Theorem proving is not for spectators; it can only be learned by doing.

Specifically, our goal is that you

- learn fundamental theory and techniques in interactive theorem proving;
- learn how to use logic as a precise language for modeling systems and stating properties about them;
- familiarize yourselves with some areas in which proof assistants are successfully applied, such as functional programming, the semantics of imperative programming languages, and mathematics;
- develop some practical skills you can apply on a larger project (as a hobby, for an MSc or PhD, or in industry);

- reach a point where you feel ready to move to another proof assistant and apply what you have learned;
- get to understand the domain well enough to start reading relevant scientific papers published at international conferences such as Certified Programs and Proofs (CPP) and Interactive Theorem Proving (ITP) or in journals such as the Journal of Automated Reasoning (JAR).

Equipped with a good knowledge of Lean, you will find it easy to move to another proof assistant based on dependent type theory, such as Agda or Coq, or to a system based on simple type theory, such as HOL4 or Isabelle/HOL.

We assume that you are familiar with typed functional programming, as embodied in the languages Haskell, OCaml, and Standard ML. If you do not see that the term g(fab) applies the (curried) function f to two arguments a and b and passes the result to the function g, or that  $\lambda n$ , n+1 is the function that maps n to n+1, we strongly recommend that you start by studying a tutorial, such as the first chapters of the online tutorial *Learn You a Haskell for Great Good!* [22]. You can stop reading once you have reached the end of the section titled "Lambdas."

An important characteristic of this guide, which it shares with Knuth's *T<sub>E</sub>Xbook* [17], is that it does not always tell the truth. To simplify the exposition, simple but false claims are made about Lean. Some of these statements are rectified in later chapters. Like Knuth, we feel that "this technique of deliberate lying will actually make it easier for you to learn the ideas. Once you understand a simple but false rule, it will not be hard to supplement that rule with its exceptions."

The Lean files accompanying this guide can be found in a public repository. The files' naming scheme follows this guide's chapters; thus, loveo6\_monads\_demo.lean is the main file associated with Chapter 6 ("Monads"), which is demonstrated in class, loveo6\_

<sup>6</sup>https://github.com/blanchette/logical\_verification\_2020

monads\_exercise\_sheet.lean is the exercise sheet, and loveo6\_monads\_homework\_sheet.lean is the homework sheet.

We have a huge debt to the authors of Theorem Proving in Lean [1] and Concrete Semantics: With Isabelle/HOL [27], who have taught us Lean and programming language semantics. Many of their ideas appear in this guide.

We thank Robert Lewis and Assia Mahboubi for their useful comments on this guide's organization and focus. We thank Kiran Gopinathan and Ilya Sergey for sharing the anecdote reported in footnote 3 of Chapter 1 and letting us share it further. We thank Daniel Fabian for designing the first tablet-optimized version of this guide. Finally, we thank Chris Bailey, Kevin Buzzard, Wan Fokkink, Antonius Danny Reyes, Robert Schütz, Patrick Thomas, Huub Vromen, and Wijnand van Woerkom for reporting typos and some more serious errors they found in earlier editions of this guide. Should *you* find some mistakes in this text, please let us know.

**Special Symbols** In this guide, we assume that you will be using Visual Studio Code and its Lean extension to edit .lean files. Visual Studio Code lets us enter Unicode symbols by entering backslash \ followed by an ASCII identifier. For example,  $\rightarrow$ ,  $\mathbb{Z}$ , or  $\mathbb{Q}$  can be entered by typing \->, \Z, or \Q and pressing the tab key or the space bar. We will freely use these notations. For reference, we provide a list of the main non-ASCII symbols that are used in the guide and, for each, one of its ASCII representations. By hovering over a symbol in Visual Studio Code while holding the control or command key pressed, you will see the different ways in which it can be entered.

\not \->  $\rightarrow$  $\exists$ \ex  $\neq$ \neq \circ 0 \intersect  $\cap$ \bigcirc  $\bigcirc$ \=>  $\Rightarrow$ \]]  $\mathbb{Q}$ \Q β \b ε \e \0 0 /3 3 \6 6

\9

9

 $\and$  $\wedge$ \<->  $\leftrightarrow$  $\leq$ \<= \~~  $\approx$  $\emptyset$ \empty  $\in$ \in \<- $\leftarrow$ \==>  $\Longrightarrow$  $\N$  $\mathbb{N}$  $\mathbb{R}$  $\R$ \g γ λ ∖la \1 1 \4 4 \7 7

\or  $\forall$ \fo  $\geq$ \>= X \x \union U \downleftharpoon \mapsto  $\mapsto$ ]]/  $\backslash Z$  $\mathbb{Z}$ \a α δ \de σ \s \2 2 \5 5 \8 8

# Part I Basics

# **Chapter 1**

# **Definitions and Statements**

We start our journey by studying the basics of Lean, without carrying out any proofs yet. We review how to define new types and functions and how to state their expected properties as lemmas.

Lean's logical foundation is a rich formalism called the *calculus of inductive constructions*, which supports *dependent types*. In this chapter, we restrict our attention to its simply (i.e., nondependently) typed fragment. It is inspired by the  $\lambda$ -calculus and resembles typed functional programming languages such as Haskell, OCaml, and Standard ML. Even if you have not been exposed to these languages, you will recognize many of the concepts from modern languages (e.g., Python, C++11, Java 8). In a first approximation:

Lean = functional programming + logic

If your background is in mathematics, you probably already know most of the key concepts underlying functional programming, sometimes under different names. For example, the Haskell program

```
fib 0 = 0
fib 1 = 1
fib n = fib (n - 2) + fib (n - 1)
```

closely matches the mathematical definition

$$fib(n) = \begin{cases} 0 & \text{if } n = 0\\ 1 & \text{if } n = 1\\ fib(n-2) + fib(n-1) & \text{if } n \geq 2 \end{cases}$$

## 1.1 Types and Terms

Simple type theory (also called higher-order logic) corresponds roughly to the simply typed  $\lambda$ -calculus [5] extended with an equality operator (=). It is an abstract, extremely simplified version of a programming language with a function-calling mechanism that prefigures functional programming. It can also be viewed as a generalization of first-order logic (also called predicate logic).

#### **1.1.1 Types**

Types are either basic types such as  $\mathbb{Z}$ ,  $\mathbb{Q}$ , and bool or total functions  $\sigma \to \tau$ , where  $\sigma$  and  $\tau$  are themselves types. Types indicate which values an expression may evaluate to. They introduce a discipline that is followed somewhat implicitly in mathematics. In principle, nothing prevents a mathematician from stating 1  $\in$  2, but a typing discipline would mark this as the error it likely is.

Semantically, types can be viewed as sets. We would normally define the types  $\mathbb{Z}$ ,  $\mathbb{Q}$ , and bool so that they faithfully capture the mathematicians'  $\mathbb{Z}$  and  $\mathbb{Q}$  and the computer scientists' Booleans, and similarly for the function arrow ( $\rightarrow$ ). But despite their similarities, Lean and mathematics are distinct languages. Lean's types may be *interpreted as* sets, but they *are not* sets.

Higher-order types are types containing left-nested  $\to$  arrows. Values of such types are functions that take other functions as arguments. Accordingly, the type  $(\mathbb{Z} \to \mathbb{Z}) \to \mathbb{Q}$  is the type of unary functions that take a function of type  $\mathbb{Z} \to \mathbb{Z}$  as argument and that return a value of type  $\mathbb{Q}$ .

#### **1.1.2** Terms

The terms, or expressions, of simple type theory consist of

- constants c;
- variables x;
- applications t u;
- $\lambda$ -expressions  $\lambda x$ , t.

Above, t and u denote arbitrary terms. We can also write t :  $\sigma$  to indicate that the term t has the type  $\sigma$ .

A constant  $c:\sigma$  is a symbol of type  $\sigma$  whose meaning is fixed in the current global context. For example, an arithmetic theory might contain constants such as  $o:\mathbb{Z}$ ,  $1:\mathbb{Z}$ , abs  $:\mathbb{Z}\to\mathbb{N}$ , square  $:\mathbb{N}\to\mathbb{N}$ , and prime  $:\mathbb{N}\to \mathrm{bool}$ . Constants include functions (e.g., abs) and predicates (e.g., prime).

A variable  $x:\sigma$  is either bound or free. A bound variable refers back to the input of a  $\lambda$ -expression  $\lambda x:\sigma$ , t enclosing it. In  $\lambda x:\mathbb{Z}$ , square (abs x) the second x is a variable that refers back to the  $\lambda$  binder's input x. By contrast, a free variable is declared in the local context—a concept that will be explained below.

An application t u, where t :  $\sigma \to \tau$  and u :  $\sigma$ , is a term of type  $\tau$  denoting the result of applying the function t to the argument u—e.g., abs  $\sigma$ . No parentheses are needed around the argument, unless it is a complex term—e.g., prime (abs  $\sigma$ ).

Given a term  $t:\tau$ , a  $\lambda$ -expression  $\lambda x:\sigma$ , t denotes the total function of type  $\sigma \to \tau$  that maps each input value x of type  $\sigma$  to the function body t, where t may contain x. For example,  $\lambda x:\mathbb{Z}$ , square (abs x) denotes the function that maps (the value denoted by) o to (the value denoted by) square (abs o), that maps t to square (abs t), and so on. A more intuitive syntax would have been t in square (abs t), but this is not supported by Lean.

Applications and  $\lambda$ -expressions mirror each other: A  $\lambda$ -expression "builds" a function; an application "destructs" a function. Although our functions are unary (i.e., they take one argument), we can build n-ary functions by nesting  $\lambda$ s, using an ingenious tech-

nique called *currying*. For example,  $\lambda x:\sigma$ ,  $(\lambda y:\tau,x)$  denotes the function of type  $\sigma \to (\tau \to \sigma)$  that takes two arguments and returns the first one. Strictly speaking,  $\sigma \to (\tau \to \sigma)$  takes a single argument and returns a function, which in turn takes an argument. Applications work in the same way: If  $K:=(\lambda x:\mathbb{Z},(\lambda y:\mathbb{Z},x))$ , then  $K:=(\lambda y:\mathbb{Z},1)$  and (K:1) 0:=1. The function K in K 1, which is applied to a single argument, is said to be *partially applied*.

Currying is so useful a concept that we will omit most parentheses, writing

$$\sigma \rightarrow \tau \rightarrow \upsilon$$
 for  $\sigma \rightarrow (\tau \rightarrow \upsilon)$   
tuv for (tu)v  
 $\lambda x : \sigma, \lambda y : \tau, t$  for  $\lambda x : \sigma, (\lambda y : \tau, t)$ 

and also

$$\lambda(x:\sigma)(y:\tau)$$
, t for  $\lambda x:\sigma$ ,  $\lambda y:\tau$ , t  $\lambda x y:\sigma$ , t for  $\lambda(x:\sigma)(y:\sigma)$ , t

In mathematics, it is customary to write binary operators in infix syntax—e.g., x + y. Such notations are also possible in Lean, as syntactic sugar for (+) x y. The parentheses around the + sign are necessary to disable the parsing of + as an infix operator. Partial application is possible with this syntax. For example, (+) 1 denotes the unary function that adds one to its argument. Other ways to write this function are  $\lambda x$ , (+) 1 x and  $\lambda x$ , 1 + x.

One way to work with Lean is to declare the types and constants we need using the constant(s) command. Consider the following declarations:

```
constants a b : \mathbb{Z} constant f : \mathbb{Z} \to \mathbb{Z} constant g : \mathbb{Z} \to \mathbb{Z} \to \mathbb{Z} #check \lambda x : \mathbb{Z}, g (f (g a x)) (g x b) #check \lambda x, g (f (g a x)) (g x b)
```

The first three lines declare four constants (a, b, f, g), which can be used to form terms. The last two lines use the #check command to type-check some terms and show their types. The # prefix identifies diagnosis commands: commands that are useful for debugging but that we would normally not keep in our Lean files.

The constant(s) command can even be used to declare types. In Lean, types are terms like any other. The next example introduces a type of "Trooleans" and three constants of that type:

```
constant trool : Type
constants trool.true trool.false trool.maybe : trool
```

The constants thus introduced have no definition. They are fully unspecified. We cannot tell whether trool.true, trool.false, and trool.maybe are equal or not, and we know nothing about trool except that it has at least one value (e.g., trool.true).

#### 1.1.3 Type Checking and Type Inference

When Lean parses a term, it checks whether the term is well typed. In the process, it tries to infer the types of bound variables if those are omitted—e.g., the type of x in  $\lambda x$ , 1 + x. Type inference lightens notations and saves some typing.

For simple type theory, type checking and type inference are decidable problems. Advanced features such as overloading (the possibility to reuse the same name for several constants—e.g.,  $o: \mathbb{N}$  and  $o: \mathbb{R}$ ) can lead to undecidability [30]. Lean takes a pragmatic, computer-science-oriented approach and assumes that numerals  $o, 1, 2, \ldots$  are of type  $\mathbb{N}$  if several types are possible.

Lean's type system can be expressed as a formal system. A formal system consists of judgments and of (derivation) rules for producing judgments. A typing judgment has the form  $C \vdash t : \sigma$ , meaning that term t has type  $\sigma$  in local context C. The local context gives the types of the variables in t that are not bound by a  $\lambda$ . The local context is used to keep track of the variables bound by  $\lambda$ 's outside t. For a function definition, it will consist of the function's

parameters. For example, in Lean, the right-hand side of the last equation of fib's above would be type-checked in a local context consisting of  $n: \mathbb{N}$ .

For simple type theory, there are four typing rules, one per kind of term:

CST if c is declared with type 
$$\sigma$$

$$C \vdash c : \sigma$$
VAR if  $x : \sigma$  is the last occurrence of  $x$  in  $C$ 

$$\frac{C \vdash t : \sigma \rightarrow \tau \quad C \vdash u : \sigma}{C \vdash t u : \tau} APP$$

$$\frac{C \vdash t : \sigma \rightarrow \tau \quad C \vdash u : \tau}{C \vdash (\lambda x : \sigma, t) : \sigma \rightarrow \tau} LAM$$

Each rule has zero or more premises (above the horizontal bar), a conclusion (below the bar), and possibly a side condition. The premises are typing judgments, whereas the side conditions are arbitrary mathematical conditions on the mathematical variables occurring in the rule. To discharge the premises, we need to continue performing a derivation upward, as we will see in a moment. As for the side conditions, we can use the entire arsenal of mathematics to show that they are true.

The first two rules, labeled CST and VAR, have no premises, but they have side conditions that must be satisfied for the rules to apply. The last two rules take one or two judgments as premises and produce a new judgment. LAM is the only rule that modifies the local context: As we enter the body t of a  $\lambda$ -expression, we need to record the existence of the bound variable x and its type to be ready when we meet x in t.

We can use this rule system to prove that a given term is well typed by working our way backwards (i.e., upwards) and applying the rules, building a *formal derivation* of a typing judgment. Like natural trees, derivation trees are drawn with the root at the bot-

tom. The derived judgment appears at the root, and each branch ends with the application of a premise-less rule. Rule applications are indicated by a horizontal bar and a label. The following typing derivation establishes that the term  $\lambda x:\mathbb{Z}$ , abs x has type  $\mathbb{Z}\to\mathbb{N}$  in an arbitrary local context C:

Reading the proof from the root upwards, notice how the local context is threaded through and how it is extended by the LAM rule. The rule moves the variable bound by the  $\lambda$ -expression to the local context, making an application of VAR possible further up the tree. If the variable x is already declared in C, it becomes shadowed by  $x:\mathbb{Z}$  after entering the  $\lambda$ -expression.

The above type system only checks that terms are well typed. It does not check that types are well formed. For example, list  $\mathbb{Z}$  is well formed, whereas  $\mathbb{Z}$  list and list list are ill-formed. For simple type theory, well-formedness is easy to check: Only declared type constructors should be used, and each n-ary type constructor should be passed exactly n type arguments.

As a side note, type inference is a generalization of type checking where the types on the right-hand side of the colon (:) in judgments may be replaced by placeholders. Lean's type inference is based on an algorithm due to Hindley [15] and Milner [25], which also forms the basis of Haskell, OCaml, and Standard ML. The algorithm generates type constraints involving type variables  $?\alpha$ ,  $?\beta$ ,  $?\gamma$ , ..., and attempts to solve them using a type unification procedure. For example, when inferring the type  $?\alpha$  of  $\lambda x$ , abs x, Lean

would perform the following schematic type derivation:

$$\frac{x:?\beta \vdash abs:?\beta \rightarrow ?\gamma}{x:?\beta \vdash x:?\beta} \frac{\text{VAR}}{x:?\beta \vdash x:?\beta} \frac{\text{APP}}{\text{APP}}$$

$$\frac{x:?\beta \vdash abs x:?\gamma}{\vdash (\lambda x, abs x):?\alpha} \text{LAM}$$

In addition, Lean would generate the following constraints to ensure that all the rule applications are legal:

- 1. For the application of LAM, the type of  $\lambda x$ , abs x must be of the form  $?\beta \rightarrow ?\gamma$ , for some types  $?\beta$  and  $?\gamma$ . Thus, Lean would generate the constraint  $?\alpha = ?\beta \rightarrow ?\gamma$ .
- 2. For the application of CST, the type of abs must correspond to the declaration as  $\mathbb{Z} \to \mathbb{N}$ . Thus, Lean would generate the constraint  $\beta \to \gamma = \mathbb{Z} \to \mathbb{N}$ .

Solving the two constraints yields  $?\alpha := \mathbb{Z} \to \mathbb{N}$ , which is indeed the type that Lean infers for  $\lambda x$ , abs x.

#### 1.1.4 Type Inhabitation

Given a type  $\sigma$ , the type inhabitation problem consists of finding an "inhabitant" of that type—a term of type  $\sigma$ —within the empty local context. It may seem like a pointless exercise, but as we will see in Chapter 3, this problem is closely related to that of finding a proof of a proposition. Seemingly silly exercises of the form "find a term of type  $\sigma$ " are good practice towards mastery of theorem proving.

To create a term of a given type, start with the placeholder \_ and recursively apply a combination of the following two steps:

1. If the type is of the form  $\sigma \to \tau$ , a possible inhabitant is an anonymous function, of the form  $\lambda x:\sigma$ , \_, where \_ is a placeholder for a missing term of type  $\tau$ . Lean will mark \_ as an error; if you hover over it in Visual Studio Code, a tooltip will

show the missing term's type as well as any variables declared in the local context.

2. Given a type  $\sigma$  (which may be a function type), you can use any constant c or variable  $x:\tau_1\to\cdots\to\tau_n\to\sigma$  to build a term of that type. For each argument, you need to put a placeholder, yielding c \_ . . . \_ or x \_ . . . \_ .

The placeholders can be eliminated recursively using the same procedure.

As an example, we will apply the procedure to find a term of type

$$(\alpha \rightarrow \beta \rightarrow \gamma) \rightarrow ((\beta \rightarrow \alpha) \rightarrow \beta) \rightarrow \alpha \rightarrow \gamma$$

Initially, only step 1 is applicable, with  $\sigma:=\alpha\to\beta\to\gamma$  and  $\tau:=((\beta\to\alpha)\to\beta)\to\alpha\to\gamma$ . (Recall that  $\to$  is right-associative:  $\sigma\to\tau\to\upsilon$  stands for  $\sigma\to(\tau\to\upsilon)$ .) This results in the term  $\lambda f$ , \_, which has the right type but has a placeholder left. Since the argument f has type  $\sigma$ , a function type, it makes sense to use the name f for it. Then we continue recursively with the placeholder, of type  $\tau$ . Again, only step 1 is possible, so we end up with the term  $\lambda f$ ,  $\lambda g$ , \_, where g has type  $(\beta\to\alpha)\to\beta$  and the placeholder has type  $\alpha\to\gamma$ . A third application of step 1 yields  $\lambda f$ ,  $\lambda g$ ,  $\lambda a$ , \_, where a has type  $\alpha$  and the placeholder has type  $\gamma$ .

At this point, step 1 is no longer possible. Let us see if step 2 is applicable. The context surrounding the placeholder contains the following variables:

$$\textbf{f}:\alpha\rightarrow\beta\rightarrow\gamma$$
 ,  $\textbf{g}:(\beta\rightarrow\alpha)\rightarrow\beta$  ,  $\textbf{a}:\alpha$ 

Recall that we are trying to build a term of type  $\gamma$ . The only variable we can use to achieve this is f: It takes two arguments and returns a value of type  $\gamma$ . So we replace the placeholder with the term  $f_{-}$ , where the two new placeholders stand for the two missing arguments. Putting everything together, we now have the term  $\lambda f$ ,  $\lambda g$ ,  $\lambda a$ ,  $f_{-}$ .

Following f's type, the placeholders are of type  $\alpha$  and  $\beta$ , respectively. The first placeholder is easy to fill, using step 2 again, by

simply supplying a, of type  $\alpha$ , with no arguments. For the second placeholder, we apply step 2 with the variable g, which is the only source of  $\beta$ s. Since g takes an argument, we must supply a placeholder. This means our current term is  $\lambda f$ ,  $\lambda g$ ,  $\lambda a$ , f a  $(g_{-})$ .

We are almost done. The only placeholder left has type  $\beta \to \alpha$ , which is g's argument type. Applying step 1, we replace the placeholder with  $\lambda b$ , \_, where \_ has type  $\alpha$ . Here, we can simply supply a. Our final term is  $\lambda f$ ,  $\lambda g$ ,  $\lambda a$ , f a  $(g(\lambda b, a))$ —i.e.,  $\lambda f$  g a, f a  $(g(\lambda b, a))$ .

The above derivation was tedious but deterministic: At each point, either step 1 or 2 was applicable, but not both. In general, this will not always be the case. For some other types, we might encounter dead ends and need to backtrack. We might also fail altogether, with nowhere to backtrack to. Notably, with an empty local context, it is impossible to supply a witness for  $\alpha$ .

The key idea is that the term should be syntactically correct at all times. The only red underlining we should see in Visual Studio Code should appear under the placeholders. In general, a good principle for software development is to start with a program that compiles, perform the smallest change possible to obtain a new compiling program, and repeat until the program is complete.

## 1.2 Type Definitions

A distinguishing feature of Lean's calculus of inductive constructions is its built-in support for inductive types. An *inductive type* is a type whose values are built by applying special constants called *constructors*. Inductive types are a concise way of representing acyclic data in a program. You may know them under some other, largely synonymous names, including algebraic data types, inductive data types, freely generated data types, recursive data types, and data types.

#### 1.2.1 Natural Numbers

The "Hello, World!" example of inductive types is the type  $\mathbb N$  of natural numbers. In Lean, it can be defined as follows:

```
inductive nat : Type
| zero : nat
| succ : nat → nat
```

The first line announces to the world that we are introducing a new type called nat, intended to represent the natural numbers. The second and third line declare two new constructors, nat.zero : nat and nat.succ : nat  $\rightarrow$  nat, that can be used to build values of type nat. Following an established convention in computer science and logic, counting starts at zero. The second constructor is what makes this inductive definition interesting—it requires an argument of type nat to produce a value of type nat. The terms

```
nat.zero
nat.succ nat.zero
nat.succ (nat.succ nat.zero)
:
```

denote the different values of type nat—zero, its successor, its successor's successor, and so on. This notation is called *unary*, or *Peano*, after the logician Giuseppe Peano. For an alternative explanation of Peano numbers in Lean (and some groovy video game graphics), see Kevin Buzzard's article "Can computers prove theorems?"

The general format of type declarations is

<sup>1</sup>http://chalkdustmagazine.com/features/can-computers-provetheorems/

```
| constructor-name<sub>n</sub> : constructor-type<sub>n</sub>
```

For the natural numbers, it is possible to convince Lean to use the familiar names  $\mathbb{N}$ , 0, 1, 2, ..., and indeed the predefined nat type offers such syntactic sugar. But using the lengthier notations sheds more light on Lean's type definitions. Besides, syntactic sugar is known to cause cancer of the semicolon [29].

In the Lean file accompanying this chapter, the definition of nat is located in a namespace, delimited by namespace my\_nat and end my\_nat, to keep its effects contained to a portion of the file. After end my\_nat, any occurrence of nat, nat.zero, or nat.succ will refer to Lean's predefined type of natural numbers. Similarly, the entire file is located in the LoVe namespace to prevent name clashes with existing Lean libraries.

We can inspect an earlier definition at any point in Lean by using the #print command. For example, #print nat within the my\_nat namespace displays the following information:

```
inductive my_nat.nat : Type
constructors:
my_nat.nat.zero : nat
my_nat.nat.succ : nat → nat
```

The focus on natural numbers is one of the many features of this guide that reveal a bias towards computer science. Number theorists would be more interested in the integers  $\mathbb Z$  and the rational numbers  $\mathbb Q$ ; analysts would want to work with the real numbers  $\mathbb R$  and the complex numbers  $\mathbb C$ . But the natural numbers are ubiquitous in computer science and enjoy a very simple definition as an inductive type. They can also be used to build other types, as we will see in Chapter 13.

#### 1.2.2 Arithmetic Expressions

If we were to specify a calculator program or a programming language, we would likely need to define a type to represent arithmetic expressions. The next example shows how this could be

```
inductive aexp : Type
| num : \mathbb{Z} \to aexp
| var : string \to aexp
| add : aexp \to aexp \to aexp
| sub : aexp \to aexp \to aexp
| mul : aexp \to aexp \to aexp
| div : aexp \to aexp \to aexp
```

Mathematically, this definition is equivalent to defining the type aexp inductively by the following formation rules:

- 1. For every integer i, we have that aexp.num i is an aexp value.
- 2. For every character string x, we have that aexp.var x is an aexp value.
- 3. If  $e_1$  and  $e_2$  are aexp values, then so are aexp.add  $e_1$   $e_2$ , aexp.sub  $e_1$   $e_2$ , aexp.mul  $e_1$   $e_2$ , and aexp.div  $e_1$   $e_2$ .

The above definition is exhaustive. The only possible values for aexp are those built using formation rules 1 to 3. Moreover, aexp values built using different formation rules are distinct. These two properties of inductive types are captured by the motto "No junk, no confusion," due to Joseph Goguen.

#### 1.2.3 Comparison with Java

It may be instructive to compare the concise Lean specification of aexp above with a Java program that achieves the same. The program consists of one interface and six classes that implement it, corresponding to the aexp type and its six constructors:

```
public interface AExp { }
public class Num implements AExp {
   public int num;
```

```
public Num(int num) { this.num = num; }
}
public class Var implements AExp {
    public String var;
    public Var(String var) { this.var = var; }
}
public class Add implements AExp {
    public AExp left;
    public AExp right;
    public Add(AExp left, AExp right)
    { this.left = left; this.right = right; }
}
public class Sub implements AExp {
    public AExp left;
    public AExp right;
    public Sub(AExp left, AExp right)
    { this.left = left; this.right = right; }
}
public class Mul implements AExp {
    public AExp left;
    public AExp right;
    public Mul(AExp left, AExp right)
    { this.left = left; this.right = right; }
}
public class Div implements AExp {
    public AExp left;
```

```
public AExp right;

public Div(AExp left, AExp right)
{ this.left = left; this.right = right; }
}
```

### 1.2.4 Comparison with C

In C, the natural counterpart of an inductive type is a tagged union. The type declarations would be as follows:

```
#include <stddef.h>
#include <stdlib.h>
enum AExpKind {
    AET NUM, AET VAR, AET ADD, AET SUB, AET MUL, AET DIV
};
struct aexp;
struct aexp num {
    int num;
};
struct aexp var {
    char var[1024];
};
struct aexp binop {
    struct aexp *left;
    struct aexp *right;
};
struct aexp {
    enum AExpKind kind;
```

```
union {
    struct aexp_num anum;
    struct aexp_var avar;
    struct aexp_binop abinop;
} data;
};
```

Corresponding to each constructor in Lean, we would need to write a function to allocate an aexp object of the right size in memory. Here is the definition of the function corresponding to the first constructor, aexp.num:

The subtle pointer arithmetic for the malloc call is needed to allocate exactly the right amount of memory.

#### 1.2.5 **Lists**

The next type we consider is that of finite lists:

```
inductive list (\alpha : Type) : Type | nil : list | cons : \alpha \rightarrow list \rightarrow list
```

The type is *polymorphic*: It is parameterized by a type  $\alpha$ , which we can instantiate with concrete types. For example, list  $\mathbb{Z}$  is the type of lists over integers, and list (list  $\mathbb{R}$ ) is the type of lists of lists of real numbers. The type constructor list takes a type as argument and returns a type. Inside the definition, since  $\alpha$  is fixed

as a parameter on the left-hand side of the colon, it is sufficient to write list to get list  $\alpha$ . Polymorphism is related to generics (in Java) and templates (in C++). The general idea in all cases is to have parameterized types.

The following commands display the constructors' types:

```
#check list.nil
#check list.cons
```

#### The output is

```
list.nil : \Pi(\alpha: \mathsf{Type}), list \alpha list.cons : \mathsf{?M}_1 \to \mathsf{list} \; \mathsf{?M}_1 \to \mathsf{list} \; \mathsf{?M}_1
```

#### Informally:

- The nil constructor takes a type  $\alpha$  as argument and produces a result of type list  $\alpha$ . The  $\Pi$  syntax for the type argument will be explained in Chapter 3.
- The cons constructor takes an element (the *head*) of some arbitrary type ?M\_1 as argument and a list over ?M\_1 (the *tail*) and produces a result of type list ?M\_1. Unlike for nil, there is no need to pass a type argument to cons—the type is inferred from the first argument. If we want to pass the type argument explicitly, we need to write an at sign (a) in front of the constant: alist.cons. We then have the type  $\Pi\alpha$ : Type,  $\alpha \to \text{list } \alpha \to \text{list } \alpha$ .

Even if we try to restrict ourselves to a fragment of Lean's language, Lean often exposes us to more advanced constructs in the output, such as  $\Pi$  above, Sort u, or Type 1. Our advice is to adopt a *sporty attitude*: Do not worry if you do not always understand everything the first time. Use your common sense and your imagination. And, above all, do not hesitate to ask.

Lean's built-in lists offer syntactic sugar for writing lists:

```
[] for list.nil x :: xs for list.cons x xs [x_1, ..., x_n] for x_1 :: ... :: x_n :: []
```

The :: operator, like all other binary operators, binds less tightly than function application. Thus, f x :: reverse ys is parsed as (f x) :: (reverse ys). It is good practice to avoid needless parentheses. They can quickly impair readability. In addition, it is important to put spaces around infix operators, to suggest the right precedence; it is all too easy to misread <math>f x :: reverse ys as f (x :: reverse) ys.

Functional programmers often use names such as xs, ys, zs for lists, although l is also common in Lean. A list contains many elements, so a plural form is natural. A list of cats might be called cats; a list of list of cats, catss. When a nonempty list is presented as a head and a tail, we usually write, say, x :: xs or cat :: cats.

## 1.3 Function Definitions

If all we want is to declare a function, we can use the constant command described above (Section 1.1.2). But usually, we want to define the function's behavior, and for this we would use def.

Going back to the arithmetic expression example (Section 1.2), if we wanted to implement an eval function in Java, we would probably add it as part of AExp's interface and implement it in each subclass. For Add, Sub, Mul, and Div, we would recursively call eval on the left and right objects.

In Lean, the syntax is very compact. We define a single function and use *pattern matching* to distinguish the six cases:

```
\begin{array}{lll} \text{def eval (env : string} \rightarrow \mathbb{Z}) : \text{aexp} \rightarrow \mathbb{Z} \\ \mid (\text{aexp.num i}) & := i \\ \mid (\text{aexp.var x}) & := \text{env x} \\ \mid (\text{aexp.add } e_1 \ e_2) := \text{eval } e_1 + \text{eval } e_2 \\ \mid (\text{aexp.sub } e_1 \ e_2) := \text{eval } e_1 - \text{eval } e_2 \\ \mid (\text{aexp.mul } e_1 \ e_2) := \text{eval } e_1 * \text{eval } e_2 \\ \mid (\text{aexp.div } e_1 \ e_2) := \text{eval } e_1 / \text{eval } e_2 \end{array}
```

The keyword def introduces the definition. The general format of definitions by pattern matching is

```
def name (params<sub>1</sub> : type<sub>1</sub>) ... (params<sub>m</sub> : type<sub>m</sub>) : type
| patterns<sub>1</sub> := result<sub>1</sub>
| patterns<sub>n</sub> := result<sub>n</sub>
```

The parentheses () around the parameters can also be curly braces {} if we want to make them implicit arguments. The parameters cannot be subjected to pattern matching, only the remaining arguments (e.g., the aexp argument of eval).

Patterns may contain variables, which are then visible in the corresponding right-hand side, as well as constructors. For example, in the second case in eval's definition, the variable x can be used in the right-hand side env x.

Some definitions do not need pattern matching. For these, the syntax is simply

```
def name (params<sub>1</sub>: type<sub>1</sub>) ... (params<sub>m</sub>: type<sub>m</sub>) : type :=
result
```

We can have pattern matching without recursion (e.g., in the aexp. num and aexp.var cases above), and we can have recursion without pattern matching.

The basic arithmetic operations on natural numbers, such as addition, can be defined by recursion:

```
\begin{array}{lll} \text{def add} & : \; \mathbb{N} \; \to \; \mathbb{N} \; \to \; \mathbb{N} \\ \mid \; \text{m nat.zero} & := \; \text{m} \\ \mid \; \text{m (nat.succ n)} & := \; \text{nat.succ (add m n)} \end{array}
```

We pattern-match on two arguments at the same time, distinguishing the case where the second argument is zero and the case where it is nonzero. Each recursive call to add peels off one nat.succ constructor from the second argument. Instead of nat.zero and nat. succ n, Lean also allows us to write o and n + 1 as syntactic sugar.

We can evaluate the result of applying add to numbers using #eval or #reduce:

```
#eval add 2 7
#reduce add 2 7
```

Both commands print 9, as expected (at least in Visual Studio Code). #eval employs an optimized interpreter, whereas #reduce uses Lean's inference kernel, which is less efficient.

Perhaps you are worried about division by zero in the definition of eval above. Let us see what #eval has to say about it:

```
#eval eval (λs, 7) (aexp.div (aexp.var "x") (aexp.num Θ))
```

The output is o. In Lean, division is conveniently defined as a total function that returns zero when the denominator is zero. For a lucid explanation of why this is not dangerous, see Buzzard's blog.<sup>2</sup>

It is good practice to provide a few tests each time we define a function, to ensure that it behaves as expected. You can even leave the #eval or #reduce calls in your Lean files as documentation.

The definition of multiplication is similar to that of addition:

The underscore (\_) stands for an unused variable. We could have put a name (e.g., m), but \_ documents our intentions better.

The #eval command below prints 14, as expected:

```
#eval mul 2 7
```

The power operation ("m to the power of n") can be defined in various ways. Our first proposal is structurally identical to the definition of multiplication:

Since the first argument, m, remains unchanged in the recursive call, power m n, we can factor it out and put it next to the function's

<sup>2</sup>https://xenaproject.wordpress.com/2020/07/05/division-by-zero-intype-theory-a-fag/

name, as a *parameter*, before the colon introducing the type of the function (excluding the parameter m):

```
\begin{array}{lll} \text{def power}_2 & (\text{m} : \mathbb{N}) : \mathbb{N} \to \mathbb{N} \\ | & \text{nat.zero} & := 1 \\ | & (\text{nat.succ n}) := \text{m} * \text{power}_2 & \text{n} \end{array}
```

Notice that the recursive call to  $power_2$  does *not* take m as argument. In the entire def block introducing  $power_2$ , m is fixed as the first argument. Outside the block,  $power_2$  is a (curried) binary function, just like power. From the outside, there is no difference between the two definitions. In fact, we already saw this syntax for the type argument  $\alpha$  of the list constructor (Section 1.2).

Yet another definition is possible by first introducing a generalpurpose iterator and then using it with the right arguments:

```
def iter (\alpha : \mathsf{Type}) (\mathsf{z} : \alpha) (\mathsf{f} : \alpha \to \alpha) : \mathbb{N} \to \alpha | nat.zero := z | (nat.succ n) := f (iter n) | def power<sub>3</sub> (m n : \mathbb{N}) : \mathbb{N} := iter \mathbb{N} 1 (\lambda \mathsf{l}, m * \mathsf{l}) n
```

Notice that the second definition is not recursive.

Recursive functions on lists can be defined in a similar way:

The append function takes three arguments: a type  $\alpha$  and two lists of type list  $\alpha$ . By passing the placeholder \_, we leave it to Lean to infer the type  $\alpha$  from the type of the other two arguments.

To make the type argument  $\alpha$  implicit, we can put it in curly braces  $\{ \}$ :

```
\mathsf{def} \ \mathsf{append_2} \ \{\alpha \ \colon \mathsf{Type}\} \ \colon \mathsf{list} \ \alpha \ \to \ \mathsf{list} \ \alpha \ \to \ \mathsf{list} \ \alpha
```

The at sign (a) can be used to make the implicit arguments explicit. This is occasionally necessary to guide Lean's parser.

We can use syntactic sugar in the definition, both in the patterns on the left-hand sides of := and in the right-hand sides:

```
\begin{array}{lll} \mathsf{def} \ \mathsf{append_3} \ \{\alpha \ : \ \mathsf{Type}\} \ : \ \mathsf{list} \ \alpha \ \to \ \mathsf{list} \ \alpha \ \to \ \mathsf{list} \ \alpha \\ | \ [] & \mathsf{ys} \ := \ \mathsf{ys} \\ | \ (\mathsf{x} \ :: \ \mathsf{xs}) \ \mathsf{ys} \ := \ \mathsf{x} \ :: \ \mathsf{append_3} \ \mathsf{xs} \ \mathsf{ys} \end{array}
```

In Lean's standard library, the append function is an infix operator called ++. We can use it to define a function that reverses a list:

#### 1.4 Lemma Statements

What makes Lean a proof assistant and not only a programming language is that we can state lemmas about the types and constants we define and prove that they hold. We will use the terms lemma, theorem, corollary, fact, property, and true statement more or less interchangeably. Similarly, propositions, logical formulas, and statements will all mean the same.

In Lean, propositions are simply terms of type Prop. This stands in contrast with most descriptions of first-order logic, where terms and formulas are separate syntactic entities. A proposition that

can be proved is called a theorem (or lemma, corollary, etc.); otherwise it is a nontheorem or false statement. Mathematicians sometimes use the term "proposition" as a synonym for theorem (e.g., "Proposition 3.14"), but in formal logic propositions can also be false.

Here are examples of true statements that can be made about the addition, multiplication, and list reversal operations defined in Section 1.3:

```
lemma add comm (m n : \mathbb{N}):
  add m n = add n m :=
sorrv
lemma add assoc (l m n : \mathbb{N}):
  add (add l m) n = add l (add <math>m n) :=
sorry
lemma mul comm (m n : \mathbb{N}):
  mul m n = mul n m :=
sorry
lemma mul assoc (l m n : \mathbb{N}):
  mul (mul l m) n = mul l (mul m n) :=
sorry
lemma mul_add (l m n : N) :
  mul l (add m n) = add (mul l m) (mul l n) :=
sorry
lemma reverse reverse \{\alpha : \mathsf{Type}\}\ (\mathsf{xs} : \mathsf{list}\ \alpha) :
  reverse (reverse xs) = xs :=
sorry
   The general format is
      lemma name (params<sub>1</sub>: type<sub>1</sub>) ... (params<sub>m</sub>: type<sub>m</sub>):
         statement :=
      proof
```

The := symbol separates the lemma's statement and its proof. The syntax of lemma is very similar to that of a def command without pattern matching, with statement instead of type and proof instead of result. In the examples above, we put the marker sorry as a placeholder for the actual proof. The marker is quite literally an apology to future readers and to Lean for the absence of a proof. It is also a reason to worry, until we manage to eliminate it. In Chapters 2 and 3, we will see how to achieve this.

The intuitive semantic of a lemma command with a sorry proof is, "This proposition should be provable, but I have not carried out the proof yet—sorry." Sometimes, we want to express a related idea, namely, "Let us assume this proposition holds." Lean provides the axiom command for this, which is often used in conjunction with constant(s). For example:

```
constants a b : Z
axiom a_less_b :
  a < b</pre>
```

After the constants command, we have no information about a and b beyond their type. The axiom specifies the desired property about them. The general format of the command is

```
axiom name (params<sub>1</sub>: type<sub>1</sub>) ... (params<sub>m</sub>: type<sub>m</sub>):
    statement
```

Axioms are dangerous, because they rapidly can lead to an inconsistent logic, in which we can derive false. For example, if we added a second axiom stating that a = b, we could easily derive b < b, from which we could derive false. The history of interactive theorem proving is paved with inconsistent axioms. An anecdote among many: At the 2020 edition of the Certified Programs and Proofs (CPP) conference, a submitted paper was rejected due to a flawed axiom, from which one of the peer reviewers derived false

.3 Therefore, we will generally avoid axioms, preferring the benign combination of def and lemma to the dangerous combination of constant(s) and axiom.

From Lean's point of view, a lemma with a sorry proof is effectively an axiom and can be used to play havoc with the logic's consistency. To prevent misunderstandings, it is better to use sorry only as a temporary measure, while developing a proof, and not as an alternative to the more explicit, and honest, axiom.

## **1.5 Summary of New Lean Constructs**

At the end of this and most other chapters, we include a brief summary of the constructs introduced in the chapter. Some syntaxes have multiple meanings, which will be introduced gradually. We refer to *The Lean Reference Manual* [3], the *Theorem Proving in Lean* [1] tutorial, and the mathlib documentation<sup>4</sup> for details.

#### **Diagnosis Commands**

#check checks and prints type of a term
#eval executes a term using an optimized interpreter
#print prints the definition of a constant

#reduce executes a term using Lean's inference kernel

#### **Declarations**

axiom states an axiom constant(s) declares unspecified new constants or types

<sup>&</sup>lt;sup>3</sup>One of the authors of that submission reported that the axiom has now been revised and derived as a theorem. The entire formal development is now axiom-free. The revised paper has been accepted at the conference Computer Aided Verification (CAV) 2020 [10].

<sup>4</sup>https://leanprover-community.github.io/mathlib\_docs/

defines a new constant

inductive introduces a type and its constructors

lemma states a lemma and its proof

namespace ... end collects declarations in a named scope

#### **Proof Commands**

stands for a missing proof or definition

## **Chapter 2**

## **Backward Proofs**

In this chapter, we see how to prove Lean lemmas using tactics, and we review the most important Lean tactics. A tactic is a procedure that operates on the goal—the proposition to prove—and either fully proves it or produces new subgoals (or fails). When we state a lemma, the lemma statement is the initial goal. A proof is complete once all (sub)goals have been eliminated using suitable tactics. The tactics described here are documented in more detail in Chapter 5 of Theorem Proving in Lean [1] and Chapter 6 of The Lean Reference Manual [3].

Tactics are a *backward* proof mechanism. They start from the goal and work backwards towards the already proved lemmas. Consider the lemmas a,  $a \to b$ , and  $b \to c$  and the goal  $\vdash c$ . An informal backward proof is as follows:

To prove c, by b  $\to$  c it suffices to prove b. To prove b, by a  $\to$  b it suffices to prove a. To prove a, we use a.

The telltale sign of a backward proof is the phrase "it suffices to." Notice how we progress from one *goal* to another ( $\vdash$  c,  $\vdash$  b,  $\vdash$  a) until no goal is left to prove. By contrast, a *forward* proof would

start from the lemma a and progress, one *theorem* at a time, towards the desired theorem c:

From a and a 
$$\rightarrow$$
 b, we have b. From b and b  $\rightarrow$  c, we have c, as desired.  $\Box$ 

A forward proof only manipulates theorems, not goals. We will study forward proofs more deeply in Chapter 3.

Informal proofs are sometimes written in a mixture of both styles. This is manageable as long as the backward steps are clearly identified, by such wording as "to prove ..., it suffices to prove ...." It may help to put the  $\vdash$  sign next to goals as a reminder that these claims have not been proved yet.

Another format for expressing proofs you may be familiar with is natural deduction. A natural deduction derivation corresponding to the above proof is given below:

When read from top to bottom, the derivation corresponds to a forward proof; when read from bottom to top, it corresponds to a backward proof. We will not use this notation in the rest of this guide.

#### 2.1 Tactic Mode

In Chapter 1, whenever a proof was required, we simply put a sorry placeholder. For a tactical proof, we will now write begin and end to enter and leave *tactic mode*. In this mode, we can invoke a sequence of tactics, separated by commas.

Tactics operate on the goal, which consists of the proposition Q that we want to prove and of a local context C. The local context

2.1. Tactic Mode 31

consists of variable declarations of the form  $c:\tau$  and hypotheses of the form h:P. We write  $C\vdash Q$  to denote a goal, where C is a list of variables and hypotheses and Q is the goal's target.

To make things more concrete, consider the following Lean example:

```
lemma fst_of_two_props : \forall a \ b : Prop, \ a \rightarrow b \rightarrow a := begin intros a b, intros ha hb, apply ha end
```

Note that the implication arrow  $\rightarrow$  is right-associative; this means that  $a \rightarrow b \rightarrow a$  is the same as  $a \rightarrow (b \rightarrow a)$ . Intuitively speaking, it has the meaning "a implies that b implies a," or equivalently "a and b imply a." In the example, three tactics are invoked, each on its own line. Let us trace their behavior:

1. Initially, the goal is simply the lemma statement:

```
\vdash \forall a \ b : Prop, \ a \rightarrow b \rightarrow a
```

2. The intros a b tactic tells Lean to fix two free variables, a and b, corresponding to the two bound variables of the same names. The tactic mimics how mathematicians work on paper: To prove a ∀-quantified proposition, it suffices to prove it for some arbitrary but fixed value of the bound variable. The goal becomes

```
ab : Prop \vdash a \rightarrow b \rightarrow a
```

We often name the free variables after the bound variables, as we did here, but this is not mandatory. Moreover, it is good practice to use unique names and avoid shadowing existing variables.

3. The intros ha hb tactic tells Lean to move the assumptions a and b to the local context and to call these hypotheses ha

and hb. Indeed, to prove an implication, it suffices to take its left-hand side as hypothesis and prove its right-hand side. The goal becomes

```
ab: Prop, ha: a, hb: b \vdash a
```

It is customary to prefix hypothesis names with h.

4. The apply ha tactic tells Lean to match the hypothesis a, called ha, against the goal ⊢ a. Since a is syntactically equal to a, we have a match, and this completes the proof.

Informally, in a style reminiscent of pen-and-paper mathematics, we could write the proof as follows:

Let a and b be propositions.

Assume (ha) a and (hb) b are true.

To prove a, we use hypothesis ha.

(Mathematicians would probably use numeric tags such as (1) and (2) for the hypotheses instead of informative names.)

Going back to the Lean proof, we can avoid the intros invocations by declaring the variables and hypotheses as parameters of the lemma, as follows:

```
lemma fst_of_two_props2 (a b : Prop) (ha : a) (hb : b) :
  a :=
begin
  apply ha
end
```

If the proof consists of a single tactic invocation, we can use by instead. The syntax by tactic abbreviates begin tactic end:

```
lemma fst_of_two_props3 (a b : Prop) (ha : a) (hb : b) :
  a :=
by apply ha
```

Here is an example with multiple applys in sequence:

2.2. Basic Tactics 33

```
lemma prop_comp (a b c : Prop) (hab : a \rightarrow b) (hbc : b \rightarrow c) : a \rightarrow c := begin intro ha, apply hbc, apply hab, apply ha end
```

Putting on our mathematician's hat, we can verbalize the last proof as follows:

П

```
Assume (ha) a is true.
```

To prove c, by hypothesis hbc it suffices to prove b.

To prove b, by hypothesis hab it suffices to prove a.

To prove a, we use hypothesis ha.

#### 2.2 Basic Tactics

We already saw the intros and apply tactics. These are the staples of tactical proofs. Other basic tactics include refl, exact, assumption, and cc. These tactics can go a long way, if we are patient enough to carry out the reasoning using them, without appealing to stronger proof automation. They can also be used to solve various logic puzzles.

Below, the thin, large square brackets [] enclose optional syntax.

## intro(s)

```
intro\ [name]
intros\ [name_1...name_n]
```

The intro tactic moves the leading  $\forall$ -quantified variable or the leading assumption  $a \rightarrow$  from the goal's target to the local context.

The tactic takes as an optional argument the name to give to the variable or to the assumption in the context, overriding the default name. The plural variant intros can be used to move several variables or assumptions at once.

#### refl

The refl tactic proves goals of the form  $\vdash l = r$ , where the two sides l and r are syntactically equal up to computation. Computation means expansion of definitions, reduction of an application of a  $\lambda$ -expression to an argument, and more. These conversions have traditional names. The main conversions are listed below together with examples, in a global context containing def double  $(n : \mathbb{N})$ :  $\mathbb{N} := n + n$ :

```
α-conversion (\lambda x, f x) = (\lambda y, f y)

β-conversion (\lambda x, f x) a = f a

δ-conversion double 5 = 5 + 5

ζ-conversion (\text{let } n : \mathbb{N} := 2 \text{ in } n + n) = 4

η-conversion (\lambda x, f x) = f

ι-conversion prod.fst (a, b) = a
```

Applying conversions repeatedly as left-to-right rewrite rules is called *reduction*; applying a conversion once in reverse is called *expansion*. Since much of Lean's machinery treats terms that are syntactically equal up to computation uniformly, it usually makes sense to tell Lean's pretty-printer to aggressively  $\beta$ -reduce its output. This is what the the command

```
set_option pp.beta true
```

near the top of the files accompanying this guide achieves.

#### apply

```
apply lemma-or-hypothesis
```

2.2. Basic Tactics 35

The apply tactic matches the goal's target with the conclusion of the specified lemma or hypothesis and adds the lemma or hypothesis's assumptions as new goals. The matching is performed up to computation.

We must invoke apply with care, because it can transform a provable goal into an unprovable subgoal. For example, if the goal is  $\vdash$  2 + 2 = 4 and we apply the lemma false  $\rightarrow$  ?a, the variable ?a is matched against 2 + 2 = 4, and we end up with the unprovable subgoal  $\vdash$  false. We say that apply is *unsafe*. In contrast, intro always preserves provability and is therefore *safe*.

#### exact

exact lemma-or-hypothesis

The exact tactic matches the goal's target with the specified lemma or hypothesis, closing the goal. We can often use apply in such situations, but exact communicates our intentions better. In the example from Section 2.1, we could have used exact ha instead of apply ha.

#### assumption

The assumption tactic finds a hypothesis from the local context that matches the goal's target and applys it to prove the goal. In the example from Section 2.1, we could have used assumption instead of apply ha.

#### CC

The cc tactic implements an algorithm known as *congruence closure* [33] to derive new equalities from those that are present in the goal. For example, if the goal contains b = a and f  $b \neq f$  a as hypotheses, the algorithm will derive f b = f a from b = a and discover a contradiction with f  $b \neq f$  a. The cc tactic is also suitable

for more pedestrian work, such as proving  $hb: b \vdash a \lor b \lor c$  or discovering a contradiction among the goal's hypotheses.

In addition, cc can be used to reason up to associativity (e.g., (a + b) + c = a + (b + c)) and commutativity (e.g, a + b = b + a). This works for binary operations that are registered as associative and commutative, such as + and \* on arithmetic types, and  $\cup$  and  $\cap$  on sets. We will see an example in Section 2.6.

At this point, you might wonder, "So what does cc do exactly?" Of course, you could look up the reference [33] given above to the scientific paper that describes its underlying algorithm, or even read the source code. But this might not be the most efficient use of your time. In truth, even expert users of proof assistants do not fully understand the behavior of the tactics they use daily. The most successful users adopt a relaxed, sporty attitude, trying tactics in sequence and studying the emerging subgoals, if any, to see if they are on the right path.

As you keep on using cc and other tactics, you will develop some intuition about what kinds of goal they work well on. This is one of the many reasons why interactive theorem proving can only be learned by doing. Often, you will not understand exactly what Lean does—why a tactic succeeds, or fails. Theorem proving can be very frustrating at times. The advice printed in large, friendly letters on the cover of *The Hitchhiker's Guide to the Galaxy* applies here: DON'T PANIC.

#### sorry

The sorry proof command we encountered in Chapter 1 is also available as a tactic. It "proves" the current goal without actually proving it. Use with care.

# 2.3 Reasoning about Connectives and Quantifiers

Before we learn to reason about natural numbers, lists, or other data types, we must first learn to reason about the logical connectives and quantifiers of Lean's logic. Let us start with a simple example: commutativity of conjunction  $(\land)$ .

```
lemma and_swap (a b : Prop) :
    a ∧ b → b ∧ a :=
begin
    intro hab,
    apply and.intro,
    apply and.elim_right,
    exact hab,
    apply and.elim_left,
    exact hab
```

At this point, we recommend that you move the cursor over the example in Visual Studio Code, to see the sequence of proof states. By putting the cursor immediately after each comma, you can see the effect of the command on that line. If you want to see the proof state after the last command, you need to move the cursor right before the e of the keyword end. Lean simply reports "goals accomplished," meaning that no subgoals remain to be proved.

The proof is a typical intro-apply-exact mixture. It uses the lemmas

```
and.intro: ?a \rightarrow ?b \rightarrow ?a \land ?b
and.elim_left: ?a \land ?b \rightarrow ?a
and.elim_right: ?a \land ?b \rightarrow ?b
```

where the question marks (?) indicate variables that can be instantiated—for example, by matching the goal's target against the conclusion of a lemma.

The three lemmas above are the introduction rule and the two elimination rules associated with conjunction. An *introduction rule* for a logical symbol (e.g.,  $\land$ ) is a lemma whose conclusion has that symbol as the outermost symbol. Dually, an *elimination rule* has the symbol in an assumption. For each logical symbol, the introduction rules tell us what we need to provide in order to justify a proposition with that symbol as the outermost position, whereas the elimination rules tell us what we may infer from such a proposition.

In the above proof, we apply the introduction rule associated with  $\land$  to prove the goal  $\vdash$  b  $\land$  a, and we apply the two elimination rules to extract b and a from the hypothesis a  $\land$  b.

Question marks can also arise in goals. They indicate variables that can be instantiated arbitrarily. In the middle of the proof above, right after the tactic apply and.elim\_right, we have the goal

$$ab : Prop, hab : a \land b \vdash ?m_1 \land b$$

The tactic exact hab matches ?m\_1 (in the target) with a (in hab). Because variables can occur both in the hypothesis or lemma that is applyd and in the target, in general the procedure used to instantiate variables to make two terms syntactically equal is called unification. Matching is a special case of unification where one of the two terms contains no variables. In practice, goals with variables are rare, so Lean's unification usually amounts to matching.

(Incidentally, whenever ?m\_ variables appear in goals, additional subgoals will also appear, with the type of each ?m\_ variable as a subgoal (e.g., ⊢ Prop). These confusing subgoals are merely a reminder that we will have to instantiate the ?m\_ variables with a term of the right type. We can usually ignore them. As soon as the variable is instantiated, typically when we solve another subgoal, its associated subgoal also goes away.)

In Lean, unification is performed up to computation. For example, the terms  $(\lambda x, ?m)$  a and b can be unified by taking ?m := b, because  $(\lambda x, b)$  a and b are syntactically equal up to  $\beta$ -conversion.

The following is an alternative proof of the lemma and\_swap:

```
lemma and_swap2 :
    ∀a b : Prop, a ∧ b → b ∧ a :=
begin
    intros a b hab,
    apply and.intro,
    { exact and.elim_right hab },
    { exact and.elim_left hab }
end
```

The lemma is stated differently, with a and b as  $\forall$ -quantified variables instead of parameters of the lemma. Logically, this is equivalent, but in the proof we must then introduce a and b in addition to hab.

Another difference is the use of curly braces  $\{ \}$ . When we face two or more goals to prove, it is generally good style to put each proof in its own block enclosed in curly braces. The  $\{ \}$  tactic combinator focuses on the first subgoal; the tactic inside must prove it. In our example, the apply and intro tactic creates two subgoals,  $\vdash$  b and  $\vdash$  a.

A third difference is that we now apply, by juxtaposition, and. elim\_right and and.elim\_left directly to the hypothesis a  $\land$  b to obtain b and a, respectively, instead of waiting for the lemmas' assumptions to emerge as new subgoals. This is a small forward step in an otherwise backward proof. The same syntax is used both to discharge (i.e., prove) a hypothesis and to instantiate a  $\forall$ -quantifier. One benefit of this approach is that we avoid the potentially confusing ?m\_1 variable.

The introduction and elimination rules for disjunction ( $\lor$ ) are as follows:

```
or.intro_left : \forall b : Prop, ?a \rightarrow ?a \vee b or.intro_right : \forall b : Prop, ?a \rightarrow b \vee ?a or.elim : ?a \vee ?b \rightarrow (?a \rightarrow ?c) \rightarrow (?b \rightarrow ?c) \rightarrow ?c
```

The  $\forall$ -quantifiers in or.intro\_left and or.intro\_right can be instantiated directly by applying the lemma name to the value we want to instantiate with, via simple juxtaposition. Thus, or.

intro\_left false corresponds to the lemma ?a  $\rightarrow$  ?a  $\vee$  false. This is the forward style.

Alternatively, we can invoke apply or.intro\_left on a goal of the form ...  $\vdash$  c  $\lor$  d. This instantiates the lemma with ?a := c and b := d. The new subgoal is ...  $\vdash$  c. This is the backward style.

Both or.intro\_left and or.intro\_right are unsafe: If you apply the wrong one of the two, or either of them too early in a proof, you might end up with an unprovable subgoal. This is easy to see if you consider the provable goal  $\vdash$  true  $\lor$  false: applying or.intro right yields the unprovable subgoal  $\vdash$  false.

The or.elim rule may seem counterintuitive at a first glance. In essence, it states that if we have  $a \lor b$ , then to prove an arbitrary c, it suffices to prove c when a holds and when b holds. You can think of  $(?a \to ?c) \to (?b \to ?c) \to ?c$  as a clever trick to express the concept of disjunction using only implication.

The introduction and elimination rules for equivalence  $(\leftrightarrow)$  are as follows:

```
\begin{array}{c} \text{iff.intro}: (?a \rightarrow ?b) \rightarrow (?b \rightarrow ?a) \rightarrow (?a \leftrightarrow ?b) \\ \text{iff.elim\_left}: (?a \leftrightarrow ?b) \rightarrow ?a \rightarrow ?b \\ \text{iff.elim\_right}: (?a \leftrightarrow ?b) \rightarrow ?b \rightarrow ?a \end{array}
```

The introduction and elimination rules for existential quantification  $(\exists)$  are

```
exists.intro: \forall w, (?p w \rightarrow (\exists x, ?p x))
exists.elim: (\exists x, ?p x) \rightarrow (\forall a, ?p a \rightarrow ?c) \rightarrow ?c
```

The introduction rule for  $\exists$  can be used to instantiate an existential quantifier with a witness. For example:

```
lemma nat_exists_double_iden :
    ∃n : N, double n = n :=
begin
    apply exists.intro ⊙,
    refl
end
```

Again, we instantiate a  $\forall$ -quantifier in a forward fashion: exists. intro  $\odot$  is the lemma ?p  $\odot \to (\exists x, ?p \ x)$ . The rule is unsafe: Choosing the wrong witness for x will result in an unprovable goal. For example, if the goal is  $\vdash \exists n, n > 5$  and we take 3 as the witness, we end up with the unprovable subgoal  $\vdash 3 > 5$ .

The elimination rule for  $\exists$  is reminiscent of  $\lor$ . Indeed, a fruitful way to think of a quantification  $\exists n$ , ?p n is as a possibly infinitary disjunction  $?p \circ \lor ?p \ 1 \lor \cdots$ . Similarly,  $\forall n$ , ?p n can be thought of as  $?p \circ \land ?p \ 1 \land \cdots$ .

For truth (true), there is only an introduction rule:

```
true.intro: true
```

Truth holds no information whatsoever. If it appears as a hypothesis, it is completely useless, and there is no elimination rule that will succeed at extracting any information from it. The clear tactic, described in Section 2.8 below, can be used to remove such useless hypotheses.

Dually, for falsehood (false), there is only an elimination rule:

```
\texttt{false.elim:false} \rightarrow \texttt{?a}
```

There is no way to prove falsehood, but if we somehow have it from somewhere (e.g., from a hypothesis), then we can derive ?anything.

In fact, negation (not) is defined in terms of implication and falsehood:  $\neg$  a abbreviates  $a \rightarrow false$ . Intuitively, these mean the same. We can think of "not a" as saying that "a would imply something absurd (hence not a)."

Lean's logic is classical, with support for the law of excluded middle and proof by contradiction:

```
classical.em : \forall a: Prop, a \lor \neg a classical.by_contradiction : (\neg ?a \to false) \to ?a
```

Finally, implication  $(\rightarrow)$  and universal quantification  $(\forall)$  are the proverbial dogs that did not bark. For both of them, the intro tactic is the introduction principle, and application is the elimination

principle. For example, given the lemmas hab:  $a \to b$  and ha: a, the juxtaposition hab ha is a lemma stating b.

For proving logic puzzles involving connectives and quantifiers, we advocate a "mindless," "video game" style of reasoning that relies mostly on basic tactics such as intro(s) and apply. Here are some strategies that often work:

- If the goal's target is an implication  $P \rightarrow Q$ , invoke intro hP to move P into your hypotheses: ..., hP :  $P \vdash Q$ .
- If the goal's target is a universal quantification  $\forall x : \sigma$ , Q, invoke intro x to move x into the local context: ...,  $x : \sigma \vdash Q$ .
- Look for a lemma or hypothesis whose conclusion has the same shape as the goal's target (possibly containing variables that can be matched), and apply it. For example, if the goal's target is Q and you have a lemma or hypothesis of the form hPQ: P → Q, try apply hPQ.
- A negated goal ⊢ ¬ P is syntactically equal to ⊢ P → false up to computation, so you can invoke intro hP to produce the subgoal hP : P ⊢ false. Expanding negation's definition by invoking rw not\_def (described in Section 2.5) is often a good strategy.
- Sometimes you can make progress by replacing the goal by false, by entering apply false.elim. As next step, you would typically apply a lemma or hypothesis of the form P → false or ¬ P.
- When you face several choices (e.g., between or.intro\_left and or.intro\_right), remember which choices you have made, and backtrack when you reach a dead end or have the impression you are not making any progress.
- If you suspect that you might have reached a dead end, check whether the goal actually is provable under the given assumptions. Even if you started with a provable lemma statement, the current goal might be unprovable (e.g., if you used unsafe rules such as or.intro\_left).

## 2.4 Reasoning about Equality

Equality (=) is also a basic logical constant. It is characterized by the following introduction and elimination rules:

```
eq.refl: \forall a, a = a
eq.symm: ?a = ?b \rightarrow ?b = ?a
eq.trans: ?a = ?b \rightarrow ?b = ?c \rightarrow ?a = ?c
eq.subst: ?a = ?b \rightarrow ?p ?a \rightarrow ?p ?b
```

The first three lemmas are introduction rules specifying that = is an equivalence relation. The fourth lemma is an elimination rule that allows us to replace equals for equals in an arbitrary context, represented by the higher-order variable ?p.

An example will show how this works. Below, we apply eq.subst to rewrite f a b to f a' b, using the equation a = a':

```
lemma cong_fst_arg \{\alpha: \mathsf{Type}\}\ (a\ a'\ b:\alpha) (f:\alpha\to\alpha\to\alpha)\ (ha:a=a'): f a\ b=f\ a'\ b:= begin apply eq.subst ha, apply eq.refl end
```

The eq.subst instance we use has ?a := a, ?b := a', and  $?p := (\lambda x, f a b = f x b)$ :

```
a = a' \rightarrow (\lambda x, f a b = f x b) a \rightarrow (\lambda x, f a b = f x b) a'
```

In β-reduced form:

```
a = a' \rightarrow f a b = f a b \rightarrow f a b = f a' b
```

The lemma's first assumption matches the hypothesis ha: a = a', which is passed as argument in the first apply invocation. The lemma's second assumption is a trivial equality that can be proved by apply eq.refl or refl. The lemma's conclusion matches the

goal's target. Notice how a higher-order variable (e.g., ?p) can represent an arbitrary context (e.g.,  $f ext{ ... b}$ ) around a term (e.g., a or a'). This works because apply unifies up to computation, including  $\beta$ -conversion.

The eq.subst lemma can be used several times in sequence, as follows:

```
lemma cong_two_args {\alpha : Type} (a a' b b' : \alpha) (f : \alpha \to \alpha \to \alpha) (ha : a = a') (hb : b = b') : f a b = f a' b' := begin apply eq.subst ha, apply eq.subst hb, apply eq.refl end
```

Since rewriting in this way is such a common operation, Lean provides a rw tactic to achieve the same result. The tactic will also notice if refl is applicable:

A note on parsing: Equality binds more tightly than the logical connectives. Thus,  $a = a' \wedge b = b'$  should be read as  $(a = a') \wedge (b = b')$ .

## 2.5 Rewriting Tactics

The rewriting tactic rw and its relative simp replace equals for equals. Unlike cc, they use equations as left-to-right rewrite rules. By default, they operate on the goal's target, but they can also be used to rewrite hypotheses specified using the at keyword:

```
\begin{array}{lll} \text{at } h_1 & \text{rewrites the target (the default)} \\ \text{at } h_1 & \text{...} & h_n \\ \text{at } * & \text{rewrites the specified hypotheses} \\ \end{array}
```

#### rw

```
rw lemma-or-constant [at position]
```

The rw tactic rewrites the goal using a single equation as a left-to-right rewrite rule. It searches for the first subterm that matches the equation's left-hand side; once found, all occurrences of that subterm are replaced by the right-hand side of the equation. If the equation contains variables, these are instantiated as necessary. To use a lemma as a right-to-left rewrite rule, put a short left arrow  $(\leftarrow)$  in front of the lemma's name.

Given the lemma  $l : \forall x, g x = f x$  and the goal  $\vdash h (f a) (g b) (g c)$ , the tactic rwl produces the subgoal  $\vdash h (f a) (f b) (g c)$ , whereas rw  $\leftarrow l$  produces the subgoal  $\vdash h (g a) (g b) (g c)$ .

Instead of a lemma, we can also specify the name of a constant. This will attempt to use one of the constant's defining equations as rewrite rules. For technical reasons, this does not work with not  $(\neg)$ , and we must use rw not\_def instead.

#### simp

```
simp [at position]
```

The simp tactic rewrites the goal using a standard set of rewrite rules, called the *simp set*, exhaustively. The simp set can be extended by putting the  $\Im[\text{simp}]$  attribute on lemmas. Unlike rw, simp can rewrite terms containing bound variables (e.g., occurrences of x in the body of  $\lambda x$ , ...,  $\forall x$ , ..., or  $\exists x$ , ...).

```
\begin{array}{c} \text{simp [lemma-or-constant}_1, \, \dots, \, \text{lemma-or-constant}_n] \\ & \left[ \text{at position} \right] \end{array}
```

For the above simp variant, the specified lemmas are temporarily added to the simp set. In the lemma list, an asterisk (\*) can be use to represent all hypotheses. The minus sign (-) in front of a lemma name temporarily removes the lemma from the simp set. A powerful incantation that both simplifies the hypotheses and uses the result to simplify the goal's target is simp [\*] at \*.

Given the lemma  $l: \forall x, g \ x = f \ x$  and the goal  $\vdash h \ (f \ a) \ (g \ b)$   $(g \ c)$ , the tactic simp [l] produces the subgoal  $\vdash h \ (f \ a) \ (f \ b)$   $(f \ c)$ , where both  $g \ b$  and  $g \ c$  have been rewritten. Instead of a lemma, we can also specify the name of a constant. This temporarily adds the constant's defining equations to the simp set.

To find out what simp does, you can enable tracing via the command

```
set_option trace.simplify.rewrite true
```

## 2.6 Proofs by Mathematical Induction

The induction' tactic performs structural induction on a value of an inductive type [21]. Structural induction means that the induction follows the structure of the inductive type. For natural numbers constructed from nat.zero and nat.succ, structural induction corresponds to standard mathematical induction: To prove p n, it suffices to prove p o and  $\forall k$ , p k  $\rightarrow$  p (k + 1). Equipped with induction', we can reason about the addition and multiplication operations we defined by recursion in Section 1.3.

Addition is defined by recursion on its second argument. We will prove two lemmas, add\_zero and add\_succ, that give us alternative equations that recurse on the *first* argument. We start with add\_zero:

```
lemma add_zero (n : N) :
  add o n = n :=
begin
  induction' n,
```

```
{ refl },
  { simp [add, ih] }
end
```

The first  $\{\ \}$  encloses the base case  $\vdash$  add  $\circ \circ = \circ$ . The second block corresponds to the induction step

```
n : \mathbb{N}, ih : add o n = n \vdash add o (nat.succ n) = (nat.succ n)
```

The local variable n in the step should not be confused with the n in the lemma statement. Like mathematicians, induction' tries to reuse variable names that are no longer needed. The name ih for the induction hypothesis, in the induction step, is also generated by the induction' tactic.

We can keep on proving lemmas by structural induction:

```
lemma add succ (m n : \mathbb{N}):
  add (nat.succ m) n = nat.succ (add m n) :=
begin
  induction' n.
  { refl },
  { simp [add, ih] }
end
lemma add comm (m n : \mathbb{N}):
  add m n = add n m :=
begin
  induction' n,
  { simp [add, add_zero] },
  { simp [add, add_succ, ih] }
end
lemma add_assoc (l m n : \mathbb{N}):
  add (add l m) n = add l (add <math>m n) :=
begin
  induction' n.
  { refl }.
  { simp [add, ih] }
end
```

Once we have proved that a binary operator is commutative and associative, it is a good idea to let Lean's automation, notably cc, know about this. The following commands achieve this for add:

```
 \begin{tabular}{ll} @[instance] & def & add.is_commutative : is_commutative $\mathbb{N}$ & add := $\{ comm := add_comm $\} $ \\ @[instance] & def & add.is_associative : is_associative $\mathbb{N}$ & add := $\{ assoc := add_assoc $\} $ \\ \end{tabular}
```

(The @[instance] mechanism will be explained in Chapter 4.) The following example uses the cc tactic to reason up to associativity and commutativity of add:

```
lemma mul_add (l m n : N) :
  mul l (add m n) = add (mul l m) (mul l n) :=
begin
  induction' n,
  { refl },
  { simp [add, mul, ih],
     cc }
end
```

Here are a few hints on how to carry out proofs by induction:

- It is usually beneficial to perform induction following the structure of the definition of one of the functions appearing in the goal. In particular, if a function is defined by recursion on its nth argument, it usually makes sense to perform the induction on that argument.
- If the base case of an induction is difficult, this is often a sign that the wrong variable was chosen or that some lemmas should be proved first.

## 2.7 Induction Tactic

#### induction'

```
induction' term [with name_1 ... name_n]
```

The induction' tactic performs structural induction on the specified term. This gives rise to as many subgoals as there are constructors in the definition of the term's type. Induction hypotheses are available as hypotheses in the subgoals corresponding to recursive constructors (e.g., nat.succ or list.cons). The optional names name<sub>1</sub>, ..., name<sub>n</sub> are used for any emerging variables or hypotheses.

Note that the standard Lean tactic for structural induction is called induction. The primed variant is provided by mathlib. It is more user-friendly and is the only one we will use.

## 2.8 Cleanup Tactics

The following tactics help us clean up the goal by allowing us to give more meaningful names to variables or hypotheses or to remove useless hypotheses. We did not need them so far, but they can be helpful during proof exploration.

#### rename

```
rename variable-or-hypothesis new-name
```

The rename tactic changes the name of a variable or hypothesis.

#### clear

```
clear variable-or-hypothesis<sub>1</sub>...
variable-or-hypothesis<sub>n</sub>
```

The clear tactic removes the specified variables and hypotheses, as long as they are not used anywhere else in the goal.

## 2.9 Summary of New Lean Constructs

#### **Command**

set\_option changes or activates tracing, syntax output, etc.

#### **Attribute**

all adds a lemma to the simp set

#### **Proof Commands**

begin ... end invokes a list of tactics in sequence

by invokes a single tactic

#### **Tactics**

apply matches the goal's target with the lemma's conclusion

assumption proves the goal using a hypothesis

cc propagates equalities up to to associativity and commuta-

tivity

clear removes a variable or hypothesis from the goal exact proves the goal using the specified lemma

induction' performs structural induction on a variable of an induc-

tive type

intro(s) moves  $\forall$ -quantified variables into the goal's hypotheses proves l = r where l and r are equal up to computation

rename renames a variable or hypothesis

rewrite rewrites once using the given lemma as a left-to-right

rewrite rule

simp rewrites using a set of preregistered rewrite rules exhaus-

tively

stands for a missing proof or definition

#### **Tactic Combinator**

 $\{\,\dots\,\}$  focuses on the first subgoal; needs to prove that goal

# **Chapter 3**

## **Forward Proofs**

Tactics are a backward proof mechanism. They start from the goal and break it down. Often it makes sense to work in a forward fashion: to start with what we already know and proceed step by step towards our goal. Structured proofs are a style that supports this kind of reasoning. It can be combined with the tactical style. Tactical proofs tend to be easier to write but harder to read; it is, after all, easier to destruct things than to construct them. Most users combine the two styles, using whichever seems the most appropriate for the situation. The higher readability of structured proofs make them popular with some users, especially in mathematical circles.

Structured proofs are syntactic sugar sprinkled over Lean's proof terms. They are built using keywords such as assume, have, let, and show that mimic pen-and-paper proofs. All Lean proofs, whether tactical or structured, are reduced internally to proof terms. We have seen some specimens already, in Chapter 2: Given hypotheses ha: a and hab:  $a \rightarrow b$ , the term hab ha is a proof term for the proposition b, and we write hab ha: b. The names of lemmas and hypotheses, such as ha and hab, are also proof terms. Pushing this idea further, given hbc:  $b \rightarrow c$ , we can build the proof term hbc (hab ha) for the proposition c. We can think of hab as a function that converts a proof of a to a proof of b, and similarly for

hbc.

Structured proofs are the default in Lean. They can be used outside tactic mode. To enter tactic mode, we need to use a begin ... end block or a by line.

The concepts covered here are described in more detail in Chapters 2 to 4 of *Theorem Proving in Lean* [1]. Nederpelt and Geuvers's textbook [26] and Van Raamsdonk's lecture notes [31] are other useful references.

## 3.1 Structured Proofs

As a first example, consider the following structured proof:

```
lemma fst_of_two_props : \forall a \ b : Prop, \ a \rightarrow b \rightarrow a := fix \ a \ b : Prop, assume ha : a, assume hb : b, show a, from ha
```

Each variable bound by a  $\forall$ -quantifier and each assumption of an implication is introduced explicitly in the proof using the fix and assume commands. Several variables or hypotheses can be introduced simultaneously. We will often omit the types of variables, especially when they can be guessed from their names; however, we will usually spell out the propositions and put them one per line, to increase readability—which is, after all, one of the main potential advantages of the structured style. The show ..., from command at the end repeats the proposition to prove, for the sake of readability, and gives the proof after the keyword from. The goal at this point is a b: Prop, ha: a, hb: b  $\vdash$  a.

Informally, we could write the proof as follows:

Fix some propositions a and b. Assume (ha) a and (hb) b are true.

We must show a. This follows trivially from ha.

Some authors would insert some qualifiers such as "arbitrary but fixed" in front of "propositions," especially in textbooks for first-year bachelor students, or they would write "Let a and b be some propositions." All these variants are equivalent. And instead of " $\Box$ ," some authors would write "QED" to conclude the proof.

The Lean proof above is atypical in that the goal's target appears among the hypotheses. Usually, we must perform some intermediate reasoning steps, essentially of the form "from suchand-such, we have so-and-so." In Lean, each intermediate step takes the form of a have command, as in the following example:

```
lemma prop_comp (a b c : Prop) (hab : a \rightarrow b) (hbc : b \rightarrow c) : a \rightarrow c := assume ha : a, have hb : b := hab ha, have hc : c := hbc hb, show c, from hc
```

#### Informally:

Assume (ha) a is true.

From ha and hab, we have (hb) b.

From hb and hbc, we have (hc) c.

We must show c. This follows trivially from hc.

Notice that this is a forward proof: It progresses one theorem at a time from the hypothesis a to the desired theorem c.

In general, the fix-assume-show skeleton simply repeats the lemma's statement. We often name the fixed variables after the bound variables from the goal, as we did here, but this is not mandatory. Moreover, it is good practice to avoid shadowing existing variables. Between the last fix or assume and the show, we may have

as many have commands as we want, depending on how detailed we want the argument to be. Details can increase readability, but providing too many details can overwhelm even the most persistent reader.

The have command has a similar syntax to lemma but appears inside a structured proof. We can think of a have as a definition. In have hb: b:= hab ha, the right-hand side hab ha is a proof term for b, and the left-hand side hb is defined as a synonym for that proof term. From then on, hb and hab ha can be used interchangeably. Since hb and hc are used only once and their proofs are very short, experts would tend to inline them, replacing hc by hbc hb and hb by hab ha, yielding

```
lemma prop_comp2 (a b c : Prop) (hab : a \rightarrow b) (hbc : b \rightarrow c) : a \rightarrow c := assume ha : a, show c, from hbc (hab ha)
```

A typical structured proof has the following fix-assume-haveshow format:

```
lemma l: \forall (c_1:\sigma_1)\ldots(c_1:\sigma_1),\; P_1\to\cdots\to P_m\to R:= fix (c_1:\sigma_1)\ldots(c_1:\sigma_1), assume h_1:P_1, \vdots assume h_m:P_m, have k_1:Q_1:=\ldots, \vdots have k_n:Q_n:=\ldots, show R, from
```

## **3.2 Structured Constructs**

The previous section presented the main commands for writing structured proofs: fix, assume, have, and show. We now review the components of structured proofs more systematically.

#### **Lemma or Hypothesis**

The simplest structured proof, apart from sorry, is the name of a lemma or hypothesis. If we have

```
lemma two_add_two_eq_four :
   2 + 2 = 4 :=
begin
   ...
end
```

then the lemma name two\_add\_two\_eq\_four can be used as a proof of 2 + 2 = 4 later. For example:

```
lemma this_time_with_feelings :
  2 + 2 = 4 :=
two_add_two_eq_four
```

We can pass arguments to lemmas to instantiate  $\forall$ -quantifiers and discharge assumptions. Suppose the lemma add\_comm (m n :  $\mathbb{N}$ ): add m n = add n m is available, and suppose we want to prove its instance add o n = add n o. This can be achieved neatly using the name of the lemma and two arguments:

```
lemma add_comm_zero_left (n : N) :
  add o n = add n o :=
add comm o n
```

This has the same effect as the tactical proof by exact add\_comm on, but is more concise. The exact tactic can be seen as the inverse of by. Why enter tactic mode if only to leave it immediately?

Like with exact and apply, the lemma or hypothesis's statement is matched with the current goal up to computation. This gives some flexibility.

#### fix

```
fix names [: type],
fix (names<sub>1</sub> [: type<sub>1</sub>])...(names<sub>n</sub> [: type<sub>n</sub>]),
```

The fix command moves  $\forall$ -quantified variables from the goal's target to the local context. It can be seen as a structured version of the intros tactic.

Note that the standard Lean tactic for fixing variables is assume, which also works in this case. We prefer to use fix, which is provided by LoVelib, as a more readable alternative and reserve assume for assumptions.

#### assume

```
assume names [:proposition], assume (names_1 [:proposition_1]) ... (names_n [:type_n]),
```

The assume command moves the leading assumptions from goal's target to the local context. It can be seen as a structured version of the intros tactic.

#### have

```
have name : proposition :=
  proof,
```

The have command lets us state and prove an intermediate lemma, which may refer to names introduced by previous fixes, assumes, and haves. The proof can be tactical or structured. Generally, we tend to use structured proofs to sketch the main argument and resort to tactical proofs for proving subgoals or uninteresting intermediate steps. Another kind of mixture arises when we pass arguments to lemma names. For example, given hab:  $a \rightarrow b$  and ha: a, the tactic exact hab ha will prove the goal  $\vdash$  b. Here, hab ha is a proof term nested inside a tactic.

#### let

```
let name [: type] := term in
```

The let command introduces a new local definition. It can be used to name a complex object that occurs several times in the proof afterwards. It is similar to have but is designed for computable data, not proofs. Expanding or introducing a let corresponds to  $\zeta$ -conversion (Section 2.2).

#### show

```
show proposition, from
  proof
```

The show command lets us repeat the goal to prove, which can be useful as documentation. It also allows us to rephrase the goal in a syntactically equal form up to computation. Instead of the syntax show proposition, from proof, we can simply write proof if we do not want to repeat the goal and do not need to rephrase it. The proof can be tactical or structured.

## 3.3 Forward Reasoning about Connectives and Quantifiers

Reasoning about the logical connectives and quantifiers in a forward fashion uses the same introduction and elimination rules as in tactic mode (Section 2.3). A few examples will show the flavor. Let us start with conjunction:

```
lemma and_swap (a b : Prop) : 
 a \wedge b \rightarrow b \wedge a :=
assume hab : a \wedge b,
have ha : a :=
and.elim_left hab,
```

```
have hb : b :=
  and.elim_right hab,
show b ∧ a, from
  and.intro hb ha
```

Even readers who do not know what and <code>.elim\_left</code> etc. means can understand that we extract a and b from a  $\wedge$  b and put them back together as b  $\wedge$  a. Mathematicians would probably have an easier time making sense of this proof than of its tactical counterpart:

```
lemma and_swap (a b : Prop) :
  a ∧ b → b ∧ a :=
begin
  intro hab,
  apply and.intro,
  apply and.elim_right,
  exact hab,
  apply and.elim_left,
  exact hab
end
```

On the other hand, the tactical proof is easier to derive mindlessly.

In general, backward proofs are easier to find, and most of the automation provided by proof assistants works backwards. The reason is simple: Pretend that you are Miss Marple or Hercule Poirot on a murder investigation. A backward investigation would start from the crime scene and try to extract clues that potentially connect a handful of suspects to the crime. In contrast, a forward investigation might start by questioning as many as seven billion people to determine whether they have an alibi. Which approach is more likely to succeed?

Our next examples concern *one-point rules*. These are lemmas that can be used to eliminate a quantifier when the bound variable can effectively take only one value. For example, the proposition  $\forall n$ ,  $n = 666 \rightarrow beast \geq n$  can be simplified to beast  $\geq 666$ . The following lemma justifies this simplification:

```
lemma forall.one_point \{\alpha : \mathsf{Type}\}\ (\mathsf{t} : \alpha)\ (\mathsf{p} : \alpha \to \mathsf{Prop}) :
   (\forall x, x = t \rightarrow p x) \leftrightarrow p t :=
iff.intro
   (assume hall : \forall x, x = t \rightarrow p x,
    show p t, from
       begin
          apply hall t,
          ref1
       end)
   (assume hp : p t,
    fix x.
    assume heq : x = t,
    show p x, from
       begin
          rw heq,
         exact hp
       end)
```

The proof may look intimidating, but it was not hard to develop. The key was to proceed one step at a time. At first, we observed that the goal is an implication, so we wrote

```
iff.intro (_) (_)
```

The two placeholders are important to make this proof well formed. We already put parentheses because we strongly suspect that these will be nontrivial proofs. Because proofs are basically terms, which are basically programs, the advice we gave in Section 1.1.4 applies here as well:

The key idea is that the [proof] should be syntactically correct at all times. The only red underlining we should see in Visual Studio Code should appear under the placeholders. In general, a good principle for software development is to start with a program that compiles, perform the smallest change possible to obtain a new compiling program, and repeat until the program is complete.

Hovering over the first placeholder makes the corresponding subgoal appear. We can see that Lean expects a proof of  $\vdash (\forall x, x = t \rightarrow p \ x \ so we provide a suitable skeleton. A structured proof of an implication consists of an assume followed by a show:$ 

```
iff.intro (assume hall : \forall x, x = t \rightarrow p x, show p t, from _) (_)
```

Each of the remaining placeholder can be replaced by a structured proof or by a tactical proof. To fill these placeholders, we can use essentially the same procedure as for exhibiting an inhabitant of a type (Section 1.1.4), interpreting implication as the function arrow and writing assume—show instead of  $\lambda$ .

Let us check that the rule actually works on our motivating example:

```
lemma beast_666 (beast : \mathbb{N}) : 
 (\forall n, n = 666 \rightarrow beast \ge n) \leftrightarrow beast \ge 666 := forall.one_point _ _
```

It works. Matching forall.one\_point t p against the statement of beast\_666 yields the instantiation t := 666 and p :=  $(\lambda m$ , beast = m).

Finally, the one-point rule for  $\exists$  demonstrates how to use the introduction and elimination rules for  $\exists$  in a structured proof:

```
lemma exists.one_point \{\alpha: \mathsf{Type}\}\ (\mathsf{t}:\alpha)\ (\mathsf{p}:\alpha\to\mathsf{Prop}): (\exists \mathsf{x}:\alpha,\,\mathsf{x}=\mathsf{t}\land\mathsf{p}\,\mathsf{x})\leftrightarrow\mathsf{p}\,\mathsf{t}:= iff.intro (assume hex: \exists \mathsf{x},\,\mathsf{x}=\mathsf{t}\land\mathsf{p}\,\mathsf{x}, show \mathsf{p}\,\mathsf{t}, from exists.elim hex (fix \mathsf{x}, assume hand: \mathsf{x}=\mathsf{t}\land\mathsf{p}\,\mathsf{x}, show \mathsf{p}\,\mathsf{t}, from by \mathsf{cc})
```

```
(assume hp : p t,

show \exists x : \alpha, x = t \land p x, from

exists.intro t

(show t = t \land p t, from

by cc))
```

Notice how we use exists.elim hex to obtain an x such that  $x = t \land p x$ . This admittedly is a bit cumbersome.

## 3.4 Calculational Proofs

In informal mathematics, we often express proofs as transitive chains of equalities, inequalities, or equivalences (e.g., a = b = c,  $a \ge b \ge c$ , or  $a \leftrightarrow b \leftrightarrow c$ ). In Lean, such *calculational proofs* are supported by the calc command. It provides a lightweight syntax and takes care of applying transitivity lemmas for preregistered relations, such as equality and the arithmetic comparison operators.

The general syntax is as follows:

```
calc termo
    op1 term1:
    proof1
... op2 term2:
    proof2
     :
... opn termn:
    proofn
```

The horizontal dots (...) are part of the syntax. Each proof; justifies the statement  $term_{i-1}$  op;  $term_i$ . The operators op; need not be identical, but they must be compatible. For example, =, <, and  $\leq$  are compatible, whereas > and < are incompatible.

A simple example follows:

```
lemma two_mul_example (m n : N) :
    2 * m + n = m + n + m :=
```

Mathematicians (assuming they would condescend to offer a justification for such a trivial result) could have written the above proofs roughly as follows:

```
2 * m + n
= (m + m) + n (since 2 * m = m + m)
= m + n + m (by associativity and commutativity of +)
```

In the Lean proof, the horizontal dots stand for the term (m + m) + n, which we would have had to repeat had we written the proof without calc:

```
lemma two_mul_example_2 (m n : \mathbb{N}) : 
 2 * m + n = m + n + m :=
have h_1 : 2 * m + n = (m + m) + n :=
by rw two_mul,
have h_2 : (m + m) + n = m + n + m :=
by cc,
show _, from
eq.trans h_1 h_2
```

Notice that with haves, we also need to explicitly invoke eq.trans and to give names to the two intermediate steps.

## 3.5 Forward Reasoning with Tactics

Most users prefer the tactic mode. But even in tactic mode, it can be useful to reason in a forward fashion, mixing forward and backward reasoning steps. The structured proof commands have, let, and calc are also available as tactics, making this possible.

The following example demonstrates the have and let tactics on a lemma we have seen several times already:

```
lemma prop_comp_3 (a b c : Prop) (hab : a \rightarrow b) (hbc : b \rightarrow c) : a \rightarrow c := begin intro ha, have hb : b := hab ha, let c' := c, have hc : c' := hbc hb, exact hc end
```

#### have

```
have [[name:] proposition] :=
  proof
```

The have tactic lets us state and prove an intermediate lemma in tactic mode. Afterwards, the lemma is available as a hypothesis in the goal state.

#### let

```
let name [: type] := term
```

The let tactic lets us introduce a local definition in tactic mode. Afterwards, the defined symbol and its definition are available in the goal state.

Observe that the syntax for the let tactic is slightly different than for the let structured proof command, with no in keyword (but a comma instead if the tactic is followed by another tactic).

#### calc

The calc tactic lets us enter a calculational proof (Section 3.4) in tactic mode. The tactic has the same syntax as the structured proof command of the same name. We can regard apply calc ..., where calc is a tactic, as an alias for apply (exact calc ...), where calc is the structured proof command described above.

## 3.6 Dependent Types

Dependent types are the defining feature of the *dependent type* theory family of logics. Although you may not be familiar with the terminology, you are likely to be familiar with the concept in some form or other.

Consider a function pick that takes a natural number n (i.e., a value from  $\mathbb{N} = \{0, 1, 2, \ldots\}$ ) and that returns a natural number between o and n. Intuitively, pick n should have the type  $\{0, 1, \ldots, n\}$  (i.e., the type consisting of all natural numbers  $i \leq n$ ). In Lean, this is written  $\{i : \mathbb{N} // i \leq n\}$ . This would be the type of pick n. Mathematically inclined readers might want to think of pick as an  $\mathbb{N}$ -indexed family of terms

```
(pick n: \{i: \mathbb{N} // i \leq n\})<sub>n:\mathbb{N}</sub>
```

in which the type of each term depends on the index—e.g., pick  $5: \{i: \mathbb{N} // i \leq 5\}$ . But what would be the type of pick itself?

We would like to express that pick is a function that takes an argument  $n:\mathbb{N}$  and that returns a value of type  $\{i:\mathbb{N}\ //\ i\leq n\}$ . To capture this, we will write

$$pick: (n:\mathbb{N}) \to \{i:\mathbb{N} // i \le n\}$$

This is a dependent type: The type of the result depends on the value of the argument n. The variable name n is immaterial.

Unless otherwise specified, a *dependent type* means a type depending on a (non-type) term, as above, with  $n:\mathbb{N}$  as the term and  $\{i:\mathbb{N}\ //\ i\leq n\}$  as the type that depends on it. This is what we mean when we claim that simple type theory does not support dependent types. But a type may also depend on another type—for example, the type constructor list, its  $\eta$ -expanded variant  $\lambda\alpha$ : Type, list  $\alpha$ , or the polymorphic type  $\lambda\alpha$ : Type,  $\alpha\to\alpha$  of functions with the same domain and codomain. A term may depend on a type—for example, the polymorphic identity function  $\lambda\alpha$ : Type,  $\lambda x:\alpha$ ,  $\alpha$ ,  $\alpha$ . And of course, a term may also depend on a term—for example,  $\lambda n:\mathbb{N}$ ,  $\alpha$ ,  $\alpha$ . In summary, there are four cases for  $\alpha$ , t:

Body (t)		Argument (x)	Description	
	depending on		Simply typed λ-expression	
	depending on		Dependent type (in the narrow sense)	
A term	depending on	a type	Polymorphic term	
A type	depending on	a type	Type constructor	

The last three rows correspond to the three axes of Henk Barendregt's  $\lambda$ -cube.<sup>1</sup>

The APP and LAM rules presented in Section 1.1.3 must be gener-

¹https://en.wikipedia.org/wiki/Lambda\_cube

alized to work with dependent types:

$$\frac{\text{C} \vdash \text{t} : (\text{x} : \sigma) \rightarrow \tau[\text{x}] \quad \text{C} \vdash \text{u} : \sigma}{\text{C} \vdash \text{t} \text{u} : \tau[\text{u}]} \text{APP}'$$

$$\frac{\text{C}, \text{x} : \sigma \vdash \text{t} : \tau[\text{x}]}{\text{C} \vdash (\lambda \text{x} : \sigma, \text{t}) : (\text{x} : \sigma) \rightarrow \tau[\text{x}]} \text{LAM}'$$

The notation  $\tau[x]$  stands for a type that may contain x, and  $\tau[u]$  stands for the same type where all occurrences of x have been replaced by u.

The simply typed case arises when x does not occur in  $\tau[x]$ . Then, we can simply write  $\sigma \to \text{instead}$  of  $(x:\sigma) \to .$  The familiar notation  $\sigma \to \tau$  is equivalent to  $(\_:\sigma) \to \tau$ . It is easy to check that APP' and LAM' coincide with APP and LAM when x does not occur in  $\tau[x]$ . The example below demonstrates APP':

$$\frac{\vdash \text{ pick}: (\text{n}:\mathbb{N}) \rightarrow \text{i}:\mathbb{N} \text{ // i} \leq \text{n} \quad 5:\mathbb{N}}{\vdash \text{ pick } 5:\text{i}:\mathbb{N} \text{ // i} \leq 5} \text{App'}$$

The next example demonstrates LAM':

$$\frac{\alpha: \mathsf{Type}, \mathsf{x}: \alpha \vdash \mathsf{x}: \alpha}{\alpha: \mathsf{Type} \vdash (\lambda \mathsf{x}: \alpha, \mathsf{x}): \alpha \to \alpha} \mathsf{LAM} \, \mathsf{or} \, \mathsf{LAM}'}{\vdash (\lambda \alpha: \mathsf{Type}, \lambda \mathsf{x}: \alpha, \mathsf{x}): (\alpha: \mathsf{Type}) \to \alpha \to \alpha} \mathsf{LAM}'}$$

The picture is incomplete because we only check that the terms—the entities on the left-hand side of a colon (:)—are well typed. The types—the entities on the right-hand side of a colon—must also be checked, using the same type system, but this time as terms. For example, the type of nat.succ is  $\mathbb{N} \to \mathbb{N}$ , whose type is Type. Types of types, such as Type and Prop, are called universes. We will study them more closely in Chapter 11.

Regrettably, the intuitive syntax  $(x:\sigma) \to \tau$  used above is not available in Lean. Instead, we must write  $\forall x:\sigma$ ,  $\tau$ . The familiar

notation  $\sigma \to \tau$  is supported as syntactic sugar for  $\forall \_: \sigma$ ,  $\tau$ . The  $\forall$  syntax emphasizes that x is a bound variable. As an alias for  $\forall$ , we can also write  $\Pi$ . The two symbols are interchangeable.

In Section 1.2, we saw the commands

```
#check list.nil
#check list.cons
```

#### and their output

```
list.nil : \Pi(\alpha: \mathsf{Type}), list \alpha list.cons : \mathsf{?M}_1 \to \mathsf{list} \; \mathsf{?M}_1 \to \mathsf{list} \; \mathsf{?M}_1
```

We can now make sense of the  $\Pi$ -type: list.nil is a function that takes a type  $\alpha$  as argument and that returns a value of type list  $\alpha$ 

## 3.7 The PAT Principle

You will likely have noticed that the same symbol  $\rightarrow$  is used both for implication (e.g., false  $\rightarrow$  true) and as the type constructor of functions (e.g.,  $\mathbb{Z} \rightarrow \mathbb{N}$ ). Similarly,  $\forall$  is used both as a quantifier and to specify dependent types. Without context, we cannot tell whether  $a \rightarrow b$  refers to the type of a function with domain a and codomain b or to the proposition "a implies b," and similarly,  $\forall x:a,b[x]$  can denote a proposition or a dependent type.

It turns out that not only the two pairs of concepts *look* the same, they *are* the same. This is called the *PAT principle*, where PAT is a double mnemonic:

```
PAT = propositions as types PAT = proofs as terms
```

Furthermore, because types are also terms, we also have that propositions are terms. However, PAT is not a quadruple mnemonic (PAT = proofs as types). Also note that not all types are propositions, and not all terms are proofs.

By using terms and types to represent proofs and propositions, dependent type theory achieves a considerable economy of concepts. The question "Is H a proof of P?" becomes equivalent to "Does the term H have type P?" As a result, inside of Lean, there is no proof checker, only a type checker.

Let us review the dramatis personae one by one. We use the metavariables  $\sigma$ ,  $\tau$  for types; P, Q for propositions; t, u, x for terms; and h, G, H, for proofs.

Starting with "propositions as types," for types, we have the following:

- 1.  $\sigma \rightarrow \tau$  is the type of (total) functions from  $\sigma$  to  $\tau$ .
- 2.  $\forall x: \sigma$ ,  $\tau[x]$  is the dependent function type from  $x: \sigma$  to  $\tau[x]$ .

In contrast, for propositions, we have the following:

- 1.  $P \rightarrow Q$  can be read as "P implies Q," or as the type of functions mapping proofs of P to proofs of Q.
- 2.  $\forall x : \sigma$ , Q[x] can be read as "for all x, Q[x]," or as the type of functions mapping values x of type  $\sigma$  to proofs of Q[x].

Continuing with "proofs as terms," for terms, we have the following:

- 1. A constant is a term.
- 2. A variable is a term.
- 3. t u is the application of function t to argument u.
- 4.  $\lambda x$ , t[x] is a function mapping x to t[x].

In contrast, for proofs (i.e., proof terms), we have the following:

- 1. The name of a lemma or hypothesis is a proof.
- 2. H t, which instantiates the leading parameter or quantifier of proof H's statement with term t, is a proof.
- 3. H G, which discharges the leading assumption of H's statement with proof G, is a proof. This operation is called *modus* ponens.

- 4.  $\lambda h : P$ , H[h] is a proof of  $P \to Q$ , assuming H[h] is a proof of Q for h : P.
- 5.  $\lambda x : \sigma$ , H[x] is a proof of  $\forall x : \sigma$ , Q[x], assuming H[x] is a proof of Q[x] for  $x : \sigma$ .

The last two cases are justified by the Lam' rule. In a structured proof, as opposed to a raw proof term, we would write assume or fix instead of  $\lambda$ , and we would probably want to repeat the conclusion using show for readability, as follows:

```
lemma case_4 : lemma case_5 : P \rightarrow Q := \forall x : \sigma, Q[x] := assume h : P, fix x, show Q, from show Q[x], from H[h]
```

Some commands are provided in Lean under two different names but are really the same by the PAT principle. This is the case for constant and axiom as well as for fix and assume; indeed, fix is defined by LoVelib as an alias for assume.


There are also pairs with slightly different behavior, such as def and lemma or let and have. The fundamental difference is this: When we define some function or data, we care not only about the type but also about the body—the behavior. On the other hand, once we have proved a lemma, the proof becomes *irrelevant*. All that matters is that there *is* a proof. We will return to the topic of proof irrelevance in Chapter 11.

The following correspondence table summarizes the differences between tactical proofs, structured proofs, and raw proof terms:

Tactical proof	Structured proof	Raw proof term
intro x,	fix x,	λx,
intro h,	assume h,	λh,
have k := H,	have k := H,	(λk,) H
let $x := t$ ,	<pre>let x := t in</pre>	(λx,) t
exact(H:P)	show P, from H	H : P
calc	calc	calc

Notice that let x := a in t is essentially a sophisticated way to write  $(\lambda x, t)$  a. Its main benefit, aside from readability, is that it is not subject to  $\beta$ -reduction.

The terminology of dependent type theory can be quite confusing, because some words have a narrow and a broad sense. The following diagram captures the various meanings of important words:



According to the broad senses, any expression is a term, any expression that may occur on the right-hand side of a typing judgment is a type, and any expression that may occur on the right-hand side of a typing judgment with a type on its left-hand side is a universe. This is consistent with the reading of t: u as "t has type u" and the notion that universes are types of types.

## 3.8 Induction by Pattern Matching

In Section 2.6, we reviewed the use of the induction' tactic to perform proofs by induction. An alternative, more flexible style relies on pattern matching and the PAT principle.

Recall the definition of the reverse of a list from Section 1.3:

In fact, reverse exists as list.reverse in Lean's standard library, but our definition is optimized for reasoning. A useful property to prove is that reverse is its own inverse: reverse (reverse xs) = xs for all lists xs. However, if we try to prove it by induction, we quickly run into an obstacle. The induction step is

```
ih: ∀xs, reverse (reverse xs) = xs ⊢ reverse (reverse xs ++ [x]) = x :: xs
```

What is unpleasant about this subgoal is the presence of ++ [x] inside the double reverse sandwich. We need a way to "distribute" the outer reverse over ++ to obtain a term that matches the induction hypothesis's left-hand side. The trick is to prove and use the following lemma:

The lemma's proof uses a more flexible style than usual. Arguably, it looks more like a recursive function definition than a proof. The patterns on the left, [] and x :: xs, correspond to the two constructors of the  $\forall$ -quantified variable xs. On the right-hand side of each := symbol is a proof for the corresponding case. The variables on which we can pattern-match are those that appear in the  $\forall$  quantifiers, in order of occurrence (here, xs and ys).

Inside the induction step's proof, the induction hypothesis is available under the same name as the lemma we are proving (reverse\_a). We explicitly pass xs as argument to the induction hypothesis. This restricts the hypothesis so that it only applies to xs but not to other lists. In particular, this ensures that the lemma will not be applied to x:: xs, which could lead to a circular argument: "To prove reverse\_append (x:: xs), use reverse\_append (x:: xs)." Lean's termination checker would notice that the argument is ill-founded and raise an error, but we want to avoid that. In addition, the explicit argument xs is useful as documentation, effectively saying, "The only recursive instance of the lemma we need is the one on xs."

For reference, the corresponding tactical proof is as follows:

```
lemma reverse_append<sub>2</sub> {α : Type} (xs ys : list α) :
  reverse (xs ++ ys) = reverse ys ++ reverse xs :=
begin
  induction' xs,
  { simp [reverse] },
  { simp [reverse, ih] }
end
```

The lemma would also be provable, and useful, if we put [y] instead of ys. But it is a good habit to state lemmas as generally as possible. This results in more reusable libraries. Moreover, this is often necessary to obtain a strong enough induction hypothesis in a proof by induction. In general, finding the right inductions and lemmas can require thought and creativity.

Simultaneous pattern matching on multiple variables is supported (e.g., xs and ys above). The patterns are then separated by spaces. This explains why Lean requires us to put parentheses around complex patterns, even when we match on a single variable. The general format is

Notice the strong similarity with the syntax of def (Section 1.3). The two commands are, in fact, almost the same, the main exception being that lemma considers the defined term or the proof opaque, whereas def keeps it transparent. Since the actual proofs are irrelevant once a lemma is proven (Section 11), there is no need to expand them later. A similar distinction exists between let and have.

By the PAT principle, a proof by induction by pattern matching is the same as a recursive proof term. When we invoke the induction hypothesis, we are really just invoking a recursive function recursively. This explains why the induction hypothesis has the same name as the lemma we prove. Lean's termination checker is used to establish well-foundedness of the proof by induction.

With the reverse\_append lemma in place, we can return to our initial goal:

Induction by pattern matching is highly popular among Lean users. Its main advantages are its convenient syntax and its support for well-founded induction, which is more powerful than structural induction as provided by the induction' tactic (Section 2.7). However, in this guide, we will not need the full power of well-founded induction. Furthermore, for subtle logical reasons, induction by pattern matching is not available for inductive predicates,

which are the topic of Chapter 5. For these reasons, we will generally prefer induction'.

## 3.9 Summary of New Lean Constructs

#### **Proof Commands**

assume states assumptions
calc combines proofs by transitivity
fix fixes variables
have states an intermediate lemma
let ... in introduces a local definition

show ..., from states the target

#### **Tactics**

calc combines proofs by transitivity states an intermediate lemma let introduces a local definition

# Part II Functional-Logic Programming

## **Chapter 4**

## **Functional Programming**

We take a closer look at the essence of typed functional programming: inductive types, proofs by induction, recursive functions, pattern matching, structures (records), and type classes. The concepts covered here are described in more detail in Chapters 7 to 9 of *Theorem Proving in Lean* [1].

## 4.1 Inductive Types

Inductive types are modeled after the data types of typed functional programming languages (e.g., Haskell, ML, OCaml). They are also reminiscent of sealed classes in Scala. Already in Chapter 1, we saw some basic inductive types: the natural numbers, the finite lists, and a type of arithmetic expressions. In this chapter, we revisit the lists and study binary trees. We also take a brief look at vectors of length n, a dependent type.

Recall the definition of natural numbers as an inductive type:

```
inductive nat : Type
| zero : nat
| succ : nat → nat
```

This definition introduces three constants, which could have been declared as follows:

```
constant nat : Type
constant nat.zero : nat
constant nat.succ : nat → nat
```

In addition, the inductive definitions asserts some properties about nat.zero and nat.succ, which is why we use the inductive command. It also introduces further constants that are used internally to support induction and recursion.

As we saw in Section 1.2, an inductive type is a type whose members are all the values that can be built by a finite number of applications of its constructors, and only those. Mottos:

- No junk: The type contains no values beyond those expressible using the constructors.
- No confusion: Values built using a different combination of constructors are different.

For natural numbers, "no junk" means that there exist no special values such as -1,  $\varepsilon$ ,  $\infty$ , or NaN that cannot be expressed using a finite combination of nat.zero and nat.succ, and "no confusion" ensures that nat.zero  $\neq$  nat.succ n for all n and that nat.succ is injective. In addition, values of inductive types are always finite. The infinite term

```
nat.succ (nat.succ (nat.succ (nat.succ ...)))
```

is not a value. Nor does there exist a value n such that nat.succ n = n, as we will show below.

Inductive types are very convenient to use, because they support induction and recursion and their constructors are well behaved, but not all types can be defined as an inductive type. In particular, mathematical types such as  $\mathbb Q$  (the rationals) and  $\mathbb R$  (the real numbers) require more elaborate constructions, based

¹We could try to cheat and axiomatize special constants using constant and axiom, but the result would be either tautological (i.e., useless) or contradictory (i.e., positively harmful). A safer but equally pointless alternative would be to locally assume the existence of such a value; this would be pointless because the assumption could never be discharged.

on quotienting and subtyping. These will be explained in Chapters 11 and 13.

## 4.2 Structural Induction

Structural induction is a generalization of mathematical induction to arbitrary inductive types. To prove a goal  $n : \mathbb{N} \vdash P[n]$  by structural induction on n, it suffices to show two subgoals, traditionally called the base case and the induction step:

```
\vdash P[0]
k : \mathbb{N}, ih : P[k] \vdash P[k+1]
```

We can of course also write nat.zero and nat.succ k.

In general, the situation is more complex. The goal might contain some extra hypotheses (e.g., Q) that do not depend on n and others (e.g., R[n]) that do. Assuming we have one hypothesis of each kind, this gives the initial goal

```
hQ:Q,n:\mathbb{N},hR:R[n]\vdash S[n]
```

Structural induction on n then produces the two subgoals

```
\begin{array}{c} \mathsf{hQ}: \mathsf{Q}, \, \mathsf{hR}: \, \mathsf{R}[\mathfrak{0}] \, \vdash \, \mathsf{S}[\mathfrak{0}] \\ \mathsf{hQ}: \mathsf{Q}, \, \mathsf{k}: \, \mathbb{N}, \, \mathsf{ih}: \, \mathsf{R}[\mathsf{k}] \rightarrow \mathsf{S}[\mathsf{k}], \, \mathsf{hR}: \, \mathsf{R}[\mathsf{k}+1] \, \vdash \, \mathsf{S}[\mathsf{k}+1] \end{array}
```

The hypothesis Q is simply carried over unchanged from the initial goal, whereas  $R[n] \vdash S[n]$  is treated almost the same as if the goal's target had been  $R[n] \to S[n]$ . This is easy to check by taking  $P[n] := R[n] \to S[n]$  in the first example above. Since this general format is very verbose and hardly informative (now that we understand how it works), from now on we will present goals in the simplest form possible, without extra hypotheses.

For lists, given a goal xs : list  $\alpha \vdash P[xs]$ , structural induction on xs yields

```
\vdash P[[]]
y: \alpha, ys: list \alpha, ih: P[ys] \vdash P[y:: ys]
```

We can of course also write list.nil and list.cons y ys. There is no induction hypothesis associated with y, because y is not of list type.

For arithmetic expressions, the base cases are

```
i: \mathbb{Z} \vdash P[aexp.num i]  x: string \vdash P[aexp.var x]
```

and the induction steps are

```
e_1 e_2 : aexp, ih_1 : P[e_1], ih_2 : P[e_2] \vdash P[aexp.add e_1 e_2]
e_1 e_2 : aexp, ih_1 : P[e_1], ih_2 : P[e_2] \vdash P[aexp.sub e_1 e_2]
e_1 e_2 : aexp, ih_1 : P[e_1], ih_2 : P[e_2] \vdash P[aexp.mul e_1 e_2]
e_1 e_2 : aexp, ih_1 : P[e_1], ih_2 : P[e_2] \vdash P[aexp.div e_1 e_2]
```

Notice the two induction hypotheses, about e<sub>1</sub> and e<sub>2</sub>.

In general, structural induction produces one subgoal per constructor. In each subgoal, induction hypotheses are available for all constructor arguments of the type we are performing the induction on.

Regardless of the inductive type  $\tau$ , the procedure to compute the subgoals is always the same:

- 1. Replace the hole in P[] with each possible constructor applied to fresh variables (e.g., y:: ys), yielding as many subgoals as there are constructors.
- 2. Add these new variables (e.g., y, ys) to the local context.
- 3. Add induction hypotheses for all new variables of type  $\tau$ .

As an example, we will prove that nat.succ  $n \neq n$  for all  $n : \mathbb{N}$ . We start with an informal proof, because these require us to understand what we are doing:

The proof is by structural induction on n.

CASE 0: We must show nat.succ  $0 \neq 0$ . This follows from the "no confusion" property of the constructors of inductive types.

CASE nat.succ k: The induction hypothesis is nat.succ  $k \neq k$ . We must show nat.succ (nat.succ k)  $\neq$  nat.succ k. By the injectivity of nat.succ, we have that nat.succ (nat.succ k) = nat.succ k is equivalent to nat.succ k = k. Thus, it suffices to prove nat.succ  $k \neq k$ , which corresponds exactly to the induction hypothesis.

Notice the main features of this informal proof, which you should aim to reproduce in your own informal arguments:

- The proof starts with an unambiguous announcement of the type of proof we are carrying out (e.g., which kind of induction and on which variable).
- The cases are clearly identified, and for each case, both the goal's target and the hypotheses are stated.
- The key lemmas on which the proof relies are explicitly invoked (e.g., injectivity of nat.succ).

Now let us carry out the proof in Lean:

```
lemma nat.succ_neq_self (n : N) :
  nat.succ n ≠ n :=
begin
  induction' n,
  { simp },
  { simp [ih] }
end
```

The routine reasoning about constructors is all carried out by simp automatically, which is usually what we want.

We can supply our own names, and reorder the cases, by using the case tactic in front of each case, together with the case's name and the desired names for the variables and hypotheses introduced by induction'. For example:

```
lemma nat.succ_neq_self<sub>2</sub> (n : \mathbb{N}) : nat.succ n \neq n :=
```

```
begin
  induction' n,
  case succ : m IH {
    simp [IH] },
  case zero {
    simp }
end
```

Instead of n and ih, we chose the names m and IH.

## 4.3 Structural Recursion

Structural recursion is a form of recursion that allows us to peel off one constructor from the value on which we recurse. The factorial function below is structurally recursive:

```
def fact : \mathbb{N} \to \mathbb{N}

\mid 0 \qquad := 0

\mid (n + 1) := (n + 1) * fact n
```

The constructor we peel off here is nat.succ (written + 1). Such functions are guaranteed to call themselves only finitely many times before the recursion stops; for example, fact 12345 will call itself 12345 times. This is a prerequisite for establishing that the function terminates, an important property to ensure both logical consistency and termination of evaluation.

With structural recursion, there are as many equations as there are constructors. Novices are often tempted to supply additional, redundant cases, as in the following example:

```
\begin{array}{lll} \text{def fact}_2 & : & \mathbb{N} \to \mathbb{N} \\ \mid & 0 & := & 0 \\ \mid & 1 & := & 1 \\ \mid & (n + 1) := & (n + 1) * \text{fact}_2 & n \end{array}
```

It is in your own best interest to resist this temptation. The more cases you have in your definitions, the more work it will be to rea-

son about them. Keep in mind the saying that one good definition is worth three theorems.

For structurally recursive functions, Lean can automatically prove termination. For more general recursive schemes, the termination check may fail. Sometimes it does so for a good reason, as in the following example:

```
-- fails def illegal : \mathbb{N} \to \mathbb{N} | n := illegal n + 1
```

If Lean were to accept this definition, we could exploit it to prove that o = 1, by subtracting illegal n on each side of the equation illegal  $n = illegal \ n + 1$ . From o = 1, we could derive false, and from false, we could derive anything, including the three-color theorem. Clearly, we do not want that.

If we had used constant and axiom, nothing could have saved us:

```
constant immoral : \mathbb{N} \to \mathbb{N}

axiom immoral_eq (n : \mathbb{N}) :
    immoral n = immoral n + 1

lemma proof_of_false :
    false :=
    have immoral \mathbb{O} = immoral \mathbb{O} + 1 :=
    immoral_eq \mathbb{O},
have immoral \mathbb{O} - immoral \mathbb{O} = immoral \mathbb{O} + 1 - immoral \mathbb{O} :=
    by cc,
have \mathbb{O} = 1 :=
    by simp [*] at *,
show false, from
    by cc
```

Another reason for preferring def is that the defining equations are used to compute. Tactics such as refl that unify up to computation become stronger each time we introduce a definition, and

the diagnosis commands #eval and #reduce can be used on defined constants.

The observant reader will have noticed that the above definitions of factorial are mathematically wrong: fact shockingly returns o regardless of the argument, and fact<sub>2</sub> o should give 1, not o. We quite literally facted up. These embarrassing mistakes remind us to test our definitions and prove some properties about them. Although flawed axioms arise now and then, what is much more common are definitions that fail to capture the intended concepts. Just because a function is called fact does not mean that it actually computes factorials.

## 4.4 Pattern Matching Expressions

Pattern matching is possible not only at the top level of a def command but also deeply within terms, via a match expression. The construct has the following general syntax:

```
match term<sub>1</sub>, ..., term<sub>m</sub> with
| pattern<sub>11</sub>, ..., pattern<sub>1m</sub> := result<sub>1</sub>
| pattern<sub>n1</sub>, ..., pattern<sub>nm</sub> := result<sub>n</sub>
end
```

The construct is vaguely reminiscent of switch in many programming languages. An English translation of the above match expression follows:

```
Consider the terms term_1, ..., term_m.
If they are respectively of the forms pattern_{11}, ..., pattern_{1m}, then yield result_1.
```

If they are respectively of the forms  $pattern_{n1}$ , ...,  $pattern_{nm}$ , then yield  $result_n$ .


Notice the differences with the pattern matching syntax in def: The patterns are separated by commas and not by spaces. This reduces the number of required parentheses. The patterns may contain variables, constructors, and nameless placeholders (\_). The result<sub>i</sub> expressions may refer to the variables introduced in the corresponding patterns.

The following function definition demonstrates the syntax of pattern matching within expressions:

```
\begin{array}{lll} \text{def bcount } \{\alpha: \mathsf{Type}\} \; (\mathsf{p}: \alpha \to \mathsf{bool}) : \mathsf{list} \; \alpha \to \mathbb{N} \\ |\; [] & := 0 \\ |\; (\mathsf{x}:: \mathsf{xs}) := \\ & \mathsf{match} \; \mathsf{p} \; \mathsf{x} \; \mathsf{with} \\ |\; \mathsf{tt} := \mathsf{bcount} \; \mathsf{xs} + 1 \\ |\; \mathsf{ff} := \mathsf{bcount} \; \mathsf{xs} \\ & \mathsf{end} \end{array}
```

The bcount function counts the number of elements in a list that satisfy the given predicate p. The predicate's codomain is bool. As a general rule, we will use type bool, of Booleans, within programs and use the type Prop, of propositions, when stating properties of programs. The two values of type bool are called ff and tt.<sup>2</sup> The connectives are called bor (infix: ||), band (infix: &&), bnot, and bxor.

The following diagram shows the interpretations of bool and and Prop:



<sup>&</sup>lt;sup>2</sup>Lean also allows us to use true and false, but these are then implicitly converted from Prop to bool. We generally recommend to avoid relying on such implicit coercions. There is unfortunately no way to disable them.

Dots represents elements, circles represents types, and the rectangle represents a type of types. We see that bool are interpreted as a set with two values, whereas Prop consists in an infinite number of propositions (the types), each of which has zero or more proofs (the elements). We will refine this picture in Chapter 11.

We cannot match on a proposition of type Prop, but we can use if—then—else instead. For example, the min operator on natural numbers operator can be defined as follows:

```
def min (a b : \mathbb{N}) : \mathbb{N} := if a < b then a else b
```

This requires a decidable (i.e., executable) proposition. This is the case for  $\leq$ : Given concrete values for the arguments, such as 35 and 49, Lean can reduce 35  $\leq$  49 to true. Lean keeps track of decidability using a mechanism called type classes, which will be explained below.

#### 4.5 Structures

Lean provides a convenient syntax for defining records, or *structures* as they are also called. These are essentially nonrecursive single-constructor inductive types but with some syntactic sugar.

The definition below introduces a structure called rgb with three fields of type  $\mathbb N$  called red, green, and blue:

```
structure rgb :=
(red green blue : N)
```

This definition has roughly the same effect as the following commands:

```
\begin{array}{l} \text{inductive } \textbf{rgb} : \textbf{Type} \\ \mid \textbf{mk} : \mathbb{N} \to \mathbb{N} \to \mathbb{N} \to \textbf{rgb} \\ \\ \textbf{def } \textbf{rgb.red} : \textbf{rgb} \to \mathbb{N} \\ \mid (\textbf{rgb.mk } \textbf{r} \_ \_) := \textbf{r} \end{array}
```

4.5. Structures 89

```
\begin{array}{l} \text{def rgb.green} : \text{rgb} \rightarrow \mathbb{N} \\ \mid (\text{rgb.mk} \_ \text{g} \_) := \text{g} \\ \\ \text{def rgb.blue} : \text{rgb} \rightarrow \mathbb{N} \\ \mid (\text{rgb.mk} \_ \_ \text{b}) := \text{b} \end{array}
```

We can define a new structure as the extension of an existing structure. The definition below extends rgb with a fourth field, called alpha:

```
structure rgba extends rgb :=
(alpha : N)
```

The general syntax to define structures is

```
structure structure-name
     (params<sub>1</sub>: type<sub>1</sub>) ... (params<sub>k</sub>: type<sub>k</sub>)
     [extends structure<sub>1</sub>, ..., structure<sub>m</sub>] :=
(field-names<sub>1</sub>: field-type<sub>1</sub>)
     :
(field-names<sub>n</sub>: field-type<sub>n</sub>)
```

The parameters  $params_1, ..., params_k$  are effectively additional fields, but unlike the fields field-names<sub>1</sub>, ..., field-names<sub>n</sub>, they are also stored in the type, as arguments to the type constructor (structure-name).

Values can be specified in a variety of syntaxes:

```
def pure_red : rgb :=
{ red := oxff,
  green := oxoo,
  blue := oxoo }

def semitransparent_red : rgba :=
{ alpha := ox7f,
  ..pure_red }
```

The definition of semitransparent\_red copies all the values from pure\_red (indicated by the ..pure\_red syntax) except for the alpha field, which it overrides.

Next, we define an operation called shuffle:

```
def shuffle (c : rgb) : rgb :=
{ red := rgb.green c,
  green := rgb.blue c,
  blue := rgb.red c }
```

The definition relies on the generated selectors rgb.red, rgb. green, and rgb.blue. Instead of rgb.red c, we could also write c.red, and similarly for the other fields. Sometimes we will see these forms in Lean's output, even if we do not use them ourselves.

Applying shuffle three times in a row is the same as not applying it at all:

```
lemma shuffle_shuffle (c : rgb) :
    shuffle (shuffle (shuffle c)) = c :=
begin
    cases' c,
    refl
end
```

The proof relies on cases', a relative of induction'. It performs a case distinction on its argument but does not generate induction hypotheses. For rgb, the invocation cases' c transforms a goal  $\vdash$  P[c] into a single subgoal, rgb:  $\mathbb{N} \vdash$  P[rgb.mk rgb]. We could also have used induction' c. If you find yourself hesitating between induction' and cases', you can always choose induction ' and see afterwards if you needed the induction hypotheses. If you did not, it is good style to document that fact by substituting cases' for induction'.

In a structured proof, we can use match expressions to perform a case distinction. For example:

```
lemma shuffle_shuffle_shuffle_(c : rgb):
    shuffle (shuffle (shuffle c)) = c :=
match c with
| rgb.mk _ _ _ := eq.refl _
end
```

Recall from Section 2.4 that eq. refl is the property  $\forall a$ , a = a.

## 4.6 Type Classes

Type classes are a mechanism that was popularized by Haskell and that is present in several proof assistants. In Lean, a *type class* is a structure type combining abstract constants and their properties. A type can be declared an instance of a type class by providing concrete definitions for the constants and proving that the properties hold. Based on the type, Lean retrieves the relevant instance.<sup>3</sup>

A simple example is the type class inhabited, which requires only a constant inhabited.default and no properties:

```
\mathfrak{D}[\text{class}] structure inhabited (\alpha : Type) : Type := (default : \alpha)
```

The syntax is the same as for structures, except with the attribute a[class] in front. The parameter  $\alpha$  represents an arbitrary type that could be a member of this class.

Any type that has at least one element can be registered as an instance of this class. For example, we can register  $\mathbb N$  by choosing an arbitrary number to be the default value:

```
\mathfrak{D}[instance] def nat.inhabited : inhabited \mathbb{N} := \{ default := \emptyset \}
```

This specifies a value called nat.inhabited of type inhabited  $\mathbb{N}$ . Thanks to the  $\mathfrak{D}[$ instance] attribute, the structure value is registered as *the* canonical instance to use whenever a structure of type inhabited  $\mathbb{N}$  is desired. In the global table storing type class instances, there is now an entry

```
inhabited \mathbb{N} \mapsto \mathsf{nat.inhabited}
```

For lists, the empty list is an obvious default value that can be constructed even if  $\alpha$  is not inhabited:

```
\mathfrak{Q}[\mathsf{instance}] def list.inhabited \{\alpha : \mathsf{Type}\} :
```

<sup>&</sup>lt;sup>3</sup>Despite its name, Lean's type class mechanism is more closely related to Scala's implicit arguments than to Haskell's type classes.

```
inhabited (list \alpha) := { default := [] }
```

This adds the following entry to the global table:

```
inhabited (list \alpha) \mapsto list.inhabited
```

Sometimes we may want to supply several instances of a given type class for the same type. Lean will then choose the first matching instance it finds.

As an example, recall that finite functions of type  $\alpha \to \beta$  can be represented by their function tables, of type  $\beta \times \cdots \times \beta$  (with  $|\alpha|$  copies of  $\beta$ ). Accordingly,  $|\alpha \to \beta| = |\beta|^{|\alpha|}$ . To make this  $\emptyset$ , we must have both  $|\beta| = \emptyset$  and  $|\alpha| \neq \emptyset$ . In other words, the type  $\alpha \to \beta$  is inhabited if either (1)  $\beta$  is inhabited or (2)  $\alpha$  is *not* inhabited. Case (1) is easy:

```
<code>@[instance] def fun.inhabited {} \alpha \beta : Type} [inhabited $\beta$] : inhabited ($\alpha \to \beta$) := { default := $\lambda a : $\alpha$, inhabited.default $\beta$ }</code>
```

The instance relies itself on an instance of the same type class but on a different type. This frequently occurs and is part of the reason why we speak of type class *search* and not lookup.

For case (2), there is no type class for *non*inhabitation, but we know that the type empty is not inhabited, so it will do:

Notice the trivial match that implements the body of a function that can never be called, because no value of type empty can be passed as argument. The definition of empty has zero introduction rules, so the corresponding match has zero cases.

Finally, the type  $\alpha \times \beta$  of pairs, also called product type, contains values of the form (a, b), where a :  $\alpha$  and b :  $\beta$ . Given a

pair ab, the first and second components can be extracted by writing prod.fst ab and prod.snd ab. To provide an inhabitant of  $\alpha \times \beta$ , we need both an inhabitant of  $\alpha$  and an inhabitant of  $\beta$ :

Using the inhabited type class, we can define the head operation on lists: the function that returns the first element of a list. Because an empty list contains no elements, there is no meaningful value we can return in that case. Given a type that belongs to the inhabited type class, we can simply return the default value:

```
\begin{array}{lll} \text{def head } \{\alpha : \text{Type}\} \text{ [inhabited } \alpha \text{] : list } \alpha \to \alpha \\ | \text{ []} & := \text{ inhabited.default } \alpha \\ | \text{ (x :: }_{}) := \text{ x} \end{array}
```

We require that  $\alpha$  belongs to inhabited by writing [inhabited  $\alpha$ ]. This allows us to access inhabited.default in the definition.

The syntax [inhabited  $\alpha$ ] adds an implicit argument to the head constant. But unlike for other implicit arguments, Lean performs not only type inference but also a type class search through all declared instances to determine the value of this argument. Thus, when running the command

```
#eval head ([] : list \mathbb{N}) -- result: 0
```

Lean will look for an inhabited  $\mathbb N$  instance and find nat.inhabited, the instance we declared above. In that declaration, we set default to be  $\mathbb O$  and hence this is what #eval prints. If multiple instances are applicable and Lean chooses the wrong one, we can use the  $\mathbb O$  syntax to transform type class arguments into explicit arguments and supply the desired type class instance. This is the main reason why we give names to the type class instances.

Let us take a closer look at inhabited.default:

```
inhabited.default (\alpha: Type) [i:inhabited \alpha]: \alpha
```

Notice that the selectors of plain structures use parentheses ( ) whereas those of type classes use square brackets [ ]. When we used inhabited default  $\alpha$  to define head, Lean looked for an instance of inhabited  $\alpha$  in the global table of registered instances and in the local context. The global table contained an entry for inhabited  $\mathbb N$  but none that matched inhabited  $\alpha$ . On the other hand, the local context contained an anonymous parameter of type inhabited  $\alpha$ , which could be used.

Lean's core library defines list.head exactly as we did. In practice, almost all types are nonempty (with the notable exceptions of empty and false), so the inhabited restriction is hardly an issue. Regardless, lists over empty types are uninteresting—the only possible value is [].

We can prove abstract lemmas about the inhabited type class, such as

```
lemma head_head \{\alpha : \mathsf{Type}\}\ [\mathsf{inhabited}\ \alpha]\ (\mathsf{xs} : \mathsf{list}\ \alpha) : \mathsf{head}\ [\mathsf{head}\ \mathsf{xs}] = \mathsf{head}\ \mathsf{xs}
```

The assumption [inhabited  $\alpha$ ] is needed to use the operator head on lists of type list  $\alpha$ . If we omit this assumption, Lean will raise an error telling us that type class synthesis failed.

There are more type classes with only a constant but no properties, including

These syntactic type classes introduce constants that are used in many different contexts with different semantics. For example, one can stand for the natural number 1, the integer 1, the real 1, the

identity matrix, and many other concepts of 1. The main purpose of these type classes is to form the foundation for a rich hierarchy of algebraic type classes (groups, monoids, ring, fields, etc.) and to allow the overloading of common mathematical symbols such as +, \*, 0, 1, and  $^{-1}$ .

Syntactic type classes do not impose serious restrictions on the types that can be declared instances, except that a type of type class has\_zero or has\_one must be inhabited. In contrast, the *semantic* type classes contain properties that restrict how the given constants behave.

In Section 2.6, we encountered the following type classes:

```
class is_commutative (\alpha : Type) (f : \alpha \to \alpha \to \alpha) := (comm : \forall a b, f a b = f b a)

class is_associative (\alpha : Type) (f : \alpha \to \alpha \to \alpha) := (assoc : \forall a b c, f (f a b) c = f a (f b c))
```

This time, the associations are not from a type to a constant, but from a type and a function to a property. Lean does not mind the abuse: Although they are called type classes, Lean's type classes are very flexible and can be used to express all sorts of constraints.

Conceptually, is\_commutative is a dependent type of triples ( $\alpha$ , f, comm), and similarly for is\_associative. The type of f depends on  $\alpha$ , and the type of comm depends on  $\alpha$  and f. Although they are parameters,  $\alpha$  and f are also stored along with comm. But the type class mechanism treats the parameters differently from the fields: The parameters are treated as *input* to the type class search, whereas the fields are *output*.

In Section 2.6, we registered our add function on  $\mathbb N$  as a commutative and associative operation:

```
@[instance] def add.is_commutative : is_commutative N add :=
{ comm := add_comm }

@[instance] def add.is_associative : is_associative N add :=
{ assoc := add_assoc }
```

Whenever we try to access @is\_commutative.comm  $\mathbb N$  add, we obtain add\_comm, and similarly for @is\_associative.assoc  $\mathbb N$  add. The cc tactic tries to look up the comm and assoc properties for all binary operators in the problem and exploits the properties whenever they are present.

The general syntax to define a type class is as follows:

```
@[class] structure class-name
          (params1: type1) ... (paramsk: typek)
        [extends structure1, ..., structurem] :=
(constant-names1: constant-type1)
          :
(constant-namesn: constant-typen)
(property-names1: proposition1)
          :
(property-namesp: propositionp)
```

The general syntax to instantiate a type class is as follows:

## 4.7 Lists

Lean provides a rich library of functions on finite lists. In this section, we will review some of them, and we will define some of our own; these are good exercises to familiarize ourselves with functional programming in Lean.

4.7. Lists 97

In the first example, we show how to exploit injectivity of constructors. The cases' tactic can be used to exploit injectivity to simplify equations in which both sides have the same constructor applied. In the proof below, the equation before simplification is x :: xs = y :: ys:

```
lemma injection_example {α : Type} (x y : α) (xs ys : list α)
    (h : list.cons x xs = list.cons y ys) :
    x = y ∧ xs = ys :=
begin
    cases' h,
    cc
end
```

As a result of the case distinction, y is replaced by x and ys is replaced by xs throughout the goal, yielding the subgoal

```
h: x :: xs = x :: xs \vdash x = x \land xs = xs
```

The clear h tactic removes the useless hypothesis h : x :: xs = x :: xs.

The cases' tactic is also useful when the constructors are different, to detect the impossible case:

```
lemma distinctness_example \{\alpha: \mathsf{Type}\}\ (y:\alpha)\ (ys: \mathsf{list}\ \alpha) (h: [] = y:: ys): false := by cases' h
```

The first operation we define is a *map function*: a function that applies its argument f—which is itself a function—to all elements stored in a container.

Notice that because f does not change in the recursive call, we put it as a parameter of the entire definition. This means we must write map xs and not map f xs in the recursive call. However, when

using the function later, we need to pass an argument for f. The alternative, which is the only option for arguments that change in recursive calls, would be as follows:

A basic property of map functions is that they have no effect if their argument is the identity function ( $\lambda x$ , x):

```
lemma map_ident {α : Type} (xs : list α) :
  map (λx, x) xs = xs :=
begin
  induction' xs,
  case nil {
    refl },
  case cons : y ys {
    simp [map, ih] }
end
```

Another basic property is that successive maps can be compressed into a single map, whose argument is the composition of the functions involved:

```
lemma map_comp \{\alpha \ \beta \ \gamma : \text{Type}\}\ (f:\alpha \to \beta)\ (g:\beta \to \gamma) (xs:list \alpha):

map g (map f xs) = map (\lambdax, g (f x)) xs:=

begin

induction' xs,

case nil \{

refl \},

case cons: y ys \{

simp [map, ih] \}
```

When introducing new operations, it is useful to show how these behave when used in combination with other operations. Here is an example: 4.7. Lists 99

```
lemma map_append \{\alpha \ \beta : \ \mathsf{Type}\}\ (f:\alpha \to \beta)\ (xs\ ys: \ \mathsf{list}\ \alpha): map f(xs ++ ys) = \mathsf{map}\ f(xs ++ \mathsf{map}\ f(xs ++
```

Remarkably, the last three proofs are textually identical. These are typical induction'-refl-simp proofs.

The next list operation removes the first element of a list, returning the tail:

For [], we simply return [] as its own tail.

The counterpart of tail is a function that extracts the first element of a list. We already reviewed one solution in Section 4.6. Another possible definition uses an option wrapper:

```
def head_opt \{\alpha : \mathsf{Type}\} : \mathsf{list} \ \alpha \to \mathsf{option} \ \alpha
| [] := option.none
| (x :: \_) := \mathsf{option.some} \ x
```

The type option  $\alpha$  has two constructors: option.none and option. some a, where a:  $\alpha$ . We use option.none when we have no meaningful value to return and option.some otherwise. We can think of option.none as the null pointer of functional programming, but unlike null pointers (and null references), the type system guards against unsafe dereferences. To retrieve the value stored in an option, we must pattern-match. Schematically:

```
match head_opt xs with
| option.none := handle_the_error
| option.some x := do_something_with_value x
end
```

We cannot simply write do\_something\_with\_value (head\_opt xs), because this would be type-incorrect. The type system forces us to think about error handling.

Using the power of dependent types, another way to implement a partial function is to specify a precondition. The callee must then pass a proof that the precondition is satisfied as argument:

The head\_pre function takes two explicit arguments. The first argument, xs, is a list. The second argument, hxs, is a proof of xs  $\neq$  []. Since the type (xs  $\neq$  []) of the second argument depends on the first argument, we must use the dependent type syntax  $\forall$ xs : list  $\alpha$ , rather than list  $\alpha \rightarrow$  to name the first argument. The result of the function is a value of type  $\alpha$ ; thanks to the precondition, there is no need for an option wrapper.

The precondition hxs is used to rule out the case where xs is []. In that case, hxs is a proof of []  $\neq$  [], which is impossible. The cc tactic derives the contradiction and exploits it to derive an arbitrary  $\alpha$ . From a contradiction, we can derive anything, even an inhabitant of  $\alpha$ .

We can then invoke the function as follows:

```
#eval head_pre [3, 1, 4] (by simp)
```

This prints 3.

Let us move on. Given two lists  $[x_1, ..., x_n]$  and  $[y_1, ..., y_n]$  of the same length, the zip operation constructs a list of pairs  $[(x_1, y_1), ..., (x_n, y_n)]$ :

The function is also defined if one list is shorter than the other. For example, zip[a, b, c][x, y] = [(a, x), (b, y)]. Notice that the

4.7. Lists 101

recursion, with three cases, deviates slightly from the structural recursion schema.

The length of a list is defined by recursion:

We can say something interesting about the length of zip's result—namely, it is the minimum of the lengths of the two input lists:

```
lemma length_zip {α β : Type} (xs : list α) (ys : list β) :
  length (zip xs ys) = min (length xs) (length ys) :=
begin
  induction' xs,
  case nil {
    refl },
  case cons : x xs' {
    cases' ys,
    case nil {
    refl },
    case cons : y ys' {
        simp [zip, length, ih ys', min_add_add] } }
end
```

The proof above teaches us yet another trick. The induction hypothesis is

```
ih : \forall \{\beta\} ys : list \beta, length (zip xs ys) = min (length xs) (length ys)
```

The syntax  $\forall \{\beta\}$  is new but suggestive: It indicates that the type  $\beta$  is an implicit argument of ih. In contrast, ys is explicit—and from it,  $\beta$  can be inferred.

Why are there  $\forall$ -quantifiers? The induction' tactic automatically generalized the lemma statement so that the induction hypothesis is not restricted to same fixed  $\beta$  and ys as the proof goal but can be used for arbitrary values of  $\beta$  and ys. Such flexibility is rarely needed for types (e.g.,  $\beta$ ), but here we need it for ys, because we want to instantiate the quantifier with ys's tail (called ys') and not with ys itself.

The proof relies on a lemma about the min function that we need to prove ourselves:

```
lemma min_add_add (l m n : N) :
    min (m + l) (n + l) = min m n + l :=
begin
    cases' classical.em (m ≤ n),
    case inl {
        simp [min, h] },
    case inr {
        simp [min, h] }
end
```

Recall the definition min a b = (if a  $\leq$  b then a else b). To reason about min, we often need to perform a case distinction on the condition a  $\leq$  b. This is achieved using cases' classical.em (a  $\leq$  b). This creates two subgoals: one with a  $\leq$  b as a hypothesis and one with  $\neg$  a < b.

Here are two different ways to perform a case distinction on a proposition in a structured proof:

```
lemma min_add_add2 (l m n : \mathbb{N}) :
    min (m + l) (n + l) = min m n + l :=
match classical.em (m \leq n) with
| or.inl h := by simp [min, h]
| or.inr h := by simp [min, h]
end

lemma min_add_add3 (l m n : \mathbb{N}) :
    min (m + l) (n + l) = min m n + l :=
if h : m \leq n then
    by simp [min, h]
else
    by simp [min, h]
```

We see again that the mechanisms that are available to write functional programs, such as match and if—then—else, are also available for writing structured proofs (which are, after all, terms). We

4.7. Lists 103

can now add a few rows to the table presented at the end of Section 3.7:

Tactical proof	Structured proof	Raw proof term
<pre>cases' t, cases' classical.em Q,</pre>	<pre>match t with _ end if Q then _ else _</pre>	<pre>match t with _ end if Q then _ else _</pre>

We conclude with a distributivity law about map and zip, expressed using the prod.fst and prod.snd selectors on pairs:

```
lemma map_zip \{\alpha \ \alpha' \ \beta \ \beta' : Type\} (f:\alpha \to \alpha') (g:\beta \to \beta') : \forall xs \ ys, map (\lambda ab : \alpha \times \beta, (f \ (prod.fst \ ab), g \ (prod.snd \ ab))) (zip \ xs \ ys) = zip \ (map \ f \ xs) \ (map \ g \ ys) | (x :: xs) \ (y :: ys) := by \ simp \ [zip, map, map_zip \ xs \ ys] | [] = = by \ refl | (::) [] = by \ refl
```

The patterns on the left correspond exactly to the patterns used in the definition of zip. This is simpler than performing the induction on xs and the case distinction on ys separately, as we did when we proved length\_zip. Good proofs often follow the structure of the definitions they are based on.

In the definition of zip and in the proof of map\_zip, we were careful to specify three nonoverlapping patterns. It is also possible to write equations with overlapping patterns, as in

```
def f {\alpha : Type} : list \alpha \rightarrow \cdots | [] := ... | xs := ... xs ...
```

Since the patterns are applied sequentially, the above command defines the same function as

```
\begin{array}{lll} \text{def } f_2 & \{\alpha : \text{Type}\} : \text{list } \alpha \rightarrow \cdots \\ | & [] & := \dots \\ | & (x :: xs) := \dots (x :: xs) \dots \end{array}
```

We generally recommend the latter, more explicit style, because it leads to fewer surprises, especially in proofs.

## 4.8 Binary Trees

Inductive types with constructors taking several recursive arguments define tree-like objects. *Binary trees* have nodes with at most two children. A possible definition of binary trees follows:

```
inductive btree (\alpha : Type) : Type | empty {} : btree | node : \alpha \rightarrow btree \rightarrow btree \rightarrow btree
```

(The  $\{\}$  annotation is often used with nullary constructors of polymorphic types. It indicates that all parameters—here,  $\alpha$ —should be implicitly derived from the context. This enables us to write tree.empty instead of tree.empty \_, or tree.empty  $\mathbb{N}$ , etc.)

With binary trees, structural induction produces two induction hypotheses, one for each subtree of an inner node. To prove a goal  $t:tree\ \alpha\ \vdash\ P[t]$  by structural induction on t, we need to show the subgoals

```
\vdash P[tree.empty]
a: \alpha, lr: tree \alpha, ih_l: P[l], ih_r: P[r] \vdash P[tree.node a l r]
```

The tree counterpart to list reversal is the mirror operation:

```
def mirror \{\alpha : \mathsf{Type}\} : btree \alpha \to \mathsf{btree}\ \alpha
| btree.empty := btree.empty
| (btree.node a l r) := btree.node a (mirror r) (mirror l)
```

Mirroring can be defined directly, without appealing to some append operation. As a result, reasoning about mirror is simpler than reasoning about reverse, as we can see below:

```
lemma mirror_mirror {α : Type} (t : btree α) :
  mirror (mirror t) = t :=
begin
  induction' t.
```

```
case empty {
   refl },
   case node : a l r ih_l ih_r {
     simp [mirror, ih_l, ih_r] }
end
```

A more detailed informal proof would be as follows:

The proof is by structural induction on t.

CASE tree.empty: We must show that mirror (mirror tree.empty) = tree.empty. This follows directly from the definition of mirror.

CASE tree.node a l r: The induction hypotheses are

To achieve the same level of detail in the Lean proof, we could use a calculational block (Section 3.4) instead of simp:

```
... = btree.node a (mirror (mirror l)) (mirror (mirror r)) :
by refl
... = btree.node a l (mirror (mirror r)) :
by rw mirror_mirror<sub>2</sub> l
... = btree.node a l r :
by rw mirror_mirror<sub>2</sub> r
```

The following lemma will be useful in Chapter 5:

## 4.9 Case Distinction and Induction Tactics

The following tactics are useful when performing case distinctions, with or without induction hypotheses.

#### cases'

```
cases' term [with name_1 ... name_n]
```

The cases' tactic performs a case distinction on the specified term. This gives rise to as many subgoals as there are constructors in the definition of the term's type. The tactic behaves roughly the same as induction' except that it does not produce induction hypotheses and it eliminates impossible cases.

The optional names  $name_1, ..., name_n$  are used for any emerging variables or hypotheses, to override the default names.

Note that the standard Lean tactic for case distinction is called cases. The primed variant is provided by mathlib. It is more user-friendly and is the only one we will use.

```
cases' hypothesis-of-the-form-l-equals-r
```

The cases' tactic can also be used on a hypothesis h of the form l=r. It matches r against l and replaces all occurrences of the variables occurring in r with the corresponding terms in l everywhere in the goal. The remaining hypothesis l=l can be removed using clear h if desired. If r fails to match l, no subgoals emerge; the proof is complete.

```
cases' classical.em (proposition) [with name, name]
```

The cases' tactic can also be used to perform a case distinction on a proposition. Two cases emerge: one in which the proposition is true and one in which it is false. The optional names name, and name, are used for hypotheses in the true and false cases, respectively.

#### case

```
case name \left[: name_1 \dots name_n\right] \{ \dots \}
```

The case tactic (with no s) can be used in tandem with cases' or induction' to select a subgoal corresponding to the constructor called name and focus on it.

The optional names  $name_1, ..., name_n$  are used for any emerging variables or hypotheses, to override the default names.

# 4.10 Dependent Inductive Types

The inductive types list  $\alpha$  and btree  $\alpha$  fall within the simply typed fragment of Lean. Inductive types may also depend on (non-type) terms. A typical example is the type of lists of length n, or vectors:

```
\begin{array}{lll} \text{inductive vec } (\alpha: \mathsf{Type}) : \mathbb{N} \to \mathsf{Type} \\ \mid \mathsf{nil} \ \{\} & : \mathsf{vec} \ \mathsf{o} \\ \mid \mathsf{cons} \ (\mathsf{a} : \alpha) \ \{\mathsf{n} : \mathbb{N}\} \ (\mathsf{v} : \mathsf{vec} \ \mathsf{n}) : \mathsf{vec} \ (\mathsf{n} + \mathbf{1}) \end{array}
```

Thus, the term vec.cons 3 (vec.cons 1 vec.nil) has type vec  $\mathbb N$  2. By encoding the vector length in the type, we can provide more precise information about the result of functions. A function such as vec.reverse, which reverses a vector, would map a value vec  $\alpha$  n to another value of the same type, with the same n. And vec.zip could require its two arguments to have the same length. Fixed-length vectors and matrices are also useful in mathematics.

Unfortunately, this more precise information comes at a cost. Dependent inductive types cause difficulties when the terms they depend on are provably equal but not syntactically equal up to computation (e.g., m + n vs. n + m). In this guide, we will generally avoid dependent inductive types. They are briefly covered in this section for completeness. To put it unambiguously: This is not exam material.

The definitions below introduce conversions between lists and vectors:

The list\_of\_vec conversion takes a type  $\alpha$ , a length n, and a vector of length n over  $\alpha$  as arguments and returns a list over  $\alpha$ . Although we do not care about the length n, it still needs to be an argument because it appears in the type of the third argument. The first two arguments are implicit, which is reasonable since they can be inferred from the type of the third argument.

The  $vec_of_list$  conversion takes a type  $\alpha$  and a list over  $\alpha$  as arguments and returns a vector of the same length as the list. Lean's type checker is strong enough to determine that the two right-hand sides have the desired type.

By the PAT principle, proofs are analogous to function definitions. Let us verify that converting a vector to a list preserves its length:

To prove a goal  $v : vec \alpha n \vdash P[v]$  by structural induction on v, we might naively think that it suffices to show the following two subgoals:

```
\vdash P[\text{vec.nil}] m: \mathbb{N}, a: \alpha, u: \text{vec } \alpha \text{ m}, \text{ ih}: P[u] \vdash P[\text{vec.cons a } u]
```

This is naive because the subgoals are not even type-correct: The hole in P[] has type vec  $\alpha$  n (the type of its original dweller, v), so we cannot simply plug vec.nil, u, or vec.cons a u—which have types vec  $\alpha$  0, vec  $\alpha$  m, and vec  $\alpha$  (m + 1)—into that hole. We must massage P each time, replacing n with 0, m, or m + 1. Using the notation P<sub>t</sub>[] for the variant of P[] where all occurrences of n are replaced by term t, we get

```
\vdash P_{o}[\text{vec.nil}] m: \mathbb{N}, a: \alpha, u: \text{vec } \alpha \text{ m, ih}: P_{m}[u] \vdash P_{m+1}[\text{vec.cons a } u]
```

Proofs by case distinction using the cases' tactic are similar, but without the induction hypothesis. Often, the length n will be not a variable but a complex term. Then the replacement of n in P[] might not be intuitively meaningful. With cases', the corresponding subgoal is silently eliminated. Thus, a case distinction on a value of type  $vec\ \alpha$  o will yield only one subgoal, of the form P[vec.nil], since o could never be equal to a term of the form P[vec.nil].

Dependently typed pattern matching is subtle, because the type of the value we match on may change according to the constructor. Given  $v: vec \ \alpha$  n, we might be tempted to write

but this is just as naive as our first induction proof attempt above. Because the term n in the type  $\text{vec } \alpha$  n may change depending on the constructor, we must pattern-match on it as well:

```
match n, v with
| 0, vec.nil := ...
| m + 1, vec.cons a u := ...
end
```

Showing the implicit arguments, we have

```
match n, v with | \  \, 0, \qquad \  \, \text{ ovec.nil } \alpha \qquad \qquad := \dots \\ | \  \, \text{m + 1, @vec.cons } \alpha \  \, \text{a m u := } \dots \\ \text{end}
```

Often, it is sufficient to put placeholders in the first column:

It may seem paradoxical to pattern-match on n only to ignore the result, but without it Lean cannot infer the second implicit argument to vec.cons. In this respect, cases' is more user-friendly than match.

Incidentally, Lean's core libraries define a type of fixed-length vectors, called vector  $\alpha$  n, as a subtype of list  $\alpha$ :

```
def vector (\alpha : Type) (n : \mathbb{N}) := \{xs : list <math>\alpha // list.length \ xs = n\}
```

In other words, the type  $vector \alpha$  n consists of the values xs of type list  $\alpha$  such that xs is of length n. This makes it possible to reuse Lean's extensive list library. Subtyping will be explained in Chapter 11.

## 4.11 Summary of New Lean Constructs

#### **Declaration**

structure introduces a structure type and its selectors

### **Attributes**

a[class] declares a structure type as a type class

a[instance] registers a structure value as a type class instance

### **Term Language**

```
if ... then ... else ...

performs a case distinction on a decidable proposition

match ... with ... end

performs pattern matching
```

#### **Tactics**

case focuses on a subgoal and names its variables and

hypotheses

cases' performs a case distinction

# **Chapter 5**

# **Inductive Predicates**

Inductive predicates, or inductively defined propositions, are a convenient way to specify functions of type  $\cdots \to \mathsf{Prop}$ . They are reminiscent of formal systems and of Prolog-style logic programming. But Lean offers a much stronger logic than Prolog, so we need to do some work to establish theorems instead of just running the Prolog interpreter. A possible view of Lean:

Lean = functional programming + logic programming + more logic

# **5.1 Introductory Examples**

Unless you have been exposed to Prolog or logic programming, you will probably wonder what inductive predicates are and why they are useful. We start by reviewing three examples that demonstrate the variety of uses: even numbers, tennis games, and the reflexive transitive closure.

#### **5.1.1 Even Numbers**

Mathematicians often define sets as the smallest set that meets some criteria. Consider this definition:

The set *E* of *even natural numbers* is defined as the smallest set *S* closed under the following rules:

- (1)  $0 \in S$ ;
- (2) for every  $k \in \mathbb{N}$ , if  $k \in S$ , then  $k + 2 \in S$ .

Such a set exists by the Knaster-Tarski theorem.

(The last sentence is often left implicit.) It is easy to convince ourselves that *E* contains all the even numbers and only those. Let us put on our mathematician's hat and prove that 4 is even:

By rule (1), we have  $o \in E$ .

Hence, by rule (2) (with k := 0), we have  $2 \in E$ .

Thus, by rule (2) (with k := 2), we have  $4 \in E$ , as desired.

By contrast, computer scientists might use a formal system consisting of two derivation rules to specify the same set:

$$\frac{k \in E}{constant} ZERO \qquad \frac{k \in E}{k + 2 \in E} ADDTWO_k$$

A proof is then a derivation tree:

$$\frac{\overline{O \in E}}{2 \in E} \begin{array}{l} \mathsf{ZERO} \\ \hline 2 \in E \\ \hline 4 \in E \end{array} \mathsf{ADDTWO}_2$$

The proof is forward if we read it downwards and backward if we read it upwards.

Inductive predicates are the logicians' way to achieve the same result. In Lean, instead of a set, we would define a characteristic predicate inductively:

```
inductive even : \mathbb{N} \to \mathsf{Prop} | zero : even o | add_two : \forall k : \mathbb{N}, even k \to \mathsf{even} (k + 2)
```

This should look familiar. We have used the exact same syntax, except with Type instead of Prop, to define inductive types. Inductive types and inductive predicates are the same mechanism in Lean.

The command above defines a unary predicate even as well as two introduction rules even.zero and even.add\_two that can be used to prove goals of the form  $\vdash$  even.... Recall that an introduction rule for a symbol (e.g., even) is a lemma whose conclusion contains that symbol (Section 3.3). By the PAT principle, even n can be viewed as a dependent inductive type like vec  $\alpha$  n (Section 4.10), and even.zero and even.add\_two as constructors like vec.nil and vec.cons.

If we translate the Lean definition back into English, we get something similar to the Knaster–Tarski-style definition above:

The following clauses define the even numbers.

- (1) o is even;
- (2) any number of the form k + 2 is even, if k is even.

All other number is not even.

It is worth nothing that inductively defined symbols have no definition. Thus, we cannot unfold their definition using simp, any more than we can unfold the definition of an inductive type such as list. The only reasoning principles available are introduction and elimination.

As a warm-up exercise, here is a proof of even 4:

```
lemma even_4 :
    even 4 :=
have even_0 : even 0 :=
    even.zero,
have even_2 : even 2 :=
```

```
even.add_two _ even_0,
show even 4, from
even.add_two _ even_2
```

For example, the proof term even.add\_two \_ even\_o has type even (o + 2), which is syntactically equal to even 2 up to computation and hence equal from the type checker's point of view. The underscore stands for o.

Thanks to the "no junk" guarantee of inductive definitions, even.zero and even.add\_two are the only two ways to construct proofs of  $\vdash$  even.... By inspection of the conclusions even o and even (k + 2), we see that there is no danger of ever proving that 1 is even.<sup>1</sup>

Another way to view the inductive definition above is as follows: The first line introduces a predicate, whereas the second and third lines introduce axioms we want the predicate to satisfy. Accordingly, we could have written

```
constant even : \mathbb{N} \to \mathsf{Prop} axiom even.zero : even o axiom even.add two : \forall k : \mathbb{N}, even k \to \mathsf{even} (k + 2)
```

replacing inductive by constant and | by axiom. But this axiomatic version, apart from being dangerous, does not give us any information about when even is false. We cannot use it to prove  $\neg$  even 1 or  $\neg$  even 17. For all we know, even could be true for all natural numbers. In contrast, the inductive definition guarantees that we obtain the least (i.e., the most false) predicate that satisfies the introduction rules even.zero and even.add\_two, and provides elimination and induction principles that allow us to prove  $\neg$  even 1,  $\neg$  even 17, or  $\neg$  even (2 \* n + 1).

Why should we bother with inductive predicates when we can define recursive functions? Indeed, the following definition is perfectly legitimate:

<sup>&</sup>lt;sup>1</sup>We could cheat and add an axiom stating even 1, but this would be inconsistent. Alternatively, we could add an assumption even 1 to our lemma and use it to prove even 1, but this would be tautological and hence pointless.

```
\begin{array}{cccc} \operatorname{def} \ \operatorname{even}_2 & : \ \mathbb{N} \ \to \ \operatorname{bool} \\ \mid \ o & := \ \operatorname{tt} \\ \mid \ 1 & := \ \operatorname{ff} \\ \mid \ (k + 2) & := \ \operatorname{even}_2 \ k \end{array}
```

Each style has its strengths and weaknesses. The recursive version forces us to specify a false case (the second equation), and it forces us to worry about termination. On the other hand, because it is equational and computational, it works well with refl, simp, #reduce, and #eval. The inductive version is arguably more abstract and elegant. Each introduction rule is stated independently. We can add or remove rules without having to think about termination or executability.

Yet another way to define even is as a nonrecursive definition, using the modulo operator (%):

```
def even<sub>3</sub> (k : \mathbb{N}) : bool := k % 2 = 0
```

Mathematicians would likely prefer this version. But the inductive version is a convenient "Hello, World!" example that resembles many realistic inductive definitions. It might be a toy, but it is a useful toy.

## **5.1.2** Tennis Games

Transition systems consists of transition rules that connect a "before" and an "after" state. As a specimen of a transition system, we consider the possible transitions in a game of tennis, starting from o-o ("Love all"). Tennis games are also a toy, but in Chapter 8, we will define the semantics of an imperative programming language as a transition system in a similar style.

The scoring rules for tennis from the International Tennis Federation's *Rules of Tennis* are reproduced below.

A standard game is scored as follows with the server's score being called first:

```
No point - "Love"
First point - "15"
Second point - "30"
Third point - "40"
Fourth point - "Game"
```

except that if each player/team has won three points, the score is "Deuce." After "Deuce," the score is "Advantage" for the player/team who wins the next point. If that same player/team also wins the next point, that player/team wins the "Game"; if the opposing player/team wins the next point, the score is again "Deuce." A player/team needs to win two consecutive points immediately after "Deuce" to win the "Game."

We first define an inductive type to represent scores:

```
\begin{array}{lll} \text{inductive score} : & \text{Type} \\ \mid \text{vs} & : & \mathbb{N} \to \mathbb{N} \to \text{score} \\ \mid & \text{adv\_srv} & : & \text{score} \\ \mid & \text{adv\_rcv} & : & \text{score} \\ \mid & \text{game\_srv} & : & \text{score} \\ \mid & \text{game\_rcv} & : & \text{score} \\ \end{array}
```

A score such as 30–15 is represented as score.vs 30 15, abbreviated to 30–15. We ignore some of the most frivolous aspects of the scoring rules, writing 0 for "Love" and 40–40 for "Deuce." If we really cared, we could introduce aliases such as def love:  $\mathbb{N}$ := score.vs 0 0.

The next stage is to introduce a binary predicate step that determines whether a transition is possible:


```
inductive step : score \to score \to Prop | srv_0_15 : \foralln, step (0-n) (15-n) | srv_15_30 : \foralln, step (15-n) (30-n) | srv_30_40 : \foralln, step (30-n) (40-n) | srv_40_game : \foralln, n < 40 \to step (40-n) score.game_srv | srv_40_adv : step (40-40) score.adv_srv
```

Let  $s \Rightarrow t$  abbreviate step s t. A game is a chain  $s_0 \Rightarrow s_1 \Rightarrow s_2 \Rightarrow \cdots \Rightarrow s_n$  where  $s_0 = 0$ —0 and no transition is possible from  $s_n$ . The predicate allows nonsensical transitions such as 15—99  $\Rightarrow$  30—99, but since the score 15—99 cannot be reached from 0—0, these transitions are harmless.

Equipped with a formal definition, we can ask, and formally answer, questions such as: Do games have a maximal length? How many different final scores are possible? Is the score 15–99 reachable from "Love all"? And can the score ever return to 0–0? Let us use Lean to disprove the last claim:

```
lemma no_step_to_o_o (s : score) :
   ¬ step s (o-o) :=
begin
   intro h,
   cases h
end
```

The diagram below summarizes which scores are reachable from which scores:



#### **5.1.3** Reflexive Transitive Closure

Are there any convincing non-toy applications of inductive predicates? The answer is yes. Consider the reflexive transitive closure  $r^*$  of a binary relation r. The star (\*) operator is often defined as a formal system:

$$\frac{(a,b)\in r}{(a,b)\in r^*}\,\mathsf{BASE}\ \frac{(a,a)\in r^*}{(a,a)\in r^*}\,\mathsf{REFL}\ \frac{(a,b)\in r^*}{(a,c)\in r^*}\,\mathsf{TRANS}$$

These rules define  $r^*$  as the smallest relation that contains r (by BASE) and that is reflexive (by REFL) and transitive (by TRANS). If we wanted to define the transitive closure  $r^*$  instead, we would simply omit the REFL rule. If we wanted the reflexive symmetric closure, we would replace the TRANS rule by a SYMM rule. With a formal system, we simply declare the properties we want to be true, without giving a thought to termination or executability.

It is straightforward to translate the above derivation rules into introduction rules of an inductive predicate:

```
inductive star \{\alpha: \mathsf{Type}\}\ (\mathtt{r}: \alpha \to \alpha \to \mathsf{Prop}): \alpha \to \alpha \to \mathsf{Prop} | base (a b : \alpha) : r a b \to star a b | refl (a : \alpha) : star a a | trans (a b c : \alpha) : star a b \to star b c \to star a c
```

We represent relations as binary predicates rather than sets of pairs, writing r a b for  $(a, b) \in r$ . The reflexive transitive closure of r is written star r. Inside the definition, r is omitted since it is declared as a parameter, on the left of a colon (:). Notice that a, b, and c are declared as parameters of the introduction rules, on the left of the colons. We could also have written

```
| base : \forall a \ b : \alpha, r a b \rightarrow star a b or at the other extreme | base (a b : \alpha) (hab : r a b) : star a b
```

and similarly for star.refl and star.trans. All these forms are logically and operationally equivalent.

The general format of inductive predicates is as follows:

where the conclusion of each proposition; must be an application of the defined predicate predicate-name to some arguments. These arguments may be arbitrary terms; they do not need to be constructor patterns. We can also use curly braces { } instead of parentheses ( ) if we wanted to make the arguments corresponding to the parameters implicit.

The above definition of star is truly elegant. If you still doubt this, try implementing it as a recursive function:

```
def star_rec {\alpha : Type} (r : \alpha \rightarrow \alpha \rightarrow \text{bool}) : \alpha \rightarrow \alpha \rightarrow \text{bool} :=
```

To summarize, each introduction rule of an inductive predicate p consists of the following components:

- an application of p to some arguments, forming a pattern;
- zero or more conditions that must be fulfilled, which may apply p recursively;
- variables that may appear in the pattern and the conditions.

Thus, for rule star.base, the pattern is star a b, the condition is r a b, and the variables are a and b.

## 5.1.4 A Nonexample

Not all inductive definitions admit a least solution. The simplest nonexample is

```
-- fails
inductive illegal₂ : Prop
| intro : ¬ illegal₂ → illegal₂
```

If Lean accepted this definition, we could use it to prove the equivalence  $illegal_2 \leftrightarrow \neg illegal_2$ , from which we could easily derive false. Fortunately, Lean rejects the definition:

```
arg #1 of 'illegal<sub>2</sub>.intro' has a non positive occurrence of the datatypes being declared
```

The nonpositive occurrence it complains about is the occurrence of illegal<sub>2</sub> under a negation. Mathematicians would reject the definition on the ground that the monotonicity condition of the Knaster–Tarski theorem is not satisfied.

# **5.2 Logical Symbols**

Although even is the first openly inductive predicate in this guide, the earlier chapters already presented other inductive predicates clandestinely. The very first of these is equality (=), introduced in Chapter 1, followed by the logical symbols  $\land$ ,  $\lor$ ,  $\leftrightarrow$ ,  $\exists$ , true, and false. Their definitions are worth studying closely:

```
inductive and (a b : Prop) : Prop 

| intro : a \rightarrow b \rightarrow and 

inductive or (a b : Prop) : Prop 

| intro_left : a \rightarrow or 

| intro_right : b \rightarrow or 

inductive iff (a b : Prop) : Prop 

| intro : (a \rightarrow b) \rightarrow (b \rightarrow a) \rightarrow iff 

inductive Exists {\alpha : Type} (p : \alpha \rightarrow Prop) : Prop 

| intro : \foralla : \alpha, p a \rightarrow Exists 

inductive true : Prop 

| intro : true 

inductive false : Prop 

inductive eq {\alpha : Type} : \alpha \rightarrow \alpha \rightarrow Prop 

| refl : \foralla : \alpha, eq a a
```

(Strictly speaking, in Lean, some of the above definitions are actually structures and not inductive predicates, but the difference is unimportant: Structures are essentially single-constructor inductive predicates with some syntactic sugar.)

The traditional notations  $\exists x:\alpha$ , p and x=y are syntactic sugar for Exists ( $\lambda x:\alpha$ , p) and eq x y. Notice how  $\lambda$  acts as an all-purpose binder. Notice also that there is no constructor for false. There are no proofs of false, just like there is no proof of even 1. With inductive predicates, we only state the rules we want to be true.

The symbol  $\forall$ , including its special case  $\rightarrow$ , is built directly into the logic and is not defined as inductive predicates. Nor does it have explicit introduction or elimination rules. The introduction

principle is  $\lambda$ -expression  $\lambda x$ , \_, and the elimination principle is function application \_ u.

As for any inductive predicates, only the introduction rules are specified. The elimination rules presented in Sections 2.3 and 2.4 must be derived manually.

## 5.3 Rule Induction

In the same way that we can perform an induction on a term of inductive type, we can perform an induction on a proof of an inductive predicate. For example, given the goal  $h : even n \vdash p n$ , we can invoke induction' h and get two subgoals, for even.zero and even.add\_two. This is called *induction on the structure of the derivation of h* or simply *rule induction*, because the induction is on the predicate's introduction rules (i.e., the constructors of the proof term).

There are two ways to look at rule induction: the "least-predicate-such-that view" and the "PAT view." To understand the least-predicate-such-that view, recall that an inductive definition introduces a symbol as the least (i.e., the most false) predicate satisfying the introduction rules. Accordingly, even is the least predicate q such that the properties q o and  $\forall k$ , q k  $\rightarrow$  q (k + 2) hold. Therefore, if we can show that p o and  $\forall k$ , p k  $\rightarrow$  p (k + 2) hold for some predicate p, then p is either even itself or greater than (i.e., more true than) even. As a result, even n implies p n, which is exactly what we need to prove the goal h: even n  $\vdash$  p n.

The least-predicate-such-that view gives a nice intuitive account of rule induction that can be used in informal arguments, such as the following proof that even n implies n % 2 = 0 for all n:

The proof is by rule induction on the hypothesis even n.

CASE even.zero: We must show 0 % 2 = 0. This follows by computation.

CASE even.add\_two k: The induction hypothesis is k % 2 = 0. We must show (k + 2) % 2 = 0. This follows by basic arithmetic reasoning.

The Lean proof has the same structure:

```
lemma mod_two_eq_zero_of_even (n : N) (h : even n) :
    n % 2 = 0 :=
begin
    induction' h,
    case zero {
       refl },
    case add_two : k hk ih {
       simp [ih] }
end
```

The PAT principle gives us another fruitful way to look at rule induction. The key idea is that rule induction on h in a goal such as h : even n  $\vdash$  P[h] is perfectly analogous to structural induction on a value of a dependent inductive type such as vec  $\alpha$  n (Section 4.10). Writing Pu[] for the variant of P[] where n is replaced by some term u, we get the subgoals

```
\vdash P_0[\text{even.zero}: \text{even o}] k: \mathbb{N}, hk: \text{even k}, \text{ih}: P_k[hk] \vdash P_{k+2}[\text{even.add\_two k hk}: \text{even (k + 2)}]
```

These are precisely the subgoals produced by induction' h with k hk ih.

Regardless of the inductive predicate q, the procedure to compute the subgoals is always the same:

- Replace h in P[h] with each possible introduction rule applied to fresh variables (e.g., even.add\_two k hk), massaging P[] to make it type-correct. This yields as many subgoals as there are introduction rules.
- 2. Add these new variables (e.g., k, hk) to the local context.
- 3. Add induction hypotheses for all new hypotheses that assert  $\mathsf{q}\ldots$

Notice the presence of hk: even k among the hypotheses above. It was absent in the least-predicate-such-that view and is not essential because it can always be recovered by strengthening P to be of the form even  $n \land \cdots$ .

In nearly all practical cases, h will not occur in P[h]. We can then simply write

```
\vdash P_0  k : \mathbb{N}, hk : even k, ih : P_k \vdash P_{k+2}
```

In rare cases, h will occur in P[h]. Proofs may appear as subterms in arbitrary terms, as we saw when we tried to extract the head of a list in Section 4.7.

When the argument to even in h is not a variable, performing an induction will destroy its structure. Fortunately, induction' (with a prime) notices this situation and preserves the information in a hypothesis called index\_eq. Thus, given the goal h: even (2 \* n + 1)  $\vdash$  false, invoking induction' h with k hk ih produces the induction step

```
k : \mathbb{N}, h : \text{even } k, n : \mathbb{N}, \text{index\_eq} : k + 1 = 2 * n, \text{ih} : \forall n, \neg k = 2 * n + 1

\vdash false
```

for even.add\_two. The trivial base case  $\neg$  even 1 is eliminated automatically by basic constructor reasoning.

The hypothesis index\_eq records the equality between even's argument in the initial goal (i.e., 2 \* n + 1) and even's argument in the even.add\_two rule (i.e., k + 2). Notice that the equation is simplified by subtracting 1 from both sides. To continue the proof, we need to instantiate n in the induction hypothesis with n - 1. Here is the complete proof:

```
lemma not_even_two_mul_add_one (n : N) :
  ¬ even (2 * n + 1) :=
begin
  intro h,
  induction' h,
  apply ih (n - 1),
```

```
cases' n,
case zero {
   linarith },
case succ {
   simp [nat.succ_eq_add_one] at *,
   linarith }
end
```

We use the lemma nat.succ\_eq\_add\_one to rewrite terms of the form nat.succ n to n + 1 and the linarith tactic to perform simple arithmetic reasoning. The lemma is very useful when we face a mixture of nat.succ and addition.

The reflexive transitive closure star r is similar to even. Given a goal h: star  $r \times y \vdash P$ , rule induction on h produces the following subgoals, where  $P_{t,u}$  denotes the variant of P where x and y are replaced by t and u, respectively:

```
\begin{array}{c} \text{a b}: \alpha, \text{ hab}: \text{r a b} \vdash P_{\text{a,b}} \\ \\ \text{a}: \alpha \vdash P_{\text{a,a}} \\ \\ \text{a b c}: \alpha, \text{ hab}: \text{star r a b, hbc}: \text{star r b c, ihab}: P_{\text{a,b}}, \text{ ihbc}: P_{\text{b,c}} \vdash P_{\text{a,c}} \\ \end{array}
```

This is where the "assistant" aspect of "proof assistant" comes into play. One of the key properties of star is idempotence—applying star to star r has no effect. This can be proved as follows in Lean, using rule induction for the  $\rightarrow$  direction of the equivalence:

```
lemma star_star_iff_star {\alpha : Type} (r : \alpha \to \alpha \to Prop) (a b : \alpha) : star (star r) a b \leftrightarrow star r a b := begin apply iff.intro, { intro h, induction' h, case base : a b hab { exact hab }, case refl : a { apply star.refl }, case trans : a b c hab hbc ihab ihbc {
```

```
apply star.trans a b,
    { exact ihab },
    { exact ihbc } } },
    { intro h,
    apply star.base,
    exact h }
```

We use the case tactic both to document which cases we are focusing on and to give intuitive names to the emerging variables. It is easy to get lost in goals containing long, automatically generated names. The cleanup tactics introduced in Section 2.8 can also be a great help when facing large goals.

We can state the idempotence property more standardly in terms of equality instead of as an equivalence:


The proof requires two lemmas that are available because Lean's logic is classical:

```
funext: (\forall x, ?f x = ?g x) \rightarrow ?f = ?g
propext: (?a \leftrightarrow ?b) \rightarrow ?a = ?b
```

Functional extensionality (funext) states that if two functions yield equal results for all inputs, then the two functions must be equal. Propositional extensionality (propext) states that equivalence of propositions coincides with equality. In these phrases, extensionality means something like "what you see is what you get." These

properties may seem obvious, and yet there exist proof assistants built on weaker, intuitionistic logics in which the properties do not generally hold.

In Chapter 4, we saw a diagram depicting the interpretation of bool and Prop side by side. The diagram suggested the existence of an infinite number of propositions, but we now know that there are exactly two propositions: false and true. Moreover, the latter admits any number of proofs. Here is a revised diagram (with only five proofs shown):



We register the lemma star\_star\_eq\_star as a simp rule, because viewed as a left-to-right rewrite rule, it genuinely replaces a complex term by a simpler term. It is hard to imagine a situation where we would not want simp to rewrite star (star ...) to star

. . . **.** 

For rule induction, we normally use the induction' tactic. For subtle logical reasons that will become clearer in Chapter 11, rule induction by pattern matching is generally not possible.

### 5.4 Linear Arithmetic Tactic

#### linarith

The linarith tactic can be used to prove goals involving linear arithmetic equalities (=), inequalities (<, >,  $\leq$ , and  $\geq$ ), and disequalities ( $\neq$ ). Linear means that multiplication and division do not occur, of if they do then one of the operands must be a numeric constant. For example, 2 \* x < y is a linear constraint (which can be rewritten to x + x < y) whereas x \* y < y is nonlinear.

### 5.5 Elimination

Given an inductive predicate q, its introduction rules typically are of the form  $\forall \ldots, \cdots \rightarrow q \ldots$  and can be used to prove goals of the form  $\vdash q \ldots$  Elimination works the other way around: It extracts information from a lemma or hypothesis h:  $q \ldots$  Elimination takes many forms: the cases' and induction' tactics, pattern matching, and elimination rules (e.g., and elim\_left).

Invoked on h: q..., the cases' h tactic performs roughly the same rule induction as induction' h but without producing any induction hypotheses. We encountered two idioms in Chapter 4 that we can finally understand.

The first idiom is when h is of the form l = r—i.e., eq l r (Section 5.2). Suppose the goal is  $h : l = r \vdash P[h]$ . The procedure presented in Section 5.3 produces the subgoal

$$a : \alpha \vdash P_{a,a}[eq.refl \ a : a = a]$$

in which  $P_{t,u}[$  ] stands for the variant of P[ ] where l and r are replaced by t and u, respectively. (Strictly speaking, the useless hypothesis h: a = a would also appear in the subgoal.) In practice, P[h] would likely not depend on h. Moreover, cases' reuses the name l instead of inventing the name a. Thus, we would get

$$1: \alpha \vdash P_{1,1}$$

In other words, all occurrences of  ${\bf r}$  in the original goal have been replaced by 1. This corresponds to the behavior we observed in Section 4.7.

The second idiom is the tactic cases' classical.em Q, where Q is a proposition. The classical.em Q part is a proof term for Q  $\vee \neg$  Q—i.e., or Q ( $\neg$  Q) (Section 5.2). Then cases' is simply applied to eliminate the  $\vee$  connective. Suppose the goal is  $\vdash$  P[classical.em Q]. By the definition of the or predicate, the new subgoals are

```
hQ: Q \vdash P[or.intro\_left hQ: Q \lor \neg Q]

hnQ: \neg Q \vdash P[or.intro right hnQ: Q \lor \neg Q]
```

5.5. Elimination 131

There is no need to massage P because or.intro\_left hQ and or.intro\_right hnQ have the same type as classical.em Q—namely,  $Q \lor \neg Q$ . In practice, the goal would be  $\vdash P$ , with no dependency on classical.em Q. We would then have

$$hQ:Q \vdash P$$
  $hnQ:\neg Q \vdash P$ 

Again, this is the behavior we observed in Section 4.7.

In structured proofs, we can use match expressions (Section 4.4) to achieve the same effect as cases'. This works well for logical symbols. However, for predicates such as even and star, with arguments that evolve through the induction, we end up with dependently typed pattern matching, which is subtle (Section 4.10). It is generally easier to let cases' figure out what the subgoals should look like than to pattern-match. We will review an example of both styles below.

It can be useful to expand a hypothesis of the form q(c...), where c is a constructor or some other constant. We can state and prove an *inversion rule* to support such eliminative reasoning. A typical inversion rule has the form

$$\forall x_1 \dots x_n, q (c x_1 \dots x_n) \rightarrow (\exists \dots, \dots \land \dots) \lor \dots \lor (\exists \dots, \dots \land \dots)$$

It can be useful to combine introduction and elimination into a single lemma, which can be used for rewriting both the hypotheses and the targets of goals. The format is the same except for the connective  $\leftrightarrow$  in the middle:

```
\forall x_1 \dots x_n, q (c x_1 \dots x_n) \leftrightarrow (\exists \dots, \dots \land \dots) \lor \dots \lor (\exists \dots, \dots \land \dots)
```

An inversion rule for even is given below:

```
lemma even_iff (n : \mathbb{N}) :

even n \leftrightarrow n = 0 \vee (\existsm : \mathbb{N}, n = m + 2 \wedge even m) :=

begin

apply iff.intro,

{ intro hn,
```

```
cases' hn,
    case zero {
      simp },
    case add two : k hk {
      apply or.intro_right,
      apply exists.intro k,
      simp [hk] } },
  { intro hor,
    cases' hor.
    case inl : heq {
      simp [heq. even.zero] },
    case inr : hex {
      cases' hex with k hand,
      cases' hand with heq hk,
      simp [heq, even.add_two _ hk] } }
end
```

As usual, the tactical proof is not particularly readable, but we see that introduction rules and the eliminative cases' tactic play a major role, for both the logical symbols and the even predicate. The simp tactic puts the final touches.

For those who have a preference for structured proofs, here is a version of the proof with dependently typed pattern matching on hn: even n:

```
match hor with
| or.inl heq :=
    show even n, from
        by simp [heq, even.zero]
| or.inr hex :=
    match hex with
| Exists.intro m hand :=
        match hand with
| and.intro heq hm :=
        show even n, from
        by simp [heq, even.add_two _ hm]
    end
end
end
```

For technical reasons, we must pattern-match on or.inl and or.inr, which are constructors and which take a single explicit argument, rather than on the variants or.intro\_left and or.intro\_right, which are defined functions that take two explicit arguments.

## **5.6 Further Examples**

Equipped with a better understanding of inductive predicates, we are now ready to review in turn four realistic applications.

#### 5.6.1 Sorted Lists

Our first example is a predicate that checks whether a list of natural numbers is sorted in increasing order:

```
inductive sorted : list N → Prop
| nil : sorted []
| single {x : N} : sorted [x]
| two_or_more {x y : N} {zs : list N} (hle : x ≤ y)
        (hsorted : sorted (y :: zs)) :
    sorted (x :: y :: zs)
```

This definition captures the following mathematical intuition:

The set of sorted lists is defined as the smallest set S closed under the following rules:

- (1) the list [] is sorted;
- (2) given a number x, the list [x] is sorted;
- (3) given two numbers x, y and a list zs, if x < y and y :: zs is sorted, then x :: y :: zs is sorted.

It is always a good idea to test our definitions by trying it on small examples. Is the list [3, 5] sorted? It would appear so:

```
lemma sorted_3_5 :
  sorted [3, 5] :=
begin
  apply sorted.two_or_more,
  { linarith },
  { exact sorted.single }
end
```

The example needs two of the introduction rules for sorted. This approach tends to work well for expressions consisting only of closed terms. A more compact proof follows, using proof terms for everything:

```
lemma sorted_3_52 :
   sorted [3, 5] :=
sorted.two_or_more (by linarith) sorted.single
```

The same idea can be used to prove that [7, 9, 9, 11] is sorted:

```
lemma sorted_7_9_9_11 :
   sorted [7, 9, 9, 11] :=
sorted.two_or_more (by linarith)
   (sorted.two_or_more (by linarith)
        (sorted.two_or_more (by linarith)
        sorted.single))
```

Conversely, we can show that some lists are not sorted. For this, we need to use elimination:

```
lemma not_sorted_17_13 :
   ¬ sorted [17, 13] :=
begin
   intro h,
   cases' h,
   linarith
end
```

From the hypothesis that the list [17, 13] is sorted, we extract the inequality 17  $\leq$  13. The cases' tactic silently eliminates the sorted.nil and sorted.single cases, because they cannot match a two-element list. Then we just need to invoke linarith to exploit the impossible hypothesis 17  $\leq$  13.

### 5.6.2 Palindromes

Palindromes are lists that read the same from left to right and from right to left. For example, [a, b, b, a] and [a, h, a] are palindromes. The following inductive predicate is true if and only if the list passed as argument is a palindrome:

```
inductive palindrome \{\alpha: \mathsf{Type}\}: \mathsf{list}\ \alpha \to \mathsf{Prop} | nil : palindrome [] | single (\mathsf{x}:\alpha): \mathsf{palindrome}\ [\mathsf{x}] | sandwich (\mathsf{x}:\alpha) (xs : list \alpha) (hxs : palindrome xs) : palindrome ([x] ++ xs ++ [x])
```

The definition distinguishes three cases: (1) [] is a palindrome; (2) for any element x, the singleton list [x] is a palindrome; (3) for any element x and any palindrome [ $y_1, \ldots, y_n$ ], the list [x,  $y_1, \ldots, y_n, x$ ] is a palindrome.

Palindromes are another example where inductive predicates come into their own. The following naive recursive definition cannot work because [x] ++ xs ++ [x] is not a constructor pattern, and the variable x is repeated:

```
-- fails  \frac{\text{def palindrome}_2}{\text{def palindrome}_2} \left\{ \alpha : \text{Type} \right\} : \text{list } \alpha \to \text{Prop}   \mid [] \qquad := \text{true}   \mid [\_] \qquad := \text{true}   \mid ([x] ++ xs ++ [x]) := \text{palindrome}_2 xs   \mid \_ \qquad := \text{false}
```

A correct recursive definition is possible but beyond the scope of this guide.

Naturally, the reverse of a palindrome is a palindrome. It is a good exercise:

```
lemma reverse_palindrome {α : Type} (xs : list α)
        (hxs : palindrome xs) :
    palindrome (reverse xs) :=
begin
    induction' hxs,
    case nil {
        exact palindrome.nil },
    case single {
        exact palindrome.single x },
    case sandwich {
        simp [reverse, reverse_append],
        exact palindrome.sandwich _ _ ih }
end
```

#### Informally:

The proof is by rule induction on the hypothesis hxs.

CASE palindrome.nil: We must show palindrome (reverse []). This follows from palindrome.nil using reverse [] = [].

CASE palindrome.single x: We must show palindrome (reverse [x]). This follows from palindrome.single using reverse [x] = [x].

CASE palindrome.sandwich x xs hxs: We must show palindrome (reverse ([x] ++ xs ++ [x])) under the

hypothesis (hxs) palindrome xs. The induction hypothesis is palindrome (reverse xs). By simplification, it suffices to show palindrome ([x] ++ reverse xs ++ [x]). By palindrome.sandwich, it suffices to show palindrome (reverse xs), which is exactly the induction hypothesis.

### 5.6.3 Full Binary Trees

Our third example is based on the type of binary trees introduced in Section 4.8:

```
inductive btree (\alpha : Type) : Type | empty {} : btree | node : \alpha \rightarrow btree \rightarrow btree \rightarrow btree
```

A binary tree is *full* if all its nodes have either zero or two children. This can be encoded as an inductive predicate:

```
inductive is_full \{\alpha: \mathsf{Type}\}: btree \alpha \to \mathsf{Prop} | empty: is_full btree.empty | node (a:\alpha) (l: \mathsf{tr:btree}\ \alpha)   (hl: is_full l) (hr: is_full r)   (hiff: l= \mathsf{btree.empty} \leftrightarrow r= \mathsf{btree.empty}): is_full (btree.node a l r)
```

The first case states that the empty tree is a full tree. The second case states that a nonempty tree is a full tree if it has two child trees that are themselves full and that are both empty or both nonempty. The two cases neatly follow the structure of the inductive type, so it is natural to reuse the names empty and node.

The tree that consists of a node with empty trees as children is a full tree. Here is a simple proof:

```
lemma is_full_singleton {α : Type} (a : α) :
   is_full (btree.node a btree.empty btree.empty) :=
begin
   apply is_full.node,
   { exact is_full.empty },
```

```
{ exact is_full.empty },
    { refl }
end
```

A somewhat more interesting property of full trees is that fullness is preserved by the mirror operation. Our first proof is by rule induction on ht: is full t:

```
lemma is_full_mirror {α : Type} (t : btree α)
    (ht : is_full t) :
    is_full (mirror t) :=
begin
    induction' ht,
    case empty {
      exact is_full.empty },
    case node : a l r hl hr hiff ih_l ih_r {
      rw mirror,
      apply is_full.node,
      { exact ih_r },
      { exact ih_l },
      { simp [mirror_eq_empty_iff, *] } }
end
```

Since is\_full's definition follows btree's definition, it is also reasonable to perform structural induction on the tree t:

```
lemma is_full_mirror<sub>2</sub> {α : Type} (t : btree α) :
    is_full t → is_full (mirror t) :=
begin
    induction' t,
    case empty {
        intro ht,
        exact ht },
    case node : a l r ih_l ih_r {
        intro ht,
        cases' ht with _ _ _ hl hr hiff,
        rw mirror,
        apply is_full.node,
        { exact ih_r hr },
```

```
{ apply ih_l hl },
    { simp [mirror_eq_empty_iff, *] } }
end
```

The key is the case distinction on the hypothesis ht: is\_full(node alr). The cases' tactic notices that the is\_full.empty introduction rule cannot have been used to derive ht, so it only produces one case, corresponding to is\_full.node. As usual, the tactical proof will make more sense if you inspect it in Visual Studio Code, moving the cursor around.

#### 5.6.4 First-Order Terms

Our last example is based on an inductive type of first-order terms the terms of predicate logic:

```
inductive term (\alpha \beta : Type) : Type | var \{\} : \beta \rightarrow term | fn : \alpha \rightarrow list term \rightarrow term
```

A first-order term is either a variable x or a function symbol f applied to a list of arguments:  $f(t_1, \ldots, t_n)$ , where the metavariables  $t_1, \ldots, t_n$  stand for the arguments, which are themselves terms. Thus,  $\sin(\max(x, y))$  is a first-order term. The parameters  $\alpha$  and  $\beta$  are the types of function symbols and variables, respectively.

Not all terms are legal. For example, the term  $\min(\cos(a), \cos(a, b))$  is considered ill-formed, because the function cos is invoked with different number of arguments (1 versus 2). Along with  $\alpha$  and  $\beta$ , we also consider the arity, represented by a function  $\operatorname{arity}: \alpha \to \mathbb{N}$  indicating how many arguments each function symbol takes. For example, a binary symbol has arity 2. The well\_formed predicate then checks whether the given term only contains function symbol applications with the specified number of arguments:

```
inductive well_formed {\alpha \beta : Type} (arity : \alpha \to \mathbb{N}) : term \alpha \beta \to \text{Prop} | var (x : \beta) : well_formed (term.var x) | fn (f : \alpha) (ts : list (term \alpha \beta))
```

```
(hargs : ∀t ∈ ts, well_formed t)
  (hlen : list.length ts = arity f) :
well formed (term.fn f ts)
```

The fn case checks that the arguments ts are recursively well formed and that the length of ts equals the specified arity for the function symbol f in question.

Another interesting property of first-order terms is whether they contain variables. This can be checked easily using an inductive predicate:

```
inductive variable_free {\alpha \beta : Type} : term \alpha \beta \to Prop | fn (f : \alpha) (ts : list (term \alpha \beta)) (hargs : \forallt \in ts, variable_free t) : variable free (term.fn f ts)
```

There is no introduction rule corresponding to term.var because variables are never variable-free.

## **5.7 Summary of New Lean Constructs**

#### Lemmas

funext functional extensionality propext propositional extensionality

#### **Tactic**

linarith invokes a procedure for linear arithmetic

# **Chapter 6**

# **Monads**

Pure functional programming can sometimes feel overly restrictive. *Effectful functional programming* provides idioms that alleviate some of these restrictions, giving us the impression of programming with side effects, exceptions, nondeterminism, and other effects.

The underlying abstraction is called *monad*. Monads generalize programs with side effects. They are popular in Haskell to write imperative programs. In Lean, they are used to express imperative programs and to reason about them. They are even useful for programming Lean itself, as we will see in Chapter 7.

These notes are inspired by Chapter 7 of *Programming in Lean* [2]. We also refer to Chapter 14 of *Real World Haskell* [28] for a general introduction to effectful functional programming.

# **6.1 Introductory Example**

Consider the following programming task:

Implement a function  $sum_2_5_7$  ns that sums up the second, fifth, and seventh items of a list ns of natural numbers. Use option  $\mathbb N$  for the result so that if the list has too few items, return option.none.

#### A straightforward solution would be as follows:

(Confusingly, list.nth and its relative list.update\_nth count elements from o.) The code is quite inelegant, because of all the pattern matching on option. Although the programming task is contrived, we can all recall writing code with nested error handling and ever increasing indentation levels.

We can do better, by concentrating all the ugliness in one function:

```
def connect \{\alpha : \mathsf{Type}\}\ \{\beta : \mathsf{Type}\}\ : option \alpha \to (\alpha \to \mathsf{option}\ \beta) \to \mathsf{option}\ \beta option.none f := option.none | (option.some a) f := f a
```

The connect function works on an option. If the value is option.none, we leave it as is. This corresponds to an error condition, and errors are "sticky." Otherwise, the value is of the form option.some a, and we apply the operation f on a—or bind f's argument to a. We can now use connect to program our sum function:

```
def sum_2_5_7_2 (ns : list \mathbb{N}) : option \mathbb{N} := connect (list.nth ns 1)  
(\lambdan2, connect (list.nth ns 4)  
(\lambdan5, connect (list.nth ns 6)
```

```
(\lambda n7, option.some (n2 + n5 + n7)))
```

Intuitively, the program performs the following steps:

- 1. Extract the second item from the list. If it is option.none, we are done. Otherwise, bind n2 to this item and continue with the next step.
- 2. Perform the same for the fifth and seventh item, mutatis mutandis
- 3. Return the sum of n2, n5, and n7 in an option. some wrapper. Semantically, our new function  $sum_2_5_7_2$  is equal to the original  $sum_2_5_7$ .

Instead of defining connect ourselves, we could have used Lean's predefined general bind operation. It takes the same arguments in the same order. Here is the new code:

```
def sum_2_5_7_3 (ns : list \mathbb{N}) : option \mathbb{N} := bind (list.nth ns 1)  
(\lambdan2, bind (list.nth ns 4)  
(\lambdan5, bind (list.nth ns 6)  
(\lambdan7, pure (n2 + n5 + n7))))
```

We also use the predefined pure function instead of option.some to convert a pure  $\alpha$  value to an option.

One of the advantages of using the predefined notions is that they provide syntactic sugar, in the form of the >>= operator:

```
def sum_2_5_74 (ns : list \mathbb{N}) : option \mathbb{N} := list.nth ns 1 >>= \lambdan2, list.nth ns 4 >>= \lambdan5, list.nth ns 6 >>= \lambdan7, pure (n2 + n5 + n7)
```

The syntax ma >>= f expands to bind ma f, where ma stands for "maybe an  $\alpha$ ."

The next-to-last version of the sum program uses heavier syntactic sugar:

```
def sum_2_5_7_5 (ns : list \mathbb N) : option \mathbb N :=
```

```
do n2 \leftarrow list.nth ns 1,
do n5 \leftarrow list.nth ns 4,
do n7 \leftarrow list.nth ns 6,
pure (n2 + n5 + n7)
```

The do notation provides a convenient syntax for effectful programs. The program do a  $\leftarrow$  ma, t is equivalent to ma >>=  $(\lambda a, t)$ . If we are not interested in the result of ma's computation, we can omit the a  $\leftarrow$  binding and write do ma, t, which expands to ma >>=  $(\lambda, t)$ .

The do notation conveniently allows multiple  $\leftarrow$  bindings in a single block. This brings us to the final version of the program:

```
def sum_2_5_7<sub>6</sub> (ns : list \mathbb{N}) : option \mathbb{N} := do

n2 \leftarrow list.nth ns 1,

n5 \leftarrow list.nth ns 4,

n7 \leftarrow list.nth ns 6,

pure (n2 + n5 + n7)
```

Each line with an arrow ← attempts to read a value. In case of failure, the entire program evaluates to option.none.

The above function can be read as an imperative program where each of the list.nth calls can throw an exception. But even though the notation has an imperative flavor, the function is a pure functional program.

## 6.2 Two Operations and Three Laws

The option type constructor is an example of a monad, called the *option monad*. In general, a monad is a unary type constructor m: Type  $\rightarrow$  Type that depends on some type parameter  $\alpha$  equipped with two distinguished operations:

```
pure \{\alpha: \mathsf{Type}\}: \alpha \to \mathsf{m} \ \alpha
bind \{\alpha \ \beta: \mathsf{Type}\}: \mathsf{m} \ \alpha \to (\alpha \to \mathsf{m} \ \beta) \to \mathsf{m} \ \beta
```

As usual, curly braces denote implicit arguments. For options, the pure operation is simply option. some, whereas bind is what we called connect.

Recall that a value of type m  $\alpha$  is an effectful program. The pure operation embeds a pure, effectless program of type  $\alpha$  in m  $\alpha$ . The bind operation composes two effectful programs, of types m  $\alpha$  and m  $\beta$ . The first program's output, of type  $\alpha$ , is passed to the second program. The second program's output is also the output of the composite program.

We can think of a monad as a box containing some data. The box captures some special effect (e.g., exceptions, a mutable state). The pure operation puts data into the box, whereas bind allows us to access the data in the box and modify it—possibly even changing its type, since the result has type m  $\beta$ , not m  $\alpha$ . There is, however, no general way to extract the data from the box—i.e., to obtain an  $\alpha$  from an m  $\alpha$ . There might not be any  $\alpha$  value in it, or there might be several.

To summarize, pure a is a box containing the value a, with no side effects, whereas bind ma f (also written ma >>= f or do a  $\leftarrow$  ma, f a) executes ma, then executes f with the boxed result a of ma. It is convenient to use names such as ma or mb for values of type m  $\alpha$  or m  $\beta$ , and a or b for data of type  $\alpha$  or  $\beta$ .

Monads are an abstract concept with many applications. The option type is only one instance among many. The following table gives an overview of some monad instances and their effects.

Туре	Effect
$id \alpha$	no effects
option $lpha$	simple exceptions
$\sigma \to \alpha \times \sigma$	threading through a state of type $\sigma$
set $\alpha$	nondeterministic computations returning $lpha$ values
$t\to \alpha$	reading elements of type t (e.g., a configuration)
$\mathbb{N} \times \alpha$	adjoining running time (e.g., to model algorithmic complexity)
$\text{string} \times \alpha$	adjoining text output (e.g., for logging)
prob $lpha$	probability (e.g., using random number generators)
io $\alpha$	interaction with the operating system
tactic $\alpha$	interaction with the proof assistant

All of the above are unary type constructors  $m: \mathsf{Type} \to \mathsf{Type}$  applied to an argument  $\alpha$ . Some effects can be combined (e.g., option  $(\mathsf{t} \to \alpha)$ ). Some effects are not executable (e.g.,  $\mathsf{prob}\ \alpha$ ); they are nonetheless useful for modeling programs abstractly. Specific type constructors m may provide further operators beyond pure and bind. For example, they may provide a way to extract the boxed value.

Monads have several benefits. They provide the highly readable do notation. They support generic operations, such as list. mmap  $\{\alpha \ \beta : \text{Type}\} : (\alpha \to \text{m} \ \beta) \to \text{list} \ \alpha \to \text{m} \ (\text{list} \ \beta)$ , that work uniformly for all monads m. To quote *Programming in Lean* [2]:

The power of the abstraction is not only that it provides general functions and notation that can be used in all these various instantiations, but also that it provides a helpful way of thinking about what they all have in common.

Besides being a useful computer science concept, monads provide a nice example of axiomatic reasoning.

The bind and pure operations are normally required to obey three laws. The bind operation combines two programs. If either of these is a pure program, we can inline it and eliminate the bind. This gives us the first two laws:

The third law is an associativity rule for bind. It allows us to flatten a nested computation:

Notice that we can use curly braces { } around the body of a do block to delimit it, instead of enclosing the entire block in parentheses ( ).

The do notation is compatible with let, which can be used for storing pure data. Instead of

```
do
    a' ← pure a,
    b ← f a',
    g b
```

we can write

```
do
    let a' := a in
        do
        b ← f a',
        g b
```

```
do
  let a' := a,
  b ← f a',
  g b
```

Earlier we compared a monad to a box. It may help to think of the box more concretely as a Swiss bank account, with  $\alpha$ := money. The first law means that if you put some money in the account, you can take it out. The second law means that nobody will notice if you take out some money and put it back afterwards. The third law means that performing two bank operations together followed by a third one is the same as performing the first one alone followed by the other two. Given the Swiss banks' reputation for secrecy, all three laws seem plausible.

# 6.3 A Type Class

Monads are a mathematical structure, so we use a type class to specify them in Lean. Recall that a type class is a structure type that is parameterized, typically by a type, but here by a type constructor  $m: \mathsf{Type} \to \mathsf{Type}$ . Whenever we use a field from the type class on a concrete m, the type class inference mechanism retrieves the relevant structure value—the type class instance.

A possible Lean definition of monads, together with the three laws, follows:

```
\begin{array}{l} \text{@[class] structure lawful\_monad} \ (\text{m}: \text{Type} \rightarrow \text{Type}) \\ \text{extends has\_bind m, has\_pure m} := \\ (\text{pure\_bind } \{\alpha \ \beta : \text{Type}\} \ (\text{a}: \alpha) \ (\text{f}: \alpha \rightarrow \text{m} \ \beta) : \\ (\text{pure a} >>= \text{f}) = \text{f a}) \\ (\text{bind\_pure } \{\alpha : \text{Type}\} \ (\text{ma}: \text{m} \ \alpha) : \\ (\text{ma} >>= \text{pure}) = \text{ma}) \\ (\text{bind\_assoc} \ \{\alpha \ \beta \ \gamma : \text{Type}\} \ (\text{f}: \alpha \rightarrow \text{m} \ \beta) \ (\text{g}: \beta \rightarrow \text{m} \ \gamma) \\ (\text{ma}: \text{m} \ \alpha) : \\ (\text{ma} >>= \text{f}) >>= \text{g}) = (\text{ma} >>= (\lambda \text{a, f} \ \text{a} >>= \text{g}))) \\ \end{array}
```

Let us study this definition step by step:

6.4. No Effects

■ We are creating a structure parameterized by a unary type constructor m—i.e., a value of type Type → Type.

- The structure inherits the fields, and any syntactic sugar, from structures called has\_bind and has\_pure. These provide the bind and pure operations on m, with the expected types and the syntactic sugar.
- Finally, three fields (pure\_bind, bind\_pure, and bind\_assoc) are added to those already provided by has\_bind and has\_pure. Each field is a proof of one of the three laws.

We call our type class "lawful monad" because the three laws are required to hold. To instantiate this definition, we must supply the type constructor m, suitable bind and pure operators, and proofs of the laws.

Lean includes a predefined concept of monads, which is essentially equivalent to the above but is more complicated. In particular, Lean distributes the definition over two type classes, monad and is\_lawful\_monad.

### 6.4 No Effects

Lean's constant id  $\{\alpha: \mathsf{Type}\}: \alpha \to \alpha$  is defined as the identity function  $\lambda x$ , x. The identity type constructor is obtained by taking  $\alpha:=\mathsf{Type}$ . We can register it as a monad:

```
intros \alpha \beta a f, reflend, bind_pure := begin intros \alpha ma, reflend, bind_assoc := begin intros \alpha \beta \gamma f g ma, reflend }
```

The *identity monad* is the simplest monad possible. It provides a simple box, with a single value in it, without any effect. It plays a similar role as o in additive arithmetic. We can think of the other monads as variations of it; for example, the option monad is an identity monad enriched with a special option.none value representing an error state.

# 6.5 Basic Exceptions

As we saw above, the option type provides a basic exception mechanism. The following code shows how to register option : Type  $\to$  Type as a lawful monad:

```
def option.pure \{\alpha: \mathsf{Type}\}: \alpha \to \mathsf{option} \ \alpha:= \mathsf{option.some}

def option.bind \{\alpha \ \beta: \mathsf{Type}\}: \mathsf{option} \ \alpha \to (\alpha \to \mathsf{option} \ \beta) \to \mathsf{option} \ \beta

option.none \mathsf{f}:= \mathsf{option.none}

option.some a) \mathsf{f}:= \mathsf{f} \ a

\mathsf{a}[\mathsf{instance}] \ \mathsf{def} \ \mathsf{option.lawful\_monad}: \ \mathsf{lawful\_monad} \ \mathsf{option}:= \{\mathsf{pure} \ := \ \mathsf{aoption.pure}, \}
```

```
bind
     := @option.bind,
pure_bind :=
  begin
    intros \alpha \beta a f,
    refl
  end.
bind pure :=
  begin
    intros \alpha ma,
    cases' ma,
    { refl }.
    { refl }
  end.
bind_assoc :=
  begin
    intros \alpha \beta \gamma f g ma,
    cases' ma,
    { refl },
    { refl }
  end }
```

The registration process requires us to provide five components: the pure and bind operations and the proofs of the three laws. For the first and second components of the record, we use the a syntax to make the type arguments explicit, in agreement with the format expected by has\_pure and has\_bind. The three proofs are straightforward.

Beyond the standard operations, it can be useful to throw and catch exceptions. This can be implemented as follows:

The option.throw operation raises an exception, leaving the program in an error state (option.none). The option.catch operation can be used to recover from an earlier exception. If the program is currently in an error state, option.catch invokes some exception-handling code (its second argument). This code might in turn raise a new exception. If option.catch is applied to a normal state (of the form option.some a), nothing happens.

As a convenient alternative to option.catch ma ma', Lean supports the syntax ma.catch ma'. Here is a schematic example that demonstrates throwing and catching with this syntax:

```
do {
    ...,
    if ... then
       option.throw
    else
       ... }
.catch do {
    ... }
```

The corresponding Java code would look as follows:

```
try {
    ...
    if (...) {
        throw new UnknownException();
    } else {
        ...
    }
} catch (UnknownException e) {
    ...
}
```

A more idiomatic approach is to register option.catch as the option type's "or else" operator <|>:

```
@[instance] def option.has_orelse : has_orelse option :=
{ orelse := @option.catch }
```

This enables us to write

6.6. Mutable State

```
do {
    ...,
    if ... then
       option.throw
    else
       ... }
<|>
do {
    ... }
```

Monads equipped with <|> are called alternative monads.

Options cater for only one kind of error state. A more general abstraction, called *error monad*, supports different errors, as with the exceptions of Java and other programming languages.

### 6.6 Mutable State

The state monad provides an abstraction corresponding to a mutable state. For some programming languages, the compiler can detect the use of the state monad and translate programs using them to more efficient imperative programs.

Admittedly, "abstraction corresponding to a mutable state" may sound somewhat abstract, so let us consider a semiconcrete example. If you have some experience with functional programming, you probably at some point or other have written code that looks very much like this fragment:

```
def welcome_new_user (user_name : string) (ctxt : context) :
    (N × string) × context :=
let
    (user, ctxt') := create_user user_name ctxt,
    (password, ctxt'') := generate_temporary_password user ctxt',
    (ok, ctxt''') := send_unencrypted_email user password ctxt''
in
    ((user, password), ctxt''')
```

(It is very important to send the password in an unencrypted email, and to ignore the ok status of the function.) The function takes some global state or context as input and invokes three functions in turn, each of which takes a context value and returns both some data and a new context. The context is effectively threaded through the program. This allows us to have a mutable state in a programming language without side effects.

The above approach is error-prone—it is all too easy to forget one prime (') and pass the wrong context to a function. It is also cluttered by all the context variables. What if we could simply write the following instead?

```
\begin{array}{l} \text{def welcome\_new\_user}_2 \text{ (user\_name : string) :} \\ \text{context} \rightarrow (\mathbb{N} \times \text{string}) \times \text{context :=} \\ \text{do} \\ \text{user} \leftarrow \text{create\_user user\_name,} \\ \text{password} \leftarrow \text{generate\_temporary\_password user,} \\ \text{ok} \leftarrow \text{send\_unencrypted\_email user password,} \\ \text{pure (user, password)} \end{array}
```

This is exactly what the state monad provides.

The state monad builds on top of a binary type constructor action, which captures the concept of computations or actions over states of type  $\sigma$  with return values of type  $\alpha$ . In Lean, action  $\sigma$   $\alpha$  is defined as equal to  $\sigma \to \alpha \times \sigma$ :

```
def action (\sigma \alpha : Type) : Type := \sigma \rightarrow \alpha \times \sigma
```

(Since types are terms, we can use def also to define type abbreviations.) For a given type  $\sigma$ , we have that action  $\sigma: \mathsf{Type} \to \mathsf{Type}$  is a monad. The type  $\sigma$  abstracts over the exact memory layout. We could use tuples or lists to represent memory, for example, and instantiate the abstract state  $\sigma$  accordingly.

<sup>&</sup>lt;sup>1</sup>This type constructor is traditionally called State or state, but these names are awfully confusing. In this case, it seems preferable to honor tradition in the breach rather than in the observance.

A stateful action is a function that takes some state and returns both a value and some new state. The  $\sigma \to \text{part}$  of action's definition gives the old state; the left component of the cartesian product,  $\alpha$ , gives the result of the computation; and the right component of the product,  $\sigma$ , gives the new state. Thus, the state is implicitly threaded through the program. As with other effectful programs, the do notation only exposes the data—the value of type  $\alpha$ —and conceals the effect—the old and new  $\sigma$  states.

The case where  $\sigma$  := unit, a type of cardinality one (similar to void in C or Java) whose unique value is written (), corresponds to the identity monad: The type unit  $\to \alpha \times$  unit is isomorphic to  $\alpha$ . This intuition can guide us when defining pure and bind.

We start by defining basic operations: two operations, read and write, to access the memory, and the standard operations bind and pure:

```
def action.read \{\sigma: \mathsf{Type}\}: action \sigma \sigma | s:=(s,s) def action.write \{\sigma: \mathsf{Type}\} (s:\sigma): action \sigma unit | | :=((),s) def action.pure \{\sigma \ \alpha: \mathsf{Type}\} (a:\alpha): action \sigma \alpha | s:=(a,s) def action.bind \{\sigma: \mathsf{Type}\} \{\alpha \ \beta: \mathsf{Type}\} (ma:action \ \sigma \ \alpha) (f:\alpha \to action \ \sigma \ \beta): action \sigma \beta | s:= match mas with | (a,s'):=fas' end
```

The read operation simply returns the current state s (in the first pair component) and leaves the state unchanged (in the second pair component). The write operation replaces the state s with t and returns (). The pure operation returns the current state s

unchanged tupled together with the given value a. The bind operation passes the initial state to the f argument, yielding a result a and a new state t. These are passed to g, which returns a new result and a new state.

To register the action type constructor as a lawful monad, we need functional extensionality (funext): To prove f = g, it suffices to show  $\forall x$ , f = g x.

```
\mathfrak{d}[instance] def action.lawful_monad \{\sigma : Type\}:
  lawful monad (action \sigma) :=
{ pure
            := @action.pure \sigma,
              :=  action.bind  \sigma,
  bind
  pure_bind :=
    begin
       intros \alpha \beta a f,
       apply funext,
       intro s,
       refl
    end.
  bind pure :=
    begin
       intros \alpha ma,
       apply funext,
       intro s,
       simp [action.bind],
       cases' ma s.
       refl
    end.
  bind_assoc :=
    begin
       intros \alpha \beta \gamma f g ma,
       apply funext,
       intro s,
       simp [action.bind],
       cases' ma s,
       refl
    end }
```

As an example, the following program removes all elements that are smaller than a previous element in the list, leaving us with a list of increasing elements. The maximal element is stored as the state  $\sigma$ .

Notice how the state is accessed by read and write.

To execute the program, we must supply an initial state. The last state is returned along with the resulting list. It corresponds to the largest element encountered in the list or the start state. Thus, the commands

```
#eval increasingly [1, 2, 3, 2] 0
#eval increasingly [1, 2, 3, 2, 4, 5, 2] 0
produce the output
([1, 2, 3], 3)
([1, 2, 3, 4, 5], 5)
```

## 6.7 Nondeterminism

Whereas the option monad stores zero or one  $\alpha$  values and the identity and state monads store exactly one  $\alpha$  value, the *set monad* supports an arbitrary, possibly infinite number of  $\alpha$  values. This is useful to model nondeterminism, as sets of possible behaviors.

Lean's type set  $\alpha$  is defined as  $\alpha \to \mathsf{Prop}$ . In other words, a set is identified with its characteristic predicate. Familiar operators such as the empty set ( $\emptyset$ ), the universal set (set.univ), union ( $\cup$ ), intersection ( $\cap$ ), and membership ( $\in$ ) are supported, as well as traditional curly brace notations such as {a}, {a, b}, and {x | p x}. Many set constructs can be simplified by simp.

The set type constructor can be registered as a lawful monad as follows:

```
def set.pure \{\alpha : \mathsf{Type}\} : \alpha \to \mathsf{set}\ \alpha
| a := \{a\}
def set.bind \{\alpha \ \beta : \mathsf{Type}\} : \mathsf{set} \ \alpha \to (\alpha \to \mathsf{set} \ \beta) \to \mathsf{set} \ \beta
| A f := \{b \mid \exists a, a \in A \land b \in f a\}
@[instance] def set.lawful monad : lawful monad set :=
            := @set.pure.
{ pure
  bind
            := @set.bind,
  pure bind :=
     begin
        intros \alpha \beta a f,
        simp [set.pure, set.bind]
     end.
  bind pure :=
     begin
        intros \alpha ma,
        simp [set.pure, set.bind]
     end.
  bind assoc :=
     begin
        intros \alpha \beta \gamma f g ma,
        simp [set.pure, set.bind],
        apply set.ext,
        simp,
        tautology
     end }
```

The pure operation simply puts the given value a in a singleton set  $\{a\}$ . The bind operation calls f on all values in the set A and returns the union of all the results. For example, if  $A := \{3, 8\}$  and  $f := (\lambda a, \{a, a + 1\})$ , then set bind A f equals  $\{3, 4, 8, 9\}$ . The last proof relies on set extensionality, which states that two sets that contain the same elements must be equal:

```
set.ext \{\alpha : \mathsf{Type}\}\ \{\mathsf{A}\;\mathsf{B} : \mathsf{set}\;\alpha\} : (\forall \mathsf{x},\;\mathsf{x}\in\mathsf{A}\leftrightarrow\mathsf{x}\in\mathsf{B})\to\mathsf{A}=\mathsf{B}
```

Another noteworthy point is the use of the tautology tactic. The goal's target is  $\forall x$ ,  $(\exists a, (\exists a\_1, a\_1 \in m \land a \in f a\_1) \land x \in g a) \leftrightarrow (\exists a, a \in m \land \exists a\_1, a\_1 \in f a \land x \in g a\_1)$ , where the two sides of  $\leftrightarrow$  are the same except for the placement of the existential quantifiers (and, confusingly, the names of the bound variables). This ugly proposition could be proved by a tedious sequence of introduction and elimination, but we deserve more automation.

# 6.8 Tautology Tactic

### tautology

The tautology tactic performs elimination of the logical symbols  $\land$ ,  $\lor$ ,  $\leftrightarrow$ , and  $\exists$  in hypotheses and introduction of  $\land$ ,  $\leftrightarrow$ , and  $\exists$  in the target, until all the emerging subgoals can be trivially proved (e.g., by refl).

# 6.9 A Generic Algorithm: Iteration over a List

Suppose we apply an effectful function f on all elements of a list using map. We then obtain a regular list of effectful values. For example:

```
def nths_fine \{\alpha : \mathsf{Type}\}\ (\mathsf{xss} : \mathsf{list}\ (\mathsf{list}\ \alpha))\ (\mathsf{n} : \mathbb{N}) : list (option \alpha) := list.map (\lambda \mathsf{xs}, list.nth \mathsf{xs} n) \mathsf{xss}
```

The function nths\_fine xss n tries to extract the (n + 1)st element of each list in xss. Running

```
#eval nths_fine [[11, 12, 13, 14], [21, 22, 23]] 2
returns [option.some 13, option.some 23, option.some 33].
```

These option.some constructors can be inconvenient. Often, we only care about whether any error arose. This leads us to our final example: a generic effectful program mmap that iterates over a list and applies an effectful function f to each element. The definition is recursive:

```
def mmap {m : Type \rightarrow Type} [lawful_monad m] {\alpha \beta : Type} (f : \alpha \rightarrow m \beta) : list \alpha \rightarrow m (list \beta) | [] := pure [] | (a :: as) := do b \leftarrow f a, bs \leftarrow mmap as, pure (b :: bs)
```

Notice that the function returns a single m value containing a list and not a list of m values. Try to work out why it is well typed and has the desired behavior.

We can now trying using mmap instead of map:

```
def nths_coarse \{\alpha: \mathsf{Type}\}\ (\mathsf{xss}: \mathsf{list}\ (\mathsf{list}\ \alpha))\ (\mathsf{n}: \mathbb{N}): option (list \alpha) := mmap (\lambda \mathsf{xs}, list.nth \mathsf{xs} n) \mathsf{xss} Now, running
```

```
#eval nths_coarse [[11, 12, 13, 14], [21, 22, 23]] 2
```

returns option.some [13, 23, 33], with a single option.some around a pure list.

The mmap function distributes over the append operator ++. The do notation is useful not only for defining functions but also for stating their properties:

```
lemma mmap_append \{m: Type \rightarrow Type\} [lawful_monad m] \{\alpha \ \beta: Type\} (f: \alpha \rightarrow m \ \beta): \forall as \ as': list \ \alpha, mmap f (as \ ++ \ as') = do bs \leftarrow mmap \ f as, bs' \leftarrow mmap \ f as', pure \ (bs \ ++ \ bs') | [] _ := by simp [mmap, lawful_monad.bind_pure, lawful_monad.pure_bind] | (a:: as) as':= by simp [mmap, mmap_append as as', lawful_monad.pure_bind, lawful_monad.bind_assoc]
```

Incidentally, Lean already includes a function called list.mmap corresponding to our mmap.

### **6.10** Summary of New Lean Constructs

#### **Notations**

do	indicates the start of an effectful program
let	introduces a local definition in an effectful program
< >	tries the left-hand side first; tries the right-hand side
	on failure
>>=	composes effectful computations

#### Lemma

set.ext set extensionality

#### **Tactic**

tautology proves tautologies involving the basic logical symbols

# **Chapter 7**

# Metaprogramming

Like most proof assistants, Lean can be extended with custom tactics and other functionality. Programming Lean itself is called metaprogramming. Lean's metaprogramming framework uses mostly the same notions and syntax as Lean's input language itself, so that we do not need to learn a different programming language. Monads are used to provide access to the proof assistant's state and handle tactic failures.

Abstract syntax trees, presented as inductive types, *reflect* internal data structures. In a metaprogram, Lean terms are viewed as an inductive type with ten constructors, but internally they are stored in a C++ data structure. The proof assistant's C++ internals are exposed through Lean interfaces, which we can use for accessing the current goal, unifying terms, querying and modifying the global context, and setting attributes (e.g., @[simp]).

Here are some example applications of metaprogramming:

- goal transformations (e.g., applying safe introduction rules, put the goal in negation normal form);
- heuristic proof search (e.g., applying unsafe introduction rules with backtracking);
- decision procedures (e.g., for linear arithmetic, propositional logic);

- definition generators (e.g., Haskell-style derive for inductive types);
- advisor tools (e.g., lemma finders, counterexample generators);
- exporters (e.g., documentation generators);
- ad hoc proof automation (to avoid boilerplate or duplication).

As mathematician and Lean user Kevin Buzzard wrote:1

If you find yourself "grinding" (to use a computer game phrase), doing the same sort of stuff over and over again because you need to do it to make progress, then you can try to persuade a computer scientist to write a tactic to do it for you (or even write your own tactic if you're brave enough to write meta Lean code).

This chapter is inspired by an article by Ebner et al. [7] and by Chapter 8 of *Programming in Lean* [2].

### 7.1 Tactics and Tactic Combinators

When programming our own tactics, we often need to repeat some actions on several goals, or to recover if a tactic fails. Tactic combinators help in such case. The most important one is repeat { tactic }. It invokes the tactic specified within curly braces repeatedly on all goals, emerging subgoals, emerging subsubgoals, and so on, until the tactic is no longer applicable. Here is an example of repeat in action:

```
lemma repeat_example :
   even 4 ∧ even 7 ∧ even 3 ∧ even 0 :=
begin
```

https://xenaproject.wordpress.com/2020/02/09/lean-is-better-forproper-maths-than-all-the-other-theorem-provers/

```
repeat { apply and.intro },
repeat { apply even.add_two },
```

After the first repeat, the proof state consists of four goals:

```
⊢ even 4 ⊢ even 7 ⊢ even 3 ⊢ even 0
```

There are no more conjunctions in sight. The second repeat, which applies the lemma even.add\_two :  $\forall k$ , even  $k \rightarrow$  even (k + 2) over and over, leaves us with these goals:

```
⊢ even o ⊢ even 1 ⊢ even o
```

The first and last goals are annoying because they correspond to the lemma even.zero. We can prove them by trying to apply even.zero whenever applying even.add\_two fails. This is achieved as follows:

```
lemma repeat_orelse_example :
    even 4 \( \) even 7 \( \) even 3 \( \) even 0 :=
begin
    repeat \( \) apply and intro \( \),
    repeat \( \)
    apply even.add_two
    <|> apply even.zero \( \),
```

The tactic combinator <|> first executes its left-hand side. If this fails, the right-hand side is executed instead. If the right-hand side also fails, the entire combinator fails. With the new tactic, we have two unprovable goals left:

```
⊢ even 1 ⊢ even 1
```

The repeat combinator is very aggressive, targeting all goals and subgoals recursively. A less ambitious variant is iterate. It works repeatedly on the first goal until it fails; then it stops. Consider this example:

```
lemma iterate_orelse_example :
    even 4 \( \) even 7 \( \) even 3 \( \) even 0 :=
begin
    repeat \( \) apply and intro \( \),
    iterate \( \)
    apply even.add_two
    <|> apply even.zero \( \),
```

After three applications of even.add\_two or even.zero, the first goal cedes its place to the second goal. The iterate combinator keeps on working on this goal, until it gets stuck:

```
⊢ even 1 ⊢ even 3 ⊢ even 0
```

The last two goals are never considered.

Next, all\_goals invokes a tactic exactly once on each goal. The combinator succeeds only if the inner tactic succeeds on *all* goals. It fails in the example below, because even.add\_two cannot be applied to the goal  $\vdash$  even o:

```
lemma all_goals_example :
   even 4 \( \) even 7 \( \) even 3 \( \) even 0 :=
begin
   repeat { apply and.intro },
   all_goals { apply even.add_two }, -- fails
```

To ignore failures of the inner tactic, we can wrap it in the try combinator:

```
lemma all_goals_try_example :
  even 4 \wedge even 7 \wedge even 3 \wedge even 0 :=
begin
  repeat { apply and.intro },
  all_goals { try { apply even.add_two } },
```

The resulting state is

```
⊢ even 2 ⊢ even 5 ⊢ even 1 ⊢ even 0
```

The construct try { tactic } is equivalent to tactic <|> skip, where skip is a tactic that succeeds without doing anything. A related tactic is done; it succeeds if there are no goals left and fails otherwise.

Another variant is the any\_goals combinator. Like all\_goals, it invokes a tactic once one each goal. But unlike all\_goals, it succeeds if the inner tactic succeeds on any goal. The example

```
lemma any_goals_example :
    even 4 \( \) even 7 \( \) even 3 \( \) even 0 :=
begin
    repeat \( \) apply and.intro \( \),
    any_goals \( \) apply even.add_two \( \),
results in the state
    \( \) even 2 \( \) even 5 \( \) even 1 \( \) even 0
```

This is the same state as in the previous example. The difference is that any\_goals can fail whereas all\_goals { try { . . . } } always succeeds.

Sometimes we want to leave a goal alone unless we can fully prove it. The solve1 combinator takes an arbitrary tactic as argument and transforms it into an all-or-nothing tactic (also called a terminal or end-game tactic). If the inner tactic does not prove the goal, the combinator fails. In the example

```
lemma any_goals_solve1_repeat_orelse_example :
    even 4 \( \) even 7 \( \) even 3 \( \) even 0 :=
begin
    repeat \{ apply and.intro \},
    any_goals \{ solve1 \{ repeat \{
        apply even.add_two
        <|> apply even.zero \} \} \},
```

the first and fourth goals are proved, and we are left with the two unprovable goals, exactly as they stand in the lemma statement:

```
⊢ even 7 ⊢ even 3
```

The combinators repeat, iterate, all\_goals, and any\_goals can lead to infinite looping. Consider this example:

```
lemma repeat_not_example :
    ¬ even 1 :=
begin
    repeat { apply not.intro },
```

The not.intro rule is  $(?a \rightarrow false) \rightarrow \neg ?a$ , so it applies once to transform the goal into  $\vdash$  even 1  $\rightarrow$  false. Because  $\neg ?a$  is defined as  $?a \rightarrow false$ , the rule applies again, yielding the same goal again. The tactic loops.

So far we have not left the comfort of Lean's interactive interface level. It is time to start with the actual metaprogramming, by coding a custom tactic. The tactic embodies the behavior we hard-coded in the solve1 example above:

```
meta def intro_and_even : tactic unit :=
do
  tactic.repeat (tactic.applyc ''and.intro),
  tactic.any_goals (tactic.solve1 (tactic.repeat
        (tactic.applyc ''even.add_two
        <|> tactic.applyc ''even.zero))),
  pure ()
```

Let us analyze this code step by step:

- The first line starts with meta def, indicating the declaration of a metafunction. These are much like regular functions declared without the meta keyword, but they do not need to pass the termination check.
- The first line declares a metafunction called intro\_and\_even of type tactic unit, where tactic α is the type of metaprograms returning a value of type α. (The terms "tactic" and "metaprogram" are often used interchangeably in Lean.) Often, metaprograms only have side effects on the state but return no value. In such cases, they can return the empty tuple (), of type unit.

- The second line, do, marks the entry into a monad: the metaprogramming monad (tactic).
- The third line is the low-level, metaprogram equivalent of the familiar tactic invocation repeat { apply and.intro }. The tactic.repeat metafunction corresponds to repeat. The tactic.applyc metafunction is one of several that correspond to apply. It applies a lemma specified by its name, of type name. The name here is ''and.intro. The two backticks resolve the name of a Lean constant and create a value of type name.
- The rest of the metaprogram follows the same principles. Notice how the monad operator <|> is used to provide the same effect as the tactic combinator <|>. In fact, the metaprogramming monad is an alternative monad. Without pure () at the end, the metaprogram would have all\_goal's result type as its type, but we want tactic unit.

Tactics and tactic combinators often have the same name at the user interface level and at the metaprogramming level, except for the tactic. prefix. A notable exception is sorry, which is called tactic.admit.

Once we have defined a custom tactic, we can invoke it as usual:

```
lemma any_goals_solve1_repeat_orelse_example2 :
    even 4 \( \times \) even 7 \( \times \) even 0 :=
begin
    intro_and_even,
```

This yields the subgoals

```
⊢ even 7 ⊢ even 3
```

A key benefit of Lean's metaprogramming framework is that we do not need to learn another programming language to write metaprograms. We can work with the same constructs and notation used to define ordinary objects in Lean's library. Everything in that library is available for metaprogramming purposes (e.g., Z,

list). Metaprograms can be written and debugged in the same interactive environment, encouraging a style where formal libraries and supporting automation are developed at the same time. All we need is some basic understanding of monads and some familiarity with Lean's metaprogramming interfaces.

We conclude this section by introducing a further tactic combinator that can be useful for metaprogramming. The "and then" operator; can be used to join two tactics. The left-hand side is executed on the first goal. The right-hand side is executed on every emerging subgoals (but not on the original second goal, third goal, etc.). Like <|>, the; operator has the same syntax at the interactive interface level and at the metaprogramming level. Thus, we can write

```
by induction n; simp [*]
instead of
begin
  induction n,
  { simp },
  { simp [n_ih] }
end
```

# 7.2 The Metaprogramming Monad

The metaprogramming monad, tactic, combines the attributes of several kinds of monads:

- tactic is a state monad providing access among others to the list of goals, the global context (including all definitions and inductive types), notations, and attributes (e.g., the list of @[simp] lemmas).
- tactic behaves like an option monad. The program tactic. fail string, where string is an error message, leaves the monad in an error state.

- tactic is an alternative monad. Namely, it provides an operator <|> for specifying alternatives.
- tactic supports tracing. We can use tactic.trace string, or display the current proof state using tactic.trace\_state.

In addition, metaprograms have access to the current goals. Each goal is represented as a metavariable ?m standing for a missing proof term. Each metavariable has a type and a local context specifying the variables and hypotheses that can be used to instantiate ?m. By the PAT principle, variables and hypotheses are not distinguished. Metaprograms have access to Lean's functions for inferring implicit arguments and types.

Let us put the metaprogramming monad to some use:

```
lemma even_14 :
    even 14 :=
by do
    tactic.trace "Proving evenness ...",
    intro and even
```

This first example is very modest: It uses the monad's tracing facilities to print the message "Proving evenness..." before invoking the intro\_and\_even tactic we developed above. The do keyword signals entry into the monad. The by keyword preceding it is necessary to enter tactic mode, as usual.

As usual in computer science, we can introduce a layer of indirection. We can package our combination of tracing and proving as its own tactic, and invoking it to prove that 16 is even:

```
meta def hello_then_intro_and_even : tactic unit :=
do
   tactic.trace "Proving evenness ...",
   intro_and_even

lemma even_16 :
   even 16 :=
by hello_then_intro_and_even
```

Notice, again, the presence of the meta keyword in front of the function definition. Any executable Lean definition can be used as a metaprogram. In addition, we can put meta in front of a definition to indicate that is a metadefinition. These need not terminate but cannot be used in non-meta contexts, because they might be logically inconsistent. Metaprograms can also call other metaprograms, some of which are implemented in C++.

Most metaprograms are designed to work on a proof state, as a tactic, but some can also be used on the top level, without a proof context, by invoking the run\_cmd command:

```
run cmd tactic.trace "Hello, Metaworld!"
```

To adapt to the situation in which they are invoked, tactics must be able to inspect the current goal. The following example outputs the local context, the target, and all goals (both the metavariables and their types):

```
meta def trace goals : tactic unit :=
do
  tactic.trace "local context:",
  ctx ← tactic.local context,
  tactic.trace ctx,
  tactic.trace "target:",
  P ← tactic.target,
  tactic.trace P,
  tactic.trace "all missing proofs:",
  Hs \leftarrow tactic.get goals,
  tactic.trace Hs,
  \tau s \leftarrow list.mmap tactic.infer type Hs,
  tactic.trace τs
lemma even_18_and_even_20 (\alpha : Type) (a : \alpha) :
  even 18 ∧ even 20 :=
by do
  tactic.applyc 'and.intro,
  trace_goals,
  intro_and_even
```

The output, which is visible by hovering over the do keyword, is as follows:

```
local context:
[\alpha, a]
target:
even 18
all missing proofs:
[?m_1, ?m_1]
[even 18, even 20]
```

The goals are a list of metavariables that must be instantiated. They are normally not displayed in the proof state; instead, only their types are shown. In the example, the goals are ?m\_1 : even 18 and ?m\_1 : even 20. Confusingly, Lean reuses the same metavariable names for different goals.

The metafunctions in the above programs have the following types:

```
\label{eq:context} \begin{array}{c} \text{tactic.trace}: \alpha \rightarrow \text{tactic unit} \\ \text{tactic.local\_context}: \text{tactic} \; (\text{list expr}) \\ \text{tactic.target}: \text{tactic expr} \\ \text{tactic.get\_goals}: \text{tactic} \; (\text{list expr}) \\ \text{tactic.infer\_type}: \text{expr} \rightarrow \text{tactic expr} \end{array}
```

where expr stores a term and  $\alpha$  is restricted by a type class (not shown) to be printable. In addition, list.mmap (Section 6.9) is typed as follows:

```
list.mmap \{\alpha \ \beta : \mathsf{Type}\} : (\alpha \to \mathsf{m} \ \beta) \to \mathsf{list} \ \alpha \to \mathsf{m} \ (\mathsf{list} \ \beta)
```

In practice, instead of the ad hoc trace\_goals tactic, we would normally use the following predefined tactic:

```
tactic.trace_state : tactic unit
```

The next example is more realistic. It implements a hypothesis tactic that, like the predefined assumption, looks for a hypothesis

of the right type (i.e., the right proposition) and applies it to prove the goal:

The main metafunction extracts all available hypotheses in the local context, which are stored as expressions of type expr, and iterates through them. For each hypothesis, we first try to invoke the predefined exact tactic. If this fails, the <|> operator passes control to the right branch, in which we continue recursively with the remaining hypotheses. If no hypotheses are left, we invoke tactic.fail.

A simple invocation follows:

```
lemma app_of_app \{\alpha: Type\} \{p: \alpha \to Prop\} \{a: \alpha\} (h: p: a): p: a:= by hypothesis
```

The tracing informs us that  $\alpha$ , p, a, and h were tried in turn before the matching h was found and successfully applied.

The example featured the metafunction

```
tactic.fail \{\alpha \ \beta : \mathsf{Type}\} : \alpha \to \mathsf{tactic} \ \beta
```

where  $\alpha$  must support printing.

# 7.3 Names, Expressions, Declarations, and Environments

The metaprogramming framework revolves around five main types:

- tactic manages the proof state, the global context, and more;
- name represents a structured name (e.g., x, even.add\_two);
- expr represents an expression—i.e., a term, type, proof, or proposition—as an abstract syntax tree;
- declaration represents a constant declaration, a definition, an axiom, or a lemma;
- environment stores all the declarations and notations that make up the global context.

The type expr is presented as follows to metaprograms:

```
meta inductive expr : Type
| var \{ \}  : nat \rightarrow expr
| sort \{\} : level \rightarrow expr
|\hspace{.05cm}\mathsf{const}\hspace{.1cm}\{\} \hspace{.5cm} :\hspace{.05cm} \mathsf{name}\hspace{.1cm} \rightarrow \hspace{.05cm} \mathsf{list}\hspace{.1cm} \mathsf{level}\hspace{.1cm} \rightarrow \hspace{.1cm} \mathsf{expr}
l mvar
                       : name 
ightarrow name 
ightarrow expr 
ightarrow expr
| local_const : name 	o name 	o binder_info 	o expr 	o expr
| app
                       : expr \rightarrow expr \rightarrow expr
                       : name \rightarrow binder_info \rightarrow expr \rightarrow expr \rightarrow expr
| lam
| pi
                       : name 	o binder_info 	o expr 	o expr 	o expr
                   : name 
ightarrow expr 
ightarrow expr 
ightarrow expr
| elet
                       : macro_def 	o list expr 	o expr
macro
```

#### In turn:

- expr.var represents a bound variable, using a notation known as De Bruijn indices. For example, expr.var o refers to the variable bound by the closest λ- or ∀-binder, expr.var 1 refers to the variable bound by the second closest binder, and so on.
- expr.sort is used to represent the types of types. For example, expr.sort level.zero represents Prop, and expr.sort (level.succ level.zero) represents Type.

- expr.const represents a constant, such as list.reverse or
   N. The second argument represents universe levels, a concept that will be explained in Chapter 11.
- expr.mvar represents a metavariable, i.e., a variable ?m with a question mark. The first name argument is a unique name for the metavariable, the second name is a "pretty" name, and the expr either stores the metavariable's type or is a dummy value.
- expr.local\_const represents a variable in the local context (e.g., x, h). The first name argument is a unique name for the variable; the second name is a "pretty" name; the binder\_info stores whether the variable is an explicit ( ), implicit { }, or type class [ ] argument; and the expr either stores the variable's type or is a dummy value.
- expr.app represents the application of a function to an argument.
- expr.lam represents a λ-expression. The name is the bound variable's name, the binder\_info is as for expr.local\_const , the first expr is the bound variable's type, and the second expr is the λ-expression's body.
- expr.pi represents a ∀-quantifier, including the degenerate case where the bound variable does not occur in the body
   (→). The arguments are as for expr.lam.
- expr.elet represents a let expression. The name is the name of the definition variable, the first expr is its type, the second expr is the right-hand side of the definition, and the third expr is the let expression's body.
- expr.macro is a low-level mechanism used to implement sorry, nameless placeholders (\_), and recursive metaprograms.

To each constructor corresponds a discriminator of type  $expr \rightarrow bool$ . For example,  $expr.is_pi$  e returns tt if e is of the form expr.pi ...; otherwise, it returns ff. The names are easy to guess, except

for expr.is\_constant, expr.is\_local\_constant, expr.is\_lambda,
and expr.is let.

A lightweight way to construct an expression uses Lean's quoting mechanism, based on backticks. Expressions are enclosed in parentheses ( ) and preceded by one, two, or three backticks (').

An expression preceded by a single backtick—'(expr)—must be fully elaborated. In other words, it may not contain placeholders, as in the second example:

```
run_cmd do
  let e : expr := '(list.map (λn : N, n + 1) [1, 2, 3]),
  tactic.trace e

run_cmd do
  let e : expr := '(list.map _ [1, 2, 3]), -- fails
  tactic.trace e
```

With two backticks—''(expr)—we have *pre-expressions*. These are Lean expressions that may contain some placeholders to be filled in later:

```
run_cmd do let e_1: pexpr := ''(list.map (\lambdan, n + 1) [1, 2, 3]), let e_2: pexpr := ''(list.map _ [1, 2, 3]), tactic.trace e_1, tactic.trace e_2
```

The type pexpr is a variant of expr. A pexpr can be elaborated into an expression using tactic.to\_expr, which has type pexpr  $\to$  tactic expr.

With three backticks—'''(expr)—we have pre-expressions without any name checking. We can refer to seattle.washington even if it does not exist when the pre-expression is parsed:

```
run_cmd do
  let e : pexpr := '''(seattle.washington),
  tactic.trace e
```

A similar quoting mechanism exists for names. The syntax consists of one or two backticks, without parentheses:

```
run_cmd tactic.trace 'and.intro
run_cmd tactic.trace 'intro_and_even
run_cmd tactic.trace 'seattle.washington

run_cmd tactic.trace ''and.intro
run_cmd tactic.trace 'intro_and_even
run_cmd tactic.trace 'seattle.washington -- fails
```

The single-backtick variant—'name—does not check whether the name exists. The two-backtick variant—'name—fully elaborates the name of a constant and returns an error if no constant of that name exists in the current global context. Whether the check and elaboration is desirable depends on the situation. In the demonstration file associated with this chapter, 'intro\_and\_even returns the name intro\_and\_even (of the metafunction we defined in Section 7.1), whereas 'intro\_and\_even expands to the fully elaborated name LoVe.intro\_and\_even, including the namespace LoVe. Tactics such as tactic.applyc expect a fully elaborated name. This is why we wrote tactic.applyc 'and.intro earlier. A name can be turned into a string using the polymorphic to\_string operation.

Sometimes we would like to embed an existing expression in a larger expression. This can be achieved using *antiquotations*. These are announced by the prefix %% followed by a name from the current context. Antiquotations are available with one, two, and three backticks:

```
run_cmd do
  let x : expr := '(2 : N),
  let e : expr := '(%%x + 1),
  tactic.trace e

run_cmd do
  let x : expr := '(@id N),
  let e : pexpr := ''(list.map %%x),
  tactic.trace e
```

```
run_cmd do
  let x : expr := '(@id N),
  let e : pexpr := '''(a _ %%x),
  tactic.trace e
```

The three commands print 2 + 1, list.map (as\_is (id.{1} nat)), and a 6.\_.13 (as\_is (id.{1} nat))—cryptic but precise.

Antiquotations can also be used to perform pattern matching:

```
lemma one_add_two_eq_three :
    1 + 2 = 3 :=
by do
    '(%%a + %%b = %%c) ← tactic.target,
    tactic.trace a,
    tactic.trace b,
    tactic.trace c,
    '(@eq %%α %%l %%r) ← tactic.target,
    tactic.trace d,
    tactic.trace l,
    tactic.trace r,
    tactic.exact '(refl _ : 3 = 3)
```

The two  $\leftarrow$  lines, corresponding to bind operations, succeed only if the right-hand side has the right shape (which it has). Given that target is 1 + 2 = 3, executing the first line binds the variables a, b, r to 1, 2, 3, respectively. The second pattern looks at 1 + 2 = 3 through different lenses—namely, without syntactic sugar, as an expression of the form  $\log q \alpha l r$ . This also succeeds, binding  $\alpha$ , l, r to  $\mathbb{N}$ , 1 + 2, and 3, respectively.

Declarations are presented as an inductive type to metaprogrammers:

```
meta inductive declaration |\ \text{defn}: \ \text{name} \to \text{list name} \to \text{expr} \to \text{expr} \to \text{reducibility\_hints} \to \text{bool} \to \text{declaration} |\ \text{thm}: \ \text{name} \to \text{list name} \to \text{expr} \to \text{task expr} \to \text{declaration} |\ \text{cnst}: \ \text{name} \to \text{list name} \to \text{expr} \to \text{bool} \to \text{declaration} |\ \text{ax}: \ \text{name} \to \text{list name} \to \text{expr} \to \text{declaration}
```

For example, in a declaration of the form declaration.defn nam  $_{\tau}$  rhs  $_{-\tau}$ , nam is the name of the defined constant,  $\tau$  is its type, and rhs is the definition's body. Given a declaration, declaration. to\_name extracts its first component, of type name.

The environment is presented as an abstract type, equipped with some operations to query and modify it. The most useful metafunction is

```
environment.fold \{\alpha : \mathsf{Type}\} : environment \to \alpha \to (\mathsf{declaration} \to \alpha \to \alpha) \to \alpha
```

A call to environment.fold env x f applies the given function f on each declaration in the global context in turn. In addition, an accumulator value is threaded through. Its initial value is x and the final value is returned at the end. For example, the following command outputs the number of global declarations:

```
run_cmd do
  env ← tactic.get_env,
  tactic.trace (environment.fold env ⊙ (λdecl n, n + 1))
```

Fold functions are a common idiom in functional programming. If you are not familiar with them, we recommend the chapter "Higher Order Functions" of *Learn You a Haskell for Great Good!* [22]. You can start from the heading "Only folds and horses."

# 7.4 First Example: A Conjunction-Destructing Tactic

In this and the next section, we review two metaprograms that accomplish well-defined tasks. The first program is a tactic called destruct\_and that automates the elimination of conjunctions in premises. Our aim is to automate proofs such as the following:

```
lemma abcd_a (a b c d : Prop) (h : a \land (b \land c) \land d) : a := and.elim_left h
```

```
lemma abcd_b (a b c d : Prop) (h : a \land (b \land c) \land d) :
    b :=
and.elim_left (and.elim_left (and.elim_right h))
lemma abcd_bc (a b c d : Prop) (h : a \land (b \land c) \land d) :
    b \land c :=
and.elim_left (and.elim_right h)
```

In each case, we would like to simply write destruct\_and h.

Our tactic relies on a helper metafunction, which takes as argument the hypothesis h to use as an expression rather than as a name:

```
meta def destruct and helper : expr \rightarrow tactic unit
l h :=
  do
    t ← tactic.infer_type h,
    match t with
    | '(%%a ∧ %%b) :=
      tactic.exact h
      <|>
      do {
        ha ← tactic.to_expr ''(and.elim_left %%h),
        destruct_and_helper ha }
      < | >
      do {
        hb ← tactic.to expr ''(and.elim right %%h),
        destruct and helper hb }
                   := tactic.exact h
    | _
    end
```

We first extract h's type (i.e., its proposition). We perform pattern matching with antiquotations to see if the type is of the form  $a \wedge b$ . If so, we try three things in turn. First, we invoke exact h. If this fails, we apply the left elimination rule on h, producing the proof term and.elim\_left h: a, and continue recursively with that proof term. The tactic.to\_expr call converts a pre-expression into an

expression, elaborating it. If our attempt at left elimination fails, we try right elimination instead.

The main metafunction has very little to do:

```
meta def destruct_and (nam : name) : tactic unit :=
do
   h ← tactic.get_local nam,
   destruct_and_helper h
```

The metafunction retrieves the hypothesis h with the specified name nam using tactic.get\_local, which has type name  $\rightarrow$  tactic expr, and calls the helper metafunction.

We can now turn to our motivating examples and prove them again, this time using our new widget:

```
lemma abc_a (a b c : Prop) (h : a \land b \land c) : 
 a := 
by destruct_and 'h 
lemma abc_b (a b c : Prop) (h : a \land b \land c) : 
 b := 
by destruct_and 'h 
lemma abc_bc (a b c : Prop) (h : a \land b \land c) : 
 b \land c := 
by destruct_and 'h
```

Unfortunately, we need to quote the name of the hypothesis with a backtick. We would need more, unpleasant-to-write code to provide proper parsing for our tactic's argument.

# 7.5 Second Example: A Provability Advisor

Sometimes we state a lemma, prove it, and later realize that the lemma already existed. This can be prevented using prove\_direct, a tool that traverses all available lemmas and checks whether one of them can prove the current goal.

We will review the code of prove\_direct in steps. The first step is a metafunction is\_theorem that returns tt if a declaration is a lemma or an axiom and ff otherwise:

```
meta def is_theorem : declaration \rightarrow bool | (declaration.defn _ _ _ _ ) := ff | (declaration.thm _ _ _ ) := tt | (declaration.cnst _ _ _ ) := ff | (declaration.ax _ _ ) := tt
```

We will use this metafunction to filter out the declarations that do not interest us.

The next metafunction returns the list of all theorem names:

```
meta def get_all_theorems : tactic (list name) :=
do
    env ← tactic.get_env,
    pure (environment.fold env [] (λdecl nams,
        if is_theorem decl then declaration.to_name decl :: nams
        else nams))
```

The tactic.get\_env metafunction yields an environment object giving the global context. We can traverse it using environment. fold, which is applied on each declaration decl in turn and on an accumulator thms. We extend the accumulator with each declaration's name that satisfies is\_theorem. At the end, environment. fold returns the accumulator.

Another component we need is a tactic that attempts to prove the goal using a theorem specified by its name:

```
meta def prove_with_name (nam : name) : tactic unit :=
do
  tactic.applyc nam
  ({ md := tactic.transparency.reducible, unify := ff }
    : tactic.apply_cfg),
  tactic.all_goals tactic.assumption,
  pure ()
```

The tactic.applyc applies the theorem named nam to the current goal. For performance, we pass an optional configuration,

{ md := transparency.reducible, unify := ff }, which instructs tactic.applyc to perform less computational expansion. This is important since this tactic will be invoked thousands of times. The tactic.all\_goals combinator, in conjunction with tactic. assumption, is used to apply the hypotheses from the local context to all remaining subgoals.

Finally, our prove\_direct tactic goes through all theorems found by get\_all\_theorems and succeeds with the first theorem that proves the current goal:

The list.mfirst effectful function invokes a tactic to each element of a list until one invocation succeeds. We use tactic.trace to output the successful theorem to the user, so that they can apply the theorem directly.

As a small refinement of the above, we propose a version of  $prove\_direct$  that also looks for equalities stated in symmetric form—for example, if the goal is l = r but the theorem is r = l:

```
meta def prove_direct_symm : tactic unit :=
prove_direct
<|>
do {
  tactic.applyc 'eq.symm,
  prove_direct }
```

Our solution is simple: We first try prove\_direct, and if that fails, we apply symmetry of equality (the property  $?l = ?r \rightarrow ?r = ?l$ ) to change the goal and try prove direct again.

This concludes our review of prove\_direct. A similar, more sophisticated tactic is available as library\_search in mathlib.

#### 7.6 A Look at Two Predefined Tactics

Quite a few of Lean's predefined tactics are implemented as metaprograms and not in C++. We can find these definitions by clicking the name of a construct in Visual Studio Code while holding the control or command key. To familiarize ourselves further with the metaprogramming framework, we conclude this chapter by looking at the implementation of two basic tactics that were introduced in Chapter 2: intro and assumption.

We can find intro's code by entering #check tactic.intro and by clicking tactic.intro while holding control or command:

```
meta def intro (n : name) : tactic expr :=
do
   t ← target,
   if expr.is_pi t ∨ expr.is_let t then
      intro_core n
   else
      whnf target >> intro core n
```

The definition is located inside the tactic namespace, explaining why tactic.intro and tactic.target are simply called intro and target. If the goal's target is a  $\forall$ -quantifier or a let expression, tactic.intro\_core is invoked. Otherwise, a tactic named tactic.whnf\_target is invoked before tactic.intro\_core, in the hope that a  $\forall$  or let will emerge. The >> operator performs sequential composition in a monad; ma >> mb is syntactic sugar for do ma, mb.

If we follow the definition of tactic.intro\_core, we reach a dead end:

```
\texttt{meta constant intro\_core} \; : \; \mathsf{name} \; \to \; \mathsf{tactic \; expr}
```

The metafunction is simply declared as a constant, without a definition. This indicates that the implementation is in C++. The core of the  $whnf_target$  tactic, which normalizes the goal by expanding definitions,  $\beta$ -reducing, and more, is also implemented in C++, for efficiency reasons.

In Section 7.2, we implemented a hypothesis tactic that mimicked assumption. Let us now have a look at the real thing:

```
meta def assumption : tactic unit :=
do {
  ctx ← local_context,
  t ← target,
  H ← find_same_type t ctx,
  exact H }
<|> fail "assumption tactic failed"
```

Again, the definition is in the tactic namespace. The assumption tactic attempts to find a hypothesis H (of type expr) among the variables declared in the local context ctx that has a type unifiable with that of the target t (i.e., a hypothesis whose statement is unifiable with t). Then H is applied to the target using tactic.exact. If anything goes wrong, the assumption tactic catches the error and reports "assumption tactic failed."

The auxiliary metafunction find\_same\_type is defined as follows:

The metafunction takes an expression e representing a type (i.e., a proposition) and a context and returns an expression. If the context is empty, an anonymous error is raised. Otherwise, we check if the local context's first hypothesis H's type t is unifiable with e.

If so, H is returned; otherwise, we continue with the remaining hypotheses Hs. The return operation is an alias for pure.

The find\_same\_type metafunction relies on two general-purpose metafunctions we had not encountered before:

```
tactic.failed \{\alpha: \mathsf{Type}\}: \mathsf{tactic}\ \alpha tactic.unify: \mathsf{expr} \to \mathsf{expr} \to \mathsf{tactic} unit
```

#### 7.7 Miscellaneous Tactics

#### skip

The skip tactic succeeds without doing anything. It is sometimes useful when writing custom tactics.

#### done

The done tactic raises a failure if there are some goals left; otherwise, it succeeds without doing anything. It is sometimes useful as a building block when writing custom tactics.

#### library\_search

The library\_search tactic searches the loaded libraries for a lemma that exactly proves the goal. On success, it suggests a tactic invocation of the form exact ..., which can be inserted in the formalization.

### **7.8 Summary of New Lean Constructs**

#### **Command**

run\_cmd

executes a metaprogram on the top level, without a proof context

#### **Declaration**

prefixes declarations of metaprograms meta

#### **Quotations**

'n	quotes literal name, with checking
''n	quotes literal name, without checking
'(e)	quotes fully elaborated expressions
111-1	aviata a mua aviana a la cal

quotes pre-expressions, with name checking ''(e) quotes pre-expressions, without name checking '''(e)

embeds an expression within a quotation %%x

#### **Tactic**

fails if there are some goals left done

searches the libraries for a lemma that proves the library search

current goal

does nothing skip

#### **Tactic Combinators**

invokes the second tactic to all subgoals from the ;

first tactic

<|> tries the left-hand side first; tries the right-hand side

on failure

invokes a tactic once on each goal, expecting all\_goals

only successes

invokes a tactic once on each goal, expecting at least any goals

one success

repeatedly invokes a tactic on the first goal until failure iterate repeatedly invokes a tactic on all goals and subgoals repeat

until failure

transforms a tactic into an all-or-nothing tactic solve1 tries to invoke a tactic; does nothing on failure try

# Part III Program Semantics

# **Chapter 8**

# **Operational Semantics**

In this and the next two chapters, we will see how to use Lean to specify the syntax and semantics of programming languages, to prove properties of the semantics, and to reason about concrete programs.

This chapter is heavily inspired by Chapter 7 of Concrete Semantics: With Isabelle/HOL [27].

#### 8.1 Formal Semantics

A formal semantics allows us to specify and reason about a programming language and about individual programs written in that language. It can form the basis of verified compilers, interpreters, verifiers, static analyzers, type checkers, and more. Without formal proofs, these tools are almost always wrong.

Consider WebAssembly, a new machine-like language for web browsers, designed as a portable target for compiling high-level languages such as C++ and Rust. A researcher, Conrad Watt [34], formalized its semantics and type system using the Isabelle/HOL proof assistant. He found many issues (our italics):

We have produced a full Isabelle mechanisation of the core execution semantics and type system of the WebAssembly language. In addition, we have created a mechanised proof for the type soundness properties stated in the working group's paper. In order to complete this proof, several deficiencies in the official WebAssembly specification, uncovered by our proof and modelling work, needed to be corrected by the specification authors. In some cases, these meant that the type system was originally unsound.

We have maintained a constructive dialogue with elements of the working group, mechanising and verifying new features as they are added to the specification. In particular, the mechanism by which a WebAssembly implementation interfaces with its host environment was not formally specified in the working group's original paper. Extending our mechanisation to model this feature revealed a deficiency in the WebAssembly specification that sabotaged the soundness of the type system.

Watt's research is only one example among many. Proof assistants are widely used for programming language research. Every year, around 10%–20% of papers presented at the Principles of Programming Languages (POPL) conference are formalized. This is possible because comparatively little machinery is needed to get started. The proofs tend to have lots of cases, which is a good match for computers. Moreover, proof assistants are extremely convenient to keep track of what needs to be changed as we extend a programming language with more features.

# 8.2 A Minimalistic Imperative Language

WHILE<sup>1</sup> is a minimalistic imperative language with the following grammar:

<sup>&</sup>lt;sup>1</sup>Fans of backronyms might enjoy this one: Weak Hypothetical Imperative Language Example.

```
S::= skip (no-op)
| x:= a (assignment)
| S; S (sequential composition)
| if b then S else S (conditional statement)
| while b do S (while loop)
```

where S stands for a *statement* (also called *command* or *program*), x for a program variable, a for an arithmetic expression, and b for a Boolean expression.

In our grammar, we deliberately leave the syntax of arithmetic and Boolean expressions unspecified. In Lean, we have the choice:

- We can use a type such as aexp from Section 1.2 and similarly for Booleans.
- We can simply decide that an arithmetic expression is a function from states to numbers (e.g., state  $\rightarrow \mathbb{N}$ ) and a Boolean expression is a predicate over states (e.g., state  $\rightarrow$  bool or state  $\rightarrow$  Prop). A state is a mapping from program variables to values. Thus, x + y + 1 would be represented by the function  $\lambda$ s: state, s "x" + s "y" + 1, and a  $\neq$  b would be represented by the predicate  $\lambda$ s: state, s "a"  $\neq$  s "b".

These two options correspond to the difference between deep and shallow embeddings. A *deep embedding* of some syntax (expression, formula, program, etc.) consists of an abstract syntax tree specified in the proof assistant (e.g., aexp) with a semantics (e.g., eval). In contrast, a *shallow embedding* simply reuses the corresponding mechanisms from the logic (e.g., functions and predicates).

A deep embedding allows us to reason about a program's syntax. A shallow embedding is more lightweight, because we can use it directly, without having to define a semantics. A shallow embedding is its own semantics.

In Chapter 6, we used a shallow embedding of effectful programs. Here, we will use a deep embedding of programs (which we find interesting and want to study closely) and a shallow em-

bedding of arithmetic and Boolean expressions (which we find less interesting). Our Lean definition of programs follows:

```
inductive stmt : Type
| skip : stmt
| assign : string \rightarrow (state \rightarrow N) \rightarrow stmt
| seq : stmt \rightarrow stmt \rightarrow stmt
| ite : (state \rightarrow Prop) \rightarrow stmt \rightarrow stmt
| while : (state \rightarrow Prop) \rightarrow stmt \rightarrow stmt
infixr ';; ': 90 := stmt.seq
```

The infix syntax S;; T abbreviates stmt.seq S T. We cannot use a single semicolon; because it means "and then" (Section 7.1).

The correspondence between the inductive type's constructors and the WHILE grammar rules should be clear. Variables are represented by strings. The type state is defined as  $\mathtt{string} \to \mathbb{N}$ , a mapping from variable names to values. For simplicity, our program variables are all of type natural number, and (somewhat unusually) all possible variable names exist in the state and are assigned a value.

The following small program illustrates deep embedding:

```
def silly_loop : stmt :=
stmt.while (λs, s "x" > s "y")
  (stmt.skip ;; stmt.assign "x" (λs, s "x" - 1))
```

# 8.3 Big-Step Semantics

An operational semantics corresponds to an idealized interpreter. There are two main variants: big-step semantics and small-step semantics. We will start by giving a big-step semantics to our WHILE language.

In a big-step operational semantics (also called natural semantics), judgments have the form  $(S, s) \Longrightarrow t$  and the following intuitive interpretation:

Starting in a state s, executing S may terminate in the state t.

For deterministic languages, since programs always have a single outcome, "may terminate" means the same as "must terminate" or "terminates."

In accordance with the definition of WHILE programs, a state s is a function of type  $string \rightarrow \mathbb{N}$ . An example judgment follows:

$$(x := x + y; y := 0, [x \mapsto 3, y \mapsto 5]) \Longrightarrow [x \mapsto 8, y \mapsto 0]$$

We use the informal notation  $[x \mapsto 3, y \mapsto 5]$  to represent the function  $\lambda v$ , if v = "x" then 3 else if v = "y" then 5 else o and similarly for  $[x \mapsto 8, y \mapsto o]$ . Intuitively, the judgment holds.

The traditional way to specify such a semantics is through a formal system of derivation rules, in the style of the typing rules presented in Sections 1.1.3 and 3.6. The derivation rules for big-step semantics judgments are given below. The rules can be seen as an idealized interpreter for WHILE programs.

$$\frac{(skip,s)\Rightarrow s}{(x:=a,s)\Rightarrow s\{x\mapsto a\,s\}}$$

$$\frac{(s,s)\Rightarrow t\quad (T,t)\Rightarrow u}{(s;T,s)\Rightarrow u}$$

$$\frac{(s,s)\Rightarrow t}{(if\,b\,then\,S\,else\,T,\,s)\Rightarrow t}$$

$$\frac{(T,s)\Rightarrow t}{(if\,b\,then\,S\,else\,T,\,s)\Rightarrow t}$$

$$\frac{(T,s)\Rightarrow t}{(if\,b\,then\,S\,else\,T,\,s)\Rightarrow t}$$

$$\frac{(s,s)\Rightarrow t\quad (while\,b\,do\,S,\,t)\Rightarrow u}{(while\,b\,do\,S,\,s)\Rightarrow u}$$

$$While-True if\,b\,s\,is\,true$$

$$\frac{}{\text{(while b do S, s)} \Rightarrow s} \text{WHILE-FALSE} \quad \text{if b s is false}$$

In the rules, a s denotes the value of arithmetic expression a in state s, and similarly for b s. Moreover, the syntax  $s\{x \mapsto n\}$  represents the state that is identical to s except that it maps the variable x to n. Formally:

$$s\{x \mapsto n\} = (\lambda v, \text{ if } v = x \text{ then } n \text{ else } s \text{ } v)$$

This syntax and its automation are provided by LoVelib.

As an exercise, let us derive the example judgment above. Let  $s := [x \mapsto 3, y \mapsto 5], t := [x \mapsto 8, y \mapsto 5],$  and  $u := [x \mapsto 8, y \mapsto 0].$  Then we have

$$\frac{1}{(x := x + y, s) \Longrightarrow t} ASN \qquad \frac{1}{(y := 0, t) \Longrightarrow u} ASN \qquad SEQ$$

The derivation rules can be read intuitively. Consider SEQ:

If (1) executing S in state s leads to state t and (2) executing T in state t leads to state u, then executing the sequential composition S; T in state s leads to state u.

The conditions (1) and (2) correspond to the two premises of SEQ.

The most complicated rule is undoubtedly WHILE-TRUE. Intuitively, it can be understood as follows:

Assume condition b is true in state s. If (1) executing S in state s leads to state t and (2) executing while b do S from state t leads to state u, then executing while b do S in state s leads to state u.

Another way to think about WHILE-TRUE is in terms of sequential composition. If the loop condition is true, while b do S should be equivalent to the compound statement S; while b do S. The two

premises of WHILE-TRUE correspond to the two premises of the instance of the SEQ rule for S; while b do S.

In Lean, a big-step semantics judgment is represented by an inductive predicate whose introduction rules closely follow the derivation rules above:

```
inductive big_step : stmt × state → state → Prop
| skip {s} :
 big step (stmt.skip, s) s
\mid assign \{x \ a \ s\}:
  big step (stmt.assign x a, s) (s{x \mapsto a s})
| seq {S T s t u} (hS : big_step (S, s) t)
    (hT : big_step (T, t) u) :
  big step (S ;; T, s) u
| ite true \{b : state \rightarrow Prop\} \{S T s t\} (hcond : b s)
    (hbody : big_step (S, s) t) :
  big_step (stmt.ite b S T, s) t
| ite_false {b : state \rightarrow Prop} {S T s t} (hcond : \neg b s)
    (hbody : big_step (T, s) t) :
  big_step (stmt.ite b S T, s) t
| while_true {b : state → Prop} {S s t u} (hcond : b s)
    (hbody : big_step (S, s) t)
    (hrest : big step (stmt.while b S, t) u) :
  big step (stmt.while b S, s) u
| while_false {b : state \rightarrow Prop} {S s} (hcond : \neg b s) :
  big step (stmt.while b S, s) s
```

We use of implicit arguments, within curly braces, for many of the variables corresponding to the derivation rule's metavariables.

Using an inductive predicate as opposed to a recursive function allows us to cope with nontermination (e.g., a diverging while) and, for languages richer than WHILE, nondeterminism. It also arguably provides a nicer syntax, closer to the judgment rules that are traditionally used in the scientific literature. If we were to attempt a recursive definition such as

```
\begin{array}{ll} \text{def eval} : \text{stmt} \rightarrow \text{state} \rightarrow \text{state} \\ | \text{stmt.skip} & \text{s := s} \end{array}
```

instead, we would face nontermination of the stmt.while case. Indeed, since the program stmt.while ( $\lambda_{-}$ , true) stmt.skip loops forever, trying to evaluate it using eval would never return. (There are other issues with the above definition, due to the use of Prop to represent Boolean expressions.)

Equipped with a big-step semantics, we can reason about concrete programs such as the one defined in Section 8.2 and prove lemmas such as the following:

```
lemma silly_loop_from_1_big_step:  (silly_loop, (\lambda_{-}, o)\{"x" \mapsto 1\}) \Longrightarrow (\lambda_{-}, o) := \\ begin \\ rw silly_loop, \\ apply big_step.while_true, \\ \{ simp \}, \\ \{ apply big_step.seq, \\ \{ apply big_step.skip \}, \\ \{ apply big_step.assign \} \}, \\ \{ simp, \\ apply big_step.while_false, \\ linarith \} \\ end \\
```

# **8.4 Properties of the Big-Step Semantics**

Equipped with a big-step semantics, we can reason about concrete programs, proving theorems relating final states with initial states. Perhaps more interestingly, we can prove properties of the programming language, such as determinism and nontermination.

We start with determinism. It may seem like a trivial property, but it is easy to mistype a rule and obtain nondeterminism. For example, in the rule for assignment, if we mistakenly write big\_step (stmt.assign x a, s) (s{y  $\mapsto$  a s}) with y instead of x, we suddenly can use the rule to modify any variable y we please. In other words, an execution of the program could modify any variable at random. So let us verify that our WHILE language really is deterministic:

```
lemma big_step_deterministic {S s l r} (hl : (S, s) \Longrightarrow l) (hr : (S, s) \Longrightarrow r) : l = r
```

The Lean proof is in the demonstration file associated with this chapter. We content ourselves with an informal proof sketch:

The proof is by rule induction over  $(s, s) \Longrightarrow l$ .

CASE SKIP: To have (skip, s)  $\Longrightarrow$  l or (skip, s)  $\Longrightarrow$  r, we need l = r = s.

CASE ASN: Similar to SKIP.

CASE SEQ: We have the hypotheses (S, s)  $\Longrightarrow$  t, (T, t)  $\Longrightarrow$  l, (S, s)  $\Longrightarrow$  t', and (T, t')  $\Longrightarrow$  r and the induction hypotheses  $\forall$ r, (S, s)  $\Longrightarrow$  r  $\rightarrow$  t = r and  $\forall$ r, (T, t)  $\Longrightarrow$  r  $\rightarrow$  l = r. From the first induction hypothesis together with (S, s)  $\Longrightarrow$  t', we derive t = t'. From the second induction hypothesis together with (T, t')  $\Longrightarrow$  r, we derive l = r.

CASE IF-TRUE: Since b s is true, (if b then S else T, s)  $\Rightarrow$  r can only have been derived using IF-TRUE and thus (S, s)  $\Rightarrow$  r. The induction hypothesis is  $\forall$ r, (S, s)  $\Rightarrow$  r  $\rightarrow$  l = r. We can apply (S, s)  $\Rightarrow$  r to it to obtain l = r.

CASE IF-FALSE: Similar to IF-TRUE.

CASE WHILE-TRUE: Similar to SEQ.

CASE WHILE-FALSE: Similar to SKIP.

Given that the WHILE language is deterministic, for the big-step semantics, termination would amount to the following:

```
lemma big_step_terminates \{S \ s\}:
\exists t, (S, s) \Rightarrow t
```

This property means that for every statement S and state s, there exists a state t such that executing S starting in s may terminate in t. Because WHILE is deterministic, "may terminate" means the same as "must terminate."

However, the property does not hold. The big-step semantics can be used to show that stmt.while ( $\lambda_{-}$ , true) S will never terminate, regardless of the start state:

```
lemma big_step_doesnt_terminate \{S \ s \ t\}:
\neg \ (stmt.while \ (\lambda_{,} \ true) \ S, \ s) \implies t :=
begin
intro hw,
induction' hw,
case while_true \{
assumption \},
case while_false \{
cc \}
```

When reasoning about an inductive predicate, it is often convenient to use inversion rules (Section 5.5). Accordingly, we prove the following rules:

```
(b s \land (S, s) \Rightarrow t) \lor (\neg b s \land (T, s) \Rightarrow t)

lemma big_step_while_iff {b S s u} :
  (stmt.while b S, s) \Rightarrow u \leftrightarrow
  (\existst, b s \land (S, s) \Rightarrow t \land (stmt.while b S, t) \Rightarrow u)
  \lor (\neg b s \land u = s)

lemma big_step_while_true_iff {b : state \rightarrow Prop} {S s u}
  (hcond : b s) :
  (stmt.while b S, s) \Rightarrow u \leftrightarrow
  (\existst, (S, s) \Rightarrow t \land (stmt.while b S, t) \Rightarrow u)

@[simp] lemma big_step_while_false_iff {b : state \rightarrow Prop}
  {S s t} (hcond : \neg b s) :
  (stmt.while b S, s) \Rightarrow t \leftrightarrow t = s
```

We add most of the rules to the simp set. We leave out big\_step\_while\_iff and big\_step\_while\_true\_iff because they could make simp loop, due to the presence on the right-hand side of a term that matches the left-hand side.

## 8.5 Small-Step Semantics

A limitation of big-step semantics is that they do not let us reason about intermediate states. From a judgment  $(S, s) \Longrightarrow t$ , all we see is the initial state s and the final state t. This is too coarsegrained to reason about multithreaded programs, where several processes can interact with each other's intermediate states. Moreover, for nondeterministic languages, big-step semantics offer no general way to express termination: A judgment indicates a possibility (executing s in state s may result in state t), not a necessity.

Small-step operational semantics provide a finer view. The transition predicate  $\Rightarrow$  has type stmt  $\times$  state  $\rightarrow$  stmt  $\times$  state  $\rightarrow$  Prop. Intuitively, (S, s)  $\Rightarrow$  (T, t) means that executing one step of program S in state s leaves the program T to be executed, in state t. If there is nothing left to be executed, we put skip.

An execution is a finite or infinite chain  $(S_0, s_0) \Rightarrow (S_1, s_1) \Rightarrow \cdots$  of "small"  $\Rightarrow$  steps. A pair (S, s) is called a *configuration*; it is *final* if no transition of the form  $(S, s) \Rightarrow (T, t)$  is possible, for any (T, t). A possible execution follows:

$$(x := x + y; y := 0, [x \mapsto 3, y \mapsto 5])$$

$$\Rightarrow (skip; y := 0, [x \mapsto 8, y \mapsto 5])$$

$$\Rightarrow (y := 0, [x \mapsto 8, y \mapsto 5])$$

$$\Rightarrow (skip, [x \mapsto 8, y \mapsto 0])$$

If we take the analogy of a computer processor, the S component of a configuration (S, s) can be thought of as a program counter, which indicates which instructions should be executed next. The step-by-step execution of a program resembles running a program in a debugger with a breakpoint at every step.

The valid small-step judgments are given as derivation rules:

$$(x := a, s) \Rightarrow (skip, s\{x \mapsto a s\})$$

$$\frac{(S, s) \Rightarrow (S', s')}{(S; T, s) \Rightarrow (S'; T, s')} SEQ-STEP$$

$$\frac{(skip; T, s) \Rightarrow (T, s)}{(skip; T, s) \Rightarrow (T, s)} SEQ-SKIP$$

$$\frac{(if b then S else T, s) \Rightarrow (S, s)}{(if b then S else T, s) \Rightarrow (T, s)} IF-FALSE if b s is false$$

(while b do S, s)  $\Rightarrow$  (if b then (S; while b do S) else skip, s)

The rules are reminiscent of the tennis game transition system of Section 5.1.2. These, too, specified small steps: from 0-0 to 15-0, to 15-15, and so on.

Unlike in the big-step semantics, there is no rule for skip in the small-step semantics. This is because a configuration of the form

(skip, s) is considered final; skip is understood as the statement whose execution is vacuous. By inspection of the rules, we can convince ourselves that a configuration is final if and only if its first component is skip.

Two rules concern sequential composition S; T. The first rule is applicable if some progress can be made executing S. But if S is skip, no progress can be made and the second rule applies.

The rules for if check the condition b and, depending on its truth value, put the then or the else branch as the remaining computation to perform.

For the while loop, there is a single unconditional rule that expands one iteration of the loop, introducing an if statement. It is then the role of the IF-TRUE and IF-FALSE rules to process the if. In the IF-TRUE case, we eventually reach the while loop again. This can continue forever for infinite loops.

In Lean, the small-step semantics is specified as follows:

```
inductive small_step : stmt × state → stmt × state → Prop
| assign {x a s} :
    small_step (stmt.assign x a, s) (stmt.skip, s{x ↦ a s})
| seq_step {S S' T s s'} (hS : small_step (S, s) (S', s')) :
    small_step (S ;; T, s) (S' ;; T, s')
| seq_skip {T s} :
    small_step (stmt.skip ;; T, s) (T, s)
| ite_true {b : state → Prop} {S T s} (hcond : b s) :
    small_step (stmt.ite b S T, s) (S, s)
| ite_false {b : state → Prop} {S T s} (hcond : ¬ b s) :
    small_step (stmt.ite b S T, s) (T, s)
| while {b : state → Prop} {S s} :
    small_step (stmt.while b S, s)
        (stmt.ite b (S ;; stmt.while b S) stmt.skip, s)
```

Based on a small-step semantics, we can *define* a big-step semantics as

```
(S, s) \Rightarrow t if and only if (S, s) \Rightarrow * (skip, t)
```

where p\* denotes the reflexive transitive closure of a binary predicate p. Alternatively, if we have already defined a big-step semantics, we can *prove* the above equivalence theorem to *validate* our definitions.

The main disadvantage of small-step semantics is that we now have two predicates,  $\Rightarrow$  and  $\Rightarrow$ \*, and the derivation rules and proofs tend to be more complicated than with big steps. This is clearly visible on the following example, which requires us to apply the lemma

```
star.head: ?r?a?b \rightarrow star?r?b?c \rightarrow star?r?a?c
```

#### once for each small step:

```
lemma silly loop from 1 small step :
  (silly loop, (\lambda, 0)\{"x" \mapsto 1\}) \Rightarrow *
  (stmt.skip, ((\lambda_{-}, 0) : state)) :=
begin
  rw silly_loop,
  apply star.head,
  { apply small_step.while },
  { apply star.head,
    { apply small_step.ite_true,
      simp }.
    { apply star.head,
      { apply small step.seq step,
        apply small step.seg skip },
      { apply star.head,
        { apply small_step.seq_step,
          apply small_step.assign },
        { apply star.head,
          { apply small_step.seq_skip },
          { apply star.head,
            { apply small_step.while },
             { apply star.head,
               { apply small_step.ite_false,
                 simp }.
               { simp } } } } }
```

end

## 8.6 Properties of the Small-Step Semantics

We can prove that a configuration (S, s) is final if and only if S = skip. This ensures that we have not forgotten a derivation rule and hence that the small-steps semantics cannot get stuck while executing a program. The lemma statement is as follows:

```
lemma small_step_final (S s): (\neg \exists T \ t, (S, s) \Rightarrow (T, t)) \leftrightarrow S = stmt.skip
```

The proof is by structural induction on S.

Like the big-step semantics, the small-step semantics is deterministic:

```
lemma small_step_deterministic {S s Ll Rr} (hl : (S, s) \Rightarrow Ll) (hr : (S, s) \Rightarrow Rr) : Ll = Rr
```

The proof is by rule induction on hl or hr.

For the small-step semantics, a configuration  $(S_0, s_0)$  terminates if all executions starting in it are finite:  $(S_0, s_0) \Rightarrow (S_1, s_1) \Rightarrow \cdots \Rightarrow (S_n, s_n)$ . It is nonterminating if there exists an infinite chain  $(S_0, s_0) \Rightarrow (S_1, s_1) \Rightarrow \cdots$ . The programming language as a whole is terminating if and only if all its configurations terminate. It is easy to show that the WHILE language is nonterminating, by taking  $S_0 := \text{stmt.while}(\lambda_-, \text{true}) \text{stmt.skip.}$  For any  $s_0$ , we then have

```
(S_0, S_0) \Rightarrow (S_0, S_0) \Rightarrow (S_0, S_0) \Rightarrow \cdots
```

We can define inversion rules about the small-step semantics, such as these:

```
lemma small_step_skip {S s t} :
\neg ((stmt.skip, s) \Rightarrow (S, t))
```

```
 \begin{aligned} & \texttt{@[simp] lemma small\_step\_seq\_iff} \; \{ \texttt{S T s Ut} \} \; : \\ & (\texttt{S}\;;\; \texttt{T},\; \texttt{s}) \; \Rightarrow \; \texttt{Ut} \; \leftrightarrow \\ & (\exists \texttt{S'}\; \texttt{t},\; (\texttt{S},\; \texttt{s}) \; \Rightarrow \; (\texttt{S'},\; \texttt{t}) \; \wedge \; \texttt{Ut} \; = \; (\texttt{S'}\;;\; \texttt{T},\; \texttt{t})) \\ & \lor \; (\texttt{S}\; = \; \texttt{stmt.skip} \; \wedge \; \texttt{Ut} \; = \; (\texttt{T},\; \texttt{s})) \\ & \texttt{@[simp] lemma small\_step\_ite\_iff} \; \{ \texttt{b S T s Us} \} \; : \\ & (\texttt{stmt.ite b S T},\; \texttt{s}) \; \Rightarrow \; \texttt{Us} \; \leftrightarrow \\ & (\texttt{b s } \wedge \; \texttt{Us} \; = \; (\texttt{S},\; \texttt{s})) \; \lor \; (\neg \; \texttt{b s } \wedge \; \texttt{Us} \; = \; (\texttt{T},\; \texttt{s})) \\ \end{aligned}
```

A more fundamental result is the equivalence between the bigstep and the small-step semantics:

```
lemma big_step_iff_star_small_step {S s t} : (S, s) \Rightarrow t \leftrightarrow (S, s) \Rightarrow * (stmt.skip, t)
```

Recall that  $\Rightarrow *$  denotes the reflexive transitive closure of the small-step predicate  $\Rightarrow$ . The theorem's proof is beyond the scope of this course. We refer to Chapter 7 of *Concrete Semantics: With Isabelle/HOL* [27] or to the demonstration file for this chapter.

## 8.7 Parallelism

The WHILE language is a sequential language: one in which statements are executed in turn, without interruptions or interference. In this section, we will consider PAR, an extension of WHILE with top-level parallelism. A PAR program consists of a list  $[S_1, \ldots, S_n]$  of WHILE programs. The individual WHILE programs are executed in parallel, as their own threads, using an interleaving semantics. This means that the semantics of the entire PAR program is given by interleaving small steps of the threads.

A PAR configuration (Ss, s) consists of a list Ss of WHILE programs and a state s. A PAR step can be defined as follows:

```
inductive par_step :
    nat → list stmt × state → list stmt × state → Prop
| intro {Ss Ss' S S' s s' i}
    (hi : i < list.length Ss)
    (hS : S = list.nth_le Ss i hi)</pre>
```

8.7. Parallelism 207

```
(hs: (S, s) \Rightarrow (S', s'))

(hS': Ss' = list.update_nth Ss i S'):

par_step i (Ss, s) (Ss', s')
```

Informally, from an arbitrary configuration (Ts ++ [S] ++ Us, s), we can make a small step to (Ts ++ [S'] ++ Us, s') whenever (S, s)  $\Rightarrow$  (S', s'). The par\_step predicate's first argument is the index of the thread that makes a transition.

To see that PAR is nondeterministic, consider the program  $[S_0, S_1]$ , with

```
S_0 := \text{stmt.assign } "x" (\lambda_1, 42) \quad S_1 := \text{stmt.assign } "x" (\lambda_1, 999)
```

and consider the start state s. Taking a par\_step o step results in the configuration ([stmt.skip,  $S_1$ ],  $s\{x \mapsto 42\}$ ), whereas taking a par\_step 1 step results in ([ $S_0$ , stmt.skip],  $s\{x \mapsto 999\}$ ).

Sometimes the nondeterminism is harmless, if the final result is the same. This is called *determinacy*. In particular, some languages enjoy the diamond property:

```
lemma par_step_diamond {i j Ss Ts Ts' s t t'}
    (hi : i < list.length Ss)
    (hj : j < list.length Ss)
    (hij : i ≠ j)
    (hT : par_step i (Ss, s) (Ts, t))
    (hT' : par_step j (Ss, s) (Ts', t')) :
∃u Us, par_step j (Ts, t) (Us, u) ∧
    par_step i (Ts', t') (Us, u)</pre>
```

The diamond property implies determinacy. However, PAR is not determinate, and hence cannot enjoy the diamond property. If we take par\_step o followed by par\_step 1, we inevitably reach the final state ([stmt.skip, stmt.skip], s $\{x \mapsto 999\}$ ), whereas if we start with par\_step 1 and then take par\_step 0, we inevitably reach ([stmt.skip, stmt.skip], s $\{x \mapsto 42\}$ ), a counterexample both to determinacy and to the diamond property.

The counterexample has two threads writing to the same variable x. If we were to restrict access to variables written by other

threads, we could salvage the diamond property. First, we need some auxiliary definitions:

```
def stmt.W : stmt \rightarrow set string
| stmt.skip
                        := 0
| (stmt.assign x _) := \{x\}
| (stmt.seg S T) := stmt.W S ∪ stmt.W T
| (stmt.ite \_ S T) := stmt.W S \cup stmt.W T
| (stmt.while S) := stmt.W S
def exp.R \{\alpha : \mathsf{Type}\} : (\mathsf{state} \to \alpha) \to \mathsf{set} \ \mathsf{string}
| f := \{x \mid \forall s \ n, \ f (s\{x \mapsto n\}) \neq f \ s\}
def stmt.R : stmt \rightarrow set string
| stmt.skip := \emptyset
| (stmt.assign \_ a) := \exp.R a
| (stmt.seg S T) := stmt.R S \cup stmt.R T
| (stmt.ite b S T) := \exp.R \ b \cup stmt.R \ S \cup stmt.R \ T
| (stmt.while b S) := exp.R b ∪ stmt.R S
def stmt.V : stmt \rightarrow set string
I S := stmt.W S \cup stmt.R S
```

The function stmt.W returns the set of variables written by a thread, stmt.R is the set of variables read by a thread, and stmt.V is the set of variables written or read by a thread. Notice the markedly semantic definition of exp.R: A variable x is read by a shallowly embedded arithmetic or Boolean expression f if there exists a state s and a value n such that setting x to n in s influences f's behavior. If we had a deep embedding of arithmetic and Boolean expressions, we could perform a syntactic analysis instead (as we do in stmt.R for programs).

Using the above definitions, we can revise the diamond property:

```
lemma par_step_diamond_VW_disjoint {i j Ss Ts Ts' s t t'}
    (hiS : i < list.length Ss)
    (hjT : j < list.length Ts)</pre>
```

8.7. Parallelism 209

```
(hij : i ≠ j)

(hT : par_step i (Ss, s) (Ts, t))

(hT' : par_step j (Ss, s) (Ts', t'))

(hWV : stmt.W (list.nth_le Ss i hiS)

\cap stmt.V (list.nth_le Ts j hjT) = \emptyset)

(hVW : stmt.V (list.nth_le Ss i hiS)

\cap stmt.W (list.nth_le Ts j hjT) = \emptyset) :

\existsu Us, par_step j (Ts, t) (Us, u) \land

par_step i (Ts', t') (Us, u)
```

This proposition looks plausible, but considering Conrad Watt's experience with WebAssembly and that of dozens of other researchers, we should not put too much faith in it until we have established it formally.

# **Chapter 9**

# **Hoare Logic**

If operational semantics corresponds to an idealized interpreter, *Hoare logic* corresponds to an idealized verifier. Hoare logic can be used to specify the semantics of a programming language, but it is particularly convenient to reason about concrete programs and prove them correct. It is named after its inventor, Charles Antony Richard (Tony) Hoare. Hoare logic is also called *axiomatic semantics*.

This chapter is heavily inspired by Chapter 12 of Concrete Semantics: With Isabelle/HOL [27].

## 9.1 Hoare Triples

Hoare logic is a framework for deducing valid correctness formulas in a mechanical way, using a set of derivation rules. It allows us to reason directly about a program's syntax, without concerning ourselves with its operational semantics. The approach is mechanical in the sense that the applicability of a derivation rule can easily be checked.

We start by introducing Hoare logic abstractly, without any connection to Lean. In a second step, we will see how we can embed Hoare logic judgments in Lean in a natural way. The basic judg-

ments of Hoare logic are called *Hoare triples*. They have the form {P} S {Q}, where S is a WHILE statement, and P and Q are logical formulas over the program variables. For the moment, we imagine the formulas as syntactic objects built using the familiar connectives and quantifiers. The intended meaning of a Hoare triple is as follows:

If the *precondition* P is true before S is executed and the execution terminates normally, the *postcondition* Q is true at termination.

This is a *partial correctness* statement: The program is correct *if* it terminates normally; otherwise, we do not know or care. For WHILE programs, the only way not to terminate normally is to enter an infinite loop. For other programming languages, infinite recursion and run-time errors such as division by zero may also result in divergence or abnormal termination.

Intuitively, all of the Hoare triples below should be valid:

The first three Hoare triples should be fairly natural. The fourth triple is vacuously true, since the precondition false can never be met. The part "If the precondition P is true" of the definition of Hoare triple is always false; hence the triple is true. The triple is equivalent to the proposition false  $\rightarrow$  b = 10, which holds for any value of b. As for the fifth triple, there are no guarantees that control will escape the loop (if i > 10 initially), but if it does escape, then the loop's condition must be false and hence we have the postcondition i = 10.

The following triples are bizarre but interesting:

9.2. Hoare Rules 213

```
{false} S {true}
{false} S {false}
{true} S {true}
{true} S {false}
```

The first two triples are true of any statement S (and therefore pointless): The precondition is never satisfied, so any postcondition holds vacuously. The third triple is always true as well, regardless of S. The fourth triple is true if S never terminates (e.g., S := while true do skip); otherwise, it is false.

## 9.2 Hoare Rules

Below we give a complete set of derivation rules for reasoning about WHILE programs:

$$\frac{}{\{P\} \text{ skip } \{P\}} \text{SKIP}}{\frac{}{\{Q[a/x]\} \text{ } x \text{ } := \text{ a } \{Q\}}} \text{ASN}}$$

$$\frac{\{P\} \text{ S } \{R\} \text{ } \{R\} \text{ } T \text{ } \{Q\}}{\{P\} \text{ S; } T \text{ } \{Q\}} \text{SEQ}}$$

$$\frac{\{P \land b\} \text{ S } \{Q\} \text{ } \{P \land \neg b\} \text{ } T \text{ } \{Q\}}{\{P\} \text{ if b then S else } T \text{ } \{Q\}} \text{IF}}$$

$$\frac{\{P \land b\} \text{ S } \{P\}}{\{P\} \text{ while b do S } \{P \land \neg b\}} \text{WHILE}}$$

$$\frac{P' \rightarrow P \text{ } \{P\} \text{ S } \{Q\} \text{ } Q \rightarrow Q'}{\{P'\} \text{ S } \{Q'\}} \text{CONSEQ}}$$

In the ASN rule, the syntax Q[a/x] denotes the condition Q in which all occurrences of x are replaced by a. The rule is sometimes pre-

sented as

$$\{Q[a]\} x := a \{Q[x]\}$$

where [x] factors out the occurrences of x in Q.

The ASN rule may seem counterintuitive because it works backwards: From the postcondition, it computes a precondition. Nevertheless, it correctly captures the semantics of the assignment statement, as illustrated below:

$$\{0 = 0\} \ x := 0 \ \{x = 0\}$$
  
 $\{0 = 0 \land y = 5\} \ x := 0 \ \{x = 0 \land y = 5\}$   
 $\{x + 1 \ge 5\} \ x := x + 1 \ \{x \ge 5\}$ 

Using elementary arithmetic, we can simplify the computed preconditions; for example, o = o is equivalent to true, and  $x + 1 \ge 5$  is equivalent to  $x \ge 4$ .

The SEQ rule requires us to come up with an intermediate condition R that holds after executing S and before executing T. Here is an example of SEQ in action:

$$\frac{\{a = 2\} \ b := a \ \{b = 2\}}{\{a = 2\} \ b := a; \ c := b \ \{c = 2\}} ASN$$

$$= \{a = 2\} \ b := a; \ c := b \ \{c = 2\}$$
SEQ

The While rule is the most intricate. The condition P is called an *invariant*. It is the pre- and the postcondition of the loop itself but also of its body. The body's precondition is strengthened with the knowledge that b must be true before executing the body. Similarly, the loop's postcondition is strengthened with the knowledge that b must be false when the loop exits.

Consider an execution of a loop with n iterations. Suppose that the initial state is  $s_o$  and that the state after the ith iteration of the loop is  $s_i$ . Then the following conditions will hold:

$$P s_0 P s_1 \wedge b s_1 \cdots P s_{n-1} \wedge b s_{n-1} P s_n \wedge \neg b s_n$$

9.2. Hoare Rules 215

If n = 0, we immediately have that  $P s_0 \land \neg b s_0$  and never enter the loop.

The Conseq rule is the only rule that has logical formulas among its premises, as opposed to Hoare triples. These conditions must be discharged, whether using pen and paper or a proof assistant. Conseq can be used to strengthen a precondition (i.e., make it more strict), weaken a postcondition (i.e., make it less strict), or both. An example derivation follows:

$$\frac{x > 8 \rightarrow x > 4}{\{x > 4\} \ y := x \ \{y > 4\}} Asn y > 4 \rightarrow y > 0 
\{x > 8\} \ y := x \ \{y > 0\}$$
Conseq

Reading the tree from top to bottom, we have strengthened the triple's precondition from x > 4 to x > 8 and weakened the postcondition from y > 4 to y > 0. We can also read the tree from the bottom up: To prove the triple  $\{x > 8\}$   $y := x \{y > 0\}$ , it suffices to prove  $\{x > 4\}$   $y := x \{y > 4\}$ , where the precondition is weakened and the postcondition is strengthened.

Except for Conseq, the rules are *syntax-driven*: We know which rule to apply in each case, simply by inspecting the statement at hand. For an assignment, we apply Asn; for a while loop, we apply While; and so on.

The rules SEQ, IF, and CONSEQ are *bidirectional*: Their conclusions are of the form {P} ... {Q}, for distinct mathematical metavariables P, Q. This can make them convenient to apply. By combining the other rules with CONSEQ, we can derive bidirectional variants:

$$\frac{P \to Q}{\{P\} \text{ skip } \{Q\}} SKIP'$$

$$\frac{P \to Q[a/x]}{\{P\} x := a \{Q\}} ASN'$$

$$\frac{ \{ \texttt{P} \ \land \ \texttt{b} \} \ \texttt{S} \ \{ \texttt{P} \} \quad \texttt{P} \ \land \ \neg \texttt{b} \ \rightarrow \ \texttt{Q} }{ \{ \texttt{P} \} \ \text{while} \ \texttt{b} \ \texttt{do} \ \texttt{S} \ \{ \texttt{Q} \} } \mathsf{WHILE}'$$

As an easy exercise, you could try to derive each of these rules from Skip, Asn, or While in conjunction with Conseq.

## 9.3 A Semantic Approach to Hoare Logic

A natural way to encode Hoare logic in Lean would be to proceed as we have done for the big- and small-step semantics: Define a syntactic notion of Hoare triple and an inductive predicate, with one introduction rule for each core Hoare rule; to represent preand postconditions, use predicates on states (state  $\rightarrow$  Prop). Then we could prove soundness with respect to the big-step semantics, meaning the following: Whenever {P} S {Q} is derivable, if P s and (S, s)  $\Longrightarrow$  t, then Q t. This is merely a logical rendition of the intuitive meaning of a Hoare triple:

If the precondition P is true before S is executed (i.e., P s) and the execution terminates normally (i.e., (S, s)  $\Rightarrow$  t), the postcondition Q is true at termination (i.e., Q t).

Instead of pursuing this approach, we propose to define Hoare triples semantically in Lean, in terms of the big-step semantics, so that they are *correct by definition*. Then we will derive the Hoare rules as lemmas, instead of stating them as introduction rules. In conjunction with the use of predicates to represent formulas, this approach is resolutely semantic.

Hoare triple (for partial correctness) are defined as follows:

```
def partial_hoare (P : state \rightarrow Prop) (S : stmt) (Q : state \rightarrow Prop) : Prop := \foralls t, P s \rightarrow (S, s) \Longrightarrow t \rightarrow Q t
```

Instead of writing partial\_hoare P S Q, we introduce some syntactic sugar to allow  $\{* P *\} S \{* Q *\}$ , which is closer to the informal syntax  $\{P\} S \{Q\}$ .

The core Hoare rules are stated as follows:

```
lemma skip intro {P} :
  {* P *} stmt.skip {* P *}
lemma assign_intro (P : state \rightarrow Prop) {x} {a : state \rightarrow N} :
  \{* \lambda s, P (s\{x \mapsto a s\}) *\} stmt.assign x a <math>\{* P *\}
lemma seg intro {P Q R S T} (hS : {* P *} S {* Q *})
    (hT : \{* Q *\} T \{* R *\}) :
  {* P *} S ;; T {* R *}
lemma ite_intro {b P Q : state → Prop} {S T}
    (hS: \{* \lambda s, P s \land b s *\} S \{* Q *\})
    (hT : \{* \lambda s, P s \land \neg b s *\} T \{* Q *\}) :
  {* P *} stmt.ite b S T {* Q *}
lemma while_intro (P : state → Prop) {b : state → Prop} {S}
    (h : \{* \lambda s, P s \land b s *\} S \{* P *\}):
  \{* P *\} stmt.while b S \{* \lambda s, P s \land \neg b s *\}
lemma consequence {P P' Q Q' : state → Prop} {S}
    (h : \{* P *\} S \{* Q *\}) (hp : \forall s, P' s \rightarrow P s)
    (hq : \foralls, Q s \rightarrow Q' s) :
  {* P' *} S {* Q' *}
```

All of the above lemmas have proofs based on the big-step semantics. Some of the triples—for example, the precondition in assign\_intro—need to refer to the state explicitly. In such case, we use a  $\lambda$ -expression to access it. Recall that P and  $\lambda s$ , P s are the same by  $\eta$ -conversion. Moreover, for a premise written informally as P  $\rightarrow$  Q, in Lean we must write  $\forall s$ , P s  $\rightarrow$  Q s. This reflects the difference between syntactic formulas and semantic predicates. As in Chapter 8, the syntax s{x  $\mapsto$  n} in the assignment rule denotes the state that is identical to s at all points except x and that maps x to n.

The following convenience rules can be derived from the core rules:

```
lemma consequence left (P' : state \rightarrow Prop) {P Q S}
     (h: \{* P *\} S \{* Q *\}) (hp: \forall s, P' s \rightarrow P s):
  {* P' *} S {* Q *}
lemma consequence_right (Q) {Q' : state → Prop} {P S}
     (h : \{* P *\} S \{* Q *\}) (hq : \forall s, Q s \rightarrow Q' s) :
  {* P *} S {* Q' *}
lemma skip intro' {PQ: state \rightarrow Prop} (h: \foralls, Ps \rightarrow Qs):
  {* P *} stmt.skip {* Q *}
lemma assign_intro' {P Q : state \rightarrow Prop} {x} {a : state \rightarrow \mathbb{N}}
     (h : \foralls, P s \rightarrow Q (s{x \mapsto a s})):
  {* P *} stmt.assign x a {* Q *}
lemma seq_intro' {P Q R S T} (hT : {* Q *} T {* R *})
     (hS : \{* P *\} S \{* Q *\}) :
  {* P *} S :: T {* R *}
lemma while_intro' {b P Q : state → Prop} {S}
     (I : state \rightarrow Prop)
     (hS: \{* \lambda s, I s \land b s *\} S \{* I *\})
     (hP: \foralls, Ps \rightarrow Is)
     (hQ : \foralls, \neg b s \rightarrow I s \rightarrow Q s) :
  {* P *} stmt.while b S {* Q *}
```

Using the bidirectional assign\_intro', we can derive a forward version of the assignment rule:

```
lemma assign_intro_forward (P) {x a} : 
 {* P *} 
 stmt.assign x a 
 {* \lambdas, \exists n_0, P (s{x \mapsto n_0}) \wedge s x = a (s{x \mapsto n_0}) *} := 
 begin 
 apply assign_intro', 
 intros s hP,
```

```
apply exists.intro (s x),
  simp [*]
end
```

The bound variable  $n_o$  stands for the value of x before the assignment. Hence, in the postcondition,  $s\{x \mapsto n_o\}$  denotes the state before the assignment. Since P is a precondition, we have P ( $s\{x \mapsto n_o\}$ ). In addition, the new value of x, given by s x, must be equal to the value of the arithmetic expression a evaluated in the old state  $s\{x \mapsto n_o\}$ .

The forward rule is less convenient than the backward rule, because the postcondition contains an existential quantifier. It is possible to state a backward rule in a similar style, revealing a hidden symmetry:

```
lemma assign_intro_backward (Q : state \rightarrow Prop) {x} {a : state \rightarrow N} : {* \lambdas, \existsn', Q (s{x \mapsto n'}) \wedge n' = a s *} stmt.assign x a {* Q *}
```

Notice that this existential quantifier can be eliminated using a one-point rule (Section 3.3). We then get the familiar backward rule assign\_intro, with  $\lambda s$ ,  $P(s\{x \mapsto a s\})$  as the precondition.

## 9.4 First Program: Exchanging Two Variables

Let us use the Hoare logic to verify our first program: a three-line program that exchanges the values of its variables a and b, using t for temporary storage. The program is defined as follows:

```
def SWAP : stmt := stmt.assign "t" (\lambda s, s "a");; stmt.assign "a" (\lambda s, s "b");; stmt.assign "b" (\lambda s, s "t")
```

The correctness statement is as follows:

end

```
lemma SWAP_correct (a_0 \ b_0 : \mathbb{N}) : \{* \ \lambda s, \ s \ "a" = a_0 \ \land \ s \ "b" = b_0 \ *\} SWAP \{* \ \lambda s, \ s \ "a" = b_0 \ \land \ s \ "b" = a_0 \ *\}
```

The logical variables  $a_o$  and  $b_o$  "freeze" the initial value of the program variables a and b so that we can refer to them in the postcondition. After all, it would make no sense to put  $\lambda s$ , s "a" = s "b"  $\wedge$  s "b" = s "a" as the postcondition.

The correctness proof follows:

```
begin
  apply partial_hoare.seq_intro',
  apply partial_hoare.seq_intro',
  apply partial_hoare.assign_intro,
  apply partial_hoare.assign_intro,
  apply partial_hoare.assign_intro',
  simp { contextual := tt }
```

The applications of the sequential composition and assignment rules are guided by the program's syntax. There are two sequential compositions and three assignments in the program and therefore as many invocations of the corresponding rules. We end up with a very aesthetically challenged subgoal:

```
 \begin{array}{l} \vdash \forall s: state, \\ s \; "a" = a_o \wedge s \; "b" = b_o \to \\ s \{ "t" \mapsto s \; "a" \} \{ "a" \mapsto s \{ "t" \mapsto s \; "a" \} \; "b" \} \\ \{ "b" \mapsto s \{ "t" \mapsto s \; "a" \} \{ "a" \mapsto s \{ "t" \mapsto s \; "a" \} \; "b" \} \; "t" \} \; "a" = b_o \wedge \\ s \{ "t" \mapsto s \; "a" \} \{ "a" \mapsto s \{ "t" \mapsto s \; "a" \} \; "b" \} \\ \{ "b" \mapsto s \{ "t" \mapsto s \; "a" \} \{ "a" \mapsto s \{ "t" \mapsto s \; "a" \} \; "b" \} \; "t" \} \; "b" = a_o \\ \end{array}
```

Fortunately, simp can reduce the subgoal dramatically, and if we enable *contextual rewriting*—an option that tells simp to use any assumptions in the goal's target (such as s "a" =  $a_o$  and s "b" =  $b_o$ ) as left-to-right rewrite rules—simp can even prove the subgoal.

The above proof could also be carried out directly in terms of the big-step semantics, without using the Hoare rules:

```
lemma SWAP_correct_2 (a_0 b_0 : \mathbb{N}) : {* \lambda s, s "a" = a_0 \wedge s "b" = b_0 *} SWAP {* \lambda s, s "a" = b_0 \wedge s "b" = a_0 *} := begin intros s th has the photograph intros s the hat ep, cases' hat ep, cases' hat ep, cases' hat ep_1, cases' hat ep_1, cases' hat ep_1, finish end
```

The finish tactic combines various techniques, including normalization of logical connectives and quantifiers, simplification, congruence closure, and quantifier instantiation. Like tautology (Section 6.8), it provides strong automation that can ruin many logical exercises if introduced too early in a course.

## 9.5 Second Program: Adding Two Numbers

Our second example computes m+n, leaving the result in m, using only these primitive operations: k+1, k-1, and  $k\neq 0$  (for arbitrary k):

```
def ADD : stmt := stmt.while (\lambdas, s "n" \neq 0) (stmt.assign "n" (\lambdas, s "n" - 1) ;; stmt.assign "m" (\lambdas, s "m" + 1))
```

Because of the presence of a while loop, the proof is more involved:

```
lemma ADD_correct (n_0 \ m_0 : \mathbb{N}) : \{ * \lambda s, s "n" = n_0 \land s "m" = m_0 * \} ADD \{ * \lambda s, s "n" = 0 \land s "m" = n_0 + m_0 * \} :=
```

```
partial_hoare.while_intro' (λs, s "n" + s "m" = n₀ + m₀)
begin
    apply partial_hoare.seq_intro',
    { apply partial_hoare.assign_intro },
    { apply partial_hoare.assign_intro',
        simp,
        intros s hnm hnz,
        rw ←hnm,
        cases s "n",
        { finish },
        { simp [nat.succ_eq_add_one],
              linarith } }
end
(by simp { contextual := true })
(by simp { contextual := true })
```

The first step is to apply the derived while rule with a loop invariant. Our invariant is that the sum of the program variables n and m must be equal to the desired mathematical result  $n_0 + m_0$ , where  $n_0$  and  $m_0$  correspond to the initial values of n and m, as required by the precondition.

How did we come up with this invariant? Even for a simple loop, finding a suitable invariant can be challenging. The difficulty is that the invariant must

- 1. be true before we enter the loop;
- 2. remain true after each iteration of the loop if it was true before the iteration;
- 3. be strong enough to imply the desired loop postcondition.

An invariant such as true meets requirements 1 and 2 but usually not 3. Similarly, false meets requirements 2 and 3 but usually not 1. In practice, suitable invariants tend to be of the form

work done + work remaining = desired result

where + stands for some appropriate operator (not necessarily addition). When we enter the loop, work done will often be  $\circ$  (or

some other appropriate "zero" value), and the invariant becomes

#### work remaining = desired result

This invariant would have to be provable at the beginning of the loop—either from the postcondition of the previous statement or from the desired precondition of the entire program if there is no previous statement. When we exit the loop, work remaining should be o (or some variant thereof), and the invariant becomes

#### work done = desired result

Often, work done takes the form of a variable in which we accumulate the result, whereas work remaining is similar to desired result but depends on program variables whose values change inside the loop and must account for work done.

For the ADD program's loop, the work done is m, the work remaining is n, and the desired result is  $n_o + m_o$ . When entering the loop, the invariant  $m + n = n_o + m_o$  holds because then  $m = m_o$  and  $n = n_o$ . (Unusually, the work done is not o for this example, because we reuse the input m as our accumulator, as an optimization.) When exiting the loop, we have that n = o, so the invariant becomes  $m = n_o + m_o$ . We can retrieve the result from m.

The while\_intro' lemma is used directly as a proof term. It gives rise to three subgoals. The proofs that the invariant is implied by the desired precondition and that it implies the desired postcondition are trivial: They consist of a call to simp with contextual rewriting enabled. The only nontrivial subgoal is the condition that executing the body maintains the invariant. There is some ugly arithmetic on natural numbers necessary due to the use of the truncated minus operator. Our proof simply performs a case distinction to isolate the zero case (where o - 1 = o) from the nonzero case.

For this example, Hoare logic really helps. Reasoning directly about the operational semantics would be inconvenient, because we would need induction to reason about the while loop. With

Hoare logic, this induction is performed out once and for all in the proof of the while\_intro rule.

## 9.6 Finish Tactic

#### finish

finish 
$$[[h_1, ..., h_m]]$$
 [using  $[k_1, ..., k_n]$ ]

The finish tactic first normalizes and simplifies the hypotheses, using the optionally supplied lemmas  $h_1, \ldots, h_m$ . Then it applies the congruence closure (cc) together with the optionally supplied lemmas  $k_1, \ldots, k_n$ . Universally quantified hypotheses and lemmas are instantiated automatically. Like cc, finish either fully succeeds or fails. The tactic is complete for propositional logic.

## 9.7 A Verification Condition Generator

Verification condition generators (VCGs) are programs that apply Hoare logic rules, producing verification conditions that must be proved manually. We can think of them as mechanical civil servants that take care of the Hoare logic bureaucracy. As users, we must provide strong enough loop invariants as annotations in our programs. Hundreds of program verification tools are based on these principles.

VCGs typically work backwards from the postcondition, using backward rules—rules stated to have an arbitrary Q as their postcondition. This works well because the central rule of Hoare logic—the assignment rule—is backward. If it were not for the assignment statement, the state would never change, and there would be no need for Hoare logic.

We can use Lean's metaprogramming framework to define a simple VCG. First, we introduce a constant called stmt.while\_inv

that carries a user-supplied invariant P in addition to the loop condition b and body S:

```
def stmt.while_inv (I b : state \rightarrow Prop) (S : stmt) : stmt :=
stmt.while b S
```

We provide two Hoare rules for the construct: a backward rule and a bidirectional rule. Both are justified in terms of the bidirectional while intro' rule:

```
lemma while_inv_intro {b I Q : state \rightarrow Prop} {S} (hS : {* \lambda s, I s \lambda b s *} S {* I *}) (hQ : \forall s, \cap b s \rightarrow I s \rightarrow Q s) : {* I *} stmt.while_inv I b S {* Q *} := while_intro' I hS (by cc) hQ lemma while_inv_intro' {b I P Q : state \rightarrow Prop} {S} (hS : {* \lambda s, I s \lambda b s *} S {* I *}) (hP : \forall s, P s \rightarrow I s) (hQ : \forall s, \cap b s \rightarrow I s \rightarrow Q s) : {* P *} stmt.while_inv I b S {* Q *} := while_intro' I hS hP hQ
```

The rules simply use the invariant annotation as their invariant. When using the framework, we will have to be careful to annotate all while loops with suitable invariants. If we specify a wrong invariant, we will face unprovable subgoals, indicating that we must revise the invariant.

The code of the VCG is fairly concise:

```
| '(stmt.assign _ _)
                            : =
    tactic.applvc
      (if expr.is_mvar P then ''partial_hoare.assign_intro
       else ''partial hoare.assign intro')
  | '(stmt.seq )
   tactic.applyc ''partial hoare.seg intro'; vcg
  '(stmt.ite _ _ _)
   tactic.applyc ''partial_hoare.ite_intro; vcg
  | '(stmt.while_inv _ _ _) :=
   tactic.applyc
      (if expr.is_mvar P then ''partial_hoare.while_inv_intro
       else ''partial_hoare.while_inv_intro');
      vcg
                            :=
   tactic.fail (to_fmt "cannot analyze " ++ to_fmt S)
 end
                           := pure ()
end
```

The VCG extracts the current goal's target and inspects it. If it is a Hoare triple, it inspects its precondition P and statement S. If the precondition is a metavariable (e.g., ?P), it applies a backward rule if there exists one (e.g., the standard assignment rule assign\_intro) because this will instantiate the metavariable. Otherwise, a bidirectional rule is used, with an arbitrary variable as its precondition, which can be matched against the target's precondition. For while loops, we only consider programs that use stmt.while\_inv, because we cannot guess the invariant programmatically.

The VCG calls itself recursively on all newly emerging subgoals (via the tactic combinator;) for the compound statements.

# 9.8 Second Program Revisited: Adding Two Numbers

Using the verification condition generator, we can revisit the correctness proof for the ADD program presented above:

```
lemma ADD_correct<sub>2</sub> (n_o m_o : \mathbb{N}):
  \{* \lambda s, s "n" = n_0 \wedge s "m" = m_0 *\}
  ADD
  \{* \lambda s, s "n" = 0 \land s "m" = n_0 + m_0 *\} :=
show \{* \lambda s, s "n" = n_0 \land s "m" = m_0 *\}
      stmt.while_inv (\lambdas, s "n" + s "m" = n_0 + m_0)
         (\lambda s, s "n" \neq o)
         (stmt.assign "n" (λs, s "n" - 1) ;;
          stmt.assign "m" (\lambda s, s "m" + 1))
      \{* \lambda s, s "n" = 0 \land s "m" = n_0 + m_0 *\}, from
  begin
     vcg; simp { contextual := tt },
     intros s hnm hnz,
     rw \leftarrow hnm,
     cases s "n",
     { finish },
     { simp [nat.succ_eq_add_one],
       linarith }
  end
```

We use show to annotate the while loop with an invariant. Recall that the show command restates the goal in a computationally equivalent way. Here, we use this facility to replace stmt.while by stmt.while\_inv, which equals stmt.while by definition. The program and its pre- and postconditions are otherwise the same as in the lemma statement.

We invoke vcg as the first proof step. This will apply all the necessary Hoare rules and leave us with some subgoals. We first simplify them with contextual rewriting enabled; this already takes care of two of the three subgoals. For the third subgoal, the proof is as in Section 9.5.

## 9.9 Hoare Triples for Total Correctness

The focus so far in this chapter has been on partial correctness. When we state the Hoare triple {P} S {Q}, we claim that the final state will satisfy Q if the program S terminates, but we say nothing when S does not terminate. In particular, we can prove any post-condition for the program while true do skip. This is admittedly too liberal: If you are asked to program a sorting algorithm at an exam, you should certainly not give while true do skip as your answer.

Total correctness is a stronger notion that asserts, in addition to partial correctness, that the program terminates normally. One reason to focus on partial correctness is that it is a necessary component of total correctness. It is also simpler, which is a benefit in the classroom.

The Hoare triples for total correctness have the form [P] S [Q], with the following intended meaning:

If the precondition P holds before S is executed, the execution terminates normally and the postcondition Q holds in the final state.

For deterministic programs, this can be expressed equivalently as follows:

If the precondition P holds before S is executed, there exists a state in which execution terminates normally and the postcondition Q holds in that state.

Here is an example Hoare triple that is valid:

```
[i < 10] while i < 10 do i := i + 1[i = 10]
```

For the WHILE language, the distinction between partial and total correctness only concerns while loops (and programs containing them). The Hoare rule for while must now be annotated by

a *variant* V—a natural number that decreases by one or more with each iteration:

$$\frac{ \left[ P \ \land \ b \ \land \ V \ = \ v_{o} \right] \ S \ \left[ P \ \land \ V \ < \ v_{o} \right] }{ \left[ P \right] \ while \ b \ do \ S \ \left[ P \ \land \ \neg b \right] } WHILE-VAR$$

Here,  $v_0$  is a logical variable that freezes V's initial value and whose scope is the entire premise, whereas V is a mathematical metavariable (like P, b, and S). For the example above, we could take 10 - i as the variant (or 50 - i or 1024 - i \* i).

Consider an execution  $s_0$ ,  $s_1$ , ...,  $s_{n-1}$ ,  $s_n$  corresponding to n loop iterations, as in Section 9.2. The following conditions will hold:

## 9.10 Summary of New Lean Constructs

#### **Tactics**

## **Chapter 10**

## **Denotational Semantics**

We review a third way to specify the semantics of a programming language: denotational semantics. Denotational semantics directly specifies the meaning of programs as a mathematical object. If operational semantics corresponds to an idealized interpreter and Hoare logic corresponds to an idealized verifier, then denotational semantics corresponds to an idealized compiler.

The core of this chapter is modeled closely after Chapter 11 of Concrete Semantics: With Isabelle/HOL [27].

## 10.1 Compositionality

A denotational semantics defines the meaning of each program as a mathematical object. Abstractly, it can be viewed as a function

$$[\![\,]\!]: \mathsf{syntax} \to \mathsf{semantics}$$

A key property of denotational semantics is *compositionality*: The meaning of a compound statement should be defined in terms of the meaning of its components. Consider the straightforward definition

$$[\![S]\!] = \{(s, t) \mid (S, s) \Longrightarrow t\}$$

in terms of the big-step semantics ( $\Longrightarrow$ ). Actually, because Lean supports only variables on the left-hand side of the vertical bar, we must write

$$[S] = \{st \mid (S, prod.fst st) \Longrightarrow prod.snd st\}$$

This definition specifies the desired semantics, but it does not qualify as a denotational semantics due to lack of compositionality: The meaning of the compound statements (sequential composition, if then else, and while) is given directly, without using the denotation of their components.

A fully compositional definition allows us to reason equationally about programs, which is often more convenient than using the introduction, elimination, and inversion principles of  $\implies$ . In essence, we want structurally recursive equations of the form

with no occurrences of S and T in the right-hand sides except as arguments to  $\[ \]$  .

An evaluation function for arithmetic expressions

$$\texttt{eval}: \underbrace{\texttt{aexp}}_{\texttt{syntax}} \to \underbrace{(\texttt{string} \to \mathbb{Z}) \to \mathbb{Z}}_{\texttt{semantics}}$$

satisfying equations such as

eval (
$$aexp.add e_1 e_2$$
) env = eval  $e_1$  env + eval  $e_2$  env

is a denotational semantics, because the semantics of aexp.add  $e_1$   $e_2$  is defined in terms of the semantics of  $e_1$  and of  $e_2$ .

Denotational semantics are a natural match for arithmetic expressions but also functional programs. Now we want a convenient denotational semantics for imperative programs. Because of while loops, which are not guaranteed to terminate, we need to develop some additional mathematical notions to reach the point where we can formulate the desired semantics.

#### 10.2 A Relational Denotational Semantics

Denotational semantics for deterministic languages are normally given as a function from prestate to poststate, but relations are more general and more convenient to manipulate. We present a relational denotational semantics.

A denotational semantics of a program will be a mathematical object of type set (state  $\times$  state). The relational approach was also used for the big-step semantics, which took the form of a predicate of type state  $\rightarrow$  state  $\rightarrow$  Prop. These two types are isomorphic, but set  $\alpha$ , defined as  $\alpha \rightarrow$  Prop, supports convenient operations and notations, such as  $\emptyset$ ,  $\cup$ ,  $\cap$ ,  $\in$ , and  $\{\ldots \mid \ldots\}$  (Section 6.7).

Our semantics is called denote, with  $[\![\,]\!]$  as syntactic sugar. We start with the first four equations of the definition, keeping while for later:

The skip statement is interpreted as the identity relation over states—i.e., the set of all tuples of the form (s, s). This captures the desired semantics of skip: The poststate is always identical to the prestate.

The semantics of assignment is the set of tuples where the second component reflects the result of the assignment.

The semantics of sequential composition is elegantly expressed as a relational composition  $\bigcirc$ , which is defined by the equation

```
r_1 \bigcirc r_2 = \{ac \mid \exists b, (prod.fst ac, b) \in r_1 \land (b, prod.snd ac) \in r_2\}
```

The definition is perhaps easier to relate to when expressed as

$$r_1 \ominus r_2 = \{(a, c) \mid \exists b, (a, b) \in r_1 \land (b, c) \in r_2\}$$

Unfortunately, Lean cannot process that form. It requires the left part of a comprehension to be a single variable.

The semantics of if then else is given as the union of the semantics of the two branches, restricted to include only the tuples whose first component makes the Boolean condition true or false, depending on the branch. The restriction operator is defined by

$$r \mid p = \{ab \mid ab \in r \land p (prod.fst ab)\}$$

A more intuitive definition would have been

$$r \mid p = \{(a, b) \mid (a, b) \in r \land p a\}$$

but again this is not supported by Lean. Note two degenerate cases: If  $p := (\lambda_{-}, \text{ true})$ , then  $r \mid p = r$ , and if  $p := (\lambda_{-}, \text{ false})$ , then  $r \mid p = \emptyset$ .

The difficulties arise when we try to define the semantics of while loops. We would like to write

```
| (stmt.while b S) := ((denote S \bigcirc denote (stmt.while b S)) \downarrow b) \cup (Id \downarrow (\lambdas, \neg b s))
```

but this is ill-founded due to the recursive call on stmt.while b S. We need something else. What we are looking for on the right-hand side is some term X that satisfies the equation

$$X = ((denote S \cap X) \mid b) \cup (Id \mid (\lambda s, \neg b s))$$

We are looking for what mathematicians call a fixpoint. The next four sections are concerned with building an operator lfp that computes the fixpoint for a given equation. Using lfp, we will be able to define the semantics of while loops by

10.3. Fixpoints 235

## 10.3 Fixpoints

A fixpoint (or fixed point) of f is a solution for X in the equation

$$X = f X$$

In general, fixpoints may not exist at all for some f (e.g., f := nat. succ), or there may be several fixpoints (e.g., f :=  $(\lambda x, x)$ ). Under some conditions on f, a unique least fixpoint and a unique greatest fixpoint are guaranteed to exist.

Consider the following fixpoint equation, with  $X : \mathbb{N} \to Prop$ :

$$X = (\lambda n : \mathbb{N}, n = 0 \vee (\exists m : \mathbb{N}, n = m + 2 \wedge X m))$$

This equation is the  $\beta$ -reduced variant of an equation with the right format:

$$X = \overbrace{(\lambda(p: \mathbb{N} \to Prop) (n: \mathbb{N}), n = o \lor (\exists m: \mathbb{N}, n = m + 2 \land p m))}^{f} X$$

A solution is X := even, the predicate that characterizes the even natural numbers. Recall that we proved the inversion rule

even 
$$n \leftrightarrow n = 0 \lor (\exists m : \mathbb{N}, n = m + 2 \land \text{even m})$$

in Section 5.5. It turns out that even is the only fixpoint. In general, the least and greatest fixpoint may be different. Consider the equation

$$X = (\lambda p, p) X$$

for X :  $\mathbb{N} \to \mathsf{Prop}$ . The least fixpoint is  $\lambda_-$ , False and the greatest fixpoint is  $\lambda_-$ , True. By convention, we have False < True and thus  $(\lambda_-$ , False) <  $(\lambda_-$ , True). Similarly,  $\emptyset$  < @set.univ  $\alpha$  for any inhabited type  $\alpha$ .

For the semantics of while loops, X will have type set (state  $\times$  state) of relations between states, and f will correspond to either taking one extra iteration of the loop (if the condition b is true) or the identity (if b is false).

Which fixpoint should we use for the semantics of while? A program should be prohibited from looping infinitely. Each time a loop is executed, it must terminate after some number n of iterations. Let  $f^n$  denote the n-fold composition of f, corresponding to performing at most n iterations. We have  $X = f^n(\emptyset)$  in this case. If we do not know the value of n ahead of time, we can always take

$$X = f^{\circ}(\emptyset) \cup f^{1}(\emptyset) \cup f^{2}(\emptyset) \cup \cdots$$

By Kleene's fixpoint theorem,<sup>1</sup> such an X is equal to the least fixpoint lfp f. The greatest fixpoint would also allow cyclic and diverging executions, which we want to rule out.

## 10.4 Monotone Functions

We claimed above that the least and greatest fixpoints are guaranteed to exist under some conditions on f. It is time to make this more precise. Let  $\alpha$  and  $\beta$  be arbitrary types, each equipped with a partial order  $\leq$ . The function  $f:\alpha\to\beta$  is monotone if  $a\le b\to f$  a  $\le f$  b for all a, b. The function  $f:\mathsf{set}\ \alpha\to\mathsf{set}\ \alpha$  admits least and greatest fixpoints if f is monotone.

Many operations on sets (e.g., union  $\cup$ ), relations (e.g., composition  $\bigcirc$ ), and functions (e.g., the identity function  $\lambda x$ , x, the constant function  $\lambda _{-}$ , k, composition  $\circ$ ) are monotone or preserve monotonicity. Of course, not all functions are monotone. Here is an example of a nonmonotone function f on set  $\alpha$ :

$$f A = (if A = \emptyset then set.univ else \emptyset)$$

If  $\alpha$  is inhabited, we have  $\emptyset \subseteq \text{set.univ}$  but  $f \emptyset = \text{set.univ} \nsubseteq \emptyset = f \text{ set.univ}$ .

¹https://en.wikipedia.org/wiki/Kleene\_fixed-point\_theorem

#### 10.5 Complete Lattices

To define lfp for sets (including relations), we need two operations: subset  $\subseteq$ : set  $\alpha \to \text{set } \alpha \to \text{Prop}$  and big intersection  $\cap$ : set (set  $\alpha$ )  $\to$  set  $\alpha$ , which can be defined as follows:

$$\bigcap X = \{a \mid \forall A, A \in X \rightarrow a \in A\}$$

If X is a finite set  $\{A_1, \ldots, A_n\}$ , then  $\bigcap X = A_1 \cap \cdots \cap A_n$ .

We can define lfp more generally so that it works not only with sets but with any instance of the algebraic structure called a complete lattice. A complete lattice  $\alpha$  is an ordered type for which each set  $\alpha$  has an infimum, also called *greatest lower bound*. A complete lattice consists of

- 1. a partial order  $\leq$  :  $\alpha \rightarrow \alpha \rightarrow \text{Prop}$  (i.e., a reflexive, transitive, and antisymmetric binary predicate);
- 2. an operator  $\square$ : set  $\alpha \to \alpha$ , called the *infimum*.

The  $\square$  operator satisfies the following two conditions:

- 1.  $\square$  A is a lower bound of A:  $\square$  A  $\leq$  b for all b  $\in$  s;
- 2.  $\prod$  A is a greatest lower bound:  $b \le \prod$  A for all b such that  $\forall a$ ,  $a \in A \to b \le a$ .

Together, conditions 1 and 2 ensure that  $\prod$  A is the unique greatest lower bound.

The lattice operators  $\leq$  and  $\prod$  generalize  $\subseteq$ : set  $\alpha \to \text{set } \alpha \to \text{Prop and } \cap :$  set  $(\text{set } \alpha) \to \text{set } \alpha$ . Be aware that  $\prod$  A need not be in A. For example, an open interval ]a,b[ over  $\mathbb R$  has infimum  $a \notin ]a,b[$ . The infimum of a set is a generalization of the notion of minimal element.

Here are some examples of complete lattices:

- set  $\alpha$  with respect to  $\subseteq$  and  $\bigcap$  for all types  $\alpha$ ;
- Prop with respect to  $\rightarrow$  and  $\lambda A$ ,  $\forall a \in A$ , a;
- enat :=  $\mathbb{N} \cup \{\infty\}$  with respect to  $\leq$  and a suitable infimum operator;

■ ereal :=  $\mathbb{R} \cup \{-\infty, \infty\}$  with respect to  $\leq$  and a suitable infimum operator.

If  $\alpha$  is a complete lattice, then  $\beta \to \alpha$  is also a complete lattice. If  $\alpha$  and  $\beta$  are complete lattices, then  $\alpha \times \beta$  is also a complete lattice. In both cases,  $\leq$  and  $\square$  are defined componentwise.

Here are some *non*examples of complete lattices:  $\mathbb{N}$ ,  $\mathbb{Z}$ ,  $\mathbb{Q}$ , and  $\mathbb{R}$  with respect to  $\leq$ . The issue is that there is no greatest element, which  $\prod \emptyset$  must be. Another nonexample: erat :=  $\mathbb{Q} \cup \{-\infty, \infty\}$ , because  $\prod \{q \mid 2 < q * q\} = sqrt 2$  is not in erat.

In Lean, it is natural to represent complete lattices using type classes:

```
\mathfrak{d}[\text{class}] structure complete_lattice (\alpha : Type)
   extends partial_order \alpha : Type :=
(Inf : set \alpha \rightarrow \alpha)
(Inf_le : \forall A b, b \in A \rightarrow Inf A \leq b)
(le_Inf : \forall A b, (\forall a, a \in A \rightarrow b \leq a) \rightarrow b \leq Inf A)
The type set \alpha is an instance of the type class:
\mathfrak{g}[\mathsf{instance}] def set.complete lattice \{\alpha : \mathsf{Type}\} :
   complete lattice (set \alpha) :=
                   := (<u></u>□).
{ le
   le_refl := by tautology,
   le trans := by tautology,
   le_antisymm :=
     begin
        intros A B hAB hBA,
        apply set.ext,
        tautology
     end,
                   := \lambda X, {a | \forall A, A \in X \rightarrow a \in A},
   Inf
```

:= by tautology,

le Inf := by tautology }

Inf\_le

#### 10.6 Least Fixpoint

Using complete lattices, we can define the least fixpoint operator:

$$lfp f = \prod \{x \mid f x \le x\}$$

The Knaster–Tarski theorem,<sup>2</sup> which we briefly mentioned in Section 5.1, gives us the following properties for any monotone function f on a complete lattice:

- lfp f is a fixpoint: lfp f = f (lfp f);
- lfp f is smaller than any other fixpoint:  $X = f X \rightarrow lfp f \leq X$ .

## 10.7 A Relational Denotational Semantics, Continued

With lfp, we can fulfill our promise and complete the definition of the denotational semantics of WHILE programs:

```
| (stmt.while b S) := 
 lfp (\lambda X, ((denote S \bigcirc X) \downarrow b) \cup (Id \downarrow (\lambda s, \neg b s)))
```

To validate our definition, we can prove the following connection between the denotational and the big-step semantics:

```
lemma denote_iff_big_step (S : stmt) (s t : state) : (s, t) \in [S] \leftrightarrow (S, s) \Longrightarrow t
```

For the proof, we refer to Chapter 11 of *Concrete Semantics: With Isabelle/HOL* [27] or to the demonstration file accompanying this chapter.

#### 10.8 Application to Program Equivalence

Based on the denotational semantics, we introduce a notion of program equivalence. Two programs are equivalent if they have the same semantics:

<sup>&</sup>lt;sup>2</sup>https://en.wikipedia.org/wiki/Knaster-Tarski\_theorem

```
def denote_equiv (S_1 S_2 : stmt) : Prop := [S_1] = [S_2]
```

We write  $S_1 \sim S_2$  as an abbreviation for denote\_equiv  $S_1 S_2$ . It is obvious from the definition that  $\sim$  is an equivalence relation.

Program equivalence can be used to replace a subprogram in a larger program with another subprogram if they have the same semantics. This is achieved by the following congruence rules:

```
lemma denote_equiv.seq_congr {S<sub>1</sub> S<sub>2</sub> T<sub>1</sub> T<sub>2</sub> : stmt}
    (hS : S<sub>1</sub> ~ S<sub>2</sub>) (hT : T<sub>1</sub> ~ T<sub>2</sub>) :
    S<sub>1</sub> ;; T<sub>1</sub> ~ S<sub>2</sub> ;; T<sub>2</sub> :=
by simp [denote_equiv, denote, *] at *

lemma denote_equiv.ite_congr {b} {S<sub>1</sub> S<sub>2</sub> T<sub>1</sub> T<sub>2</sub> : stmt}
    (hS : S<sub>1</sub> ~ S<sub>2</sub>) (hT : T<sub>1</sub> ~ T<sub>2</sub>) :
    stmt.ite b S<sub>1</sub> T<sub>1</sub> ~ stmt.ite b S<sub>2</sub> T<sub>2</sub> :=
by simp [denote_equiv, denote, *] at *

lemma denote_equiv.while_congr {b} {S<sub>1</sub> S<sub>2</sub> : stmt}
    (hS : S<sub>1</sub> ~ S<sub>2</sub>) :
    stmt.while b S<sub>1</sub> ~ stmt.while b S<sub>2</sub> :=
by simp [denote_equiv, denote, *] at *
```

A congruence rule is a lemma that lifts an equivalence relation over some context (here, ~ over ; ; , stmt.ite, and stmt.while).

Notice how the denotational semantics leads to short proofs by rewriting. This should not be surprising, given that it is designed to be equational and compositional. If we had used the big-step semantics as the basis for program equivalence instead, these proofs would have been much more complicated.

We can now prove simple programs equivalent by equational reasoning:

```
lemma denote_equiv.skip_assign_id {x} :
   stmt.assign x (λs, s x) ~ stmt.skip :=
by simp [denote_equiv, denote, Id]
```

```
lemma denote_equiv.seq_skip_left {S : stmt} :
    stmt.skip ;; S ~ S :=
by simp [denote_equiv, denote, Id, comp]
lemma denote_equiv.seq_skip_right {S : stmt} :
    S ;; stmt.skip ~ S :=
by simp [denote_equiv, denote, Id, comp]
```

We defined the semantics of while using the lfp operator, but who knows whether monotonicity—which guarantees the existence of a least fixpoint—is met? To quell such doubts, we prove the following lemma:

```
lemma denote_equiv.ite_seq_while {b} {S : stmt} :
    stmt.ite b (S ;; stmt.while b S) stmt.skip ~ stmt.while b S :=
begin
    simp [denote_equiv, denote],
    apply eq.symm,
    apply lfp_eq,
    apply monotone_union,
    { apply sorry_lemmas.monotone_restrict,
        apply sorry_lemmas.monotone_comp,
        { exact monotone_const _ },
        { exact monotone_id } },
        { apply sorry_lemmas.monotone_restrict,
        exact monotone_id } },
}
```

The lemma gives us a convenient way to expand or contract one iteration of a loop. The second apply invokes the lemma lfp\_eq : lfp f = f (lfp f), stating that lfp is a fixpoint. The rest of the proof is fairly monotonous: It is all about convincing Lean that lfp 's argument is monotone.

## 10.9 A Simpler Approach Based on an Inductive Predicate

Lean's inductive predicates correspond to least fixpoints, but they are built into Lean's logic (the calculus of inductive constructions), without using a fixpoint operator like lfp. Most of this section was dedicated to the construction of an lfp operator. Could we have used an inductive predicate instead and saved ourselves this work?

The answer is yes. First, recall that the types set (state  $\times$  state) and state  $\rightarrow$  state  $\rightarrow$  Prop are equivalent representations of binary relations in Lean. So we can build the following awhile predicate, which resembles the reflexive transitive closure but stops only if the given condition p turns false:

```
inductive awhile (b : state → Prop)
   (r : set (state × state)) :
   state → state → Prop
| true {s t u} (hcond : b s) (hbody : (s, t) ∈ r)
   (hrest : awhile t u) :
   awhile s u
| false {s} (hcond : ¬ b s) :
   awhile s s
```

With this operator, the definition of the definitional semantics becomes

The awhile predicate's introduction rules closely resembles the while rules of the big-step operational semantics. Behind their different façades, operational and denotational semantics are not so dissimilar after all.

# Part IV Mathematics

## **Chapter 11**

## Logical Foundations of Mathematics

In this chapter, we dive deeper into the logical foundations of Lean. Most of the features described here are especially relevant for defining mathematical objects and proving theorems about them. For even more details, we refer to Carneiro's MSc thesis [6].

#### 11.1 Universes

In dependent type theory, not only all terms have a type, but also all types have types themselves. We have already seen some occurrences of this principle. The PAT principle tells us to view proofs as terms and propositions as types. For example, the lemma

$$@and.intro: \forall a b, a \rightarrow b \rightarrow a \land b$$

is really a term @and.intro of type  $\forall a\ b$ ,  $a \to b \to a \land b$ , which in turn has a type:

$$\forall a b, a \rightarrow b \rightarrow a \land b : Prop$$

What is the type of Prop then? Prop has the same type as virtually all other types we have constructed so far:

Prop : Type

What is the type of Type? The simplest solution would be to let Type: Type, but this choice leads to Girard's paradox, the type theory equivalent of Russel's paradox. To avoid inconsistencies, we need a fresh type to contain Type, which we call Type 1. Type 1 itself has type Type 2, and so on:

Type : Type 1
Type 1 : Type 2
Type 2 : Type 3
:

In fact, Type with no argument is an abbreviation for Type o. If we want to incorporate Prop in this hierarchy, we can use the syntax Sort u, where Sort o is an alias for Prop and Sort (u + 1) is an alias for Type u. The hierarchy is captured by the following typing judgment:

All of these types containing other types are called *universes*, and the u in the expression Sort u is a *universe level*. Although universe levels look like terms of type  $\mathbb{N}$ , they are in fact nor even terms. By convention, universes are capitalized—hence we have Prop but bool.

Instead of using Type everywhere, you can make your theorems slightly more general without having to think about universe levels by writing Type \_ or Type\*. It leaves it to Lean to figure the right universe level or to create a fresh universe variable. This helps maintain the illusion that we work in a convenient logic where Type: Type holds, but without introducing any paradoxes. In practice, Type o is large enough for most of computer science and mathematics.

#### 11.2 The Peculiarities of Prop

Although Prop may seem to fit nicely into the universe hierarchy, it differs from the other universes in several important respects.

#### 11.2.1 Impredicativity

When constructing a new type from other types (e.g.,  $\alpha \to \beta$  from  $\alpha$ : Type u and  $\beta$ : Type v), the newly constructed type is more complex than each of its components, and it is natural to put it into the largest universe involved (e.g.,  $\alpha \to \beta$ : Type (max u v)). This is exactly what Lean does. The following typing rule expresses this idea generally for dependent types:

$$\frac{\mathsf{C} \vdash \alpha : \mathsf{Type} \; \mathsf{u} \quad \mathsf{C}, \; \mathsf{a} : \alpha \vdash \beta \; \mathsf{a} : \mathsf{Type} \; \mathsf{v}}{\mathsf{C} \vdash (\forall \mathsf{a} : \alpha, \; \beta \; \mathsf{a}) : \mathsf{Type} \; (\mathsf{max} \; \mathsf{u} \; \mathsf{v})} \mathsf{FORALL-TYPE}$$

This behavior of the Type universes is called *predicativity*. In general, predicativity means that an object *must not* be defined in terms of a quantifier ranging over that same object.

However, it is convenient to have Prop behave differently. We would like the expression  $\forall a: Prop$ ,  $a \to a$  to be of type Prop—it is, after all, a proposition. Unfolding the syntactic sugar  $\to$ , this expression is the same as  $\forall (a: Prop) \ (x:a)$ , a. If we had Type u instead of Prop, the above typing rule would yield

```
(\forall (a : Type u) (x : a), a) : Type (u + 1)
```

because Type u: Type (u+1) and max(u+1) (maxuu) = u+1. Thus, the universe level is increased by one when typing this expression. To force expressions such as  $\forall a:$  Prop,  $a \rightarrow a$  to be of type Prop anyway, we need a special typing rule for  $\forall$ -expressions with a Prop body:

```
\frac{\mathsf{C} \vdash \alpha : \mathsf{Sort} \; \mathsf{u} \quad \mathsf{C}, \; (\mathsf{a} : \alpha) \vdash \beta \; \mathsf{a} : \mathsf{Prop}}{\mathsf{C} \vdash (\forall \mathsf{a} : \alpha, \beta \; \mathsf{a}) : \mathsf{Prop}} \mathsf{FORALL\text{-}PROP}
```

The rule yields

```
(\forall (a : Prop) (x : a), a) : Prop
```

as desired. The two typing rules for  $\forall$ -expressions above can be summarized as the single rule

```
\frac{\mathsf{C} \vdash \alpha : \mathsf{Sort} \; \mathsf{u} \quad \mathsf{C}, \; \mathsf{a} : \alpha \vdash \beta \; \mathsf{a} : \mathsf{Sort} \; \mathsf{v}}{\mathsf{C} \vdash (\forall \mathsf{a} : \alpha, \; \beta \; \mathsf{a}) : \mathsf{Sort} \; (\mathsf{imax} \; \mathsf{u} \; \mathsf{v})} \mathsf{FORALL}
```

where imax u o = o and imax u (v + 1) = max u (v + 1). This behavior is called *impredicativity* of Prop. In general, impredicativity means that an object *may* be defined via a quantifier ranging over itself.

#### 11.2.2 Proof Irrelevance

A second difference between Prop and Type is *proof irrelevance*. It means that any two proofs of the same proposition a are equal:

```
lemma proof_irrel {a : Prop} (h<sub>1</sub> h<sub>2</sub> : a) :
    h<sub>1</sub> = h<sub>2</sub> :=
by refl
```

In Lean, this equality is a syntactic equality up to computation, allowing us to use the refl tactic. When viewing a proposition as a type and a proof as an element of that type, proof irrelevance means that a proposition is either an empty type or has exactly one inhabitant. If it is empty, it is false. If it has exactly one inhabitant, it is true. Proof irrelevance is very helpful when reasoning about dependent types.

In Chapter 5, we saw a diagram depicting the interpretation of bool and Prop side by side. The diagram did not take proof irrelevance into account and showed two distinct proofs for the same proposition. We now know this is not accurate. Here is a revised

diagram:



If we mentally connect Prop's two types (false and true) with bool's two elements (ff and tt), we see that Prop is nearly the same as bool, except that propositions of type Prop can store proofs, in the spirit of the PAT principle. The following table summarizes the situation:

false: Prop true.intro: true: Prop ff : bool: Type tt : bool: Type

To make proof irrelevance work, Lean must give up the "no confusion" property for inductive predicates. Indeed, various proofs of the same proposition should definitely be "confused." This only concerns inductive predicates, not inductive types in general.

An unfortunate consequence of proof irrelevance is that it prevents us from performing rule induction by pattern matching. Induction by pattern matching relies on a "measure"—a function to  $\mathbb N$  that assigns a size to the arguments. Without large elimination, the measure cannot be defined meaningfully. This explains why we always use the induction' tactic for rule induction.

Other systems and logics make different choices. For example, Coq is proof-relevant by default but compatible with proof irrelevance. Homotopy type theory and other constructive or intuitionistic type theories build on data in equality proofs and are therefore incompatible with proof irrelevance.

#### 11.2.3 No Large Elimination

A further difference between Prop and Type is that Prop does not allow *large elimination*: It is generally impossible to extract infor-

mation from a proof of a proposition and use it in a program (i.e., a value of a type from Type). After all, because of proof irrelevance, all proofs of a given proposition are equal, so they cannot carry specific information that would distinguish them.

Imagine we could extract information from proofs inside a program. We could for instance use the match construct in function definitions such as the following:

```
-- fails def unsquare (i : \mathbb{Z}) (hsq : \existsj, i = j * j) : \mathbb{Z} := match hsq with | Exists.intro j _ := j end
```

The unsquare takes a square number i and a proof hsq that i is actually a square number and returns the number j before squaring, extracted from the proof. Lean raises the error

```
induction tactic failed, recursor 'Exists.dcases_on'
can only eliminate into Prop
```

If it accepted the definition, we could derive false as follows. Let

```
hsq<sub>1</sub> := Exists.intro 3 (by linarith)
hsq<sub>2</sub> := Exists.intro (-3) (by linarith)
```

be two proofs of  $\exists j$ ,  $(9 : \mathbb{Z}) = j * j$ . Notice that they use different witnesses for j (3 versus -3). We then have unsquare  $9 \text{ hsq}_1 = 3$  and unsquare  $9 \text{ hsq}_2 = -3$ . Yet, by proof irrelevance,  $\text{hsq}_1 = \text{hsq}_2$ . Hence, unsquare  $9 \text{ hsq}_2$  equals 3. But we already determined that it equals -3. This means 3 = -3, a contradiction.

As a compromise, Lean allows *small elimination*, which eliminates only into Prop—whereas large elimination can eliminate into an arbitrary large universe Sort u. This means we can use match to analyze the structure of a proof, extract existential witnesses, and so on, as long as the match expression itself is a proof. We have seen several examples of this in Section 5.5.

As a further compromise, Lean allows large elimination for syntactic subsingletons: types in Prop for which it can be established syntactically that they have cardinality 0 or 1. These are propositions such as false and a  $\land$  b that can be proved in at most one way. For example, false has no proof, and all proofs of a  $\land$  b have the form and intro \_ \_. (Recursively, there might be several ways to prove a and b.) More precisely, a syntactic subsingleton is an inductive definition with at most one constructor whose arguments are either Prop or appear as immediate arguments in the result type. When we match h: a  $\land$  b against the pattern and intro ha hb, no information is leaked about h.

#### 11.3 The Axiom of Choice

Lean's logic includes the axiom of choice, which makes it possible to obtain an arbitrary element of any nonempty type. Consider the following predicate:

```
inductive nonempty (\alpha : Sort u) : Prop | intro (val : \alpha) : nonempty
```

The predicate states that  $\alpha$  has at least one element. To prove nonempty  $\alpha$ , we must provide an  $\alpha$  value to nonempty.intro:

```
lemma nat.nonempty :
  nonempty N :=
nonempty.intro o
```

Since nonempty lives in Prop, large elimination is not available, and thus we cannot extract the element that was used from the proof of nonempty  $\alpha$ .

In Lean, the axiom of choice takes the form of a function that returns an arbitrary  $\alpha$ -typed value given a proof of nonempty  $\alpha$ :

```
constant classical.choice \{\alpha : Sort \ u\} : nonempty \ \alpha \rightarrow \alpha
```

We have no way to know whether this is the same element that was used to prove nonempty  $\alpha$ . It will just be an arbitrary element

of  $\alpha$ , thereby avoiding inconsistencies. Using the axiom of choice, we can state properties that are unprovable but whose negation is also unprovable—e.g., classical.choice nat.nonempty = 42.

The constant classical.choice is noncomputable. If we ask Lean for its value using #reduce or #eval, it will refuse to compute it. In other words, proofs may be terms, but they are not necessarily programs. This is one of the reasons why some logicians prefer to work without this axiom. By contrast, the vast majority of mathematicians have no objections against the axiom.

Unlike proof irrelevance and large and small elimination, the axiom of choice is not built into the Lean kernel; it is only an axiom in the core library, giving us the freedom to work without it. Lean requires us to mark definitions using the noncomputable keyword if they use classical.choice to define constants in Type, much in the same way as meta for metadefinitions (Chapter 7).

The following tools rely on classical.choice:

The function classical.some, often called Hilbert's choice operator, can help us find a witness of ∃a: α, p a, if we do not care which one. Its companion classical.some\_spec gives a proof that the witness is indeed a witness.

```
classical.some : (\exists a:\alpha,p\:a)\to\alpha classical.some_spec : \forall h:(\exists a:\alpha,p\:a), p (classical.some h)
```

Intuitively, the choice operator tells us, "Convince me that there exists an element satisfying p and I will give you such an element."

• We can also derive the traditional axiom of choice:

```
classical.axiom_of_choice (\alpha \beta : Type) {r : \alpha \rightarrow \beta \rightarrow Prop} : (\forall x : \alpha, \exists y : \beta, r \times y) \rightarrow (\exists f, \forall x, r \times (f \times))
```

From the axiom of choice and propositional and functional extensionality (propext, funext), we can derive the law of excluded middle:

```
classical.em: \forall a : Prop, a \lor \neg a
```

11.4. Subtypes 253


With the law of excluded middle, every proposition is decidable. This means that we can construct proofs based on a case distinction on whether a certain proposition is true. We introduced this technique in Chapters 4 and 5.

#### 11.4 Subtypes

Inductive types are a very convenient definitional mechanism when they are applicable, but lots of mathematical constructions do not fit that mold. Lean provides two alternatives to cater for these: subtypes and quotients.

Subtyping is a mechanism to create new types from existing ones. Given a predicate on the elements of a base type, the *subtype* contains only those elements of the base type that fulfill this property. More precisely, the subtype contains element–proof pairs that combine an element of the base type and a proof that it fulfills the property.

The following diagram depicts a subtype that was created by keeping two of the base type's five elements:



Subtyping is useful for those types that cannot be defined as an inductive type. For example, any attempt at defining the type of finite sets along the lines of

```
inductive finset (\alpha : Type) : Type | empty : finset | insert : \alpha \rightarrow finset \rightarrow finset
```

is doomed, because a given set may have multiple representations. For example, {1, 2} can be represented in any of the following ways, and more:

```
finset.insert 1 (finset.insert 2 empty)
    finset.insert 2 (finset.insert 1 empty)
finset.insert 1 (finset.insert 2 empty))
```

Using a subtype, we can choose one canonical representation for each value and filter out the remaining ones. Subtypes have the syntax

```
{variable : base-type // property-applied-to-variable}
```

We saw an example in Section 3.6, namely,  $\{i : \mathbb{N} // i \le n\}$ , which consists of the natural numbers i fulfilling  $i \le n$ , where n is fixed in the context. The base type is  $\mathbb{N}$ , and the property is  $\lambda i$ ,  $i \le n$ . A less suggestive but perhaps less confusing syntax for the same type is  $\mathfrak{A}$  subtype  $\mathbb{N}$  ( $\lambda i$ ,  $i \le n$ ).

#### 11.4.1 First Example: Full Binary Trees

To illustrate subtypes, we will define a type of full binary trees, building on the type btree from Section 4.8. In Section 5.6.3, we introduced a predicate is\_full that is true if each node of a tree has either zero or two child nodes. Based on this type and this predicate, we can construct a subtype full\_btree, containing only full binary trees, as follows:

```
def full_btree (\alpha : Type) : Type := {t : btree \alpha // is_full t}

This is syntactic sugar for def full_btree (\alpha : Type) : Type := @subtype (btree \alpha) is_full where subtype is defined as follows:
```

11.4. Subtypes 255

```
inductive subtype \{\alpha : \mathsf{Type}\}\ (\mathsf{p} : \alpha \to \mathsf{Prop}) : \mathsf{Type}\ | \ \mathsf{mk} : \forall \mathsf{x} : \alpha, \ \mathsf{p} \ \mathsf{x} \to \mathsf{subtype}
```

Elements of full\_btree are essentially dependently typed pairs, where the first component is a tree t and the second component is a proof that t is full:

```
def empty_full_btree : full_btree N :=
subtype.mk btree.empty is_full.empty

def full_btree_6 : full_btree N :=
subtype.mk (btree.node 6 btree.empty btree.empty)
  begin
    apply is_full.node,
    apply is_full.empty,
    apply is_full.empty,
    refl
end
```

From a given element of full\_btree, we can retrieve its two components via subtype.val and subtype.property:

```
#reduce subtype.val full_btree_6
#check subtype.property full_btree_6
```

The most appealing aspect of subtypes is that we can *lift* the operations from the base type to the subtype instead of having to build a library from scratch, as long as they preserve the subtype property. We only need to define "wrappers" around constants from the base type. In general, defining such a wrapper given an operation f on the base type involves three steps:

- 1. extract the base type values from the wrapper arguments;
- 2. invoke f on those base type values;
- encapsulate the result using subtype.mk, together with a proof that the resulting base type value fulfills the subtype property.

Using this procedure, we can lift btree functions to full\_btree functions if they preserve the property is\_full. To lift the mirror

operation from the type  $\mathsf{btree} \to \mathsf{btree}$  to the type  $\mathsf{full\_btree} \to \mathsf{full}$  btree, we must

- 1. extract the btree from the wrapper argument;
- 2. invoke mirror on that btree;
- 3. encapsulate the result using subtype.mk, together with a proof that the resulting btree fulfills is\_full.

For step 3, we must extract the proof of is\_full from the argument and use the lemma is\_full\_mirror proved in Section 5.6.3. Putting everything together, we get

```
def full_btree.mirror {α : Type} (t : full_btree α) :
  full_btree α :=
subtype.mk (mirror (subtype.val t))
begin
  apply is_full_mirror,
  apply subtype.property t
end
```

The input is an element t of the subtype full\_btree. We decompose t into subtype.val t: btree and subtype.property t: is\_full t. We use the mirror function to reverse the tree component of t and use the lemma is\_full\_mirror, together with the proof component of t, to show the condition is\_full (mirror (subtype.val t)).

Finally, we build a pair containing the resulting tree and the proof that this tree is full using subtype.mk. The subtype.mk constructor can be seen both as a pair-like constructor and as a cast from btree to full\_btree, with a second argument that guarantees that the cast is safe.

For proofs about subtypes, the following lemma is useful:

```
subtype.eq : subtype.val ?a = subtype.val ?b \rightarrow ?a = ?b
```

It states that two subtype values are equal if their subtype.val components are equal. It is convenient that proofs are irrelevant in Lean, because we would not want to have spurious duplicate

11.4. Subtypes 257

values in the subtype, differing only in their proofs. Here is how subtype.eq can be used to prove that the mirror image of the mirror image of a full btree is the full btree itself:

```
lemma full_btree.mirror_mirror {α : Type} (t : full_btree α) :
  (full_btree.mirror (full_btree.mirror t)) = t :=
begin
  apply subtype.eq,
  simp [full_btree.mirror],
  apply mirror_mirror
end
```

Applying the lemma subtype.eq and then unfolding the definition of full\_btree.mirror yields the subgoal

```
mirror (mirror (subtype.val t)) = subtype.val t
```

which matches the statement of the lemma mirror\_mirror from Section 4.8.

#### 11.4.2 Second Example: Vectors

As a second example, consider the definition of vectors from Lean's core libraries:

```
def vector (\alpha : Type) (n : \mathbb{N}) : Type := \{xs : list <math>\alpha // list.length \ xs = n\}
```

A vector is defined as a list of a given length. With lists, there is only one type for a list of all lengths. For vectors, we have one dedicated type for every length of a vector. The advantage of this scheme is that some operations, such as addition or scalar product of vectors, rely on the two involved vectors having the same length. We saw a morally equivalent but less practical definition of vectors in Section 4.10.

Vectors can be built from lists using subtype.mk:

```
def vector_123 : vector \mathbb{Z} 3 := subtype.mk [1, 2, 3] (by refl)
```

Basic operations on vectors can be defined by decomposing them with subtype.val and subtype.property, manipulating the underlying lists, and then recomposing them with subtype.mk. For example, we can define the componentwise negation of an integer vector as follows:

```
def vector.neg {n : N} (v : vector Z n) : vector Z n :=
subtype.mk (list.map int.neg (subtype.val v))
begin
   rw list.length_map,
   exact subtype.property v
end
```

We use the function list.map to negate every entry of the underlying list and the lemma list.length\_map to show that this operation does not change the list's length.

Using subtype.eq, we can prove the following lemma about vector.neg:


```
lemma vector.neg_neg (n : N) (v : vector Z n) :
  vector.neg (vector.neg v) = v :=
begin
  apply subtype.eq,
  simp [vector.neg]
end
```

The application of subtype.eq reduces the goal to showing the corresponding property on the underlying lists. We can then use simp to finish the proof.

#### 11.5 Quotient Types

Quotients are a powerful construction in mathematics used to define  $\mathbb{Z}$ ,  $\mathbb{R}$ , and many other sets. Lean supports *quotient types*, an analogous mechanism on types. Like subtyping, quotienting constructs a new type from an existing type. Unlike a subtype, a quotient type contains all of the elements of the base type, except that some elements that are different in the base type may

be considered equal in the quotient type. In mathematical terms, the quotient type is isomorphic to a partition of the base type. The diagram below shows a quotient type built from a three-way partition:



The depicted quotient type has only three elements, represented by gray ellipses. Each of these elements corresponds to one or more base type elements.

The prerequisites to construct a quotient type are a base type  $\tau$  and an equivalence relation  $\mathbf{r}:\tau\to\tau\to\mathsf{Prop}$  specifying which elements of the base type will be considered equal in the quotient. To construct the quotient type, we first need to prove that  $\mathbf{r}$  is an equivalence relation on  $\tau$ . A type  $\tau$  equipped with an equivalence relation is called a *setoid*. In Lean, setoid is a type class. We can declare an instance as follows:

In addition, this instance declaration introduces the notation a  $\approx$  b for r a b. More importantly, we can now use the quotient type quotient  $\tau.$  setoid.

Every element a :  $\tau$  belongs to some element in quotient  $\tau$  .setoid, given by

```
quotient.mk: \tau \rightarrow quotient \tau.setoid
```

As an abbreviation for quotient.mk a, we can write [a].

The following axiom guarantees that pairs of elements for which  ${\bf r}$  holds are indeed equal in the quotient type:

```
quotient.sound {a b : \tau} : a \approx b \rightarrow [a] = [b]
```

A lemma states the converse:

quotient.exact {a b : 
$$\tau$$
} :  $[\![a]\!]$  =  $[\![b]\!] \to a \approx b$ 

Finally, we can lift functions of type  $\tau \to \upsilon$ , where  $\upsilon$  is some arbitrary type, to quotient  $\tau$ .setoid  $\to \upsilon$  using quotient.lift, which has the following computation rule. Given some  $f:\tau \to \upsilon$  such that  $h:\forall a\ b,\ a\approx b\to f\ a=f\ b$ , we have

quotient.lift 
$$f h [a] = f a$$

for all  $a:\tau$ . The argument h is a proof that f is *compatible* with  $\approx$ ; in other words, it does not distinguish between  $\approx$ -equivalent arguments.

#### 11.5.1 First Example: Integers

As an example for a quotient type, we will construct the integers. A convenient approach is to construct a quotient over pairs of natural numbers. The idea is that a pair (p, n) of natural numbers represents the integer p - n. In this way, we can represent all nonnegative integers p by (p, o) and all negative integers p by (p, n). In addition, we get many more representations of the same integers; for example, (7, o), (8, 1), (9, 2), and (10, 3) all represent the integer p.

First, we need to register the equivalence relation that we want to use. We want two pairs  $(p_1, n_1)$  and  $(p_2, n_2)$  to be equal when  $p_1 - n_1$  and  $p_2 - n_2$  yield the same integer. However, the condition  $p_1 - n_1 = p_2 - n_2$  does not work because subtraction on  $\mathbb N$  is ill-behaved (e.g., o - 1 = o). Instead, we use the condition  $p_1 + n_2 = p_2 + n_1$ , which relies on addition.

We provide the definition of our equivalence relation, followed by a proof that it is reflexive, symmetric, and transitive:

```
@[instance] def int.rel : setoid (\mathbb{N} \times \mathbb{N}) := { r := \lambda pn_1 pn_2 : \mathbb{N} \times \mathbb{N},
```

```
prod.fst pn<sub>1</sub> + prod.snd pn<sub>2</sub> = prod.fst pn<sub>2</sub> + prod.snd pn<sub>1</sub>,
iseqv :=
begin
  repeat { apply and.intro },
  { intro pn,
    refl },
  { intros pn<sub>1</sub> pn<sub>2</sub> h,
    rw h },
  { intros pn<sub>1</sub> pn<sub>2</sub> pn<sub>3</sub> h<sub>12</sub> h<sub>23</sub>,
    apply @add_right_cancel _ _ _ (prod.snd pn<sub>2</sub>),
    cc }
end }
```

We can now write  $\approx$  for the equivalence relation:

```
lemma int.rel_iff (pn<sub>1</sub> pn<sub>2</sub> : \mathbb{N} \times \mathbb{N}) : pn<sub>1</sub> \approx pn<sub>2</sub> \leftrightarrow prod.fst pn<sub>1</sub> + prod.snd pn<sub>2</sub> = prod.fst pn<sub>2</sub> + prod.snd pn<sub>1</sub> := by refl
```

Using the name int.rel of our setoid instance that we registered above, we can then define the integers as

```
def int : Type :=
quotient int.rel
```

We can define the integer zero as

```
def int.zero : int :=
[(0, 0)]
```

In fact, any term of the form [(m, m)] represents the same integer:

```
lemma int.zero_eq (m : N) :
  int.zero = [(m, m)] :=
begin
  rw int.zero,
  apply quotient.sound,
  rw int.rel_iff,
  simp
end
```

Next, we define addition on our new integers. To define functions on a quotient type, we cannot simply define them by pattern matching, as for inductive types. Instead, we define the function on the base type first and then *lift* the definition to the quotient. To achieve this, we must prove that the definition of the function f does not depend on the chosen representative of an equivalence class (i.e.,  $a \approx b \rightarrow f \ a = f \ b$ ). The functions quotient.lift (for unary functions) and quotient.lift<sub>2</sub> (for binary functions) can be used to lift functions in this way.

Addition can be defined as adding the pairs of natural numbers componentwise. We then need to provide a proof that this can be lifted to a function on the quotient by showing that  $pn_1 \approx pn_1$  and  $pn_2 \approx pn_2$  imply

 $[(prod.fst pn_1 + prod.fst pn_2, prod.snd pn_1 + prod.snd pn_2)]$ 

```
= [(\text{prod.fst pn}_1' + \text{prod.fst pn}_2', \text{prod.snd pn}_1' + \text{prod.snd pn}_2')]

Formally:

def int.add : int \rightarrow int \rightarrow int := quotient.lift<sub>2</sub>

(\lambdapn<sub>1</sub> pn<sub>2</sub> : \mathbb{N} \times \mathbb{N},

[(prod.fst pn<sub>1</sub> + prod.fst pn<sub>2</sub>,

prod.snd pn<sub>1</sub> + prod.snd pn<sub>2</sub>)])

begin

intros pn<sub>1</sub> pn<sub>2</sub> pn<sub>1</sub>' pn<sub>2</sub>' h<sub>1</sub> h<sub>2</sub>,

apply quotient.sound,

rw int.rel_iff at *,

linarith
end
```

The resulting function int.add has the intended behavior:

```
lemma int.add_eq (p<sub>1</sub> n<sub>1</sub> p<sub>2</sub> n<sub>2</sub> : \mathbb{N}) : int.add [(p_1, n_1)] [(p_2, n_2)] = [(p_1 + p_2, n_1 + n_2)] := by refl
```

It would be very convenient if Lean let us enter this lemma as the definition of int.add in the first place, presumably with the following syntax:

```
-- fails def int.add : int \rightarrow int \rightarrow int \mid \lceil (p_1, n_1) \rceil \rceil \lceil (p_2, n_2) \rceil := \lceil (p_1 + p_2, n_1 + n_2) \rceil
```

This would be a nice and intuitive syntax, but without a proof that the definition is compatible with  $\approx$ , we could define nonsensical functions and use them to derive false. For example, we could define

```
-- fails def int.fst : int \rightarrow \mathbb{N} | [(p, n)] := p
```

Observe that int.fst [(0, 0)] = 0 and int.fst [(1, 1)] = 1. However, since [(0, 0)] = [(1, 1)], we get 0 = 1, a contradiction.

We can use the characteristic lemma int.add\_eq to prove other lemmas about int.add, such as

```
lemma int.add_zero (i : int) :
   int.add int.zero i = i :=
begin
   apply quotient.induction_on i,
   intro pn,
   rw int.zero,
   cases' pn with p n,
   rw int.add_eq,
   apply quotient.sound,
   simp
end
```

The somewhat ill-named lemma quotient.induction\_on is applied to perform a case distinction on i, replacing i by [pn], where pn is an arbitrary value of the base type  $\mathbb{N} \times \mathbb{N}$ . In general:

```
quotient.induction_on : \forall q : quotient ?s, (\forall a, ?p [a]) \rightarrow ?p q
```

A more intuitive syntax for the first proof step would have been cases' i, but this is not supported by Lean.

#### 11.5.2 Second Example: Unordered Pairs

Unordered pairs are pairs for which no distinction is made between the first and second components. They are usually written  $\{a, b\}$ . We will introduce the type upair  $\alpha$  of unordered pairs over  $\alpha$  as the quotient of pairs (a, b) with respect to the relation "contains the same elements as":

```
\begin{array}{lll} \text{@[instance] def upair.rel} & (\alpha: \mathsf{Type}): \mathsf{setoid} & (\alpha \times \alpha) := \\ \{ r & := \lambda \mathsf{ab_1} \ \mathsf{ab_2} : \alpha \times \alpha, \\ & (\{\mathsf{prod.fst ab_1}, \ \mathsf{prod.snd ab_1}\} : \mathsf{set} \ \alpha) = \\ & (\{\mathsf{prod.fst ab_2}, \ \mathsf{prod.snd ab_2}\} : \mathsf{set} \ \alpha), \\ & \mathsf{iseqv} := \mathsf{by} \ \mathsf{repeat} \ \{ \ \mathsf{apply} \ \mathsf{and.intro} \ \}; \ \mathsf{finish} \ \} \\ \\ \mathsf{lemma} \ \ \mathsf{upair.rel\_iff} \ \{\alpha: \ \mathsf{Type}\} \ (\mathsf{ab_1} \ \mathsf{ab_2} : \alpha \times \alpha) : \\ & \mathsf{ab_1} \approx \mathsf{ab_2} \leftrightarrow \\ & (\{\mathsf{prod.fst ab_1}, \ \mathsf{prod.snd ab_1}\} : \mathsf{set} \ \alpha) = \\ & (\{\mathsf{prod.fst ab_2}, \ \mathsf{prod.snd ab_2}\} : \mathsf{set} \ \alpha) := \\ & \mathsf{by} \ \mathsf{refl} \\ \\ \mathsf{def upair} \ (\alpha: \ \mathsf{Type}) : \ \mathsf{Type} := \\ & \mathsf{quotient} \ (\mathsf{upair.rel} \ \alpha) \\ \end{array}
```

We define upair.mk a b to construct an unordered pair containing a and b:

```
def upair.mk \{\alpha : \mathsf{Type}\}\ (\mathsf{a}\;\mathsf{b}\; :\; \alpha) : \mathsf{upair}\; \alpha := \llbracket (\mathsf{a},\;\mathsf{b}) \rrbracket
```

It is easy to prove that upair.mk is symmetric, meaning that our pairs are really unordered:

```
lemma upair.mk_symm {α : Type} (a b : α) :
    upair.mk a b = upair.mk b a :=
begin
    unfold upair.mk,
    apply quotient.sound,
    rw upair.rel_iff,
    apply set.union_comm
end
```

Another representation of unordered pairs is as sets of cardinality 1 or 2. The following operation converts upair  $\alpha$  values to that representation:

```
def set_of_upair {\alpha : Type} : upair \alpha \to \text{set } \alpha := quotient.lift (\lambdaab : \alpha \times \alpha, {prod.fst ab, prod.snd ab}) begin intros ab<sub>1</sub> ab<sub>2</sub> h, rw upair.rel_iff at *, exact h end
```

## 11.5.3 Alternative Definitions via Normalization and Subtyping

Each element of a quotient type correspond to a class of  $\approx$ -equivalent elements of the base type. If there exists a systematic way to obtain a canonical representative for each  $\approx$ -equivalence class, we can use a subtype instead of a quotient, keeping only the canonical representatives and filtering out the other elements.

Consider the quotient type int of integers constructed above. We observed that (7, 0), (8, 1), (9, 2), and (10, 3) all represent the integer 7, but intuitively (7, 0) seems preferable to the others. We will say that a pair (p, n) is canonical if p or n (or both) is 0:

```
inductive int.is_canonical : \mathbb{N} \times \mathbb{N} \to \mathsf{Prop} | nonpos \{n : \mathbb{N}\} : int.is_canonical (0, n) | nonneg \{p : \mathbb{N}\} : int.is_canonical (p, 0)
```

The integers consist of the canonical pairs of natural numbers:

```
def int : Type :=
{pn : \mathbb{N} \times \mathbb{N} // \text{ int.is_canonical pn}}
```

Clearly, each integer can be represented in one and only one way. Operations on integers, such as addition and multiplication, must provide canonical results. Fortunately, normalizing pairs of natural numbers is easy:

```
def int.normalize : \mathbb{N} \times \mathbb{N} \to \mathbb{N} \times \mathbb{N}
| (p, n) := if p \geq n then (p - n, 0) else (0, n - p)

lemma int.is_canonical_normalize (pn : \mathbb{N} \times \mathbb{N}) :
   int.is_canonical (int.normalize pn)
```

For unordered pairs, there is no obvious normal form, except to always put the smaller element first (or last). This requires a linear order < on  $\alpha$ :

```
def upair.is_canonical \{\alpha: \text{Type}\}\ [\text{linear\_order }\alpha]: \alpha \times \alpha \to \text{Prop} \ |\ (a, b):=a \le b def upair (\alpha: \text{Type})\ [\text{linear\_order }\alpha]: \text{Type}:= \{ab: \alpha \times \alpha \ // \ upair.is\_canonical \ ab\}
```

Returning to the int.is\_canonical predicate, notice that there are two proofs that (0, 0) is canonical, using either nonpos or nonneg. By proof irrelevance, these proofs must be equal. But this would seem to contradict the "no confusion" property of inductive definitions. Fortunately, as we saw in Section 11.2.2, this property does not apply to inductive predicates, so there is reason to worry.

#### 11.6 Summary of New Lean Constructs

#### **Declaration**

noncomputable prefixes noncomputable declarations

#### **Constants**

classical.choice function that returns an arbitrary element of a

nonempty type

classical.some function that returns a witness given a proof of

an existential

quotient function that creates a quotient type of a given

setoid instance

quotient.lift function that lifts a unary function to a quotient type function that lifts a binary function to a quotient type

setoid type class for a type with an equivalence relation on it

Sort u universe at level u

#### **Notations**

 $\{x : \alpha // P[x]\}$  subtype of all x in  $\alpha$  fulfilling P[x]

pprox equivalence relation on a setoid (used for quotienting)

Prop abbreviation for Sort o
Type u abbreviation for Sort (u + 1)

#### Lemmas

classical.axiom\_of\_choice

traditional axiom of choice

classical.some\_spec

characteristic property of classical.some

quotient.exact equality on quotient type implies  $\approx$  on base type

quotient.induction\_on

case distinction on a quotient type value

quotient.sound  $\approx$  on base type implies equality on quotient type subtype.eq equality on base type implies equality of subtype

### **Chapter 12**

#### **Basic Mathematical Structures**

In this chapter, we introduce definitions and proofs about basic mathematical structures such as groups, fields, and linear orders.

#### 12.1 Type Classes over a Single Binary Operator

In mathematics, a *group* is a set G with a binary operator  $\bullet : G \times G \rightarrow G$  fulfilling the following properties, called group axioms:

- Associativity: For all a, b,  $c \in G$ , we have  $(a \bullet b) \bullet c = a \bullet (b \bullet c)$ .
- Identity element: There exists an element  $e \in G$  such that for all  $a \in G$ , we have  $e \bullet a = a$ .
- Inverse element: For each  $a \in G$ , there exists an inverse element denoted  $a^{-1}$  such that  $a^{-1} \bullet a = e$ .

In Lean, a type class for groups could be defined as follows:

```
\begin{array}{lll} \text{@[class] structure group } (\alpha : \text{Type}) := \\ (\text{mul} & : \alpha \rightarrow \alpha \rightarrow \alpha) \\ (\text{one} & : \alpha) \\ (\text{inv} & : \alpha \rightarrow \alpha) \\ (\text{mul\_assoc} & : \forall \text{a b c, mul (mul a b) c = mul a (mul b c))} \end{array}
```

```
(one_mul : \forall a, mul one a = a)

(mul_left_inv : \forall a, mul (inv a) a = one)
```

However, this is not the official definition. Groups are part of a larger hierarchy of algebraic structures.

Group operations can be written multiplicatively (with operator \*, identity element 1, and inverse element  $a^{-1}$ ) or additively (with operator \*, identity element 0, and inverse element -a). This is why Lean offers two type classes for groups: the multiplicative group and the additive add\_group. They are essentially the same but use different names for their constants and properties.

Any type that satisfies the group axioms can be registered as a group or add\_group. To illustrate this, we will define the type int2 of integers modulo 2, also known as  $\mathbb{Z}/2\mathbb{Z}$ , and register it as an add\_group. The type int2 has two elements:

```
inductive int2 : Type
| zero
| one
```

Addition is defined as follows:

Running #print add\_group tells us all the constants and properties we need to provide:

```
structure add_group : Type u \rightarrow Type u fields: add_group.add : \Pi\{\alpha\} [add_group \alpha], \alpha \rightarrow \alpha \rightarrow \alpha add_group.add_assoc : \forall \{\alpha\} [add_group \alpha] (a b c : \alpha), a + b + c = a + (b + c) add_group.zero : \Pi(\alpha) [add_group \alpha], \alpha add_group.zero_add : \forall \{\alpha\} [add_group \alpha] (a : \alpha), o + a = a add_group.add_zero : \forall \{\alpha\} [add_group \alpha] (a : \alpha), a + o = a add_group.neg : \Pi\{\alpha\} [c : add_group \alpha], \alpha \rightarrow \alpha add_group.add_left_neg : \forall \{\alpha\} [add_group \alpha] (a : \alpha), -a + a = o
```

The constants add\_group.add, add\_group.zero, and add\_group.neg correspond to the binary operator •, the identity element e, and the inverse -1. The properties add\_group.add\_assoc, add\_group. zero\_add, and add\_group.add\_left\_neg correspond to the three group axioms. For technical reasons, we must also prove the redundant property add group.add zero.

The type int2 can be registered as a group as follows:

With this instance declaration, we can now write o, +, and -:

```
#reduce int2.one + \circ - \circ - int2.one
```

We can also use any lemma that has been proved for add\_group:

```
lemma int2.add_right_neg:
    ∀a : int2, a + - a = 0 :=
add_right_neg
```

The algebraic hierarchy contains further type classes with one binary operator. The main ones are listed below:

Type class	Properties	Examples
semigroup monoid left_cancel_semigroup	associativity of * semigroup with unit 1 semigroup with $c * a = c * b \rightarrow a = b$	$\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$ , $\mathbb{N}$ $\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$ , $\mathbb{N}$
right_cancel_semigroup group	semigroup with a * c = b * c $\rightarrow$ a = b monoid with inverse $^{-1}$	

Most of these structures have commutative versions (where  $a \cdot b = b \cdot a$  for all elements a, b), prefixed with comm\_: comm\_semigroup, comm\_monoid, comm\_group. All of these type classes have additive counterparts prefixed with add\_: add\_semigroup, add\_monoid, add\_group, add\_comm\_semigroup, add\_comm\_monoid, add\_comm\_group, add\_left\_cancel\_semigroup, add\_right\_cancel\_semigroup:

Type class	Properties	Examples
add_semigroup	associativity of +	$\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$ , $\mathbb{N}$
add_monoid	add_semigroup with unit o	$\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$ , $\mathbb{N}$
add_left_cancel_semigroup	add_semigroup <b>with</b>	$\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$ , $\mathbb{N}$
	$c + a = c + b \rightarrow a = b$	
add_right_cancel_semigroup	add_semigroup <b>with</b>	$\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$ , $\mathbb{N}$
	$a + c = b + c \rightarrow a = b$	
add_group	add_monoid with inverse -	$\mathbb{R}$ , $\mathbb{Q}$ , $\mathbb{Z}$


Although the additive type classes are isomorphic to their multiplicative counterparts, they are crucial when constructing algebraic structures with more than one binary operator such as rings and fields. To avoid duplicating all lemmas and definitions based on the multiplicative type classes, the copying process is automated by a tactic (transport\_multiplicative\_to\_additive) and an attribute (@[to\_additive]).

An example instance of add\_monoid is the type list with the empty list [] as zero and the append operator ++ as addition:

We could continue and register list with [] and ++ as an add

left\_cancel\_semigroup and an add\_right\_cancel\_semigroup.

The graph below illustrates the relationships between some of these type classes. An arrow from X to Y means "X inherits all properties from Y."



## 12.2 Type Classes over Two Binary Operators

The additive and multiplicative structures are amalgamated to form more complex type classes over two binary operators. One of these is field. A *field F* is defined by the following properties:

- F forms a commutative group under an operator +, called addition, with identity element o.
- F\{o} forms a commutative group under an operator \*, called multiplication.
- Multiplication distributes over addition—i.e., a \* (b + c) = a \* b + a \* c for all  $a, b, c \in F$ .

By running #print field, we can display all constants and properties required by field. Again, the type class includes some redundant properties such as left\_distrib and right\_distrib due to its construction.

We will now show that int2 is a field by instantiating the field type class with it. First, we must define multiplication on int2:

To declare int2 a field, we can reuse the instance int2.add\_group that we declared above using the syntax ..int2.add\_group. We can prove the remaining properties as follows:

```
@[instance] def int2.field : field int2 :=
{ one
                 := int2.one,
  mul
                 := int2.mul,
  inv
                 := \lambda a. a.
                 := by intros a b; cases' a; cases' b; refl,
  add\_comm
  exists_pair_ne :=
    by apply exists.intro int2.zero;
      apply exists.intro int2.one; finish,
                 := by intros a; cases' a; refl,
  one_mul
  mul one
                 := by intros a; cases' a; refl,
  mul inv cancel := by intros a h; cases' a; finish,
  inv zero
                 := by refl,
  mul assoc
    by intros a b c; cases' a; cases' b; cases' c: refl.
                 := by intros a b; cases' a; cases' b; refl,
  mul comm
  left distrib
                 : =
    by intros a b c; cases' a; cases' b; cases' c; refl,
  right distrib
    by intros a b c; cases' a; cases' b; cases' c; refl,
  ..int2.add_group }
```

With this declaration in place, we can now use the notations 1, \*, /, and more:

```
#reduce (1 : int2) * 0 / (0 - 1)
```

The command prints int2.zero. The type annotation: int2 is necessary here to tell Lean that we want to calculate in int2 and not in  $\mathbb{N}$ , the default. We can even use arbitrary numerals in int2. For example, the numeral 3 is interpreted as 1 + 1 + 1, which is the same as 1 in int2:


```
#reduce (3 : int2)
```

The command prints int2.one.

Besides field, there are many more type classes for structures with two binary operators. These are the main ones:

Type class	Properties	Examples
semiring comm_semiring ring comm_ring division_ring field	monoid and add_comm_monoid with distributivity semiring with commutativity of * monoid and add_comm_group with distributivity ring with commutativity of * ring with multiplicative inverse division_ring with commutativity of *	R, Q, Z, N R, Q, Z, N R, Q, Z R, Q, Z R, Q R, Q

The property  $\forall n$ ,  $n \neq 0$  = 0 required by the field type class is simply a convention to make division a total function. Mathematicians would regard division as a partial function. The graph below illustrates the relationships between these type classes.



The hierarchy between ring and field is more complex than depicted and includes type classes for domains, integral domains, Euclidean rings, and more.

Once we have instantiated the type classes with types of interest, we can use the ring and ring\_exp tactics [4] to normalize terms containing operators on those types. For example:

```
lemma ring_example (a b : int2) :
  (a + b) ^ 3 = a ^ 3 + 3 * a ^ 2 * b + 3 * a * b ^ 2 + b ^ 3 :=
by ring
```

```
lemma ring_exp_example (a b : int2) (n : ℕ):
```

```
(a + b) ^ (2 + n) =
(a + b) ^ n * (a ^ 2 + 2 * a * b + b ^ 2) :=
by ring_exp
```

These tactics are available for any type that is declared as a field or more generally as a commutative semiring. The distinguishing feature of ring\_exp is that it can reason about exponents containing variables.

For commutative monoids and groups, there is a similar normalization tactic called abel:

```
lemma abel_example (a b : \mathbb{Z}) :
a + b + o - (b + a + a) = -a := by abel
```

## 12.3 Coercions

When combining numbers from  $\mathbb{N}$ ,  $\mathbb{Z}$ ,  $\mathbb{Q}$ , and  $\mathbb{R}$  in the same lemmas, we may want to cast from one type to another. For example, given a natural number, we may need to convert it to an integer. Consider the following lemma and notice the type of the arguments of multiplication:

```
lemma neg_mul_neg_nat (n : \mathbb{N}) (z : \mathbb{Z}) :

(-z) * (-n) = z * n :=

neg_mul_neg_z n
```

Surprisingly, this statement does not lead to an error, although negation - n is not defined on  $n:\mathbb{N}$ , multiplication of  $z:\mathbb{Z}$  and n :  $\mathbb{N}$  is not defined, and the lemma neg\_mul\_neg does not apply to natural numbers.

The diagnosis command #print neg\_mul\_neg\_nat tells us what happened:

```
theorem neg_mul_neg_nat : \forall (z : \mathbb{Z}) \ (n : \mathbb{N}), \ -z * -\uparrow n = z * \uparrow n := \lambda(z : \mathbb{Z}) \ (n : \mathbb{N}), \ neg_mul_neg \ z \uparrow n
```

12.3. Coercions 277

Lean has a mechanism to introduce coercions, denoted by  $\uparrow$  or coe, when necessary. This coercion operator can be set up to provide implicit conversions between arbitrary types. Many coercions are already in place, including these:

- coe :  $\mathbb{N} \to \alpha$  casts  $\mathbb{N}$  to another semiring  $\alpha$ ;
- coe :  $\mathbb{Z} \to \alpha$  casts  $\mathbb{Z}$  to another ring  $\alpha$ ;
- coe :  $\mathbb{Q} \to \alpha$  casts  $\mathbb{Q}$  to another division ring  $\alpha$ .

In some cases, Lean is unable to figure out where to place the coercions. We can then provide some type annotations to guide Lean's inference algorithm, as in the following example:

```
lemma neg_nat_mul_neg (n : \mathbb{N}) (z : \mathbb{Z}) : 
 (- n : \mathbb{Z}) * (- z) = n * z := 
 neg_mul_neg n z
```

In proofs involving coercions, the tactic norm\_cast can be convenient. It can help with goals such as  $\vdash$  m n :  $\mathbb{N}$ , h :  $\uparrow$ m =  $\uparrow$ n  $\vdash$  m = n in the proof

```
lemma norm_cast_example_1 (m n : \mathbb{N}) (h : (m : \mathbb{Z}) = (n : \mathbb{Z})) : m = n := begin norm_cast at h, exact h end or \vdash m n : \mathbb{N} \vdash \uparrowm + \uparrown = \uparrow(m + n) in the proof lemma norm_cast_example_2 (m n : \mathbb{N}) : (m : \mathbb{Z}) + (n : \mathbb{Z}) = ((m + n : \mathbb{N}) : \mathbb{Z}) := by norm_cast
```

The norm\_cast tactic relies on lemmas such as the following:

```
nat.cast_add : \forall a \ b : \mathbb{N}, \ \uparrow(a + b) = \uparrow a + \uparrow b
int.cast_add : \forall a \ b : \mathbb{Z}, \ \uparrow(a + b) = \uparrow a + \uparrow b
rat.cast_add : \forall a \ b : \mathbb{Q}, \ \uparrow(a + b) = \uparrow a + \uparrow b
```

## 12.4 Normalization Tactics

The algebraic tactics ring, ring\_exp, and abel and the coercion tactic norm\_cast all work by normalization: They rewrite expressions in the hope that they become syntactically equal, at which point equality is trivial to prove. Like rw and simp, they produce a subgoal when they make some progress but do not fully succeed.

The optional position argument is as for the rewriting tactics (Section 2.5).

## ring and ring\_exp

```
ring [at position]
ring_exp [at position]
```

The ring and ring\_exp tactics prove equalities over commutative rings and semirings (such as  $\mathbb{N}$ ,  $\mathbb{Z}$ ,  $\mathbb{Q}$ , and  $\mathbb{R}$ ) by normalizing expressions and syntactically comparing the result. The ring\_exp variant performs more aggressive normalization in the presence of variables in exponents.

### abel

```
abel [at position]
```

The abel tactic proves equalities over additive commutative monoids and groups (such as  $\mathbb{N}$ ,  $\mathbb{Z}$ ,  $\mathbb{Q}$ , and  $\mathbb{R}$ ) by normalizing expressions and syntactically comparing the result.

### norm\_cast

```
norm_cast [at position]
```

The norm\_cast tactic moves coercions towards the inside of expressions, as a form of simplification.

## 12.5 Lists, Multisets, and Finite Sets

We have seen many examples of how lists can be used in previous chapters. But when making a new definition or stating a new lemma, we should also reflect on alternatives such as multisets and finite sets.

Consider the following definition, based on the binary trees we introduced in Section 4.8:

This function returns a list of all the elements occurring in the tree. It traverses the tree depth first, from left to right. But for some applications, we might not care about the order of the elements.

This is where multisets can help us. For multisets, we have  $\{3, 2, 1, 2\} = \{1, 2, 2, 3\}$ , whereas the two lists [3, 2, 1, 2] and [1, 2, 2, 3] are different. Multisets are defined as the quotient type over lists up to reordering. We can redo the above definition using multisets as follows:

```
def multiset.elems : btree \mathbb{N} \to \text{multiset } \mathbb{N}
| btree.empty := \emptyset
| (btree.node a l r) := {a} \cup (multiset.elems l \cup multiset.elems r)
```

Using this definition, we can prove that multiset.elems t = multiset.elems (mirror t), whereas list.elems t = list.elems (mirror t) does not hold.

For some applications, we might want to go a step further and ignore not only the order but also how often each element occurs in the tree, distinguishing only between occurrence and nonoccurrence. This is where finite sets, or *finsets*, come into play. On finsets, we have  $\{3, 2, 1, 2\} = \{1, 2, 3\}$ . Finsets are defined as the subtype of multisets that do not contain any repeated elements. (Another possible definition would have been as the subtype of

sets that are finite.) We can redo the definition above using finsets as follows:

```
def finset.elems : btree \mathbb{N} → finset \mathbb{N} | btree.empty := \emptyset | (btree.node a l r) := {a} \cup (finset.elems l \cup finset.elems r)
```

For all three of these structures, Lean offers sum and product operators to add or multiply all of the elements:

```
#eval list.sum [2, 3, 4] -- result: 9 #eval multiset.sum (\{2, 3, 4\} : multiset \mathbb{N}) -- result: 9 #eval finset.sum (\{2, 3, 4\} : finset \mathbb{N}) (\lambdan, n) -- result: 9 #eval list.prod [2, 3, 4] -- result: 24 #eval multiset.prod (\{2, 3, 4\} : multiset \mathbb{N}) -- result: 24 #eval finset.prod (\{2, 3, 4\} : finset \mathbb{N}) (\lambdan, n) -- result: 24
```

The operators finset.sum and finset.prod take a second argument, a function that maps the values of the finset to the values that should be added or multiplied. If we put the identity function  $\lambda n$ , n for that second argument, they behave like the corresponding operators for lists and multisets.

These operators require the type of the elements to be declared as an instance of add\_monoid for addition or of monoid for multiplication. For multiset and finset, they additionally require an instance declaration for add\_comm\_monoid or comm\_monoid because the result cannot depend on the order of adding or multiplying the elements.

Lean's mathlib provides a collections of lemmas describing how the big operators behave under reorderings, basic operators on list, multiset, and finset, homomorphisms, order relations, and more.

## 12.6 Order Type Classes

Many of the structures introduced above can be ordered. For example, the well-known order on the natural numbers can be defined

as

```
inductive nat.le : \mathbb{N} \to \mathbb{N} \to \mathsf{Prop} | refl : \forall a : \mathbb{N}, nat.le a a | step : \forall a b : \mathbb{N}, nat.le a b \to nat.le a (b + 1)
```

This is an example of a linear order. A linear order (or total order) is a binary relation  $\leq$  such that for all a, b, and c, the following properties hold:

- Reflexivity:  $a \le a$ .
- Transitivity: If  $a \le b$  and  $b \le c$ , then  $a \le c$ .
- Antisymmetry: If  $a \le b$  and  $b \le a$ , then a = b.
- Totality:  $a \le b$  or  $b \le a$ .

If a relation has the first three properties, it is a *partial order*. An example is the subset relation  $\subseteq$  on sets, finite sets, or multisets. If a relation has the first two properties, it is a *preorder*. An example is comparing lists by their length.

In Lean, there are type classes for the different kinds of orders: linear\_order, partial\_order, and preorder. The preorder class has the fields

```
le: \alpha \to \alpha \to \mathsf{Prop}
le_refl: \forall a : \alpha, le a a
le_trans: \forall a \ b \ c : \alpha, le a b \to le b c \to le a c
```

The partial order class has the additional field

```
le_antisymm : \foralla b : \alpha, le a b \rightarrow le b a \rightarrow a = b
```

and linear\_order has the additional field

```
le total : \foralla b : \alpha, le a b \vee le b a
```


We can declare the preorder on list  $\alpha$  that compares lists by their length as follows:

This type class instance gives us access to the infix syntax  $\leq$  and to the corresponding relations >, <, and >:

```
lemma list.length.preord_example \{\alpha: Type\}\ (c:\alpha): [c] > [] := dec trivial
```

dec\_trivial is a special lemma that relies on type class inference to prove trivial decidable goals.

Complete lattices, which we discussed in Chapter 10, are formalized as another type class, complete\_lattice, which inherits from partial\_order. In the following diagram, an arrow from X to Y means "X inherits all properties from Y."



Finally, Lean provides type classes that combine orders and algebraic structures: ordered\_cancel\_comm\_monoid, ordered\_comm\_group, ordered\_semiring, linear\_ordered\_semiring, linear\_ordered\_comm\_ring, linear\_ordered\_field. All these mathematical structures relate  $\leq$  and < with the constants o, 1, +, and \* by monotonicity rules (e.g.,  $a \leq b \rightarrow c \leq d \rightarrow a + c \leq b + d$ ) and cancellation rules (e.g.,  $c + a < c + b \rightarrow a < b$ ).

## **12.7 Summary of New Lean Constructs**

### **Notation**

 $\uparrow$  coercion operator coe

#### Lemma

dec\_trivial decidable truth (e.g., a true closed executable expression)

### **Tactics**

abel normalizes terms of commutative monoids and groups

norm\_cast normalizes coercions

ring normalizes ring expressions

ring\_exp normalizes ring expressions including exponents

with variables

## **Chapter 13**

## **Rational and Real Numbers**

We have seen how the natural numbers  $\mathbb N$  can be defined as an inductive type and how the integers  $\mathbb Z$  can be defined as a quotient over  $\mathbb N \times \mathbb N$ . In this chapter, we review the construction of the rational numbers  $\mathbb Q$  and the real numbers  $\mathbb R$ . The tools used for these constructions are inductive types, subtypes, and quotients.

This procedure can be followed to construct types with specific properties:

- 1. Create a new type that is large enough to represent all elements, but not necessarily in a unique manner.
- 2. Take the quotient of this representation, equating elements as needed.
- 3. Define operators on the quotient type by lifting functions from the base type, and prove that they are compatible with the quotient relation.

We used this approach to construct the type  $\mathbb Z$  in Section 11.5.1. It can be used for  $\mathbb Q$  and  $\mathbb R$  as well.

## **13.1 Rational Numbers**

A rational number is a number that can be expressed as a fraction n/d of integers n and d, where  $d \neq 0$ :

```
structure fraction :=
(num : \mathbb{Z})
(denom : \mathbb{Z})
(denom_ne_zero : denom \neq \circ)
```

The number n is called the numerator, and the number d is called the denominator. The representation of a rational number as a fraction is not unique. For example, the rationals 1/2, 2/4, and -1/-2 are all equal. This representation as a fraction will serve as the base type from which we will derive a quotient.

Two fractions  $n_1/d_1$  and  $n_2/d_2$  represent the same rational number if the ratio between numerator and denominator are the same:  $n_1*d_2=n_2*d_1$ . To construct the quotient over the type fraction with respect to this relation, we show that the relation is an equivalence relation. This is achieved by declaring fraction an instance of the setoid type class:

```
namespace fraction
```

```
by refl
end fraction
```

Then we can define the type of rationals as the quotient over this setoid:

```
def rat : Type :=
quotient fraction.setoid
```

To define zero, one, addition, multiplication, and other operations, we first define them on the fraction type. To add two fractions, we convert the fractions to a common denominator and add the numerators. The easiest common denominator to use is simply the product of the two denominators:

We register these operations directly as instances of the syntactic type classes such as has\_add to be able to use convenient notation such as + on fraction. Similarly, we define zero as 0 := 0/1, one as 1 := 1/1, and multiplication as the pairwise multiplication of numerators and denominators.

To lift these operations to  $\mathtt{rat}$ , we must prove them compatible with  $\approx$ :

```
namespace fraction

@[simp] lemma add_num (a b : fraction) :
   num (a + b) = num a * denom b + num b * denom a :=
by refl

@[simp] lemma add_denom (a b : fraction) :
   denom (a + b) = denom a * denom b :=
by refl
```

```
lemma add_equiv_add {a a' b b' : fraction} (ha : a \approx a')
    (hb : b \approx b') :
  a + b \approx a' + b' :=
begin
  simp [setoid iff, add denom, add num] at *,
  calc (num a * denom b + num b * denom a)
          * (denom a' * denom b')
      = num a * denom a' * denom b * denom b'
          + num b * denom b' * denom a * denom a':
    by simp [add_mul, mul_add]; ac_refl
  ... = num a' * denom a * denom b * denom b'
          + num b' * denom b * denom a * denom a' :
    by simp [*]
  ... = (num a' * denom b' + num b' * denom a')
          * (denom a * denom b):
    by simp [add_mul, mul_add]; ac_refl
end
```

#### end fraction

(The ac\_refl tactic is a variant of refl that reasons up to associativity and commutativity, much like cc.) Then we can use quotient.lift (2) to define the rat operations, and we can instantiate the relevant syntactic type classes, such as

```
namespace rat

@[instance] def has_add : has_add rat :=
{ add := quotient.lift₂ (λa b : fraction, [a + b])
    begin
    intros a b a' b' ha hb,
    apply quotient.sound,
    exact fraction.add_equiv_add ha hb
    end }

end rat
```

From here, we can proceed and prove all the properties needed to make rat an instance of field or, if we are more ambitious, discrete linear ordered field.

Alternative Definitions of the Rational Numbers In mathlib, another approach is used to defined the rationals. The type rat is defined as a subtype of fraction, with the requirement that the denominator is positive and that the numerator and the denominator are coprime (i.e., they have no common divisors except 1 and -1):

We discussed this general strategy in Section 11.5.3. In this way, no quotient is required, computation is more efficient, and more properties are syntactic equalities up to computation. A disadvantage is that function definitions become more complicated due to the need to normalize fractions.

Another alternative is to define all elements syntactically, including the desired operations:

```
inductive pre_rat : Type
| zero : pre_rat
| one : pre_rat
| add : pre_rat → pre_rat → pre_rat
| sub : pre_rat → pre_rat → pre_rat
| mul : pre_rat → pre_rat → pre_rat
| div : pre_rat → pre_rat → pre_rat
```

Then we define an equivalence relation and take its quotient to enforce congruence rules and the field axioms:

```
inductive rat.rel : pre_rat → pre_rat → Prop
| add_congr {a b c d : pre_rat} :
    rat.rel a b → rat.rel c d →
    rat.rel (pre_rat.add a c) (pre_rat.add b d)
| add_assoc {a b c : pre_rat} :
    rat.rel (pre_rat.add a (pre_rat.add b c))
        (pre_rat.add (pre_rat.add a b) c)
| zero_add {a : pre_rat} :
    rat.rel (pre_rat.add pre_rat.zero a) a
-- etc.

def rat : Type :=
quot rat.rel
```

This construction does not require  $\mathbb{Z}$  and that it can easily be registered as a field. Moreover, the same approach can be used for other algebraic constructions, such as free monoids and free groups. A drawback is that the definition of orders and lemmas about them become more complicated.

### 13.2 Real Numbers

Some sequences of rational numbers seem to converge because the numbers in the sequence get closer and closer to each other, and yet do not converge to a rational number. The sequence

```
a_0 = 1
a_1 = 1.4
a_2 = 1.41
a_3 = 1.414
a_4 = 1.4142
\vdots
```

where  $a_n$  is the largest number with n digits after the decimal point such that  $a_n^2 < 2$ , is such a sequence. It seems to converge be-

cause each  $a_n$  is at most  $10^{-n}$  away from any of the following numbers, but the limit is  $\sqrt{2} \notin \mathbb{Q}$ . In that sense, the rational numbers are incomplete, and the reals are their *completion*. To construct the reals, we need to fill in the gaps that are revealed by these sequences that seem to converge but do not.

Cauchy sequences capture the notion of a sequence that seems to converge. A sequence  $a_0, a_1, \ldots$  is *Cauchy* if for any  $\varepsilon > 0$ , there exists an  $N \in \mathbb{N}$  such that for all  $m \geq N$ , we have  $|a_N - a_m| < \varepsilon$ . In other words, no matter how small we choose  $\varepsilon$ , we can always find a point in the sequence from which all following numbers deviate by less than  $\varepsilon$ .

We formalize sequences of rational numbers as functions  $f:\mathbb{N}\to\mathbb{Q}$  and denote the absolute value  $|\ |$  by abs. This yields the following Lean definition of Cauchy sequences:

```
def is_cau_seq (f : \mathbb{N} \to \mathbb{Q}) : Prop := \forall \varepsilon > 0, \exists N, \forall m > N, abs (f N - f m) < \varepsilon
```

Not every sequence is a Cauchy sequence:

```
lemma id_not_cau_seq :
   ¬ is_cau_seq (λn : N, (n : Q)) :=
begin
  rw is_cau_seq,
  intro h,
  cases' h 1 zero_lt_one with i hi,
  have hi_succi :=
    hi (i + 1) (by simp),
  simp [←sub_sub] at hi_succi,
  assumption
end
```

We define a type of Cauchy sequences as a subtype:

```
def cau_seq : Type := \{f : \mathbb{N} \to \mathbb{Q} // \text{ is_cau_seq f} \}
```

It will be convenient to have an auxiliary function that extracts the actual sequence from a cau\_seq:

```
\mathsf{def} \ \mathsf{seq\_of} \ (\mathsf{f} : \mathsf{cau\_seq}) : \mathbb{N} \to \mathbb{Q} := \mathsf{subtype.val} \ \mathsf{f}
```

The basic idea of the construction is to represent the real numbers by Cauchy sequences. Each Cauchy sequence represents the real number that is its limit; for example, the sequence  $a_n = 1/n$  represents the real number o, and the sequence 1, 1.4, 1.41, ... represents the real number  $\sqrt{2}$ .

Two different Cauchy sequences can represent the same real number; for example, the sequence  $a_n = 1/n$  and the constant sequence  $b_n = 0$  both represent 0. Therefore, we need to take the quotient over sequences representing the same real number. Two sequences represent the same real number when their difference converges to zero:

```
namespace cau_seq
@[instance] def setoid : setoid cau_seq :=
            := \lambda f g : cau_seq,
     \forall \varepsilon > 0, \exists N, \forall m \geq N, abs (seq_of f m - seq_of g m) < \varepsilon,
   iseqv :=
     begin
         apply and.intro,
         \{ \text{ intros } f \in h\epsilon, \}
           apply exists.intro o,
           finish },
         apply and.intro,
         { intros f g hfg \varepsilon h\varepsilon,
           cases' hfg \varepsilon h\varepsilon with N hN,
           apply exists.intro N,
           intros m hm.
           rw abs sub.
           apply hN m hm },
         { intros f g h hfg hgh \varepsilon h\varepsilon,
           cases' hfg (\varepsilon / 2) (half_pos h\varepsilon) with N<sub>1</sub> hN<sub>1</sub>,
           cases' hgh (\varepsilon / 2) (half_pos h\varepsilon) with N<sub>2</sub> hN<sub>2</sub>,
           apply exists.intro (max N_1 N_2),
            intros m hm,
```

```
calc abs (seq_of f m - seq_of h m)  \leq \text{abs (seq_of f m - seq_of g m)} \\ + \text{abs (seq_of g m - seq_of h m)} : \\ \text{by apply abs_sub_le} \\ \dots < \varepsilon \ / \ 2 + \varepsilon \ / \ 2 : \\ \text{add_lt_add (hN_1 m (le_of_max_le_left hm))} \\ \text{(hN_2 m (le_of_max_le_right hm))} \\ \dots = \varepsilon : \\ \text{by simp } \} \\ \text{end } \} \\ \\ \text{lemma setoid_iff (f g : cau_seq) :} \\ f \approx g \leftrightarrow \\ \forall \varepsilon > o, \ \exists N, \ \forall m \geq N, \ \text{abs (seq_of f m - seq_of g m)} < \varepsilon := \\ \text{by refl} \\ \\ \text{end cau_seq} \\ \\ \end{aligned}
```

Using this setoid instance, we can now define the real numbers:

```
def real : Type :=
quotient cau_seq.setoid
```

Like for the rational numbers, we need to define zero, one, addition, multiplication, and other operators. We define them on cau\_seq first and lift them to real afterwards. For the constants o and 1, we can define them simply as constant sequences. Any constant sequence is a Cauchy sequence:

```
namespace cau_seq  \begin{tabular}{lll} def & const & (q:\mathbb{Q}): cau_seq := \\ subtype.mk & (\lambda_-:\mathbb{N}, q) & (by rw is_cau_seq; intros $\epsilon$ $h\epsilon$; finish) \\ We can declare real instances of the syntactic type classes $has_zero$ and $has one: \\ \end{tabular}
```

```
@[instance] def has_zero : has_zero real :=
{ zero := [cau_seq.const o] }
```

```
@[instance] def has_one : has_one real :=
{ one := [cau_seq.const 1] }
```

Defining addition (and, similarly, multiplication) of real numbers requires a little more effort. We define addition on Cauchy sequences by adding the elements of the sequence pairwise:

```
@[instance] def has_add : has_add cau_seq := 
{ add := \lambda f g : cau_seq,
    subtype.mk (\lambda n : \mathbb{N}, seq_of f n + seq_of g n) sorry }
```

This definition requires a proof that the result is a Cauchy sequence, given that f and g are Cauchy sequences. It is omitted.

Next, we need to show that this addition is compatible with  $\approx$ :

```
lemma add_equiv_add {f f' g g' : cau_seq} (hf : f \approx f')
    (hg : g \approx g') :
  f + g \approx f' + g' :=
begin
  intros \varepsilon_0 h\varepsilon_0,
  simp [setoid iff],
  cases' hf (\epsilon_{\rm o} / 2) (half_pos h\epsilon_{\rm o}) with Nf hNf,
  cases' hg (\varepsilon_0 / 2) (half pos h\varepsilon_0) with Ng hNg,
  apply exists.intro (max Nf Ng),
  intros m hm.
  calc abs (seq_of (f + g) m - seq_of (f' + g') m)
       = abs ((seq of f m + seq of g m)
            - (seq_of f' m + seq_of g' m)) :
    by refl
  ... = abs ((seq_of f m - seq_of f' m)
            + (seq_of g m - seq_of g' m)):
    begin
       have arg_eq:
         seq_of f m + seq_of g m - (seq_of f' m + seq_of g' m) =
         seq of f m - seq of f' m + (seq of g m - seq of g' m),
         by linarith.
       rw arg eq
    end
  \dots \leq abs (seq_of f m - seq_of f' m)
```

```
+ abs (seq_of g m - seq_of g' m):
by apply abs_add
... < \varepsilon_0 / 2 + \varepsilon_0 / 2:
add_lt_add (hNf m (le_of_max_le_left hm))
    (hNg m (le_of_max_le_right hm))
... = \varepsilon_0:
by simp
end
end cau seq
```

To prove that f + g  $\approx$  f' + g', we are given an  $\varepsilon_{o}$  > o and must show that there exists a number N such that

```
\forallm, m \geq N \rightarrow abs (subtype.val (f + g) m - subtype.val (f' + g') m) < \varepsilon_0
```

To obtain N, we use  $f \approx f'$  and  $g \approx g'$ . The equivalence  $f \approx f'$  gives us for any  $\epsilon > 0$  a number Nf such that abs (seq\_of f m - seq\_of f' m) <  $\epsilon$  for all m  $\geq$  Nf. The fact  $g \approx g'$  gives us a number Ng with a similar property. For the calculations to work out in the end, we take the numbers Nf and Ng for  $\epsilon := \epsilon_o / 2$ . Then we choose N to be the maximum of Nf and Ng, so that we get the inequalities for any m  $\geq$  N. The calc block at the end of the proof establishes that, for all m  $\geq$  N,

```
abs (seq_of (f + g) m - seq_of (f' + g') m) < \varepsilon_0
```

Using this lemma, we can define addition on real:

```
namespace real

@[instance] def has_add : has_add real :=
{ add := quotient.lift₂ (λa b : cau_seq, [a + b])
    begin
    intros a b a' b' ha hb,
    apply quotient.sound,
    exact cau_seq.add_equiv_add ha hb,
    end }
```

end real

In summary, real numbers are defined as a quotient over Cauchy sequences, which are in turn defined as a subtype of  $\mathbb{N} \to \mathbb{Q}$ .

Alternative Definitions of the Real Numbers In mathlib, the construction of the real numbers is essentially as described above. Only some definitions are stated in a more general fashion to allow construction of other algebraic structures, such as the *p*-adic numbers [20].

Alternatively, the real numbers can be defined using Dedekind cuts. A number  $r:\mathbb{R}$  is then represented as the set of numbers  $x:\mathbb{Q}$  such that x < r. Another alternative, which does not depend on  $\mathbb{Q}$ , is to define  $\mathbb{R}$  using binary sequences  $\mathbb{N} \to \mathsf{bool}$ . The elements of the sequence represent the number's digits. This works particularly well if we only need the real interval [0,1].

## **13.3 Final Exhortation**

We have now reached the end of this course. You now know fundamental theory and techniques in interactive theorem proving as well as some application areas. Although we used Lean, your skills should help you with other systems, especially those based on simple or dependent type theory. You should also be able to read many if not most scientific papers in the area.

Even if you do not choose to pursue a career in theorem proving, the authors hope you will bring proof assistants with you and use them when it makes sense, either because of their high trustworthiness or because of their convenience for keeping track of complex proof goals.

If you continue using Lean, the natural next step would be to familiarize yourself with mathlib and its documentation. If you use Lean outside a class context, you will often find yourself looking

up lemmas and tactics. The #find command will surely be useful.<sup>1</sup> And the Lean Zulip chat<sup>2</sup> is the meeting place of Lean users.

## **13.4 Summary of New Lean Constructs**

#### **Command**

#find looks up a lemma by pattern matching

### **Tactic**

ac\_refl proves equalities l = r up to associativity and

commutativity

https://leanprover-community.github.io/mathlib\_docs/commands. html#find

<sup>2</sup>https://leanprover.zulipchat.com/

# **Bibliography**

- [1] J. Avigad, L. de Moura, and S. Kong. Theorem Proving in Lean: Release 3.20.0. 2020. https: //leanprover.github.io/theorem\_proving\_in\_lean/.
- [2] J. Avigad, L. de Moura, and J. Roesch. Programming in Lean. 2016. https://leanprover.github.io/programming\_in\_lean/.
- [3] J. Avigad, G. Ebner, and S. Ullrich. The Lean Reference Manual: Release 3.3.0. 2018. https://leanprover.github.io/reference/.
- [4] A. Baanen. A Lean tactic for normalising ring expressions with exponents (short paper). In N. Peltier and V. Sofronie-Stokkermans, editors, IJCAR 2020, volume 12167 of Lecture Notes in Computer Science, pages 21–27. Springer, 2020. https://lean-forward.github.io/ring\_exp/paper.pdf.
- [5] H. P. Barendregt, W. Dekkers, and R. Statman. *Lambda Calculus with Types*. Perspectives in logic. Cambridge University Press, 2013.
- [6] M. Carneiro. The type theory of Lean. MSc thesis, Carnegie Mellon University, 2019. https://github.com/digamao/lean-type-theory/releases/download/v1.0/main.pdf.
- [7] G. Ebner, S. Ullrich, J. Roesch, J. Avigad, and L. de Moura. A metaprogramming framework for formal verification. *PACMPL*,

- 1(ICFP):34:1-34:29, 2017.
  https://leanprover.github.io/papers/tactic.pdf.
- [8] G. Gonthier. Formal proof—the Four-Color Theorem. Notices AMS, 55(11):1382–1393, 2008. https://www.ams.org/notices/200811/tx081101382p.pdf.
- [9] G. Gonthier, A. Asperti, J. Avigad, Y. Bertot, C. Cohen, F. Garillot, S. L. Roux, A. Mahboubi, R. O'Connor, S. O. Biha, I. Pasca, L. Rideau, A. Solovyev, E. Tassi, and L. Théry. A machine-checked proof of the Odd Order Theorem. In S. Blazy, C. Paulin-Mohring, and D. Pichardie, editors, ITP 2013, volume 7998 of Lecture Notes in Computer Science, pages 163–179. Springer, 2013.

https://hal.inria.fr/hal-00816699/document.

- [10] K. Gopinathan and I. Sergey. Certifying certainty and uncertainty in approximate membership query structures. In S. Lahiri and C. Wang, editors, *CAV 2020*, Lecture Notes in Computer Science. Springer, 2020. https://ilyasergey.net/papers/ceramist-draft.pdf.
- [11] R. Gu, Z. Shao, H. Chen, X. N. Wu, J. Kim, V. Sjöberg, and D. Costanzo. CertiKOS: An extensible architecture for building certified concurrent OS kernels. In K. Keeton and T. Roscoe, editors, OSDI 2016, pages 653–669. USENIX Association, 2016. https://www.usenix.org/system/files/conference/osdi16/osdi16-gu.pdf.
- [12] T. C. Hales, M. Adams, G. Bauer, D. T. Dang, J. Harrison, T. L. Hoang, C. Kaliszyk, V. Magron, S. McLaughlin, T. T. Nguyen, T. Q. Nguyen, T. Nipkow, S. Obua, J. Pleso, J. Rute, A. Solovyev, A. H. T. Ta, T. N. Tran, D. T. Trieu, J. Urban, K. K. Vu, and R. Zumkeller. A formal proof of the Kepler conjecture. *CoRR*, abs/1501.02155, 2015. http://arxiv.org/abs/1501.02155.

- [13] J. Harrison. Formal verification at Intel. In *LICS* 2003, pages 45–54. IEEE Computer Society, 2003. https://ieeexplore.ieee.org/document/1210044.
- [14] J. Harrison, J. Urban, and F. Wiedijk. History of interactive theorem proving. In J. H. Siekmann, editor, *Computational Logic*, volume 9 of *Handbook of the History of Logic*, pages 135–214. Elsevier, 2014. https://www.cl.cam.ac.uk/~jrh13/papers/joerg.pdf.
- [15] J. R. Hindley. The principal type-scheme of an object in combinatory logic. Trans. Amer. Math. Soc., 146:29-60, 1969. https://www.jstor.org/stable/pdf/1995158.pdf?casa\_ token=zeSppXjXzUsAAAAA:fzvl6NKVjcLoiVXjyrMSbh\_ bqosxtY4iPeoX5NMC6Fvl0dMDxYG1paIiqrc6P2Q7UTaT46PVHByMTWrpox4SKwOhbcxjLIm-FxWRDx-FtdyWwHaoueG.
- [16] G. Klein, J. Andronick, K. Elphinstone, G. Heiser, D. Cock, P. Derrin, D. Elkaduwe, K. Engelhardt, R. Kolanski, M. Norrish, T. Sewell, H. Tuch, and S. Winwood. seL4: Formal verification of an operating-system kernel. *Commun. ACM*, 53(6):107–115, 2010.
  - https://www.researchgate.net/profile/Gernot\_Heiser/publication/220910193\_SeL4\_Formal\_verification\_of\_an\_OS\_kernel/links/09e4150f00292a0329000000/SeL4-Formal-verification-of-an-OS-kernel.pdf.
- [17] D. E. Knuth. *The T<sub>E</sub>Xbook*. Addison-Wesley, 1986. http://www.ctex.org/documents/shredder/src/texbook.pdf.
- [18] R. Kumar, M. O. Myreen, M. Norrish, and S. Owens. CakeML: A verified implementation of ML. In S. Jagannathan and P. Sewell, editors, *POPL 2014*, pages 179–192. ACM, 2014. https://cakeml.org/popl14.pdf.

- [19] X. Leroy. A formally verified compiler back-end. *J. Automated Reasoning*, 43(4):363-446, 2009. https://arxiv.org/pdf/0902.2137.pdf.
- [20] R. Y. Lewis. A formal proof of Hensel's lemma over the p-adic integers. In A. Mahboubi and M. O. Myreen, editors, CPP 2019, pages 15–26. ACM, 2019. https://arxiv.org/pdf/1909.11342.pdf.
- [21] J. Limperg. A novice-friendly induction tactic for Lean. In C. Hritcu and A. Popescu, editors, *CPP '21*, pages 199–211. ACM, 2021. https://arxiv.org/pdf/2012.08990.pdf.
- [22] M. Lipovaca. Learn You a Haskell for Great Good!: A Beginner's Guide. No Starch Press, 1st edition, 2011. http://learnyouahaskell.com/.
- [23] A. Lochbihler. Verifying a compiler for Java threads. In A. D. Gordon, editor, ESOP 2010, volume 6012 of Lecture Notes in Computer Science, pages 427–447. Springer, 2010. https://link.springer.com/content/pdf/10.1007/978-3-642-11957-6\_23.pdf.
- [24] The mathlib Community. The Lean mathematical library. In J. Blanchette and C. Hriţcu, editors, *CPP* 2020, pages 367–381. ACM, 2020. https://arxiv.org/pdf/1910.09336.pdf.
- [25] R. Milner. A theory of type polymorphism in programming. J. Comput. Syst. Sci., 17(3):348-375, 1978. https://www.sciencedirect.com/science/article/pii/0022000078900144/pdf?md5=cdcf7cdb7cfd2e1e4237f4f779caodf7&pid=1-s2.0-0022000078900144-main.pdf&\_valck=1.
- [26] R. Nederpelt and H. Geuvers. *Type Theory and Formal Proof:* An Introduction. Cambridge University Press, 2014.

- [27] T. Nipkow and G. Klein. *Concrete Semantics: With Isabelle/HOL.*Springer, 2014. http://www.concretesemantics.org/concrete-semantics.pdf.
- [28] B. O'Sullivan, D. Stewart, and J. Goerzen. *Real World Haskell*. O'Reilly, 2008. http://book.realworldhaskell.org/read/.
- [29] A. J. Perlis. Epigrams on programming. SIGPLAN Notices, 17(9):7-13, 1982. https://dl.acm.org/doi/pdf/10.1145/947955.1083808? casa\_token=Z\_grmOD\_yAYAAAAA:dvMu7thQc-8XmGJCZ1ARQ24\_XB1qel3M1jYUmBPAKfuBaZrryxmKiKValMrwpunJodv9vhr2kDvMxA.
- [30] B. C. Pierce. *Types and Programming Languages*. MIT Press, 2002.
- [31] F. van Raamsdonk. Logical Verification: Course Notes. 2011. https://www.cs.vu.nl/~jbe248/lv2017/notes.pdf.
- [32] D. M. Russinoff. A mechanically checked proof of correctness of the AMD K5 floating point square root microcode. Formal Methods in System Design, 14(1):75–125, 1999. https://link.springer.com/content/pdf/10.1023/A: 1008669628911.pdf.
- [33] D. Selsam and L. de Moura. Congruence closure in intensional type theory. In N. Olivetti and A. Tiwari, editors, IJCAR 2016, volume 9706 of Lecture Notes in Computer Science, pages 99–115. Springer, 2016. https://arxiv.org/pdf/1701.04391.pdf.
- [34] C. Watt. Mechanising and verifying the WebAssembly specification. In J. Andronick and A. P. Felty, editors, *CPP 2018*, pages 53–65. ACM, 2018. https://www.cl.cam.ac.uk/~caw77/papers/mechanising-and-verifying-the-webassembly-specification.pdf.